
前言

如果你已经学习了VB6、VBA或VBScript，或者只是工作需要和爱好VB编程，而现在需要向.NET平台开发前进，本书都会帮助你快速达到目标，能够实现真正有用的软件功能。

本书内容包括了VB.NET编程语言、软件结构、数据库应用，以及大量的.NET Framework类库应用；并对一系列开发资源进行了有效地封装，方便在项目中重复利用这些代码。

无论是学习还是工作过程中，都可以快速参考相应的主题，查询功能实现的代码和思路；在这些基础之上，开发者可以修改或扩展代码库，并在项目中灵活应用它们。

学以致用！正是本书的目的所在。

祝大家学习、工作愉快！





1.概述

本章，我们将开始进入Visual Basic.NET的开发世界，为了有一定的共同语言，我们首先约定一些概念。

.NET是微软公司一系列战略的总称，其中关于软件开发和运行的环境称为.NET Framework。

Visual Studio是微软公司的一个集成开发工具，包括一些特定的版本，如专业版、旗舰版、Express版等等。在本书中使用的Visual Studio 2013中，包括Desktop和Web版本。从第2章开始，我们将统一使用缩写VS表示Visual Studio。

Visual Basic是Visual Studio中重要的一员，从第2章开始，我们也会使用一些缩写，如VB6表示Visual Basic 6及更早的版本，VB.NET表示Visual Basic.NET，而VB则是它们的统称，请大家根据实际情况加以区分。

1.1. VB一家亲

对于上世纪从DOS开始玩编程的朋友，一定对BASIC不陌生，它和汇编、C、C++、dBase、Foxpro等技术一起成为了一个时代的代表。

很快，我们进入了Windows时代。在基于窗口的应用程序开发中，使用传统的C、C++进行工作，相信不少程序员都会很多难忘的经历。而此时，微软推出了Visual Basic，它成为Visual Studio大家庭的一员，和Visual C++、Visual Foxpro，以及同时代的Delphi一起，成为Windows软件开发的强大工具。

Visual Basic大大简化了窗口的创建工作，我们不需要写一行代码就可以创建一个窗口界面，工作效率有多高，相信大家都能理解！而对于越来越多的基于ActiveX技术组件的出现，也让Visual Basic有了更加强大的后援团，众多开发资源的支持，也让Visual Basic几乎无敌于那个时代的Windows桌面应用开发领域。

时过境迁，转眼到了二十一世纪，互联网高速发展，开发技术突飞猛进。微软也放出大招，全面推进.NET战略，无论是Windows操作系统，还是开发工具，都全面进化。此时，Visual Basic也没有被放弃，而是一起进入了.NET大家庭。

在早期的Visual Studio.NET开发环境中，Visual Basic.NET和VC++和C#一起成为了开发语言的三驾马车，而Visual Basic.NET也完全融入了.NET世界，与前一代的Visual Basic相比……，算了，还是别比了，没有什么可比性了，完全不是一回事，要说有那么一点关系，那只是字面上的，如编程语言名称和其中所使用的关键字，看上去差不多，但本质已改变。

当然，这种改变是积极的、有效的和划时代的，完全融入.NET世界的好处就是，在Visual Basic.NET中，可以和C#程序一样，完整地使用.NET Framework中的强大开发资源，这也是基于.NET平台进行软件开发的最大特点和优势之一，无论你熟悉哪种编程语言，都可以有效地使用.NET Framework中的开发资源来提高开发效率。

说到Visual Basic，我们不得不提起它的一个近亲，这就是VBA（Visual Basic Application），如果大家在Microsoft Office中用代码写过宏（Macro），一定对VBA不会陌生了。对于VBA，如果我们熟练掌握了Visual Basic或Visual Basic.NET编程，那么，再熟悉一下Word、Excel等文档的对象结构，就可以很快掌握Microsoft Office的编程工作，对于在Excel经常需要进行大规模计算的工作者来，掌握VBA可以极大地提高工作效率。

Visual Basic的另外一个亲戚就是VBScript，在ASP（不是ASP.NET）开发Web应用的时候，VBScript真的是动态语言的不二选择了。实际上，VBScript已深深地置入到了Windows操作系统，不信？你可以创建一个文本文件，把扩展名修改为.vbs，然后输入以下内容，保存文件，再双击执行，看看会有什么结果。


MsgBox "没有开发工具也能写VB代码"



对于功能简单的计算代码，VBScript就可以了，甚至不需要任何特殊的开发工具；不过，这并不是Visual Basic.NET代码。

好的，现在，我们接下来真正要做的就是讨论Visual Basic.NET编程，相关的主题包括：


	编程语言。介绍Visual Basic.NET编程语言的数据处理、逻辑控制等基础知识。

	代码结构。我们不在乎是过程式编程或是面向对象编程；在Visual Basic.NET中，我们可以更加自由地选择。无论我们是使用模块还是使用类在组织代码，都一样能够完成各种工作，而我们需要做的就是在了解这些内容后，根据项目实际需求做出自己的选择。

	功能实现。学习编程，我们都知道只有基本的代码是不会做出强大功能的软件的，而对于开发资源的利用则可以让我们的软件功能有着无限的可能。所以，学习.NET Framework中的强大开发资源，或者是第三方开发的资源，再或者是我们自己创建的组件，这都是我们实现应用程序功能中所不能缺少的重要组成部分。

	封装。封装的作用主要有两种，即简化使用和重复利用。很多开发资源的作用是对于基本的、纯技术性的，以及系统级的功能实现，如文件处理、网络连接、多媒体应用、数据库操作等等。但我们在开发应用程序中，更注重的是用户的需求及软件的功能，如填写表单、保存数据、搜索、图片转换、从网络中获取文件等等。如何将开发资源和软件功能有机的结合，以及如何能够提高使用开发资源的效率。我想，这就是封装的本质所在。



1.2. VB.NET编程基础

本节，我们将介绍关于使用Visual Basic.NET开发的一些基本知识，以便在学习中对代码进行有效地测试。

1.2.1. 开发工具

对于从事任何工作的人来讲，好的工具和方法都是必须的，正所谓“工欲善其事，必先利其器”。软件开发更是这样，如果我们要真正的高效开发软件，就必须有着强有力的开发工具作为支撑。接下来，我们就来讨论关于本书中开发与测试工作中所使用的开发环境及相关工具。

本书中的开发环境，操作系统使用了Windows 7，项目代码都基于.NET Framework 4.0或更高版本，不过，如果大家安装了以下介绍的开发工具，这些都不会再是问题了。

本书中主要使用的代码编辑和测试的开发工具同要是Visual Studio家族中的成员，它们都可以从微软公司网站免费下载使用。我们使用的开发工具主要包括：


	VS Express 2013 for Desktop，主要用于开发运行于单机的软件类型，如Windows窗体应用程序或命令行应用程序。

	VS Express 2013 for Web，主要用于开发Web项目，如ASP.NET项目。



此外，本书的数据库主要使用了SQL Server 2012 Express，我们可以从微软公司网站免费下载使用。除了以SQL Server 2012为主介绍数据库的使用，我们还同时考虑了Access和MySQL数据库的应用开发，在桌面或Web应用中，这两种数据库都是比较常用的。

1.2.2. 项目类型

本书会使用三种基本的项目类型，即：命令行程序、窗口应用项目和ASP.NET项目。其中，前两项将在VS Express 2013 for Desktop中开发与测试；ASP.NET项目则在VS Express 2013 for Web中开发与测试；此外，对于正式一些的Web项目，我更喜欢直接使用IIS进行测试。

在介绍Visual Basic.NET编程语言时，我们会更多地使用命令行程序类型，因为这样可以简化很多工作，也使代码更加简洁明了，更易演示最重要的内容。更多功能的介绍过程中，我们会根据需要选择使用窗口应用项目或ASP.NET项目类型进行相关的测试工作。

1.2.3. 第一个程序

说了这么多，大家一定迫不及待地想飚代码了，好的，现在，我们就打开VS Express 2013 for Desktop，通过菜单“新建”->“文件项目”打开项目的创建对话框，如下图：

[image: 图像说明文字]


我们在左栏模板分类中选择“Visual Basic”中的“Windows”，然后，在模板中选择“控制台应用程序”，也就是我们所说的命令行应用程序。在对话框的下方，我们可以设置项目的名称及项目保存的路径；此外，对于简单的项目，“为解决方案创建目录”选项也可以取消。最后，我们单击“确定”按钮完成项目的创建工作。

项目创建后，我们可以看到Module1.vb文件的内容，如下图所示。

[image: 图像说明文字]


接下面，我们就可以在Sub Main()和End Sub之间书写测试代码了，如下面的代码，我们将在窗口中显示一行文字，如传统而经典的“Hello, World!”。


Module Module1
    Sub Main()
        Console.WriteLine("Hello, World!")
    End Sub
End Module



我们可以通过工具栏中的“启动”按钮或直接使用键盘上的F5键执行此程序，不过，程序窗口会一闪而过；所以，在测试这类命令行程序时，我们使用键盘组合键Ctrl+F5，此时，程序窗口会等待我们按任意键后再关闭，这样，我们就可以有效地观察程序执行的结果了。如下图，就是本程序执行的窗口。

在Visual Basic.NET开发的项目中，命令行项目会从一个名为Main的子程序（Sub）开始，也就是代码中显示的内容；而窗口程序会从一个窗体（Form）开始，这些都可以通过项目的属性进行设置，如我们通过菜单“项目”->“(项目名称)属性”就可以打开这个项目的属性页，其中，我们可以设置命令行程序的启动模块，此例中默认为Module1，如下图。

下面，我们就练习一下在项目中创建文件，比如，我们再创建一个模块文件。通过菜单“项目”->“添加模块”，我们可以打开添加模块文件的对话框，如下图。

[image: 图像说明文字]


新的模块文件名称为“Module2.vb”，我们暂时不需要修改，单击“添加”按钮确定创建；然后，我们可以看到它的内容，如下面的代码。


Module Module2

End Module



然后，我们可以在这个模块中再添加一个名为Main的子程序，如下面的代码（Hello/Module2.vb文件）。


Module Module2
    Sub Main()
        Console.WriteLine("Module2 is working.")
    End Sub
End Module



最后，我们修改项目属性，将“启动对象”设置为“Module2”，如下图所示。

现在，我们再使用Ctrl+F5启动程序就可以看到，程序会从Module2模块中的Main方法启动，如下图所示。

1.2.4. 语句

在代码中，语句是基本的执行单位，在VB中，语句是以行作为基本形式定义的，也就是说，默认情况下，一行代码就是一条语句，如下图就是前面所执行代码的说明。

我们可以看到，在VB中的语句条理非常清楚，而且，代码的结构也都是使用自然语言（英语）定义，所以，也非常容易理解。如果说学习编程还能学习一点英语，那么，VB还真的有一点这个功能！^_^

虽然在VB代码中，我们也可以通过“ _”符号将语句分写在两行，或者是使用冒号（:）将两个语句连接在一行，但出于代码的可读性考虑，我们并不建议这样使用。

不过，有时候由于一行代码过长，我们会在列出代码时使用" _"将代码分行书写，这是一个空格和一个下划线，如下面的代码。


Dim num As Integer  _
= 99



它和以下代码的含义是相同的。


Dim num As Integer = 99



除了以行为基本语句单位，我们还可以看到，在VB的语句中，空格的作用也非常重要，它用于对关键字、标识符、数据等内容进行有效的分隔。

也许你已经发现了，在输入VB.NET代码时，字母的大小写其实并不重要，这也是VB代码的默认方式，我们也可以通过修改修改参数来改变这一规则；实际上，合理地利用这一点是可以减少代码的输入错误的，这是VB给程序员的福利，如果没有特殊要求，我们也不需要去修改VB的默认设置了。

1.2.5. 注释

在代码中添加注释是一项很常见，同时，也是很重要的一项工作，在VB.NET代码中，我们可以使用三种方式添加注释。


	第一种是使用一个英文单引号作为开始，单引号后面的内容，一直到本行结束都作为注释。

	第二种方式是使用REM语句，即REM加一个空格，然后，后面的内容直到本行结束也都作为注释内容。

	第三种是使用三个英文单引号开始，可以通过一系列的标记来格式化注释内容，这种注释最大的功能就是可以通过工具自动生动代码的说明文档。



我们知道，实际上，对于代码功能最好的解释就是代码本身，但有些时候，代码的逻辑可能真的很复杂，这时候，我们使用一个注释来说明代码的目的，会帮助代码的阅读者（包括代码的原创者）很好地理解代码的功能。但在使用注释时，我们也应注意一些问题，如：


	注释应该简洁、明确，只需要说明代码的目的，而不需要对代码进行重复性的说明。

	应注意注释与代码一起更新。

	我们还可以使用空的注释行有效地分隔代码的功能区，使代码的的结构更加清晰。



1.2.6. 命名空间

命名空间（Namespace）是对代码进行组织和管理工作的非常有效的方法。在.NET Framework中，开发资源都是以命名空间的形式进行管理和组织的，在介绍相关内容时，我们会给出它们在.NET Framework中定义的命名空间，以方便大家查阅详细的说明。如.NET Framework中标准数据类型都定义在System命名空间中。

在VB.NET代码文件中，如果我们需要使用哪个命名空间中的开发资源，可以在文件的顶部使用Imports语句导入相应的命名空间，如：


Imports System
Imports System.Data
Imports System.Data.SqlClient



当我们创建自己的代码库时，也可以使用命名空间，此时，我们使用Namespace...End Namespace语句结构来定义命名空间。如果命名空间有多级，我们可以直接在名称中指定或使用嵌套的格式，如下面的两种形式。


Namespace vbchef.webx
    ' 
End Namespace



或：


Namespace vbchef
    Namespace webx
        '
    End Namespace
End Namespace



此外，本书中封装的代码位于vbchef命名空间。

1.3. 小结

本章，我们了解了如何使用用Visual Studio.NET开发环境来编写Visual Basic.NET代码，并进行测试；同时，也了解了VB.NET中代码的一些基本结构。下一章，我们将开始讨论更多的VB.NET开发知识。





2. 基本数据类型

实际上，使用VB.NET进行软件开发的一个事实是，和C#等开发语言一样，其数据类型都是定义在.NET Framework中的标准数据类型的别名。 如果我们打开开发环境的Help Library，可以在.NET Framework类库中的System命名空间中看到这些数据类型。

软件开发中，对于基本类型数据的处理是一项非常基础，也是非常重要的工作，无论我们是进行数据计算，还是进行数据的传递，基本类型的数据都是随处可见的；所以，在本章的一开始，我们就给出在VB.NET中常用的数据类型，然后，我们会介绍它们的运用特点；大家也可以在工作中很快的翻到这里参考使用这些数据类型。

下表就是VB.NET与.NET Framework中定义的基本数据类型的对比及取值范围表。

[image: 图像说明文字]


其中一些数据类型，我们会在后续的内容中详细讨论，如Object类型将在第8章有进一步的讨论、第15章深入讨论String及相关内容、Date类型则会在第16章详细讨论等。

接下来，我们就来讨论基本数据类型的使用，主要包括：


	变量与常量

	整数

	浮点数与Decimal

	算术运算

	布尔类型

	比较运算

	字符与字符串

	类型转换

	更多数据处理



2.1. 变量与常量

在程序中处理数据，我们总不能总是到处写数字吧，学数学还有那么多的符号可以使用呢。代码中，我们同样可以使用符号来表示数据，而且可以分为两种基本类型，即变量和常量。

变量是指在代码运行过程中可以改变其数据的标识符，而常量则是指在代码运行过程中数据不会被修改的标识符。

在子程序（Sub）、函数（Function）或属性（Property）等代码块中，我们使用Dim关键字定义变量，其基本格式如下：


Dim <变量名> As <变量类型>



比如，我们使用Integer类型表示32位的整数，那么，我们定义一个名为num的Integer类型变量，就可以使用如下代码：


Dim num As Integer



定义变量时，我们还可以同时给它们赋值，如：


Dim num As Integer = 99



那么，你一定会想，如果不赋值，num变量的值会是什么呢？

在VB.NET中，如果变量没有被赋值，那么它的值就是其指定类型的默认值，如数值类的默认值是0、布尔类型数据的默认值是False、Object类型默认值为Nothing等等。

如果我们忘记了指定变量类型呢？那么变量的类型就会自动定义为Object，不过，大多数情况下都应该明确指定变量的数据类型。 此外，如果几个变量的类型相同，我们还可以同时定义它们，如：


Dim num1, num2, num3 As Integer



稍后，我们会看到更多数据类型的使用。

定义常量时，我们使用Const关键字，如：


Const PI As Double = 3.1415926



请注意，常量同样需要指定数据类型，必须同时指定其数据，而且在指定数据以后就不能再更改了。

2.2. 整数

整数用于处理没有小数部分的数据，可以分为有符号整数和无符号整数。其中，有符号整数可以处理负整数、零和正整数，而无符号整数则只能处理零和正整数。

在本章的一开始，我们就给出了在VB.NET可用的整数类型，并给出了它们的取值范围，大家可以参考使用。随后的内容，我们更多的以32位整数为例来讨论整数的应用问题，我们使用的类型包括Integer和UInteger，而其它类型整数的处理，可以参考本部分介绍的内容进行操作。

2.3. 浮点数与Decimal

浮点数包括两种，即Single和Double，和Decimal类型数据一样，它们都可以处理带有小数部分的数据。只是它们的取值范围、运算精度上会有所区别，当然，在内存占用和执行的效率上，也会存在一些不同。

2.4. 算术运算

在VB.NET中，可用的算术运算包括：


	加运算，使用+运算符。

	减运算，使用-运算符。

	乘运算，使用*运算符。

	整除运算，使用\运算符，运算结果为长整数（Long）。运算执行前，将运算数尝试转换为Long类型，转换规则为“四舍六入五成双”原则，即小数部分最高位是4时直接舍去、是6时整数部分加1、是5时整数部分变为最接近的偶数。

	除运算，使用/运算符，运算结果将保留相应精度的小数部分。

	取模运算，也就是取余数运算，使用Mod运算符。保留整除后的部分，可能是整数或浮点数，这和C风格的取模运算方法是不同的。

	求幂运算，使用^运算符，如3^2=9。请注意，此运算符的运算结果总是为Double类型。



下面的代码，我们演示了几种运算符的使用。


Module Module2
    Sub Main()
        Dim num1 As Integer = 100
        Dim num2 As Integer = 33
        Console.WriteLine("{0}+{1}={2}", num1, num2, num1 + num2)
        Console.WriteLine("{0}-{1}={2}", num1, num2, num1 - num2)
        Console.WriteLine("{0}*{1}={2}", num1, num2, num1 * num2)
        Console.WriteLine("{0}\{1}={2}", num1, num2, num1 \ num2)
        Console.WriteLine("{0}/{1}={2}", num1, num2, num1 / num2)
        Console.WriteLine("{0} Mod {1}={2}", num1, num2, num1 Mod num2)
    End Sub
End Module



使用Ctrl+F5执行本代码，我们会得到如下图所示的运行结果。

虽然每一种编程语言都定义了运算符的运算优先级，但完全无误地记住它们并不是一件容易的事情，所以，在处理运算顺序时，使用小括号比完全依靠默认的运算符优先级会更安全。

2.5. 布尔类型

布尔（Boolean）类型数据，又称为逻辑类型数据，只包括两个可能的值，即True（真）和False（假），而它的运算方法却不少。如：


	与运算，使用And运算符，如m And n，当m和n都为True，运算结果才为True，否则运算结果为False。

	或运算，使用Or运算符，如m Or n，当m或n有一个为True时，运算结果就为True，m和n都为False时，运算结果才为False。

	取反运算，使用Not运算符，如Not True运算结果为False，Not False运算结果为True。

	异或运算，使用Xor运算符，当两个运算数不同时，运算结果为True，否则运算结果为False。



如下面的代码，演示了这些逻辑运算符的使用。


Console.WriteLine(True And False)
Console.WriteLine(True Or False)
Console.WriteLine(Not True)
Console.WriteLine(True Xor True)



执行此代码，我们可以得到如下图所示的结果。

除上上述的基本逻辑运算，在VB.NET中，我们不可以使用如下一些逻辑运算符，如：


	AndAlso，具有短路功能的与运算符，如m AndAlso n中，当m的值为False时，不再计算n的值，直接返回运算结果False。很多情况下，这种方法可以有效提高代码的执行效率，特别是当计算n的值需要消耗较多时间和系统资源，或者m为必要条件的时候。

	OrElse，具有短路功能的或运算符，如m OrElse n中，当m的值为True时，不再计算n的值，直接返回运算结果True。



2.6. 比较运算

比较数据的大小，相信大家都不会陌生，在VB.NET中，我们可以使用的比较运算符主要有：


	=运算符，如m=n判断m和n是否相等。请注意，在VB中等于和赋值都使用=运算符，大家应该根据代码功能加以区分。

	<>运算符，如m<>n判断m和n是否不等。

	<运算符，如m<n判断m是否小于n。

	>运算符，如m>n判断m是否大于n。

	<=运算符，如m<=n判断m是否小于或等于n。

	>=运算符，如m>=n判断m是否大于或等于n。



比较运算的结果是一个布尔（Boolean）类型的数据，其结果只有两个，即True值表示成立，False值表示不成立。

2.7. 字符与字符串

在VB.NET中，字符和字符串内容都使用一对英文字符的双引号包含，如"A"、"ABC"，那么，我们如何区分它们是字符还是字符串呢？

当我们需要处理一个字符时，应该使用字符（Char）类型，当我们指定一个字符内容时，可以在其后面使用字母c来告诉编译器，如：


Dim ch As Char
ch = "A"c



对于字符串，现在，我们只需要知道使用双引号可以定义一个字符串，而且，可以使用&运算符来连接两个字符串，如下面的代码：


Dim str1 As String = "abcd"
Dim str2 As String = "defg"
Console.WriteLine(str1 & str2)



在第15章，我们会看到更多关于字符串应用的内容。

2.8. 类型转换

在实际开发工作中，我们经常会遇到将一种数据类型转换成另外一种数据类型的工作，在VB.NET中，我们会有很多方法来完成这些转换方法，本节，我们就来讨论其中一些基本的方法。

一般来讲，在代码中进行数据类型的转换可以分为隐式转换和显式转换（也称为强制转换）。

隐式转换。本章开始的时候，我们已经看到了各种类型数据所支持的数值范围，当我们将取值范围小的类型赋值给取值较大的数据类型时，即可以完成隐式的类型转换；反之，则可能出问题，因为取值小的数据类型无法正确保存大于其取值范围的数据。

代码会自动进行隐式的数据类型转换，大多数情况并不会有问题，但如果我们需要确切的数据类型时，就应该手动进行显式的转换。

显式转换。显式的数据类型转换就是指我们在代码中使用一定的转换方法，将一种数据类型明确地转换为另外一种数据类型。

下面，我们就来介绍一些在VB.NET中常用的数据类型转换方式。

2.8.1. VB类型转换函数

使用过VB6的朋友，一定都知道一系列由大写字母C（Convert）开头的转换函数，在VB.NET中，我们依然可以使用这些函数，如：


	CByte()函数，将参数转换为Byte类型数据。

	CShort()函数，将参数转换为Short类型数据。

	CUShort()函数，将参数转换为UShort类型数据。

	CInt()函数，将参数转换为Integer类型数据。

	CUInt()函数，将参数转换为UIntger类型数据。

	CLng()函数，将参数转换为Long类型数据。

	CULng()函数，将参数转换为ULong类型数据。

	CSng()函数，将参数转换为Single类型数据。

	CDbl()函数，将参数转换为Double类型数据。

	CDec()函数，将参数转换为Decimal类型数据。在VB.NET中已不再使用Currency，所以不再提供CCur()函数。

	CBool()函数，将参数转换为Boolean类型的数据。

	CDate()函数，将参数转换为Date类型的数据，和DateTime结构类型相同。

	CStr()函数，将参数转换为String类型的数据。

	CObj()函数，将参数转换为Object类型，这个函数实际意义并不大，因为在VB.NET中，任何类型的数据都可以直接赋值为Object类型的变量。



请注意，当参数不能正确转换为目标类型时，会产生异常（Exception），在代码中，我们必须正确处理这些异常才能保存代码的运行。不过，我们会从整个项目中数据处理工作的全局来考虑类型转换和异常处理的问题，在本书后续的内容中，我们会逐渐了解相关方法。

2.8.2. Convert类

Convert类定义在System命名空间，它的功能与VB类型转换函数相同，即将指定的数据转换成另外一种数据类型。与一系列VB类型转换函数不同的只是对于类中方法的命名。

在VB类型转换函数的名称中，都使用一定的缩写形式，而在Convert类中定义的转换方法，统一使用了“To+<.NET Framework数据类型名称>”格式，如我们需要将一个数据转换为32位整数时，使用VB类型转换函数可以写成：


Dim str As String = "123"
Dim num As Integer = CInt(str)



而使用Convert类，我们则需要写成如下代码形式：


Dim str As String = "123"
Dim num As Integer = Convert.ToInt32(str)



请注意，使用Convert类中的方法转换数据类型时，如果参数不能正确转换成目标类型，同样会产生异常，这和VB类型转换函数的表现是相同的。

关于更多的.NET Framework数据类型名称，可以参考本章开始时的对照表。

2.8.3. TryParse()方法

TryParse()方法定义在基本数据类型中，它的功能是尝试将字符串类型转换为指定的数据类型。与上述两种数据类型转换方式不同，TryParse()方法不会产生异常，如下面的代码。


Dim str As String = "123"
Dim result As Integer
If Integer.TryParse(str, result) Then
    Console.WriteLine("转换成功")
Else
    Console.WriteLine("转换失败")
End If



虽然TryParse()方法不会产生异常，但它也有一些不足，比如，它的源数据类型必须是字符串（String），它的目标类型只能是值类型（如基本数据类型中除String和Object以外的数据类型）。

2.9. 更多数据处理

在一些行业软件中，我们可能需要很专业的运算方法，比如，在System命名空间中的Math类为我们定义了常用的数学计算方法及数据。 此外，在Microsoft.VisualBasic命名空间中的Financial模块中定义了一些财务计算函数，我们可以在VB.NET代码中直接使用这些函数。

相信这些数据处理的资源在专业软件中会大有用处，大家可以参考MSDN Library了解详细的使用方法。

更多类型的数据处理，如字符串、枚举、结构、类、数组等内容，我们会在后续内容中逐渐展开讨论。





3. 程序流程控制

通过流程控制语句，让我们对代码的执行有着更多、更精确的控制，可以根据各种条件执行不同的程序逻辑。在VB.NET中，我们使用的流程控制语句主要包括条件语句结构、选择语句结构和循环语句结构。

本章，我们就来讨论相关内容，包括：


	条件语句结构

	选择语句结构

	循环语句结构

	Exit语句

	Continue语句

	Goto语句与标签



3.1. 条件语句结构

在VB.NET中，条件语句结构的基本格式如下：


If <条件> Then
    <语句块>
End If



当<条件>成立（True值）时，执行<语句块>的代码，然后执行End If语句后面的内容；如果<条件>不成立，则直接执行End If后面的代码。对于这种格式的If...End If语句，如果<语句块>只有一条语句，我们还可以简写为一行，如下面的格式：


If <条件> Then <语句>



如下面的代码会判断一个整数是否为偶数。


Dim num As Integer = 10
If num Mod 2 = 0 Then
    Console.WriteLine("{0}是偶数", num)
End If



或者是：


Dim num As Integer = 10
If num Mod 2 = 0 Then Console.WriteLine("{0}是偶数", num)



一般情况下，我们应该处理条件成立或不成立的两种情况，此时，我们应该在If...End If语句结构中使用Else语句块，如下面的代码。


Dim num As Integer = 10
If num Mod 2 = 0 Then
    Console.WriteLine("{0}是偶数", num)
Else
    Console.WriteLine("{0}不是偶数", num)
End If



当然，在实际应用中，我们还可以使用更复杂的条件结构，如：


If <条件1> Then
    <语句块1>
ElseIf <条件2> Then
    <语句块2>
ElseIf <条件n> Then
    <语句块n>
Else
    <语句块n+1>
End If



其中，我们设定1到n个条件，当条件成立时就执行相应的语句块，如果条件都不成立就执行<语句块n+1>。

3.2. 选择语句结构

在VB.NET中的选择语句结构，我们使用Select Case语句，它的基本使用格式如下：


Select Case <表达式>
    Case <值1>
        <语句块1>
    Case <值2>
        <语句块2>
    Case <值n>
        <语句块n>
    Case Else
        <语句块n+1>
End Select



在选择语句结构中，确定代码执行的条件只有一个，即<表达式>的值；然后，每一个Case语句都对应指定值，如果<表达式>的值与其对应，则执行相应的代码块。如果没有相对应的值，则执行<语句块n+1>；虽然Case Else语句是可选的，但能够处理任何情况总是比较安全的。

如下面的代码，我们将根据不同的颜色英文名称显示其中文名称。


Dim colorEn As String = "red"
Select Case colorEn
    Case "red"
        Console.WriteLine("红色")
    Case "green"
        Console.WriteLine("绿色")
    Case "blue"
        Console.WriteLine("蓝色")
    Case Else
        Console.WriteLine("未知颜色")
End Select



我们使用Select Case语句结构时，如果你学习过C风格的switch语句结构，它们之间会有一些不同，如：

在Select Case语句中，不存在自动向下贯穿，也就是说每个Case语句执行完成后就会自动结束整个Select Case语句结构，不需要C语言中的break等语句来中断。

Case语句后面的值可以是一个，也可以有多个，并不像C语言中每个case语句只能指定一个值。如下面的代码，将根据年份和月份给出月份中的总天数。


Dim iYear As Integer = 2016
Dim iMonth As Integer = 2
Dim iDay As Integer = 0
Select Case iMonth
    Case 1, 3, 5, 7, 8, 10, 12
        iDay = 31
    Case 4, 6, 9, 11
        iDay = 30
    Case 2
        iDay = IIf(Date.IsLeapYear(iYear), 29, 28)
End Select
Console.WriteLine("{0}年{1}月共有{2}天", iYear, iMonth, iDay)



实际上，以上代码的功能在Date中的DaysInMonth()方法就可以实现，我们在这里只是为了演示Case语句同时指定多个值的情况。关于日期和时间的更多处理，会在第16章有更多讨论。

此外，当月份为2时，我们使用了IIf()函数，它的格式是：


IIf(<条件>, <值1>, <值2>)



当<条件>成立时，返回<值1>，否则返回<值2>。如果你在C风格的编程语言中使用过?:运算符，它们的功能是相同的。此外，这也可以看作是一个简单的If...Else...End If语句结构。

3.3. 循环语句结构

循环语句结构的功能就是在满足一定条件的情况下，可以多次执行相同或相似的代码，在VB.NET中，常用的循环语句结构包括：


	For...Next

	Do...Loop

	While...End While

	For Each...Next



接下来，我们就详细讨论它们的使用。

3.3.1. For...Next

使用For...Next语句结构，最基本的格式如下：


For <循环控制变量> = <初始值> To <终止值> Step <步长>
    <代码块>
Next



如下面的代码，我们将计算1到100整数的和。


Dim iSum As Integer = 0
For i = 1 To 100 Step 1
    iSum += i
Next
Console.WriteLine("计算结果是{0}", iSum)



在这个For...Next语句结构中，我们使用了循环控制变量i，它的初始值是1，最大值是100，步长（Step）是1；也就是说，i的值从1（最小值）开始，每一次循环后会自动加1（步长），当i的值大于100（最大值）时，执行循环语句后就终止整个For...Next语句结构。

请注意，当步长（Step）为1时，我们可以省略Step关键字及步长值，如下面的代码：


For i = 1 To 100
    iSum += i
Next



如果我们需要计算1到100中偶数的和，可以使用如下代码。


For i = 2 To 100 Step 2
    iSum += i
Next



在For...Next语句结构中，我们还可以指定循环控制变量的类型，如：


For i As Integer = 2 To 100 Step 2
    iSum += i
Next



实际应用中，循环控制变量可以使用整型、浮点型或Decimal类型。

此外，在循环语句中，我们使用了“iSum += i”语句，了解C风格编程语言的一定很熟悉它的功能了，它的功能和“iSum = iSum + i”语句是相同的。对于VB.NET中其它的一些运算符，我们也可以使用这样的组合，如/=、*=、-=等。

3.3.2. Do...Loop

Do...Loop的功能很强，使用起来也很灵活，比如，我们可以将循环的控制条件放在Do关键字后，也可以放在Loop关键字后，它们的区分在于对循环条件判断的时机不同。

我们先看下面的应用格式：


Do While <条件>
    <语句块>
Loop



当<条件>成立时，会执行<语句块>的内容。但我们应该注意，在<语句块>中，应该有改变<条件>的语句，否则，循环结构就无法结束，从而导致死循环，最终会导致应用的崩溃。

如下面的代码，我们使用Do...Loop语句结构来完成1到100的累加计算。


Dim iSum As Integer = 0
Dim i As Integer = 1
Do While i <= 100
    iSum += i
    i += 1
Loop
Console.WriteLine("计算结果是{0}", iSum)



请注意，在循环执行的代码中，我们使用了“i += 1”语句，在每次执行循环后，i的值都会加1。实际上，在本例中，我们同样是使用变量i作为循环的控制条件，只是需要我们的代码来改变它的值，而不是像For...Next语句结构中那样自动改变它的值。

使用Do...Loop语句结构的另一种格式是将<条件>放在Loop语句后面，如下面的格式。


Do
    <语句块>
Loop While <条件>



与前面一种使用方式的不同在于，当我们将循环条件放在Do语句后时，会在每一次执行循环代码之前检查条件是否成立；而将循环条件放在Loop语句后时，会在每一次执行循环代码之后检查条件是否成立，也就是说，当条件在Loop语句后时，循环结构中的<语句块>最少会执行一次。

我们应注意，在没有对条件进行检查之前就执行循环代码并不是一个好的方法，因为条件有可能在一开始就不成立，此时执行一次循环语句都是不正确的；所以，我们还是应该更多地使用将条件放在Do语句后面的循环结构，即在循环执行之前进行条件的判断。

在Do...Loop循环结构中，除了使用While语句指定条件，我们还可以使用Until语句来指定条件。从字面上，我们也可以看出它们的区别：


	While语句指定条件时，当条件成立时才会执行循环结构中的代码。

	Until语句指定条件时，指定的条件成立时就会退出循环结构的执行。



此外，Until语句指定条件，同样可以放在Do语句或Loop语句后。如下面的语句，同样是计算1到100的和。


Do Until i > 100
    iSum += i
    i += 1
Loop



3.3.3. While...End While

While...End While语句结构的使用格式如下：


While <条件>
    <语句块>
End While



实际上，我们可以看到，While...End While语句结构就是Do While...Loop语句结构的简化形式，如下面的代码，我们使用While...End While语句重写1到100累加计算。


Dim iSum As Integer = 0
Dim i As Integer = 1
While i <= 100
    iSum += i
    i += 1
End While
Console.WriteLine("计算结果是{0}", iSum)



3.3.4. For Each...Next

For Each...Next语句结构可以帮助我们快速的访问集合（Collection）中的所有成员，在第14章，我们还会有详细的讨论，现在，我们先来看一个简单示例。


Dim arr As New ArrayList
arr.Add("abc")
arr.Add("def")
arr.Add("ghi")
For Each item As String In arr
    Console.WriteLine(item)
Next



我们首先创建了一个数组列表对象arr，然后添加了三个字符串成员；最后，我们使用For Each...Next语句逐一访问这些成员，并输出到控制台。从这个示例中，我们可以了解For Each...Next语句结构的基本应用格式。


For Each <成员变量及类型> In <集合对象>
    <代码块>
Next



<成员变量及类型>中，我们定义了表示集合中一个成员的变量，并可以指定其类型（也可以不指定，默认为Object类型）；不过，当我们指定了成员变量的类型以后中，在循环结构内就可以更精确地操作它们。

在第14章，学习了更多的集合类型以后，我们会看到集合类资源在开发中的功能和作用都是非常强大的。

3.4. Exit语句

在循环语句结构中，我们可以在满足某些特定条件时就立即终止循环的执行，此时，我们使用Exit语句，它有几个变形：


	Exit For语句，用于退出For...Next或For Each...Next循环结构。

	Exit Do语句，用于退出Do...Loop循环结构。

	Exit While语句，用于退出While...End While循环结构。



比如下面的代码，其功能是，当item的内容等于"ghi"时就退出循环，不再显示内容。


For Each item As String In arr
    If item = "ghi" Then Exit For
    Console.WriteLine(item)
Next



Exit语句除了可以使用在循环语句中，还可以使用在其它语句结构中，如选择语句结构（Select Case），或者是我们在后续内容将要学习到的子程序（Sub）、函数（Function）、属性（Property）或Try语句结构。

3.5. Continue语句

Continue语句用于循环语句，它的功能是中断当前循环，并开始下一次的循环（当满足条件时）。如下面的代码，我们将利用Continue语句来完成1到100偶数的累加计算。


Dim iSum As Integer = 0
For i = 1 To 100
    If i Mod 2 <> 0 Then Continue For
    iSum += i
Next
Console.WriteLine("计算结果{0}", iSum)



我们可以看到，Continue语句同样使用了变形，即Continue For语句，它用于中断并执行下一个For循环。相应的Continue变形还有：


	Continue Do语句，用于中断并执行下一个Do循环。

	Continue While语句，用于中断并执行下一个While循环。



3.6. Goto语句与标签

首先，我们必须声明，非万不得已，不要使用Goto语句，因为它会造成代码结构的混乱，并极可能产生不易查找的错误。

在VB.NET代码中，我们可以在需要的位置定义标签，其格式如下：


<标签名称>:



然后，我们可以使用Goto语句跳转到指定的标签位置，如：


Goto <标签名称>



Goto语句和标签的是早期BASIC中的产物，在VB.NET保留它们是为了兼容早期的VB代码（也许只是情怀在作怪^^）。下面的代码，我们将使用标签和Goto语句模拟循环操作，以此向BASIC致敬！


Dim iSum As Integer = 0
Dim iCounter As Integer = 1
Tag_Add:
iSum = iSum + iCounter
iCounter = iCounter + 1
If iCounter <= 100 Then Goto Tag_Add
Console.WriteLine(iSum)



代码中，我们定了一个标签Tag_Add，然后进行累加操作，当计数变量iCounter小于等于100时，就跳转回Tag_Add标签，然后进行累加和计数变量加1的操作；当iCounter大于100时，会显示计算结果，即1到100累加的结果。

到这里，相信大家已经可以编写基本的代码了，下一章，我们将讨论如何在项目中组织和封装自己的代码。





4.代码组织与封装

通过前面的内容，我们已经可以处理基本的数据类型，以及使用一些语句结构来控制代码的执行，但我们知道，要想开发出功能强大的应用软件，我们还有很多工作要做。

本章，我们将讨论对于自定义代码的封装和组织方式，主要内容包括：


	模块

	函数与子程序

	参数

	封装自已的转换函数



4.1. 模块

模块（Module）应该是VB.NET的一个特点了，对于一些非对象类代码，我们完全可以将它们定义在模块中，这样，只要引用了模块所在的命名空间，我们就可以使用模块中的内容，而不需要像面向对象代码那样，必须通过类或其实例才能使用其中的代码。

前面的示例中，我们也可以看到，实际上，VB.NET的程序就从一个指定模块中的Main()子程序中启动的；而在开发中，我们也可以定义自己的模块，下面，我们通过菜单“项目”->“添加模块”打开添加新项目的对话框，我们将模块的名称重新命名为“MDemo.vb”，然后，我们就可以在资源管理器中看到新的模块文件了，如下图。

[image: 图像说明文字]


点击MDemo.vb，我们可以看到其中的内容，如下面的代码。


Module MDemo

End Module



接下来，我们就可以在这里添加自己的代码了，比如函数和子程序。

4.2. 函数与子程序

函数（Function）和子程序（Sub）是VB中最基本的代码封装单位，我们可以将特定功能的代码定义在函数或子程序中，然后可以在项目的其它地方调用。那么，它们有什么区别呢？其实很简单，我们先看函数的定义。


Function <函数名> (<参数>) As <返回值类型>
' 函数代码
End Function



我们可以看到，函数定义在Function和End Function之间。在Function语句中，我们需要指定函数的名称、函数返回值的类型，以及所需要的参数，当然，参数并不是必须的。如下面的代码，我们定义一个Adder()函数，它的功能是计算参数一到参数二的累加结果。


Module MDemo
    ''计算累加
    Function Adder(iMin As Integer, iMax As Integer) As Integer
        Dim iSum As Integer = 0
        For i = iMin To iMax
            iSum += i
        Next
        Return iSum
    End Function
End Module



我们可以看这个函数的一些元素：


	函数名是Adder。

	参数定义在参数名后面的一对小括号()中，我们共定义了两个参数（iMin和iMax），它们都是Integer类型的。请注意，如果没有参数，也同样需要这对小括号。

	这个函数的返回值类型是Integer，使用As关键字指定。

	函数的返回值。在函数内部，我们需要Return语句返回函数执行的结果，在VB6中它是通过赋值给函数的名称来完成的。此外，执行Return语句后，会立即结束函数，也就是说不会执行Return语句后面的代码，请注意灵活应用这一特性。



接下来，我们就可以在Main()子程序中调用这个函数了，如下面的代码。


Sub Main()
    Dim iSum As Integer
    iSum = Adder(1, 100)
    Console.WriteLine("1到100累加结果是{0}", iSum)
End Sub



通过对代码的封装，我们可以简化很多工作，特别是在开发中需要多次使用的代码，这可以极大地提高项目开发及维护等工作的效率。

再来看一看子程序，它到底和函数有什么不同呢？实际上，子程序与函数的不同只在于：（1）子程序没有返回值；（2）没了。^_^

定义子程序时，我们使用Sub关键字，它的基本结构如下：


Sub <子程序名> (<参数>)
' 子程序代码
End Sub



如下面的代码，我们定义了一个SayHello()子程序。


Sub SayHello (sName As String)
    Console.WriteLine("Hello, {0}.", sName)
End Sub



在这个子程序中，我们定义了一个参数sName，它的类型是字符串（String），在子程序中，我们在控制台输出一个文本，在Main()子程序中，我们可以使用如下代码来调用这个子程序。


SayHello ("Tom")



接下来，既然大家已经知道了子程序和函数的共同点和那么一点点的区别，我们也就可以省点事，从现在开始，我们将子程序和函数统称为“方法”，这是一个面向对象编程的概念，稍后我们就会看到，在类中定义的方法就是子程序或函数。

4.3. 参数

本节，我们将单独讨论方法中参数的定义和使用。

首先，我们需要了解，在方法中的参数，可以没有，也可以有一个或多个；每一个参数定义都应包含参数名称和参数类型，多个参数应使用逗号（,）分隔；除了这些，我们再来看一看非常重要的几个关于参数的概念，如按值或按引用传递、可选参数和参数数组。

4.3.1. 按值或按引用传递

按值传递时，会传递变量的副本，在方法中，当我们修改这个数据时，实际上是在修改副本的数据，而原数据不会改变。我们如下面的示例，首先，我们定义一个方法（Hello/MDemo.vb），其功能是对传入的数据进行加一运算。


Sub AddOne(ByVal iNum As Integer)
    iNum += 1
End Sub



接下来，我们在Main()方法中调用它，如下面的代码。


Dim iNum As Integer = 99
AddOne(iNum)
Console.WriteLine(iNum)



从代码上看，我们是想对iNum进行加1操作，但是，如果大家执行代码就会发现，iNum并没有被加1，结果显示依然是99。

其原因就是，我们定义的AddOne()方法，其参数定义为按值传递（ByVal），这样，在方法中加1的操作并不能反映到方法外定义的iNum变量上，因为在方法内只是在对iNum的复本在进行操作。

与按值传递相对应的是按引用（ByRef）传递，当我们将AddOne()方法的参数修改为按引用传递时，可以再次执行代码，就会发现，变量iNum的值才真正进行了加1的操作。

那么，我们在定义参数的传递方式时，应该如何选择呢？我们还是应该根据按值或按引用传递的特点，以及代码的需求、性能等方面综合考虑。

首先，我们看按值传递参数。按值传递会完全复制变量的数据，所以，可以有效保护原数据的安全；但是，如果原数据很大（比如一个包含了很多数据的对象），那么，数据的复制操作就会影响整个程序的性能，而我们必须要考虑，是需要高性能，还是需要对数据的绝对保护。

再看按引用传递参数。按引用传递参数，传递的效率非常高，因为是直接将数据内存区域的引用进行传递，并不对数据进行复制；也正是因为这样，在方法内，我们可以对原数据直接进行修改，这样，就有可能意外地影响原数据的内容。

所以，我们应该从代码的目的和功能、执行效率，以及安全性等方面综合考虑方法的参数是按值传递，或者是按引用传递。

4.3.2. 可选参数

当我们设置方法的参数时，有些参数可以设置为一个合适的值，这样，在调用方法时，我们就可以不设置此参数的值。在方法中设置可选参数时应注意：


	使用Optional关键字指定可选参数。

	可选参数必须在必选参数的后面。



如下面的代码，我们在AddPrefix()方法的第二个参数设置了可选参数。


Function AddPrefix(str1 As String, Optional sPrefix As String = "")
    Return sPrefix & str1
End Function



接下来，我们可以通过下面的代码来测试运行结果。


Console.WriteLine(AddPrefix("1","No."))
Console.WriteLine(AddPrefix("2"))



4.3.3. 参数数组

参数数组可以帮助我们动态地定义参数的数量，但有一些限制，即只能定义在参数列表的最后。定义一个参数数组时，我们使用ParamArray关键字，如下面的代码，我们定义了一个方法PrintStrings()，它的功能是在控制台中显示参数所带入的内容。


Sub PrintStrings(ParamArray msg() As String)
    For Each item As String In msg
        Console.WriteLine(item)
    Next
End Sub



接下来，我们可以测试以下几种调用参数数组的方式。


PrintStrings()
PrintStrings("abc")
PrintStrings("abc", "def", "ghi")



我们可以看到，参数数组可以支持带入零个或多个参数，这对某些功能，特别是在开发阶段，不能确定参数个数的情况下，使用参数数组是一个比较常用，而且有效的选择。

4.4. 封装自已的转换函数

关于子程序和函数的定义，实际并不复杂，接下来，我们就来点实际的操作。在第2章的学习中，我们已经讨论了关于数据类型的转换问题，对于VB.NET中的数据类型转换，我们可以分为两个主要类型：


	VB内置转换函数和Convert类方法。它们的表现是一致的，即当参数可以转换为目标类型时，就返回转换后的数据；如果参数不能转换成目标类型，则会产生异常。

	ToParse()方法，只能将字符串形式的数据转换为特定的值数据类型，虽然不会产生异常，但支持的数据类型范围有限，而且使用起来的代码也不够简洁。



综合以上两种情况，我们将封装自己的数据类型转换函数，其指导原则是：


	不会产生异常，也就是说，无论如何转换都会有一个结果；我们约定，如果不能正确转换为目标类型，则返回目标类型的默认值。

	使用VB风格的数据类型缩写，但转换函数名都以“To”开头。



此外，对于本书的封装代码，我们会创建在一个名为“VBChef”的项目中，当VB代码是原料时，我们就是制作出各种软件的厨师（Chef）。^^

下面，我们就来看看将Object类型转换为String类型的函数定义（VBChef/MCommon.vb文件）。


''转换为String类型
Function ToStr(ByRef obj As Object) As String
    If obj Is Nothing Then
        Return ""
    Else
        Return obj.ToString()
    End If
End Function



对于代码中的一些内容，你也许还不太明白，不过没关系，这个函数的功能就是；当参数obj是空对象（Nothing）时，就返回空字符串；否则将对象的ToString()方法定义返回其字符串形式。

关于Object类型和ToString()方法，我们会在面向对象编程相关内容中继续讨论，大家不必着急。

对于返回数值类型的转换方法，我们以Integer类型为例，如下面的函数，就是将Object类型数据转换为Integer类型的函数。


''转换为Integer类型
Function ToInt(ByRef obj As Object) As Integer
    If obj Is Nothing Then Return 0
    Dim result As Integer = 0
    If Integer.TryParse(obj.ToString(), result) Then
        Return result
    Else
        Return 0
    End If
End Function



代码中，当obj为空时，就返回0值；然后，我们使用obj的字符串形式（obj.ToString()）来进行转换，转换工作使用了Integer.TryParse()方法，转换成功即返回转换结果，否则返回0值。

对于以上两个转换函数，我们可以看到，无论是否转换成功，我们都可以得到一个目标类型的数据，在某些场合下，这比处理转换失败时的异常情况要简单地多，而且在大多情况下，这样做也是比较合理的。

不过，如果在程序中要严格区分空值和有效数据时，比如数据库中0值和空值（NULL）就是严格区分的两个不同概念。此时，我们使用另外一个版本的转换函数，即返回结果为可空类型；如下面的代码，我们还以Integer类型为例。


''转换为Integer类型或Nothing
Function ToIntNullable(ByRef obj As Object) As Integer?
    If obj Is Nothing Then Return Nothing
    Dim result As Integer = 0
    If Integer.TryParse(obj.ToString(), result) Then
        Return result
    Else
        Return Nothing
    End If
End Function



接下来，我们可以在Main()方法中测试这个函数。如下面的代码。


Imports VBChef.vbchef

Module Module1
    Sub Main()
        Dim str1 As String = "123"
        Dim result As Integer? = ToIntNullable(str1)
        If result Is Nothing Then
            Console.WriteLine("转换失败")
        Else
            Console.WriteLine(result)
        End If
    End Sub
End Module



请注意，由于我们将封装的代码定义了vbchef命名空间，所以，在使用之前必须使用Imports语句引用它。

关于可空类型，它是由不可空类型添加问号（?）生成的一种类型，可以保存原类型的数据，也可以是空值（Nothing）。如代码中使用的Integer?类型就是可空Integer类型，它可以保存32位整数，也可以是Nothing，在使用这种类型的变量时，必须首先判断它是否为空值，然后才能正确使用。

关于空值，我们必须要区分的是Nothing表示空引用（空对象），而数据库中空值（NULL）的概念，在.NET Framework中定义为DBNull.Value值。





5.枚举

我们先看一个例子，当我们需要保存一个性别数据时，会考虑使用什么数据类型呢？较早的编程语言中，一般会使用整数，如1表示男、2表示女，但这种数据处理方法必须在严格的约定下才会有效；再进一步，大家会使用常量，如SEX_MALE、SEX_FAMALE等，但使用常量表示特定值也有问题，即我们保存数据的变量依然是整数类型，这样，当我们不使用常量或赋值了有效范围以外的整数时，代码并不会产生任何异常，但数据的含义已经不正确了。

枚举的功能就是，通过直观的标识来表示特定的数据类型及其成员；比如，我们可以通过枚举定义性别类型，如下面的代码。


Enum ESex As Integer
    Unknow = 0
    Male = 1
    Female = 2
End Enum



代码中，我们使用Enum关键字定义了ESex枚举类型，并指定它的成员类型为Integer；然后，我们定义了三个枚举成员，即Unknow、Male和Female，并分别指定了这些成员所对应的数值。实际上，以下代码定义的枚举类型和上面的代码功能是一样的。


Enum ESex
    Unknow
    Male
    Female
End Enum



这是什么情况呢？在定义枚举类型时，其成员都会有一个对应的数值，默认情况下，成员相对应的数值会是Integer类型，并且是从0开始逐渐递增的。

此外，枚举成员的值也可以不从0开始，只要我们设置了第一个成员的值就可以指定开始的数值了；另一方面，在同一个枚举中，我们完全可以定义不同标识，但数值相同的成员，如：


Enum ESex As Integer
    Unknow = 0
    Male = 1
    Man = 1
    Boy = 1
    Mr = 1
    Female = 2
    Woman = 2
    Girl = 2
    Mrs = 2
End Enum



接下来，在代码中，我们如何使用枚举类型呢？实际上，这和使用基本数据类型的方式是一样的，如下面的代码。


Dim tomSex As ESex
tomSex = ESex.Male
Console.WriteLine(tomSex)



代码会显示1，即枚举成员Male对应的数值。

那么，如果我们直接给tomSex设置数值会怎么样呢？答案是，一些正常，即使是超出了ESex枚举类型指定的数值范围也不会出现异常。

但是，我们相信没有人会去自己跟自己过不去，既然我们定义了枚举类型，应该严格使用枚举成员来进行数据的操作，而不是图简单而直接使用数值。

使用枚举另一个需要注意的问题是，为特殊情况做好准备。如我们定义的ESex枚举类型中，定义了一个Unknow成员，其值定义为0；这是一个习惯用法，以保存枚举值可以处理特殊情况，比如，人们想对自己的性别保密的时候。

最后，需要说明的是，枚举属于值类型，所以，在赋值或参数传递等操作时，默认会进行数据的完全复制操作。





6.结构

结构是对基本数据类型的更进一步扩展，在VB.NET的结构中，我们不但可以将多个类型的数据组织成一个整体，而且还可以定义它们的相关操作（方法）；在结构中，我们可以处理的常用成员类型包括：变量、属性和方法。

接下来，我们就首先讨论结构的成员。

6.1. 结构成员

在结构中的变量，和我们在子程序或函数中定义的变量基本是一样的，不过，除了可以使用Dim关键字，我们还可以使用Public或Private关键字，它们的区别在于，在结构类型中，使用Dim或Public关键字定义的变量可以通过结构类型的变量来访问，而Private关键字定义的变量，只能在结构的内部分使用。

由于Public和Private关键字更清晰的说明了成员的使用范围，所以，我们更建议使用这两个关键字来定义结构成员的访问级别（作用域）。但我们应该了解的，在结构中的成员，其默认访问级别是公共的（Public）。

下面的代码就是我们定义的SDemo结构类型。


Structure SDemo
    Private iAge As Integer
    Public Name As String
End Structure



我们可以看到，在VB.NET中定义结构使用了Structure关键字，在SDemo结构类型中，我们定义了两个变量，即iAge和Name，它们分别被定义为私有的（Private）和公共的（Public）。接下来，我们在代码中使用这个结构类型。


Dim tom As SDemo
tom.Name = "Tom"
Console.WriteLine(tom.Name)



我们接下来看一看结构类型中属性（Property）的定义，如下面的代码。


Structure SDemo
    Private iAge As Integer
    Public Name As String
    '' Age属性
    Public Property Age As Integer
        Get
            Return iAge
        End Get
        Set(value As Integer)
            If value >= 0 And value <= 150 Then
                iAge = value
            End If
        End Set
    End Property
End Structure



代码中，我们使用Property...End Property定义了一个属性结构，并使用As关键字指定属性值的类型。在这个Age属性中，我们定义了两个部分：


	Get...End Get结构，定义了属性的读取部分，这和函数的定义相似，我们需要Return语句返回属性的值，本例中，我们直接返回了iAge变量的值。

	Set...End Set结构，定义了属性的设置部分，我们使用value参数（这也是习惯用法）带入属性值；本例中，当属性值在0到150之间时，会赋值给iAge变量，否则就什么也不做。



在使用属性时，我们会发现，它几乎与变量的使用方式是一样的，如下面的代码。


Dim tom As SDemo
tom.Age = 35
Console.WriteLine(tom.Age)



那么，我们使用属性的意义是什么呢？它的意义就在于，在获取数据或设置数据时，我们有机会做出更多的操作，而不像变量那么只能直接获取或设置其数据。比如，在Age属性的设置操作（Set...End Set）中，我们就对设置的数据进行了有效性判断，当数据为有效值时才真正地赋值给相应的变量，否则就什么也不做（或者给用户一些错误信息）。

在了解了结构类型中的变量和变量的使用以后，我们再来看看另一个重要成员的使用，即方法。

我们说过，方法是子程序和函数的总称，在结构中依然是这样，如下面的代码，我们在SDemo结构中定义了一个SayHello()方法。


Sub SayHello()
    Console.WriteLine("Hello, {0}", Me.Name)
End Sub



然后，我们可以通过以下代码调用这个方法。


Dim tom As SDemo
tom.Name = "Tom"
tom.SayHello()



6.2. 结构初始化

在VB.NET中的结构中，我们还可以定义特殊的方法来对结构中的数据进行初始化，这个特殊的方法名称就是New()，它的功能就是在创建结构变量时，可以同时对数据进行初始化操作。

如下面的代码，就是我们在SDemo结构中定义的New()方法。


Public Sub New(sName As String, Optional iAge As Integer = 0)
    Me.Name = sName
    Me.iAge = iAge
End Sub



请注意，在结构的New()方法中，我们可使用可选参数，但必须至少有一个非可选参数。

接下来，我们可以使用如下的代码来使用New()方法。


Dim tom As New SDemo("Tom", 35)
Dim jerry As New SDemo("Jerry")
Console.WriteLine(tom.Age)
Console.WriteLine(jerry.Age)



当我们使用New关键字定义结构时，就可以在结构名称后使用一对小括号来包含相应的参数，而此时，这些参数对应的就是我们在结果中定义的New()方法中的参数。

代码中，我们定义了两个SDemo结构类型的变量，并分别使用两个和一个参数来初始化它们。大家可以通过执行代码来观察运行的结果，我们可以看到，tom的年龄设置为35，而jerry的年龄则默认设置为0。

最后需要说明的是，结构依然是值类型，如果我们需要更多、更复杂的数据操作，接下来的面向对象编程（OOP）将会有所帮助。





7. 面向对象编程

使用VB.NET开发的另一个小秘密是，这实际上是一个完全的面向对象世界，就算我们使用的最基本的数据类型都是一个对象，比如，我们可以在程序中写如下的代码。


Console.WriteLine(1.ToString())



代码中，我们直接在数字1上使用了ToString()方法，在很多传统编程语言中，这是一个非法的调用；但在VB.NET中，这是完全合法的，因为在这里，数字1就是一个数字字面量对象，它的ToString()方法可以将数字转换为字符串类型。

那么，什么是面向对象编程呢？在VB.NET中，我们如何实现面向对象编程呢？这些就是本章的主题，相关内容包括：


	类与对象

	构造函数与析构函数

	值类型与引用类型

	类的成员

	变量

	属性与默认属性

	方法

	运算符重载

	分部类



7.1. 类与对象

面向对象编程是相对于传统的过程式编程，在没有面向对象编程概念之前，我会使用过程式的编程方法。比如，在游戏中，我们需要对一辆坦克进行操作，就会定义一系列的操作方法，如tank_set_speed()、tank_move()、tank_fire()等等；这样编写的代码最大的不足就是缺少有机的联系，代码维护和使用都很不方便。

而面向对象编程会把操作的主体当成一个对象，如坦克对象tank，我们就可以通过tank.speed属性获取或设置它的速度，然后使用tank.move()和tank.fire()等方法来执行相应的操作。

就我们给出的例子来讲，一定是使用面向对象编程的方法更直观，在编码和维护工作中也更高效。

但我们也应该明白，任何一种技术都不万能的，对于某些情况，面向对象编程反而增加了代码的复杂度。比如，我们只需要将一个对象obj转换成Integer类型的数据时，使用.NET Framework中的Convert类的代码是Convert.ToInt32(obj)；而使用VB内置转换函数的代码则是CInt(obj)，哪个更简单，相信不用多说。

不过，在对于复杂的数据类型的封装、计算等操作，面向对象编程依然是我们手中强大的开发工具。

实际应用时，我们可以将类当做是一个复杂的数据类型，此类型的变量就称为类的实例，也就是我们常说的对象，创建对象的过程就称为类的实例化。

在VB.NET中创建类时，我们会使用Class关键字，如下面的代码，我们就定义了一个名为CDataItem的类（VBChef/CDataItem.vb）。


Public Class CDataItem
' 类定义代码
End Class



下面的代码，我们创建一个名为item的对象，它的类型就是CDataItem类。


Dim item As CDataItem



请注意，代码中的item对象只是被定义为CDataItem类型，而并不能真正的使用，因为它还没有被实例化，此时，它的值是Nothing，即空对象（空引用）。

如果我要正确地使用一个对象，就必须使用New关键字对其进行实例化，如下面的代码。


Dim item As New CDataItem()
Console.WriteLine(item.ToString())



现在，我们了解了类与对象的基本知识以后，就来看看如何在类中创建更多的成员。

7.2. 构造函数与析构函数

构造函数用于在对象实例化时进行对象的初始化操作，默认情况下，类都会自动生成一个没有参数的构造函数；我们也可以在类中自己定义这个构造函数，此时，我们需要定义一个名为New()的子程序，如下面的代码。


Public Class CDataItem
    Public Sub New()
        Console.WriteLine("对象正被创建")
    End Sub
End Class



然后，我们使用New关键字创建CDataItem类的实例，就可以看到这条输出信息。如下面的代码。


Dim item As New CDataItem()



执行代码，我们会看到如下图的输出结果。

[image: 图像说明文字]


在VB.NET定义的类中，我们可以同时定义多个构造函数，分别定义不同的参数以便进行相应的初始化工作，如下面的代码，就是CDataItem类中的另一个构造函数。


Public Sub New(sName As String, oValue As Object)
    myName = sName
    myValue = oValue
End Sub



不过，我们还可以使用更强大的构造函数定义，即在构造函数中使用可选参数，这样，就可以达到一次创建多个版本构造函数的目的，如下面的代码。


Public Sub New(Optional sName As String = "", _
               Optional oValue As Object = Nothing)
    myName = sName
    myValue = oValue
End Sub



这样一来，我们定义一个构造函数，就可以使用以下三种形式来创建CDataItem对象。


Dim item1 As New CDataItem()
Dim item2 As New CDataItem("name")
Dim item3 As New CDataItem("name", "Tom")



析构函数的功能正好与构造函数相反，我们需要在析构函数中对一些特殊的资源进行释放。

我们知道，在.NET Framework中的CLR（Common Language Runtime）有着很好的内存自动回收机制，决大多数情况下，我们并不需要手动回收系统资源（如对象所占用的内存）。但是，在一些类中，我们可能需要调用外部资源，此时，能够及时地释放它们就是一项很重要的工作了。

在VB.NET中，Finalize()子程序用于定义类的析构函数；如果我们需要在自己的类中定义析构函数，应该使用如下格式的代码。


Protected Overrides Sub Finalize()
    '本类的析构代码
    Console.WriteLine("对象正被释放")
    '
    MyBase.Finalize()
End Sub



如果你对Protected、Overrides、MyBase等关键字的用途不太明白，不用着急，完成本章和下一章内容的讨论后，这些都不是问题了。

现在，再次执行如下代码。


Dim item As New CDataItem()



我们会看如下图所示的运行结果。

从运行结果中可以看到，我们只是创建了一个对象而已，却输出了两条语句，这就是构造函数和析构函数的运行机制。创建对象时会自动调用相应的构造函数，而对象释放时会自动调用析构函数。

好的，问题来了。我们看到了对象创建的代码，却没有释放对象的代码，对象是什么时候被释放的呢？答案是，这正是CLR的内存回收机制在工作，对象的释放和回收都是自动执行的。当一个对象不再被引用时，CLR就会自动对其进行回收，此时，就会自动调用对象的Finalize()方法。

7.3. 值类型与引用类型

是时候明确值类型和引用类型的概念了！

我们已说过，在VB.NET中，是一个一切皆对象的世界。那么，如何区分值类型和引用类型呢？这取决于在.NET Framework内部对于数据的处理方式，我们可以从两个方面来讨论，即数据的存储方式及数据的传递方式。

在VB.NET中，基本的数值类型、布尔类型，以及枚举、结构等类型被定义为值类型；当我们使用这些类型的变量时，会直接对它们所处的内存区域进行操作，这就是值类型数据的操作特点。值类型的优势就是处理速度较快，因为会直接访问内存。

我们所使用的String类型、Object类型，以及其它定义的类（Class）类型都属于引用类型。引用类型在内存中的管理方式称为间接访问，当我们使用对象时，对象只是指向了对象体所处的内存区域的开始位置，当我们调用对象的成员时，对象才会去访问相应的位置。

除了在内存的存储的形式不同，值类型和引用类型在默认的数据复制操作的表现中也不同，值类型会完全复制数据，而引用类型默认情况下只会传递引用（即对象体在内存的位置）。实际应用中，String类型的表现会有些不一样，我们会在第15章详细讨论。

下面，我们首先使用上一章创建的SDemo结构来看一看值类型的复制操作。


Dim s1 As New SDemo("s1", 10)
Dim s2 As SDemo = s1
s2.Name = "s2"
s2.Age = 99
Console.WriteLine("{0}, {1}", s1.Name, s1.Age)
Console.WriteLine("{0}, {1}", s2.Name, s2.Age)



代码执行会显示如下图所示的结果。

代码中，当我们将s1赋值给s2时，会完全复制其内容，这样一样，当我们修改s2的值时，就与s1无关，显示的执行结果证实了这一点。

再看本章定义的CDateItem对象的复制操作，如下面的代码。


Dim item1 As New CDataItem("name", "Tom")
Dim item2 As CDataItem = item1
item2.Value = "Jerry"
Console.WriteLine("{0}, {1}", item1.Name, item1.Value)
Console.WriteLine("{0}, {1}", item2.Name, item2.Value)



执行此代码，我们会看到如下图所示的结果。

我们可以看到，代码中只是修改了item2的值，但结果是item1的值也改变了；其原因就是，CDateItem定义为类，作为引用类型，当我们将对象item1赋值给item2时，实际复制的只是item1对象的内存地址（浅复制），也就是说，赋值后item1和item2指向同一对象体（内存区域）；这样一来，当我们通过任意一个对象修改数据后，都会反映到相关的所有对象。

那么，如果我们真的需要完全复制一个对象，即对象进行深复制时应该怎么办呢？我们会在第9章讨论相关内容。

此外，在参数中按值或按引用传递的相关内容，我们已在第4章讨论过，这里不再多说了。

7.4. 类的成员

本节，我们将讨论类成员的各种情况，如访问级别（作用域）、访问类型，以及成员的类型等内容。

7.4.1. 成员的访问级别

成员的访问级别，也称为成员的作用域或可访问性。在VB.NET中，类成员的基本访问级别有四种：


	Public（公共的），成员可以在被类的使用者调用。

	Private（私有的），成员只能在类的内部使用。

	Protected（受保护的），成员只能在本类或其子类中使用。关于类的继承相关内容会在下一章讨论。

	Friend（友元的），成员只能在本程序集中使用。根据需要，我们可以将Protected和Friend同时使用。请注意，模块成员的默认访问级别就是Friend。



其中，Public和Private的使用，相信大家也已经见过了，我们在结构中就已经使用了。而Protected则是在继承关系中使用，我们会在下一章讨论。Friend一般用于对项目代码的保护，即代码只能在当前程序集中使用，而不能被外部代码调用。

7.4.2. 共享成员与实例成员

根据成员的访问方式来分，类的成员可以分为共享成员和实例成员。其中：


	共享（Shared）成员，在很多编程语言中称为静态成员，需要使用类直接访问这些成成员。

	实例成员，也称为对象成员，我们必须使用类的实例，即对象来访问。



定义共享成员时，需要在成员前使用Shared关键字，我们可以将类的任何成员定义为共享的。如下面的代码，我们在CDemo类中定义了共享变量成员Counter。


Public Class CDemo
    Public Shared Counter As Integer = 0
End Class



然后，我们可以代码中直接使用类CDemo来调用这个成员，如下面的代码。


Console.WriteLine(CDemo.Counter)



在类中，没有使用Shared关键字的成员都定义为实例成员，我们必须使用类的实例（对象）来访问它们。

7.4.3. 成员类型

和结构的使用相似，在类中，我们同样可以定义变量、属性、方法，但在类中，我们还可以定义更多的类型成员，如默认属性、事件等。接下来的内容，我们就逐渐了解这些类成员的创建和使用。

7.5. 变量

和结构中的变量定义方法相同，如下面的代码，我们会在CDataItem类中定义两个私有变量，分别用于保存数据项的名称和值（VBChef/CDataItem.vb文件）。


Public Class CDataItem
    Private myName As String
    Private myValue As Object
End Class



如果我们只是简单使用变量表示一个数据，可以直接定义为公共的（Public）成员。这些变量成员也称为类的字段（Field）。

7.6. 属性与默认属性

本节，我们将讨论属性读写操作，以及一个特殊的属性成员，即默认属性。

7.6.1. 属性

在类中，属性（Property）的操作同样分为读取和写入两种。一般情况下，我们使用如下格式定义一个可读写的属性。


Property <属性名> As <数据类型>
    Get
        '属性读取操作
    End Get
    Set(value As <数据类型>)
        '属性写操作
    End Set
End Property



在结构的讨论中，我们已经看到了属性的定义，在类中的定义方式是相同的，如下面的代码，就是我们定义的CDataItem类中的Name和Value属性（VBChef/CDataItem.vb文件）。


' 属性Name
Public Property Name As String
    Get
        Return myName
    End Get
    Set(value As String)
        myName = value
    End Set
End Property
' 属性Value
Public Property Value As Object
    Get
        Return myValue
    End Get
    Set(value As Object)
        myValue = value
    End Set
End Property



除了可读写属性，我们还可以定义只读属性或只写属性。如下面的代码，我们在CDemo类中定义了一个只读属性TimeStamp（Hello/CDemo.vb文件）。


Public Class CDemo
    '只读属性
    Public ReadOnly Property TimeStamp As Long
        Get
            Return Date.Now.Ticks
        End Get
    End Property
End Class



定义只读属性时，我们需要在Property关键字前使用ReadOnly关键字，并且只需要实现Get...End Get语句结构；相对应的，如果我们需要定义只写属性，则需要使用WriteOnly关键字，同时，在属性内，只需要创建Set...End Set语句结构。

7.6.2. 默认属性

默认（Default）属性可以使用对象名直接访问，而不需要使用属性名；不过，默认属性也有一些要求，即它必须带有参数。

实际上，默认属性的功能是对应于C#中的索引器（indexer）功能，在类中定义了默认属性以后，我们就可以通过索引（默认属性的参数）来访问对象中的数据。在VBChef项目中定义的CDataCollection类中，就将Item属性定义成了一个默认属性（VBChef/CDataCollection.vb文件）。


Public Class CDataCollection
    Private myList As List(Of CDataItem)
    '
    Public Sub New()
        myList = New List(Of CDataItem)
    End Sub
    '' 默认属性，数据集合项目，数值索引
    Default Public Property Item(index As Integer) As CDataItem
        Get
            If index >= 0 And index < myList.Count Then
                Return myList(index)
            Else
                Return New CDataItem()
            End If
        End Get
        Set(value As CDataItem)
            myList(index) = value
        End Set
    End Property
    '
End Class



接下来，我们来看在代码中如何使用默认属性。


Dim item1 As New CDataItem("name", "Tom")
Dim dColl As New CDataCollection()
dColl.Item(0) = item1
Console.WriteLine(dColl(0).Value)



代码中，在设置索引为0的数据项时，我们使用了dColl.Item(0)，大家可以看到，这是在显式的调用Item属性；而显示数据时，我们直接使用了dColl(0)，这就是默认属性在起作用，而此代码和调用dColl.Item(0)的作用是完全相同的。

7.7. 方法

类中的方法，不是子程序就是函数了，相信大家对它们也不再陌生；只是需要注意方法定义的访问级别和访问类型。在这里，我们就不浪费篇幅和大家的时间了。^_^

7.8. 运算符重载

对于VB.NET编程语言中定义的运算符，我们可以修改它们的表现，从另一方面，我们也可以通过在类中定义运算符，让对象进行有意义的“运算”。

接下来，我们创建一个CTeam类，表示团队对象，如下面的代码（Hello/CTeam.vb文件）。


Public Class CTeam
    Public member As New ArrayList()
    Public Sub AddMember(sName As String)
        member.Add(sName)
    End Sub
    Public Sub ShowList()
        For Each item As String In member
            Console.WriteLine(item)
        Next
    End Sub
End Class



现在，假如我们有两个团队对象，需要将他们合并，我们会什么方法呢？如果能直接使用加号（+）是不是可以呢？

要使用加号，我们就需要在CTeam类中定义这个运算符的工作方式，如下面的代码（Hello/CTeam.vb文件）。


'加运算
Public Shared Operator +(ByVal team1 As CTeam, _
                         ByVal team2 As CTeam) As CTeam
    Dim result As New CTeam()
    For Each item As String In team1.member
        result.AddMember(item)
    Next
    For Each item As String In team2.member
        result.AddMember(item)
    Next
    Return result
End Operator



重载或定义运算符时，我们使用Operator...End Operator结构，而且必须定义为公共的（Public）和共享的（Shared）。在运算符后面的一对小括号中，如果有一个参数，我们定义的就是一元运算符（需要一个运算数）；如果有两个参数，则定义的就是二元运算符（需要两个运算数）。

接下来，我们使用如下代码来测试这个运算符。


Dim team1 As New CTeam()
team1.AddMember("Tom")
team1.AddMember("Jerry")
Dim team2 As New CTeam()
team2.AddMember("John")
team2.AddMember("Smith")
'
Dim team As CTeam
team = team1 + team2
team.ShowList()



执行此代码，我们会看到如下图所示的结果。

[image: 图像说明文字]


7.9. 分部类

有时候，一个类需要不同的开发者同时进行开发工作，此时，我们可以将类的代码分为两个文件同时进行编辑和维护，而在编译时能够自动合并。比如，在ASP.NET页面中，我们就可以将HTML内容和VB.NET代码分别在两个文件中编辑，这样，前端设计师和VB.NET开发者就可以各尽其职，工作也不会互相干扰了。

在VB.NET中，我们可以使用Partial关键字实现，我们可以看如下两个文件的内容。

（Hello/ClassX1.vb文件）


Partial Public Class ClassX
    Public Sub Work1()
        Console.WriteLine("ClassX.Work1() is working.")
    End Sub
End Class



（Hello/ClassX2.vb文件）


Partial Public Class ClassX
    Public Sub Work2()
        Console.WriteLine("ClassX.Work2() is working.")
    End Sub
End Class



我们可以看到，在这两个文件中定义的都是ClassX类，但在类的定义时都使用Partial关键字；这样，在编译代码时，它们会被当作一个类来处理。如下面的代码。


Dim x As New ClassX()
x.Work1()
x.Work2()



执行代码，我们会看到如下图所示的结果。

8. 类的继承

在面向对象编程中，继承是非常重要的一个概念，同时，也是代码复用的重要方式。本章，我们就讨论VB.NET中关于类的继承问题，这些内容包括：


	继承概述

	扩展

	重载

	重写

	隐藏

	抽象类与抽象方法

	对象类型判断与转换



8.1. 继承概述

继承是一个相对关系，在这一关系中，必须有继承者和被继承者。假如我们已经定义了一个类ClassA，然后，我们希望ClassB可以使用ClassA类中已经定义的一些内容，就可以让ClassB来继承ClassA类，此时，ClassB类就是继承者，ClassA类则是被继承者。

在面向对象编程的概念中，被继承者称为父类（parent）、基类（base）或超类（super）；而继承者则称为子类（subclass）。本书中，我们将统一使用父类和子类的概念。

进一步讨论继承的问题之前，我们必须了解一个前提，即类必须能够被继承，我们才能创建它的子类，在VB.NET中能够被继承是类的默认表现；但是，如果在类定义时使用了NotInheritable关键字，那么，这个类就是不能被继承的。

接下来，我们首先新建一个ClassA类，其定义如下面的代码（Hello/ClassA.vb）。


Public Class ClassA
    Private myName As String
    Private myAge As Integer
    '
    Public Sub New()
        Me.New("", 0)
    End Sub
    Public Sub New(sName As String, iAge As Integer)
        myName = sName
        myAge = iAge
    End Sub
    '
    Public Property Name As String
        Get
            Return myName
        End Get
        Set(value As String)
            myName = value
        End Set
    End Property
    '
    Public Property Age As Integer
        Get
            Return myAge
        End Get
        Set(value As Integer)
            If value >= 0 And value <= 150 Then
                myAge = value
            End If
        End Set
    End Property
    '
    Private Sub SayHi()
        Console.WriteLine("Hi, {0}", myName)
    End Sub
    '
    Public Sub SayHello()
        Console.WriteLine("Hello, {0}", myName)
    End Sub
    '
    Public Overridable Sub SayBye()
        Console.WriteLine("Bye-Bye, {0}", myName)
    End Sub
End Class



接下来，我们定义它的子类ClassB，如下面的代码（Hello/ClassB.vb）。


Public Class ClassB
    Inherits ClassA
End Class



代码中，我们使用Inherits关键字指定子类将要继承的父类名称。

那么，接下来，我们在ClassB中可以使用哪些ClassA类的成员呢？又有哪些成员不能被继承呢？

8.1.1. 构造函数

在子类中，构造函数不能被继承，但会自动包含一个没有参数的空构造函数；所以，我们必须在ClassB类中重新定义所需要的构造函数，如下面的代码（Hello/ClassB.vb文件）。


Public Class ClassB
    Inherits ClassA
    '构造函数
    Public Sub New(Optional sName As String = "", _
                   Optional iAge As Integer = 0)
        Me.Name = sName
        Me.Age = iAge
    End Sub
End Class



或者，我们可以利用父类中的构造函数，如下面的代码（Hello/ClassB.vb文件）。


Public Class ClassB
    Inherits ClassA
    '构造函数
    Public Sub New(Optional sName As String = "", _
                   Optional iAge As Integer = 0)
        MyBase.New(sName, iAge)
    End Sub
End Class



接下来，我们就可以通过构造函数创建ClassB对象了，如下面的代码。


Dim obj As New ClassB("Tom", 35)
Console.WriteLine("{0}, {1}", obj.Name, obj.Age)



8.1.2. 其它成员

除了构造函数，子类中都会按照父类中成员的访问级别设置进行访问，如：


	父类中的Private成员不能被子类访问。

	父类中的Public、Protected、Friend和Protected Friend成员，可以被子类访问。



在我们定义的ClassA类中的成员，我们可以看到，其中，除了SayHi()方法，其它的方法和属性都可以在ClassB中访问，我们可以通过一些代码来测试这一点，如前面我们已经在ClassB对象中使用过Name和Age属性，下面的代码会调SayHello()方法：


Dim obj As New ClassB("Tom", 35)
obj.SayHello()



8.1.3. 区分父类与子类成员

我们说过，继承是一个相对的关系，那么，在子类中，如果需要区分父类和本类的成员，我们主要可以使用以下两个关键字。


	Me关键字，表示当前对象（实例）。请注意，在没有继承关系的类中，我们依然可以使用Me关键字调用当前对象的成员。

	MyBase关键字，表示父类，如我们在ClassB类中调用父类的构造函数，可以使用MyBase.New()语句。



8.2. 扩展

当我们在子类中定义了父类中不存在的成员时，即可以完成对父类功能的扩展。

如下面的代码（Hello/ClassB.vb文件），我们在ClassB类中添加一个新的MoveTo()方法。


Public Sub MoveTo(x As Single, y As Single)
    Console.WriteLine("{0}移动到({1}, {2})", Me.Name, x, y)
End Sub



然后，我们可通过以下代码调用这个方法。


Dim obj As New ClassB("Tom", 35)
obj.MoveTo(100, 200)



实际上，我们可以发现，在子类中扩展成员，就是创建新的成员。

8.3. 重载（Overload）

在子类中，如果我们创建了与父类成员同名，但参数设置或返回值类型不一样的方法时，我们就是在重载此方法的新版本。重载方法时，我们应该使用Overloads关键字，如下面的代码（Hello/ClassB.vb文件）。


 '重载方法
Public Overloads Sub SayHello(sPrefix As String)
    Console.WriteLine("Hello, {0} {1}", sPrefix, Me.Name)
End Sub



然后，我们通过如下面代码调用这个方法。


Dim obj As New ClassB("Tom", 35)
obj.SayHello("Mr.")



此外，在子类中，我们重载的成员只能是方法或属性。

8.4. 重写（Override）

重写是指在子类中创建了与父类中定义完全一样的成员，用于替换父类成员的实现。

默认情况下，父类中的成员是不允许重写的，因为那样就破坏了代码封装和复用的一些基本原则；不过，有些时候，父类中的功能实现可能真的需要在子类中更加精确地完成，此时，我们需要在定义父类的成员时使用Overridable关键字，用于说明此成员是可以被重写的。

在我们定义的ClassA类中，SayBye()方法就被定义为可重写方法。下面，在ClassB类中，我们就来重写此方法（Hello/ClassB.vb）。


'重写方法
Public Overrides Sub SayBye()
    'MyBase.SayBye()
    Console.WriteLine("Goodbye.")
End Sub



代码中，在子类中重写父类成员时，应该在成员定义时使用Overrides关键字。

此外，我们可以看到代码中的MyBase.SayBye()语句被修改为注释，因为在这里，我们重写的方法中不需要调用父类中的同名方法。但大家应注意，在子类中重写的成员，很多时候实际上是在对父类成员功能的扩展，在这种情况下，我们应该首先使用MyBase关键字执行父类中的代码，然后再执行本类成员中特有的代码。

重写父类成员时，还有一些问题需要我们注意：


	在子类中重写的父类成员同样只能是属性或方法。

	子类中重写的成员定义必须与父类的成员完全一样。

	重写父类成员后，我们可以分别使用Me或MyBase关键字区分和调用它们。

	如果我们子类中重写了父类的成员，但不想让子类的子类再次重写，可以使用NotOverridable关键字定义它，如下面的代码。




Public NotOverridable Overrides Sub SayBye()
    'MyBase.SayBye()
    Console.WriteLine("Goodbye.")
End Sub



这样，假如有ClassC类，它继承了ClassB类，并且想重写SayBye()方法，如下面的代码（Hello/ClassC.vb文件）。


Public Class ClassC
    Inherits ClassB
    '重写方法
    Public Overrides Sub SayBye()
        'MyBase.SayBye()
        Console.WriteLine("Goodbye.")
    End Sub
End Class



此代码会报错，因为ClassB类中的SayBye()方法使用了NotOverridable关键字定义为不能被重写。

如果我们需要在子类重新实现一个不能被重写的方法，可以使用成员的隐藏功能。

8.5. 隐藏（Shadow）

在子类中，我们可以重新定义一个与父类成员同名的成员，无论是参数、数据类型，甚至成员类型是否一样，我们都可以通过隐藏父类成员达到重新定义的目的。

如前面示例中ClassC类中，我们无法重写SayBye()方法，因为它在ClassB类中被定义为不能被重写（NotOverridable）。不过，我们可以使用Sharows关键重新定义SayBye()方法，这样也就隐藏了父类中的同名方法。如下面的代码（Hello/ClassC.vb文件）。


Public Class ClassC
    Inherits ClassB
    '构造函数
    Public Sub New(Optional sName As String = "", _
                   Optional iAge As Integer = 0)
        MyBase.New(sName, iAge)
    End Sub
    ' 隐藏父类方法
    Public Shadows Sub SayBye()
        Console.WriteLine("再见，{0}", Me.Name)
    End Sub
End Class



然后，我们可以通过以下代码调用这个方法。


Dim obj As New ClassC("Tom", 35)
obj.SayBye()



8.6. 抽象类与抽象方法

代码中，我们不能创建抽象（abstract）类的对象（实例），而抽象类的作用就是为其子类提供一个基本模块，或者是一些通用的代码，而更具体的工作完全由子类来完成。

在VB.NET中，定义抽象类时需要使用MustInherit关键字。如下面的代码，我们定义了ClassD类，它被指定为一个抽象类，并定义了一个抽象方法DoWork()。


Public MustInherit Class ClassD
    Public MustOverride Sub DoWork()
End Class



我们可以看到，在VB.NET中定义一个抽象方法时，应该使用MustOverride关键字，即此方法是必须要被重写的。我们可以看到，抽象方法是一个空方法，我们不需要定义它的内容，所以，也就不需要End语句。

请注意，当一个类中有一个成员被定义为抽象时，这个类就必须也被定义为抽象类。

接下来，我们定义ClassE类作为ClassD抽象类的子类，如下面的代码（Hello/ClassD.vb）。


Public Class ClassE
    Inherits ClassD
    '重写抽象方法
    Public Overrides Sub DoWork()
        Console.WriteLine("ClassE.DoWork() is working.")
    End Sub
End Class



在使用抽象类时，类成员不要求一定都是抽象的，我们可定义一些普通的类成员，在子类中，我们就可以直接使用它们，当然也可以根据需要修改它们。在抽象类中定义一些具体的实现代码时，我们经常会将这些成员定义为可重写的（Overridable），这样就便于在子类修改它们。如下面的代码，我们就在ClassD类中添加了一个可重写方法DoWorkA()。


'可重写方法
Public Overridable Sub DoWorkA()
    Console.WriteLine("ClassD.DoWorkA() is working.")
End Sub



在ClassE类中，如果我们不需要修改它的实现代码，则可以直接使用，如果我们需要修改或扩展它的代码，则可以重写这个方法，如下面的代码（Hello/CalssE.vb）。


'重写方法
Public Overrides Sub DoWorkA()
    MyBase.DoWorkA()
    Console.WriteLine("ClassE.DoWorkA() is working.")
End Sub



接下来，我们可以使用如下代码测试DoWorkA()方法的使用。


Dim obj As New ClassE()
obj.DoWorkA()



此代码会显示如下图所示的结果。

8.7. 对象类型判断与转换

8.7.1. Object类

现在，我们可以回想一下，在前面的测试代码中，我们使用过ToString()方法，而我们却重来没有定义过它，这个方法是怎么回事呢？实际上，最基本的ToString()方法就定义在Object类中，那么，我们怎么可以在任何一个对象中都可以使用呢？

这就是.NET Framework所有类结构中的一个小秘密，即，除了Object类，其它所有类的终极父类都是Object类，这样一来，我们就可以在任何类中使用Object类中定义的成员了，这其中就包括最常用的ToString()方法。当然，在不同的类中，我们可以重写ToString()，以便使用最合适的格式显示数据内容。Object类定义在System命名空间，我们可以通过帮助查询它的完整定义。

8.7.2. 装箱与拆箱

实际应用中，可以使用Object对象保存任何类型的数据，但这并不意味着我们就要大规模地使用它。但是，通过Object类，可以帮助我们简化很多代码的编写工作（如数据的传递操作）；在后续的学习中，我们会看到很多Object类型的应用场景。

那么，当在Object类型与值类型数据之间转换时会怎么样呢？如我们需要传递Integer类型的数据，也可以使用Object类型的参数进行传递；在需要的时候，我们又会将数据从Object类型转换成Integer类型。这两个转换操作就是装箱与拆箱操作。

装箱操作，即将值类型的数据转换为Object类型。如下面的代码。


Dim iNum As Integer = 99
Dim obj As Object = iNum



我们知道，Object是“万能类型”，所以，当一个值类型赋值到Object对象时，是没有任何问题的。

但相反的操作，即拆箱操作就得小心了，因为对象数据未必就能正确地转换为所需要的值类型。不过，在第4章，我们已经封装了一系列将Object转换成值类型的方法了，所以在这里就不再重复了。

8.7.3. TypeOf...Is语句

TypeOf...Is语句用于判断一个对象是否为某个类的实例。接下来，我们以ClassA、ClassB和ClassC类来进行测试，但别忘了，在它们最顶层还有Object，那么，这四个类的继承关系就如下图所示。

下面的代码，我们将测试TypeOf...Is语句的作用。


Dim obj As New ClassB()
Console.WriteLine(TypeOf obj Is Object)
Console.WriteLine(TypeOf obj Is ClassA)
Console.WriteLine(TypeOf obj Is ClassB)
Console.WriteLine(TypeOf obj Is ClassC)



请注意，我们将obj对象定义为ClassB类的实例。那么，它们的判断结果都是什么呢？


	第一个，判断obj是否为Object类的实例。我们知道，Object是万能类了，我们可以将任何数据保存到Object对象中，所以，显示为True。

	第二个，判断obj是否为ClassA的实例。我们知道，ClassB是ClassA的子类，也就是说，我们在obj对象上是可以使用ClassA中的所有非私有成员；实际上，也正是这个原因，此语句判断结果也是True。

	第三个，我们定义的obj就是ClassB类的实例，所以判断结果一定会是True了。

	第四个，显示为False，也就是说，一个对象不是其类型子类的实例。



好的，我们现在可以确定，使用TypeOf...Is语句判断一个对象的类型时，当类型是其定义类型或其父类时，返回结果True；否则返回False。

此外，当对象定义为Object类型，而又赋值为其它类型的对象时，其判断结果则不会遵守这一原则，下面讨论相关内容。

8.7.4. CType()方法

CType()方法用于对象类型的转换工作，我们在应用窗体中的Button按钮控件时，在单击事件响应方法中需要使用到Button控件的一些属性，如获取按钮名称，则可以使用如下代码。


Private Sub Button1_Click(sender As Object, e As EventArgs) _
                          Handles Button1.Click
    Dim btnName As String = CType(sender, Button).Name
End Sub



其中，sender参数的类型是Object，所以，我们必须将其转换为Button类型才能使用Name属性。请注意，上述情况是我们已经了解sender一定会是Button控件时，才能确保这样的转换会顺利完成。

我们需要注意，sender参数虽然定义为Object类型，但它在传递过程中的内容是由具体的控件类型转换而来，在sender对象中实际保留了对象的原始类型信息，所以，我们可以使用CType()成功地将sender对象转换为控件的原始类型。

对于sender此类的Object类型的对象，当我们使用TypeOf...Is语句判断其类型时，是可以判断出它的原始类型的，如下面的代码。


Private Sub Button1_Click(sender As Object, e As EventArgs) _
                          Handles Button1.Click
    MsgBox(CStr(TypeOf sender Is Button))
End Sub



当我们单击Button1按钮时，此代码会显示一个包含“True”的消息对话框。则说明sender参数实现传递的就是一个Button控件，而我们也可以使用CType()方法将其转换为Button对象来使用。如：


Private Sub Button1_Click(sender As Object, e As EventArgs) _
                          Handles Button1.Click
    Dim btn As Button = CType(sender, Button)
End Sub







9. 接口

接口（Interface）可以用来解决一些使用类和对象时的问题，如多继承问题。我们知道，在VB.NET中，一个类是不能同时继承多个父类（基类）的，但在有些时候，我们又会需要使用这种关系，此时，我们在代码结构中使用接口就可以解决这个问题。

另一方面，接口的应用也给我们的软件结构带来了极大的灵活性，比如，我们有一个坦克类，它有大炮，而且有车辆的特性；那么，在类的继承关系中，它是作为大炮的子类呢？还是作为车辆的子类呢？对于这一类的问题，恐怕类的继承关系并不易处理，而此时，我们就可以使用接口来处理这样的关系。

本章，我们就来讨论接口在VB.NET中的应用问题，主要内容包括：


	定义接口

	实现接口

	接口的继承

	IDisposable接口与Using语句

	对象的复制



9.1. 定义接口

在VB.NET中定义定义接口，我们使用Interface...End Interface语句结构，如下面的代码，我们就定义了一个名为IWork的接口类型（Hello/IWork.vb文件）。


Interface IWrok
    Property Name As String
    Sub WorkA()
    Sub WorkB()
End Interface



其中，我们定义了属性Name，以及WorkA()和WorkB()方法。在接口成员的定义中，我们不需要使用访问级别限制，如Public、Private等。而且，接口中不会有任何的具体实现代码。

在接口中，我们可以定义的成员类型包括属性、方法、事件等。

9.2. 实现接口

独立的接口是无法使用的，我们必须使用类来实现它才会做具体的工具，如下面的CWorkA类，我们就使用它实现（Implement）了IWork接口（Hello/CWorkA.vb文件）。


Public Class CWorkA
    Implements IWork
    ''
    Private myName As String
    ''
    Public Property Name As String Implements IWork.Name
        Get
            Return myName
        End Get
        Set(value As String)
            myName = value
        End Set
    End Property
    ''
    Public Sub WorkA() Implements IWork.WorkA
        Console.WriteLine("CWorkA.WorkA() is working.")
    End Sub
    ''
    Public Sub WorkB() Implements IWork.WorkB
        Console.WriteLine("CWorkA.WorkB() is working.")
    End Sub
End Class



我们可以看到，当一个类实现一个接口时，我们需要使用Implements关键字指定接口类型；在接口成员的实现时，也必须使用Implements关键字指定成员所实现的接口及成员名称。

9.2.1. 使用接口类型

现在，我们已经使用CWorkA类实现了IWork接口，那么，在代码中，我们就可以使用CWorkA类了，如下面的代码。


Dim w As New CWorkA()
w.WorkA()
w.WorkB()



代码中，我们将w对象定义为CWorkA类型，实际上，我们还可以将其定义为IWork接口类型，如：


Dim w As IWork



但此时，对象w并不能使用，我们必须指定具体的实例类型，如：


w = New CWorkA()



现在，我们的代码中又添加了CWorkB类，它同样实现了IWork接口，如下面的代码（Hello/CWorkB.vb）。


Public Class CWorkB
    Implements IWork
    ''
    Private myName As String
    ''
    Public Property Name As String Implements IWork.Name
        Get
            Return myName
        End Get
        Set(value As String)
            myName = value
        End Set
    End Property
    ''
    Public Sub WorkA() Implements IWork.WorkA
        Console.WriteLine("CWorkB.WorkA() is working.")
    End Sub
    ''
    Public Sub WorkB() Implements IWork.WorkB
        Console.WriteLine("CWorkB.WorkB() is working.")
    End Sub
End Class



那么，我们在代码中，就不需要修改对象w的调用方法，而只是修改它的创建代码，就可以实现不同的执行结果，如下面的代码。


w = New CWorkB()
w.WorkA()
w.WorkB()



这样做有什么用呢？当我们接触更多关于软件结构性问题时，就可以逐渐理解了。

9.2.2. 实现多个接口

一个类可以同时实现多个接口，如下面的代码，我们定义了两个接口类型，即I1和I2（Hello/I1.vb）。


'
Interface I1
    Sub WorkA()
End Interface
'
Interface I2
    Sub WorkA()
    Sub WorkB()
End Interface



我们可以看到，在I1和I2接口中存在相同的成员WorkA()方法，在实现的类中的，我们区分它们呢？如下面的代码（Hello/CDemoA.vb）。


Public Class CDemoA
    Implements I1, I2
    '
    Public Sub WorkA() Implements I1.WorkA
        Console.WriteLine("I1.WorkA() is working.")
    End Sub
    '
    Public Sub WorkA1() Implements I2.WorkA
        Console.WriteLine("I2.WorkA() is working.")
    End Sub
    '
    Public Sub WorkB() Implements I2.WorkB
        Console.WriteLine("I2.WorkB() is working.")
    End Sub
End Class



我们知道，在一个类中，不允许有两个定义完全一样的成员，所以，在代码中，自动将实现I2.WorkA()的方法命名为WorkA1()。实际上，我们可以自己修改其实现成员，只需要使用Implements关键字将其与所要实现的接口成员关联即可，如：


Public Sub DoWorkA() Implements I2.WorkA
    Console.WriteLine("I2.WorkA() is working.")
End Sub



如果在两个或多个接口中定义的成员名称一样，功能也完全一样，我们可以在类中只使用一个成员来实现，如下面的代码就是修改后的CDemo1类的代码。


Public Class CDemoA
    Implements I1, I2
    '
    Public Sub WorkA() Implements I1.WorkA, I2.WorkA
        Console.WriteLine("I1&I2.WorkA() is working.")
    End Sub
    '
    Public Sub WorkB() Implements I2.WorkB
        Console.WriteLine("I2.WorkB() is working.")
    End Sub
End Class



我们可以看到，CDemoA类中的WorkA()方法，同时实现了I1和I2接口中的WorkA()方法。

9.2.3. 判断对象是否实现某一接口

当我们需要在对象中使用某一接口类型中定义的成员时，可以通过TypeOf...Is语句判断这个对象是否实现了指定的接口，如下面的代码。


Dim w As New CWorkA()
Console.WriteLine(TypeOf w Is IWork)



此外，我们还可以通过CType()方法，将对象转换为接口类型后再调用其成员，如下面的代码。


Dim w As New CWorkA()
CType(w, IWork).WorkA()



9.3. 接口的继承

我们说过，接口支持多继承，即一个接口可以继承多个接口，如下面定义的I3接口，就同时继承了前面创建的I1和I2接口。


Interface I3
    Inherits I1, I2
End Interface



接下来，当一个类来实现I3接口时，我们可以使用如下代码。


Public Class CDemoB
    Implements I3
    '
    Public Sub WorkA() Implements I1.WorkA

    End Sub
    '
    Public Sub WorkA1() Implements I2.WorkA

    End Sub
    '
    Public Sub WorkB() Implements I2.WorkB

    End Sub
End Class



我们可以看到，在CDemoB类中，我们实现了I3接口，但实际上，我们需要实现I1和I2接口中的所有成员，这和直接实现I1和I2接口的操作是相同的。

9.4. IDisposable接口与Using语句

在System命名空间中，定义了一些常用的接口，而它们大多与.NET Framework和编程语言有着密切的关系，接下来，我们就来了解其中的两个，即IDisposable和ICloneable接口，首先来看IDisposable接口类型。

实现了IDisposable接口的类，我们可以在代码中使用Using语句结构，其功能是在对象调用结束后可以自动进行一些处理，如对外部资源的释放。下面，我们先看一看一个类如何实现IDisposable接口。

当我们在自己的类中确定要实现IDisposable接口时，VS开发环境会自动为我们添加所需要的代码，如：


Public Class CDemoD
    Implements IDisposable

    Private disposedValue As Boolean

    Protected Overridable Sub Dispose(disposing As Boolean)
        If Not Me.disposedValue Then
            If disposing Then
                '
            End If
        End If
        Me.disposedValue = True
    End Sub

    Public Sub Dispose() Implements IDisposable.Dispose
        Dispose(True)
        GC.SuppressFinalize(Me)
    End Sub
End Class



代码中，我们去除了注释，如果大家想看到完整的内容，可以在自己的开发环境中输入Implements IDispolable语句后回车。

实际上，在IDisposable接口中被编程语言自动调用的只是Dispose()方法，只不过，在微软建议的处理方式中，将所有具体的处理代码都放在了Dispose(Boolean)方法、GC.SuppressFinalize()方法和Finalize()析构函数中；在Dispose()方法中只是调用了相关方法。

下面，我们简单地演示Dispose()方法的调用，现在修改代码如下面的内容。


Public Class CDemoD
    Implements IDisposable
    '
    Public Sub New()
        Console.WriteLine("对象创建...")
    End Sub
    '
    Public Sub Dispose() Implements IDisposable.Dispose
        Console.WriteLine("对象释放...")
    End Sub
End Class



那么，Dispose()方法会在什么时候调用呢？和析构函数不同，Dispose()方法并不是在对象释放时调用，而是用于Using语句结构，如下面代码。


Using obj As New CDemoD()
    ' 使用obj
End Using



执行此代码，我们会看如下图的结果。

我们会看到，创建对象时调用了构造函数，而在Using语句结构结束时自动调用了Dispose()方法。请注意，如果我们不使用Using语句结构，那么，Dispose()方法是不会被调用的。

应用中，IDisposable接口和Using语句块的配合有什么作用呢？实际上，Using语句定义了一个隐式的语句结构，即：


Try
    ' 执行Using...End Using之间的代码
Finally
    ' 自动调用Dispose()方法
End Try



这样一来，无论Using...End Using之间的代码是否正确执行，都会调用Dispose()方法，这样就为代码进行必要的清理工作提供了保障。如连接SQL Server数据库SqlConnection类就实现了IDisposable接口，当我们打开数据库连接后，就会进行一系列的数据操作，但无论数据操作正确或出现异常，使用Using语句结构都会自动关闭数据库连接，如下面的代码。


Using cnn As New SqlConnection(dbCnnStr)
    cnn.Open()
    ' 数据操作
End Using



9.5. 对象的复制

前面已经讨论过对象的传递与复制，我们知道，对于对象的赋值操作，默认情况下执行的是浅复制，即只是复制对象的引用，而不是对象的全部数据。但有时候，我们会需要完全复制一个对象，即执行对象的深复制，此时，我们可以主要考虑两个方法，使用ICloneable接口，或使用序列化。

9.5.1. 实现ICloneable接口

在.NET Framework中，如果一个类的对象需要进行深复制操作，这个类就应该实现ICloneable接口，其关键成员就是Clone()方法。如下面的代码，我们定义的CCloneDemo类就实现了这个方法（Hello/CCloneDemo.vb）。


Public Class CCloneDemo
    Implements ICloneable
    Public Name As String
    Public Age As Integer
    Public Phone As String
    '
    Public Function Clone() As Object Implements ICloneable.Clone
        Dim obj As New CCloneDemo()
        obj.Name = Me.Name
        obj.Age = Me.Age
        obj.Phone = Me.Phone
        Return obj
    End Function
End Class



我们可以看到，在Clone()方法中，我们重新创建了一个新的CCloneDemo对象，然后，会将当前实例（Me）中的数据逐一赋值到新的对象，最终，Clone()方法会返回一个新的对象，即完成对象的完全复制工作。下面的代码演示了Clone()方法的使用。


Dim obj1 As New CCloneDemo()
obj1.Name = "Tom"
obj1.Age = 35
obj1.Phone = "12345"
Dim obj2 AS CCloneDemo = obj1
Dim obj3 As CCloneDemo = obj1.Clone()
obj2.Name = "Jerry"
obj3.Name = "Smith"
Console.WriteLine("{0}, {1}, {2}", obj1.Name, obj1.Age, obj1.Phone)
Console.WriteLine("{0}, {1}, {2}", obj2.Name, obj2.Age, obj2.Phone)
Console.WriteLine("{0}, {1}, {2}", obj3.Name, obj3.Age, obj3.Phone)



请注意，当我们修改obj2的数据时，obj1也发生了变化，转为它们是浅复制；而obj3对象是通过Clone()方法复制的对象，所以，修改它的数据时，与obj1对象无关。

请注意，在上述的代码中，所有的类成员本身都是可以直接进行深复制的，包括值类型和String类型；但是，如果在类成员数据中包括了对象，我们就需要更深入的复制工作。如下面的代码，我们使用CCloneDemoA类演示这一过程，在类中，我们只定义了一个对象字段obj（Hello/CCloneDemoA.vb）。


Public Class CCloneDemoA
    Implements ICloneable
    '
    Public obj As ClassA
    '
    Public Function Clone() As Object Implements ICloneable.Clone
        Dim result = New CCloneDemoA()
        result.obj = Me.obj
        Return result
    End Function
End Class



实际上，我们通过Clone()方法并不能正确地进行CCloneDemoA对象的深复制操作，如下面的代码。


Dim obj1 As New CCloneDemoA()
obj1.obj = New ClassA("Tom", 35)
Dim obj2 As CCloneDemoA = obj1.Clone()
obj2.obj.Name = "Jerry"
Console.WriteLine("{0}, {1}", obj1.obj.Name, obj1.obj.Age)
Console.WriteLine("{0}, {1}", obj2.obj.Name, obj2.obj.Age)



代码执行结果如下图所示。

我们可以看到，即使我们使用Clone()方法，但实际在Clone()方法中对于对象的复制却还中浅复制，所以，在这里的Clone()方法并不能真正地达到对象深复制的目的。

如果想CCloneDemoA类中的Clone()方法能够对obj进行深复制，我们必须做一些修改，如下面的代码。


Public Class CCloneDemoA
    Implements ICloneable
    Public obj As ClassA
    Public Function Clone() As Object Implements ICloneable.Clone
        Dim result = New CCloneDemoA()
        'result.obj = Me.obj
        result.obj = New ClassA()
        result.obj.Name = Me.obj.Name
        result.obj.Age = Me.obj.Age
        Return result
    End Function
End Class



再次执行前面的测试代码，我们会看到不一样的结果，如下图所示。

这一次代码真的执行了深复制操作。那么，如果ClassA类的成员又定义了对象类型呢？

结果是，无论多少级包含对象成员，我们都必须完成它们的成员的深复制工作，才能最终完成全部数据的深复制，否则，就不是完整的深复制操作。这的确是一个很大的问题，不过，我们还有一种方法实现对象的深复制操作，即使用序列化。

9.5.2. 序列化

序列化（Serialize）包括两个方向相反的操作，即将对象转换为字节数组（序列化），以及将字节数组还原成对象（反序列化）。此外，一个对象如果可以被序列化，在它的类定义是必须使用SerializableAttribute特性（定义在System命名空间），我们定义的CDataItem等类型就是这样，如下面的代码。


<Serializable>
Public Class CDataItem
    '
End Class



此外，在.NET Framework中定义的很多类，特别是容器类的类型，都使用了SerializableAttribute特性，所以，我们可以放心使用序列化对它们的对象进行深复制操作。

关于序列化和反序列化的操作，我们都封装在vbchef命名空间中的MCommon模块中，如下面的代码（VBChef/MCommon.vb）。


Imports System.IO
Imports System.Runtime.Serialization
Imports System.Runtime.Serialization.Formatters.Binary

Namespace vbchef
    Module MCommon
        '' 将对象序列化成字节数组
        Public Function ToBytes(obj As Object) As Byte()
            If obj Is Nothing Then Return Nothing
            Using s = New MemoryStream()
                'IFormatter
                Dim f As New BinaryFormatter()
                f.Serialize(s, obj)
                Return s.GetBuffer()
            End Using
        End Function

        '' 将字节数组反序列化成对象
        Public Function ToObject(bytes As Byte()) As Object
            Using s = New MemoryStream(bytes)
                Dim f = New BinaryFormatter()
                Return f.Deserialize(s)
            End Using
        End Function
    End Module
End Namespace



代码中，ToBytes()函数用于将一个对象序列化成字节数组，ToObject()函数则用于将一个字节数组还原成对象。接下来，我们定义一个Clone()函数，将使用这两个函数完成对象的复制工作，如下面的代码（VBChef/MCommon.vb文件）。


'' 对象复制
Public Function Clone(obj As Object) As Object
    If obj Is Nothing Then Return Nothing
    Dim arrByte As Byte() = ToBytes(obj)
    Return ToObject(arrByte)
End Function



下面，我们可使用Clone()函数来测试对象的深复制操作，我们还是在Hello项目中，以CCloneDemoA对象为例，请注意，为了能够正确序列化，我们必须在CCloneDemoA类和ClassA中使用SerializableAttribute特性。

下面的就是使用Clone()函数复制CCloneDemoA对象的代码。


Dim obj1 As New CCloneDemoA()
obj1.obj = New ClassA("Tom", 35)
Dim obj2 As CCloneDemoA = CType(Clone(obj1), CCloneDemoA)
obj2.obj.Name = "Jerry"
Console.WriteLine("{0}, {1}", obj1.obj.Name, obj1.obj.Age)
Console.WriteLine("{0}, {1}", obj2.obj.Name, obj2.obj.Age)







10. 设计模式基础

在软件开发中，设计模式的应用并不是一项多么高深莫测的技术，也许你的代码结构中已经使用了一些设计模式而自己却没有察觉。那么，设计模式是什么呢？

设计模式是在软件开发过程中，针对特定的问题而使用的特定的解决方案；适用性是正确使用设计模式的唯一标准。本章，我们会讨论在VB.NET代码中应用设计模式的基础知识，主要包括两个目的截然相反的设计模式，它们是：


	策略模式，让代码结构更加灵活的基本模式。

	单件模式，让一个类只能生成一个实例的模式。



10.1. 策略模式

策略模式（Strategy Pattern），定义了一系列的算法，它们独立封装，并可以相互替换。

10.1.1. 面向对象的解决方案

首先，我们考虑一个开发问题。在生产中，我们将每个工作岗位称为一个工作站（Workstation），在工作站中，有工作的工人（Worker），以及所要进行的工作，那么，使用传统的面向对象的方法，我们应该如何定义工作站呢？

一般来讲，我们会将名词性的内容定义为字段或属性，而动词性的内容定义为方法，如下面的代码。


Public Class CWorkstation
    ''
    Public Worker As CWorker
    ''
    Public Sub DoWork()
    End Sub
End Class



代码中，我们定义了Worker字段用于表示工作站中的工人（假设已创建CWorker类），而DoWork()方法则表示在工作站中进行的工作过程。下面，如果已存在CWorker类，我们就可以创建使用它的一个工作站类型，如下面的代码。


Public Class CWorkstationA1
    Inherits CWorkstation
    Public Sub New()
        Me.Worker = new CWorker()
    End Sub
    ''
    Public Sub DoWork()
        Console.WriteLine("Worker(#1) is working.")
    End Sub
End Class



接下来，我们假设有5名工人和10项工作要做，那么工作站的设置有多少种可能呢？如果我们分别创建这些工作站对象，全部的可能包括50个（5*10），虽然，在实际应用中，我们可能不会完全使用到这些组合，但谁又说得准呢？

更可怕的是，如果增加到10名工人，20项工作要做时，工作站对象的数量会呈几何级递增。

在这里，我们定义的工作站中的元素只有两个，如果有更多的元素在其中，传统的面向对象编程方法所创建类型的数量很可能会让项目根本无法承受。

是时候改变思路了！

10.1.2. 应用策略模式

与类的继承相对应的概念是“组合”，而策略模式，或者说灵活软件架构的基本思路就是“多用组合，少用继承”。

我们还从工作站的创建来看，其中的工人属性实际上是可以任意赋值的，即我们使用不同的工人对象；如果工作内容，即DoWork()方法也能像Worker属性一样自由切换，我们创建工作站对象是不是可以灵活一些呢？接下来，我们就开始“组合”工作站对象。

当我们需要组合时，类的成员类型就不应该使用特定的类型，而是使用接口；下面就是我们在本例中所要使用的三个接口类型（Hello/PatternInterface.vb文件）。


'' 工人
Public Interface IWorker
    Property Name As String
    Property Age As Integer
End Interface

'' 工作类型
Public Interface IWorkType
    Sub DoWork()
End Interface

'' 工作站
Public Interface IWorkstation
    Property Worker As IWorker
    Property WorkType As IWorkType
    Sub DoWork()
End Interface



当然，只有接口也做不了什么工作，接下来，我们就从CWorker类开始，它实现了IWorker接口，如下面的代码（Hello/CWorker.vb）。


Public Class CWorker
    Implements IWorker
    ''
    Private myName As String
    Private myAge As Integer
    '' 年龄
    Public Property Age As Integer Implements IWorker.Age
        Get
            Return myAge
        End Get
        Set(value As Integer)
            If value >= 14 And value <= 70 Then
                myAge = value
            End If
        End Set
    End Property
    '' 名称
    Public Property Name As String Implements IWorker.Name
        Get
            Return myName
        End Get
        Set(value As String)
            myName = value
        End Set
    End Property
End Class



CWorker类的定义，相信并不难理解。

下面的代码是CWorkTypeBase类，它实现了IWorkType接口，请注意，我们将其定义为MustInherit，即抽象类，它不能创建对象（实例），只能被其它类继承使用。（Hello/CWorkTypeBase.vb）


Public MustInherit Class CWorkTypeBase
    Implements IWorkType

    Public MustOverride Sub DoWork() Implements IWorkType.DoWork
End Class



接一来的工作，我们需要创建一系列的工作类型，先来三个吧。如下面的代码（Hello/CWorkType.vb）。


''
Public Class CWorkTypeA
    Inherits CWorkTypeBase
    Public Overrides Sub DoWork()
        Console.WriteLine("WorkTypeA is working.")
    End Sub
End Class

''
Public Class CWorkTypeB
    Inherits CWorkTypeBase
    Public Overrides Sub DoWork()
        Console.WriteLine("WorkTypeB is working.")
    End Sub
End Class

''
Public Class CWorkTypeC
    Inherits CWorkTypeBase
    Public Overrides Sub DoWork()
        Console.WriteLine("WorkTypeC is working.")
    End Sub
End Class



在CWorkTypeA、CWorkTypeB和CWorkTypeC三个类中，我们重写了DoWork()方法，分别显示自己的相关信息。

最后，终于到了工作站类型的创建了，如下面的代码（Hello/CWorkstation.vb文件）。


Public Class CWorkstation
    Implements IWorkstation
    ''
    Private myWorker As IWorker
    Private myWorkType As IWorkType
    ''
    Public Sub DoWork() Implements IWorkstation.DoWork
        Console.Write("{0} : ", Me.Worker.Name)
        Me.WorkType.DoWork()
    End Sub
    ''
    Public Property Worker As IWorker Implements IWorkstation.Worker
        Get
            Return myWorker
        End Get
        Set(value As IWorker)
            myWorker = value
        End Set
    End Property
    ''
    Public Property WorkType As IWorkType  _
        Implements IWorkstation.WorkType
        Get
            Return myWorkType
        End Get
        Set(value As IWorkType)
            myWorkType = value
        End Set
    End Property
End Class



以CWorkstation类中，Worker和WorkType属性的定义很简单了，而DoWork()方法则是重点，我们可看到，在这个方法中调用了WorkType属性的DoWork()方法。

接下来，我们使用已存在的Worker和WorkType对象组合成一个CWorkstation对象，如下面的代码。


Dim worker1 As New CWorker()
worker1.Name = "Tom"
Dim worker2 As New CWorker()
worker2.Name = "Jerry"
Dim wt1 As New CWorkTypeA()
Dim wt2 As New CWorkTypeB()
Dim wt3 As New CWorkTypeC()
' 自由组合
Dim ws As New CWorkstation()
ws.Worker = worker2
ws.WorkType = wt3
' 执行工作
ws.DoWork()



代码中，我们首先创建了两个IWorker对象（worker1和worker2）和三个IWorkType对象（wt1、wt2和wt3）；然后，在CWorkstation对象（ws）中，我们可以通过不同的IWorker对象和IWorkType对象组合出不同内容的，这样，不需要修改ws对象DoWork()方法的情况下，也可以有着不同表现。以上代码执行的结果如下图。

[image: 图像说明文字]


我们可以修改ws.Worker和ws.WorkType的值，然后看看执行结果的不同。

如果我们有很多的工人和工作类型，只有创建相应的类型，就可以直接应用到CWorkstation对象中，此时，我们并不需要新的CWorksation对象，只需要使用不同的组件去组合就可以了。

以上就是策略模式的应用，我们可以看到，使用策略模式可以体现代码结构最大的灵活性，我们只需要使用最基本的组件类，就可以动态组合出所有可能的大型组件。下图中就是定义的一系列代码的的基本结构：

[image: 图像说明文字]


10.2. 单件模式

如果说策略模式体现的代码结构的灵活性，那么，单件模式（Singleton Pattern）就意味着限制，通过单件模式，我们只能使用某个类的唯一实例。

比如，在生产过程中的主调度工作，如果有多个调度，情况可能就有点混乱了；所以，对于一些重要的工作，如工作调度、系统资源分配等工作，只能有一个“当家人”，这就是单件的目的。

下面就是我们使用VB.NET代码创建的单件类的基本代码（Hello/CSingleton.vb文件）。


Public Class CSingleton
    '' 测试字段
    Public Name As String
    '' 单件对象
    Private Shared myInstance As CSingleton
    '' 私有构造函数，不能创建对象
    Private Sub New()
    End Sub
    '' 空的共享构造函数
    Shared Sub New()
    End Sub
    '' 获取单件对象的唯一方法
    Public Shared Function GetInstance() As CSingleton
        If myInstance Is Nothing Then
            myInstance = New CSingleton()
        End If
        Return myInstance
    End Function
    '' 其它成员
End Class



在CSingleton类中，字段Name用于测试工作，而其它成员则是保证单件正常工作的代码。其中，私有的（Private）构造函数和空的共享构造函数可以保证我们无法使用常规的方法创建CSingleton类的实例。

私有的myInstance字段就是唯一的实例，而我们则必须使用GetInstance()共享方法获取它。在GetInstance()方法中，如果myInstance没有实例化，我们就使用New关键字创建新的对象，如果已经实例化，则直接返回myInstance对象。

下面的代码，我们用于测试单件的有效性。


Dim s1 As CSingleton = CSingleton.GetInstance()
s1.Name = "Tom"
Dim s2 As CSingleton = CSingleton.GetInstance()
Dim s3 As CSingleton = CSingleton.GetInstance()
Console.WriteLine(s1.Name)
Console.WriteLine(s2.Name)
Console.WriteLine(s3.Name)



我们可以看到，其中的s1、s2和s3对象都由CSingleton.GetInstance()方法获取，那么，它们指向的是不是一个实例呢？代码中，我们只是将s1.Name设置为Tom，执行的结果可以看到，s2和s3的Name也都是Tom，这就说明，这三个对象实际是指向的同一对象，即我们的单件对象。

11. 事件与委托

在介绍面向对象编程时，我们没有介绍类中的一个重要成员，即事件；本章，我们就将补上这些内容。实际上，在.NET Framework中的事件，是通过委托来实现的，所以，本章我们会在讨论委托的相关主题以后，再讨论事件的实现。

11.1. 委托（Delegate）

委托是一种类型，在代码中，通过委托，可以在组件中动态地关联自定义代码，其功能类似C++中的函数指针。如下面的代码，我们定义了一个简单的委托类型DSayHello和两个方法（Hello/Delegate.vb文件）。


Module MDelegate
    Delegate Sub DSayHello()
    ''
    Sub SayHello1()
        Console.WriteLine("Hello")
    End Sub
    ''
    Sub SayHello2()
        Console.WriteLine("您好")
    End Sub
    ''
End Module



Delegate关键字用于创建委托类型，我们可以创建Sub或Function类型的委托；我们可以看到，除了Delegate关键字的使用，其它部分和创建方法的格式是一样的。代码中，我们创建的DSayHello是一个没有返回值、没有参数的委托类型。

随后的两个子程序SayHello1()和SayHello2()，我们可以看到，它们和DSayHello委托的返回值和参数设置是一样的。

接下来，我们就可以通过DSayHello委托对象来调用这两个方法，如下面的代码。


Dim d As New DSayHello(AddressOf SayHello1)
d() 'd.Invoke()



执行代码，我们可以看到输入“Hello”，即调用了SayHello1()方法。然后，我们可以将SayHello1修改为SayHello2，然后执行代码，我们可以看到会输出“您好”，即调用了SayHello2()方法。

代码中，请注意d()语句，它实际调用的就是SayHello1()方法，我们是通过AddressOf运算符将这个方法与DSayHello委托对象关联的。AddressOf运算符用于传递子程序和函数的“地址”，实际上，就是在使用“函数指针”。

我们定义的DSayHello委托可能是一个最简单的委托了，当我们需要使用包括返回值的委托时，可以使用类似如下的代码。


'' 有返回值的委托
Delegate Function DRandomBuilder() As Integer
''
Function GetRandomNumber() As Integer
    Return (New Random()).Next()
End Function



然后，我们通过下面的代码来使用RandomBuilder委托。


Dim rndNumber As New DRandomBuilder(AddressOf GetRandomNumber)
Console.WriteLine(rndNumber())



通过上面两个委托类型的使用，我们可以看到，当我们定义一个委托时，和定义子程序或函数的方法是相似的，我们需要指定返回值（或没有）和参数（或没有）；然后，我们在使用委托时，委托方法的返回值和参数设置必须与委托中定义的完全一致。

我们再来看一看在类中如何使用委托，如下面我们定义的CDelegateDemo类（Hello/Delegate.vb文件）。


Public Class CDelegateDemo
    Public Delegate Sub DSayHello()
    Public SayHello As [DSayHello]
    Public Sub DoSayHello()
        SayHello()
    End Sub
End Class



首先，我们在类的内部定义了委托类型DSayHello；然后，我们定义了一个DSayHello类型的对象SayHello；最后，在方法DoSayHello()方法中，我们调用了委托对象SayHello。

下面的代码测试了CDelegateDemo类的使用。


Dim hello As New CDelegateDemo()
hello.SayHello = AddressOf SayHello2
hello.DoSayHello()



应用中，我们可以使用AddressOf运算符根据需要切换SayHello委托对象的调用方法，这样，就在对象的外部创建了一种可以修改对象功能的方式。如果你了解了了设计模式，这实际上就是访问者模式的一种形式，访问者模式的定义就是：可以在不改变类型定义的前提下重新定义操作的具体实现。

接下来，我们就来看看在.NET Framework中关于委托的一个典型应用，即事件（Event）。

11.2. 事件（Event）

在窗体中使用控件时，我们一定不会对事件陌生，如按钮的Click事件，当我们单击按钮时，就会触发此事件，而事件中会有什么操作，则完全由开发时定义的委托方法来实现；这个过程实际上就是应用了委托。

我们先从Windows窗体控件Button中关于Click事件的定义看起吧，如下面的代码。


Public Event Click As EventHandler



其中，Event关键字说明Click是一个事件；那么，EventHandler是什么类型呢？我们接着看。


Public Delegate Sub EventHandler ( _
    sender As Object, _
    e As EventArgs _
)



我们可以看到，EventHandler是一个没有返回值的委托类型，但定义了两个参数。现在，我们在Windows窗体中看一看Click事件是怎么工作的。我们创建一个新的项目，选择“Windows窗体应用程序”模板，并命名为WinFormDemo；项目创建后，会自动创建一个名为Form1的窗体，我们接下来的代码就们于这个窗体。

首先，我们从工具箱中选择Button按钮，然后双击或拖动到Form1窗体中，在窗体中，我们双击Button1按钮，输入如下面代码。


Public Class Form1
    Private Sub Button1_Click(sender As Object, e As EventArgs) _
        Handles Button1.Click
        MsgBox("Click事件响应")
    End Sub
End Class



执行程序，并单击Button1按钮，会弹出如下图所示的消息对话框。

代码中，我们可以看到，Button1_Click()方法的定义与EventHandler委托的定义是一致的（两个参数，以及没有返回值）。

请注意代码Handles Button1.Click。在VB.NET中，Handles关键字用于将方法关联到事件（委托对象），在这里就是将Button1_Click()方法作为Button1按钮的Click事件的响应方法，这是开发工具为我们自动生成的事件响应方法，其命名格式为：


<控件名称>_<事件名称>(<参数>)



只要返回值和参数与委托的定义一致，方法的名称可以随意；但在一般情况下，我们还是建议使用默认的事件委托方法名，毕竟它比较直观。 不过，如果多个控件的响应代码有关联性时，我们也可以使用一个方法同时处理多个控件的事件。比如，我们在Form1窗体中再创建一个按钮Button2，然后，修改代码如下：


Public Class Form1
    Private Sub Button1_Click(sender As Object, e As EventArgs) _
        Handles Button1.Click, Button2.Click
        MsgBox(String.Format("{0}控件的Click事件响应", _
                             CType(sender, Button).Name))
    End Sub
End Class



再次运行代码，并分别单击Button1按钮和Button2按钮，我们会看到如下图所示的两个消息对话框。

接下来，我们自己定义一个事件来看一看，如下面的代码（CEventDemo.vb文件）。


Public Class CEventDemo
    ''
    Public Event SayHello(sName As String)
    ''
    Public Sub DoSayHello(sName As String)
        RaiseEvent SayHello(sName)
    End Sub
End Class



我们可以看到，定义事件时需要使用Event关键字，而在触发事件的时候，我们使用RaiseEvent关键字。下面，我们测试这个事件的使用。


Module Module2
    Sub SayHello1(sName As String)
        Console.WriteLine("Hello, " & sName)
    End Sub

    Sub SayHello2(sName As String)
        Console.WriteLine("您好, " & sName)
    End Sub

    Sub Main()
        Dim hello As New CEventDemo()
        AddHandler hello.SayHello, AddressOf SayHello1
        AddHandler hello.SayHello, AddressOf SayHello2
        hello.DoSayHello("Tom")
    End Sub
End Module



我们先看看代码执行的结果，如下图所示。

我们可以看到，调用DoSayHello()方法显示了两条信息，而这两条信息分别定义在SayHello1()和SayHello2()方法中的，它们是如何一起工作的呢？。真正起作用的还是SayHello事件，只是在Main()方法中，我们把两个方法同时绑定到了这个事件上。

代码中，可能大家还没使用过的就是AddHandler语句，它的功能就是将方法（参数二）绑定到指定的事件上（参数一）。请注意，参数二中，我们同样使用AddressOf运算符获取方法的“指针”。

我们使用两个AddHandler语句，在这里我们需要清楚的一个概念就是，这执行的实际上是一个“累加”的操作，而不是“替换”；也就是说，我们可以将多个方法都添加到一个事件的响应中，这样，事件实际上就形成了一个响应方法的列表，然后，事件触发时，就会逐一执行这个列表中的方法。

在这里，事件中的方法列表实际上是形成了一个多路播放委托（简称多播委托，MulticastDelegate）。不过，有一点应该注意，那就是形成多播委托的委托类型必须是没有返回值的，也就是关联的方法应该是Sub定义的子程序。





12. 泛型

泛型（Generic）是.NET Framework 2.0中添加的，用于定义算法及功能模板，可以简化很多问题，以提高开发效率；在代码编译时，会绑定具体的数据类型，提高执行效率。本章，我们讨论在VB.NET中使用泛型的相关问题，主要内容包括：


	泛型方法

	泛型类

	约束



12.1. 泛型方法

假如我们需要一个方法，其功能是交换两个变量的值，如：


Sub Swap(ByRef num1 As Integer, ByRef num2 As Integer)
    Dim temp As Integer = num1
    num1 = num2
    num2 = temp
End Sub



我们可以看到，此方法只能交换Integer类型的数据，那么，还有众多的数据类型，如果它们都需要使用交换功能呢？如果不使用泛型，我们有两个方案可以使用：


	将参数定义为Object，然后进行必要的转换（装箱和拆箱），但这样的执行效率会很低。

	重载多个数据类型的Swap()方法，代码编写工作量会很大。如




Sub Swap(ByRef num1 As Decimal, ByRef num2 As Decimal)
    Dim temp As Decimal = num1
    num1 = num2
    num2 = temp
End Sub
'' 更多类型的方法重载版本......



还记得我们关于泛型的说明吗？就在本页最上面。^_^

使用泛型同时解决了代码编写效率和执行效率两方面的问题，何乐而不为呢？如下面的代码，就是我们定义的关于交换数据的泛型方法。


Sub Swap(Of T)(ByRef num1 As T, ByRef num2 As T)
    Dim temp As T = num1
    num1 = num2
    num2 = temp
End Sub



变化不是很大，但这一个方法就可以处理所有类型的数据交换工作了。注意其中的几点：

在方法名后面的一对小括号里，我们使用Of关键字定义了一个类型标识；请注意，这只是个标识，并不是真正的类型，所以，你可以使用自己喜欢的标识名称。此外，如果在方法中需要使用多个类型标识，可以使用逗号将它们分隔，如(Of T,U)。

定义了类型标识以后，我们就可以在参数、返回值和方法的内部使用这个标识。不过，在这里应注意，因为在开发时并不知道真正的数据类型，所以，也就不能执行一些特殊的操作，如算术运算等。

第三点是个题外话。请注意Swap()方法的参数，它们定义为按引用传递（ByRef），为什么呢？因为我们在方法中真的要交换它们的值，如果按值传递就不能完成这个艰巨的任务了。

接下来，我们看一看如何使用这个泛型方法。


Dim num1, num2 As Integer
num1 = 10
num2 = 99
Console.WriteLine("num1={0}, num2={1}", num1 , num2)
Swap(Of Integer)(num1, num2)
Console.WriteLine("num1={0}, num2={1}", num1 , num2)



代码执行结果如下图。

我们可以看到，在调用泛型方法时，需要使用Of关键字将类型标识指定为具体的数据类型，这样在执行过程中，在泛型方法中就会自动绑定实际的数据类型。

12.2. 泛型类

定义类时，我们同样可以使用泛型，如下面的代码。


Public Class CGenericDemo(Of T)
    Private myData As New ArrayList()
    ''
    Public Sub Add(item As T)
        myData.Add(item)
    End Sub
    ''
End Class



我们可以看到，定义泛型类也很简单，在这里，我们定义的CGenericDemo类中，会统一使用数组列表对象（ArrayList）中的成员类型（T）；下面，我们就来测试这个泛型类的使用。


Dim obj As New CGenericDemo(Of Integer)
Dim w As New CWorkA()
obj.Add(w)



执行此代码会出现编译错误，因为我们指定CGenericDemo中的类型应该应该是Integer，而在obj.Add()方法中使用了CWorkA类型的对象。如果我们将此obj.Add()方法的参数设置为一个Integer类型的数据就不会有问题了。

实际上，在System.Collections.Generic命名空间中，已经为我们定义了大量的泛型集合类型，比起System.Collections命名空间中的非泛型集合类型，泛型集合的执行效率要高的多，因为在运行过程中不需要在Object和实际类型之间进行转换。

12.3. 约束

在使用泛型类的过程中，我们可以添加一起约束，比如要求类型必须实现了某一个或几个接口类型，如下面的代码。


Public Class CGenericDemo(Of T As IWorker)
    '' 类定义代码
End Class



代码中，我们指定泛型类中使用的类型必须是实现了IWorker接口的类型。





13. 异常处理

恐怕没有人敢说自己的软件百分之百没有问题，不过，有问题不怕，重要的是在软件执行过程中能够正确的、合理的，以及智能地处理问题。本章，我们就讨论相关的处理方法，主要包括：


	VB6风格的异常处理

	Try...End Try语句结构

	Throw语句



13.1. VB6风格的异常处理

使用过VB6的朋友应该都知道相关的错误处理方式，主要包括两种：


	On Error Goto <标签名称>，当出现错误时转换到指定标签的位置。

	On Error Resume Next，当出现错误时继续执行下一行代码。



在VB.NET中，我们依然可以使用这两种语句来处理错误。先来看看“On Error Goto <标签>”语句的的应用，它一般放在代码块的顶部，用于说明本代码块中的错误处理方法，如下面的代码。


Sub Main()
    On Error GoTo MainError
    Dim num1, num2, result As Integer
    num1 = 16
    num2 = 0
    result = num1 / num2
    Exit Sub
MainError:
    Console.WriteLine("出错错误: " & Err.Description)
End Sub



代码中，我们将除数设置为0，这样在执行除法运算时就会产生错误。再看几个关于错误处理的相关语句：

On Error Goto MainError语句，其中，MainError是我们定义的一个标签。

Exit Sub语句，此语句很重要，如果没有此语句，我们会发现，即使执行没有错误，也会执行MainError标签后面的代码。所以，在这种错误处理机制中，一定不要忘了在需要执行的代码结束时使用一条Exit语句来结束当前代码块，代码块可以是Sub、Function、Property。

Err对象，Err对象是VB.NET内置的一个对象，用于保存程序执行过程中的最后一个错误的相关信息，其的Description属性就是对于错误的描述信息。其常用的成员还有：


	Err.Clear()方法，清除所有错误信息。如果错误已被正确处理，我们可以调用此方法进行清理。

	Err.Number属性，获取错误号。



另一个VB6风格的错误处理语句就是“On Error Resume Next”，其功能就是指定，当代码出现错误时，跳过错误代码，继续执行其下面的代码。一般来讲，我们并不应该忽略代码执行中的错误，不过，如果错误是可以预知的，并且错误不会影响后续代码执行的时候，可以选择此方法进行处理。

13.2. Try...End Try语句结构

在VB.NET中，增加了新的异常处理结构，我们可以使用Try...End Try语句，其基本使用格式如下。


Try
    <代码块1>
Catch ex As Exception
    <代码块2>
Finally
    <代码块3>
End Try



其中：


	<代码块1>是我们要执行的主代码，执行过程可能会出现异常哦！

	<代码块2>是通用的异常处理代码，我们使用Catch语句捕捉一个异常类型，代码中的Exception类就是一个通用的异常类，也是其它异常类的的父类；Exception类定义在System命名空间，大家可以在Help中查看它的完整定义。此外，如果只需要捕捉异常，而不需要处理它，可以使用一条Catch语句，不需要包含异常对象。如：




Try
    '主代码块
Catch
    ' 异常后执行的代码块
End Try




	<代码块3>和Finally语句，无论<代码块1>中是否出现异常，都会执行此处的代码；在这里，我们可以对一些资源进行释放操作。而我们前面提到的Using...End Using语句结构实际上就是一个隐式的Try...Finally...End Try语句结构，只不过Finally部分的代码是通过Dispose()方法自动调用的。请注意，如果没需要清理的代码，则Finally语句部分可以没有。



下面的代码，我将使用Exception类来处理除0的错误。


Try
    Dim num1, num2, result As Integer
    num1 = 15
    num2 = 0
    result = num1 / num2
    Console.WriteLine("{0}/{1}={2}", num1, num2, result)
Catch ex As Exception
    Console.WriteLine(ex.Message)
Finally
    Console.WriteLine("全部工作完成，不知道对不对！^_^")
End Try



代码执行结果如下图所示。

我们可以看到，当Try语句后的代码出现错误后，会使用Catch语句后的代码处理，我们使用Exception对象的Message属性显示了异常的描述信息；最后，无论结果如何，都会执行Finally语句后面的代码，我们可以修改num2的值来观察代码执行的结果。

此外，在.NET Framework框架中，还定义了很多具体的异常类型，它们都是Exception类的子类；我们可以使用多个Catch语句来处理具体的异常类型，以便针对特定异常做出更精确的处理工作。在处理异常时，有以下常用成员获取异常的信息：


	Message属性，String类型，异常的描述信息。

	Source属性，String类型，描述异常产生的位置信息。

	HelpLink属性，String类型，对于异常的帮助资源链接。可以是URL地址，也可以是本地文件，如“file:///c:\help\1.html”。



13.3. Throw语句

Throw语句用于抛出一个异常，如：


Throw New Exception("随便抛个异常玩玩")



如果是在Catch语句块中，如果我们无法处理异常，可以再次向上级（如操作系统）中抛出这个异常，此时，我们只需要一个单独的Throw语句就可以了，如：


Try
    ' 可能产生错误的代码
Catch ex As Exception
    Throw
End Try







14. 数组与集合

本章，我们将讨论集合类型处理的相关内容，主要包括：


	VB风格数组

	Array类

	ArrayList与List泛型类

	Hashtable与Dictionary泛型类



14.1. VB风格数组

在VB6中，数组的处理是其比较强大的，但和其它编程语言相比，也有些不同，接下来，我们就来看看到了VB.NET中，数组是如何处理的。

14.1.1. 定义和访问数组

首先，在VB中，数组的定义是使用一对小括号，而不是C风格编程语言中的一对中括号，如下面的代码，我们就定义一个10个Integer类型成员的数组。


'定义数组，10个成员
Dim arrInt(9) As Integer
'赋值
For i = 0 To 9
    arrInt(i) = i + 1
Next
'显示
For i = 0 To 9
    Console.WriteLine(arrInt(i))
Next



请注意，VB中的数组与C风格数组最大的不同点就是，在定义时，VB数组的下标定义的是数组中最大的下标，而C风格数组的下标定义的是数组成员的数量。如上述代码中，我们定义的arrInt数组的下标是9，那么这个数组的索引值就是0到9，共10个成员。

此外，我们还可以在定义数组时不指定下标，而直接定义其成员，如下面的代码。


Dim arrInt() As Integer = {1, 1, 2, 3, 5, 8, 13}



那么，在使用这样的数组时，我们如何判断它的索引值范围呢？此时，我们可以使用LBound()和UBound()方法。

请注意，与VB6中的数组不同，在VB.NET中，数组的索引只能从0开始，而不能自己定义索引的范围。

14.1.2. LBound()和UBound()方法

当我们使用VB数组时，经常使用类似下面的代码。


Dim arrInt() As Integer = {1, 1, 2, 3, 5, 8, 13}
Dim minIndex As Integer = LBound(arrInt)
Dim maxIndex As Integer = UBound(arrInt)
For i = minIndex To maxIndex
    Console.WriteLine(arrInt(i))
Next



代码中，我们使用LBound()方法获取数组的最小索引值，使用UBound()方法获取数组的最大索引值，然后，我们再通过循环访问数组成员，这样就保证数组成员的访问不会超出索引范围了。

14.1.3. ReDim语句和Preserve关键字

ReDim语句用于重新指定数组中成员的数量，如下面的代码。


Dim arrInt() As Integer = {1, 1, 2, 3, 5, 8, 13}
ReDim arrInt(3)



使用ReDim语句时有以下几点需要注意：


	重置数组大小时，不能改变成员类型。

	括号中的数值同样是指定数组的最大索引值。

	重新指定数组的成员数量后，数组成员的数据会丢失，并恢复到成员类型的默认值。

	如果新的成员数量比原数组成员数量小，则会从数组的后部截断。

	如果新的成员数量比原数组成员数量大，则会在原数组的后面添加成员，新成员的数据为成员类型的默认值。



如果我们在重新指定数组成员数量时需要保留原数组成员的数据，可以在ReDim语句中使用Preserve关键字，如下面的代码。


Dim arrInt() As Integer = {1, 1, 2, 3, 5, 8, 13}
ReDim Preserve arrInt(3)



此时，arrInt数组会包含4个成员，即1、1、2和3。请注意，即使使用了Preserve关键字，如果指定的新成员数量小于原数组成员数量，也会从数组的后部截断。如果新成员数量大于原数组成员数量，则会在原数组后部添加相应数量的成员，新成员的数据为成员数据类型的默认值。

14.1.4. 多维数组

前面，我们使用的数组称为一维（Rank）数组，但在开发中，我们可以使用多维数组来表示相应的数据结构，如我们表示一个二维矩阵结构，就可以使用一个二维数组，如下面的代码，我们定义了一个5行，3列的矩阵结构，然后赋值并显示。


 '定义数组，10个成员
Dim matrix(4, 2) As String
' 定义数组成员内容
For row = 0 To 4
    For col = 0 To 2
        matrix(row, col) = String.Format("({0},{1})", row, col)
    Next
Next
'显示
For row = 0 To 4
    For col = 0 To 2
        Console.Write(matrix(row, col))
    Next
    Console.WriteLine()
Next



执行此代码，我们可以得到如下图所示的结果。

需要更多的数组操作，我们还是先看看Array类吧。

14.2. Array类

实际上，在.NET Framework中的数组都反映为一个Array对象，我们使用的VB.NET中的数组也是这样，这样一来，我们就可以使用Array对象中定义的成员来操作数组了。Array类定义在System命名空间，大家可在Help中查看它的详细定义。

下面，我们就来了解一些Array类的常用成员。

14.2.1. Rank属性

Rank属性返回数据的秩，即数组的维度，如：


Dim arrInt(4, 2) As Integer
Console.WriteLine(arrInt.Rank)



此代码会显示2，即这里定义的arrInt数组是一个二维数组。

14.2.2. Length属性与GetLength()方法

Length属性返回数组中所有成员的数量，如果是多维数组，则是各个维度的成员数量相乘，如下面的代码。


Dim arrInt(4, 2) As Integer
Console.WriteLine(arrInt.Length)
Dim arrStr(3) As String
Console.WriteLine(arrStr.Length)



第一行输出代码会显示15，即5*3的结果；第二行代码显示4，即arrStr数组定义了4个成员。

如果我们需要得到多维数组中某一维的成员数量，应该使用GetLength()方法，此方法有一个参数，用于指定数组的维度，这个维度和索引一样是从0开始的。如下面的代码。


Dim arrInt(4, 2) As Integer
Console.WriteLine(arrInt.GetLength(0))



代码会显示5，即第一维包括5个成员（0到4）。

14.2.3. GetLowerBound()和GetUpperBound()方法

GetLowerBound()和GetUpperBound()方法分别是用于获取指定维度的最小索引和最大索引值，它们的参数同样是从0开始的维度值，如下面的代码。


Dim arrInt(4, 2) As Integer
Console.WriteLine(arrInt.GetLowerBound(1))
Console.WriteLine(arrInt.GetUpperBound(1))



代码会分别显示0和2，即数组的第二维索引值是从0到2。

14.2.4. Resize()共享方法

Resize()方法用于修改数组的成员数量，如下面的代码。


Dim arrInt() As Integer = {1, 2, 3}
Array.Resize(arrInt, 5)
For i = 0 To 4
    Console.WriteLine(arrInt(i))
Next



我们可以看到，使用Array.Resize()方法修改数组成员数量时，会保留原数组成员的数据，这和使用ReDim Preserve arrInt(4)语句的效果是一致的；而这往往也是我们所期望的处理方法。

Array.Resize()方法有很多重载版本，我们可以参考Help了解它们的使用。

14.2.5. Sort()共享方法

Sort()方法用于对数组成员进行排序，如下面的代码。


Dim arrInt() As Integer = {4, 6, 5, 1, 2, 3}
Array.Sort(arrInt)
For i = 0 To 5
    Console.WriteLine(arrInt(i))
Next



代码会显示排序后的数组成员，即1、2、3、4、5、6。Array.Sort()方法同样有很多重载版本，我们可以参考Help了解它们的使用。

14.2.6. Reverse()共享方法

Reverse()方法用于将数组中的成员反向排列，如下面的代码。


Dim arrInt() As Integer = {4, 6, 5, 1, 2, 3}
Array.Reverse(arrInt)
For i = 0 To 5
    Console.WriteLine(arrInt(i))
Next



代码会显示3、2、1、5、6、4。

14.3. ArrayList类与List泛型类

我们可看到，虽然VB风格的数组和Array类使用起来是挺简单的，但某些功能还是不易完成，如我们需要在指定的位置添加一个新成员，或者是删除、修改某个数组成员。还好，在.NET Framework中为我们提供了不少这一类的开发资源，它们主要定义在如下命名空间中：


	System.Collections命名空间，定义了一系列集合类操作资源，如ArrayList、Hashtable、Queue、Stack等类型。

	System.Collections.Generic命名空间，定义了一系列泛型集合操作资源，如Dictionary、List、Queue、Stack等泛型类。



本章接下来内容，我们将介绍几种常用的集合和泛型集合类型的使用，首先看一看ArrayList类和List泛型类。

ArrayList类和List泛型类主要的功能就是对一个有序的集合（列表）进行管理和操作，我们可以通过索引访问和操作集合成员。下面的代码，我们先看看ArrayList类的使用。


Dim arrList As New ArrayList()
arrList.Add("abc")
arrList.Add(1)
arrList.Add(New ClassA())
For Each item In arrList
    Console.WriteLine(item)
Next



代码中，我们使用ArrayList对象的Add()方法添加了三个成员，分别是字符串、整数和ClassA对象；这里，我们可看到，ArrayList类的使用是非常灵活的，实际上，它的成员类型被定义为Object，也就是说，我们可以将任何类型的数据添加到ArrayList对象中。

不过，有讨论泛型的时候，我也提过，如果将Object类型作为“万能类型”的话，是存在一定的性能问题的。而泛型则可以在一定程度上解决这个问题，不过，前提是集合中的成员类型是一致的，此时，我们可以使用List泛型类来操作，如下面的代码。


Dim lst As New List(Of String)
lst.Add("abc")
lst.Add("def")
lst.Add("ghi")
For Each item As String In lst
    Console.WriteLine(item)
Next



实际上，如果我们打开Help文档就可以发现，ArrayList类（System.Collections命名空间）和List泛型类（System.Collections.Generic命名空间）的成员定义非常相似，只是在用途上的些区别，即：


	当列表集合中的成员类型一致时，我们应该使用List泛型类，以提高代码开发和执行时的效率。

	当列表集合中的成员类型不一致时，应使用ArrayList对象来处理，以保证成员处理的灵活性。



了解了ArrayList类和List泛型类的基本用途和区别，我们再来看一看在这两个类型中都有哪些常用成员。

14.3.1. 访问列表成员

首先，我们可以使用Count属性获取列表中成员的数量。

在访问列表成员时，我们可以使用从0开始的索引值，此时实际调用的是Item属性，它被定义为默认属性，如lst(0)。

此外，如果我们需要快速访问每一个成员，可以使用For Each...Next语句语句结构，前面的代码中，我们已经看到了相关应用。或者，我们可以使用索引值循环，如下面的代码。


Dim maxIndex As Integer = lst.Count - 1
For i = 0 To maxIndex
    Console.WriteLine(lst(i))
Next



14.3.2. 添加成员

Add()和Insert()方法用于向列表集合添加新的成员。

其中，Add()方法会在列表的末尾添加一个成员，前面，我们已经看它的使用了。Insert()方法用于在指定的索引位置插入一个集合成员，如下面的代码。


Dim arrList As New ArrayList()
arrList.Add("abc")
arrList.Add("ghi")
arrList.Insert(1, "def")
For Each item In arrList
    Console.WriteLine(item)
Next



如果我们需要将一个列表添加到另外一个列表，可以使用InsertRange()方法，其中，第一个参数为插入开始的索引值，第二个参数为需要插入的集合，如下面的代码。


Dim lst1 As New ArrayList()
lst1.Add("abc")
lst1.Add("def")
Dim lst2 As New ArrayList()
lst2.Add("ghi")
lst2.Add("jkl")
'
lst1.InsertRange(lst1.Count, lst2)
For Each item In lst1
    Console.WriteLine(item)
Next



代码中，我们分别定义了两个ArrayList对象（lst1和lst2），然后，我们将lst2的内容全部添加到lst1列表的末尾。

请注意InsertRange()方法的第一个参数，我们使用了lst1.Count属性，这样插入的位置就是它的末尾。

14.3.3. 删除成员

Clear()方法用于清除列表中的所有内容。

如果我们需要删除指定索引位置的成员，可以使用RemoveAt()方法，其参数就是需要删除成员的索引值，请注意，索引值是从0开始的。

Remove()方法同样用于删除成员，只不过它的参数指定需要删除的成员内容，并且只会删除第一个匹配的成员。

另一个删除成员的方法是RemoveRange()方法，它的第一个参数是删除开始的索引值，第二个参数是删除的成员数量。

14.4. Hashtable类与Dictionary泛型类

Hashtable类和Dictionary泛型类用于处理无序集合对象，集合的每个成员都是一个“键(Key)/值(Value)”结构。实际应用中，我们可以使用“键”来标识一个数据项（值），也就是通过“键”索引来访问数据项，如下面的代码演示了Hashtable对象的使用。


Dim worker As New Hashtable()
worker.Add("Name", "Tom")
worker.Add("Age", 35)
For Each key In worker.Keys
    Console.WriteLine("{0} = {1}", key, worker(key))
Next



代码中，和我们访问列表对象不同，我们首先通过For Each...Next语句逐一访问Hashtable对象worker中所有的键（key)，然后，通过键访问相应的数据，如worker(key)。

使用Dictionary泛型类时就注意，只在键的类型和值的类型都分别相同时才应该使用。接下来，我就以真正的字典对象来演示Dictionary泛型类的使用，如下面的代码。


Dim dict As New Dictionary(Of String, String)
dict.Add("sun", "太阳")
dict.Add("earth", "地球")
dict.Add("moon", "月亮")
For Each key As String In dict.Keys
    Console.WriteLine("{0} : {1}", key, dict(key))
Next



从Hashtable类和Dictionary泛型类的应用代码中，我们可以看到，它们的使用方法非常相似，区别只在于对于键和值类型的控制。Hashtable对象中，键和值都定义为Object类型，而在Dictionary对象中，我们需要在开发时分别确定键和值的具体类型，这样的代码执行效率会更高。

接下来，我就来看看Hashtable类和Dictionary泛型类中有哪些常用成员。

14.4.1. 成员访问

Count属性可以返回成员的数量。

Item属性定义为默认属性，其参数为“键（Key）”，如worker("Name")；其返回的属性值即为成员数据（值，Value）。

Keys属性返回所有键的集合，相应的，我们可以使用Values属性获取所有值的集合。它们分别是ICollection接口类型的对象，可以使用For Each语句快速访问所有成员。

Contains()和ContainsKey()方法的功能相同，都是判断集合中是否存在参数所指定的键。

ContainsValue()方法用于判断集合中是否存在所指定的值。

14.4.2. 添加成员

前面我们已经使用过Add()方法，它的功能就是在集合添加一个成员，其使用格式为：


Add(<键>, <值>)



请注意，Hashtable类和Dictionary泛型类中的成员是无序的，我们无法使用固定的数值索引来访问它们，实际应用中，我们也不能假设成员在集合的顺序。

14.4.3. 删除成员

Clear()方法会清除所有的成员。

Remove()方法用于删除指定键的成员，如dict.Remove("moon")。





15. 字符串与散列算法

本章，我们讨论字符串与散列函数的相关内容，主要包括：


	VB中的字符串运算符

	String类型

	StringBuilder类

	散列算法

	获取GUID



15.1. VB中的字符串运算符

VB.NET中，字符串字面量使用一对双引号包含，我们可以使用&运算符连接字符串。如:


Dim str1 As String = "abc"
Dim str2 As String = "def"
Console.WriteLine(str1 & str2)



另一个比较实用的是Like运算符，它可以帮助我们进行简单地字符串匹配测试，其基本格式如下：


<字符串> Like <匹配模式>



Like运算符的运算结果是Boolean类型，<字符串>中的内容与<匹配模式>相匹配时，运算结果返回True，否则返回False。其中，<匹配模式>也是字符串，但我们可以使用一些通配符来匹配相应的内容，如：


	?符号，表示一个任意字符。

	*符号，表示零或多个字符。

	#符号，表示0到9的一个数字。

	[<字符列表>]，表示<字符列表>中的任意一个字符。可以使用连接号（-）指定字符或数字范围。

	[!<字符列表>]，表示<字符列表>以外的任意一个字符。



如下面的代码。


Console.WriteLine("9" Like "#")
Console.WriteLine("abc123" Like "*123")
Console.WriteLine("abcdef" Like "ab*")
Console.WriteLine("caohuayu@sina.com" Like "*@*.*")



代码中会显示四个True。

在VB代码中，对于简单的匹配，我们完全可以使用Like运算符，而不是复杂的正则表达式。

15.2. String类型

VB.NET中的字符串实际上就是一个String对象；而String类是引用类型，但在应用过程中，它的表示却更像一个值类型，这是为什么呢？

在.NET Framework中的System命名空间定义的String类，其功能是定义一个不可变字符串对象；也就是说，一个字符串一旦被创建，我们就不能修改它的内容，而对于字符串的任何修改都会生成一个新的字符串对象。所以，当我们需要对文本进行大量的操作时，就需要考虑它的性能问题了。

既然所有的字符串都是String对象，那么，我们就可以使用String类中定义的成员来操作字符串了，我们接下来就看看常用的字符串操作有哪些。

在我们操作字符串之前，可以使用它的Length属性获取其中的字符数据，Length定义为只读属性，返回值为Integer类型。

15.2.1. 合并

Join()方法用于将一个字符串数组合并为一个字符串，如下面的代码。


Dim arrStr() As String = {"abc", "def", "ghi", "jkl"}
Dim str1 As String = String.Join(",", arrStr)
Console.WriteLine(str1)  ' abc,def,ghi,jkl



代码会输出abc,def,ghi,jkl。我们从中可以看到Join()方法的参数设置：


	参数一设置用于连接字符串数组成员的符号，如果不需要分隔符号，可以使用空字符中""。

	参数二是需要合并的字符串数组对象。

	我样还可以设置Join()方法的第三和第四个参数，用于设置合并字符串数组的部分成员；其中，第三个参数为需要合并的起始索引值，第四个参数为合并的成员数量。



Insert()方法用于在指定的索引位置插入内容；其中，参数一指定插入位置的索引值（从0开始）；参数二指定需要插入的内容。如下面的代码。


Dim str1 As String = "abc"
Dim str2 As String = str1.Insert(1,"***")
Console.WriteLine(str2) ' a***bc



15.2.2. 截取与分割

Substring()方法用于截取字符串的某一部分，第一个参数为开始截取的索引值，参数二为截取的字符数量，如果截取开始位置以后所有内容，可以不设置第二个参数。如下面的代码。


Dim str1 As String = "abcdefg"
Console.WriteLine(str1.Substring(1,3))  'bcd
Console.WriteLine(str1.Substring(3))  'defg



Split()方法用于分割字符串，并返回分割后的字符串数组；常用的参数是第一个，即用于分割的字符（Char类型），如下面代码。


Dim str1 As String = "abc,def,ghi,jkl"
Dim arrStr() As String = str1.Split(","c)
For Each item As String in arrStr
    Console.WriteLine(item)
Next



Chars默认属性，可以根据索引值返回相应的字符（Char类型）。如下面的代码。


Dim str1 As String = "abcdefg"
Console.WriteLine(str1(1))  'b



如果我们需要字符串中的字符数组，可以使用ToCharArray()方法，如果没有使用参数则返回所有字符的数组，如果需要某一部分，可以在参数一指定开始的索引位置，参数二指定需要的字符数量。

15.2.3. 大小写转换

ToLower()方法用于将字符串中的字母转换为小写，并返回新的字符串对象。ToUpper()方法用于将字符串中的字母转换为大写，并返回新的字符串对象。

ToLowerInvariant()和ToUpperInvariant()方法完成同样的工作，只是会考虑系统中区域性设置中的大小写相关规则。

15.2.4. 查询与替换

Contains()方法用于判断字符串中是否存在指定的内容，返回值类型为Boolean，如下面的代码。


Dim str1 As String = "abcdefg"
Console.WriteLine(str1.Contains("bcd"))  'True



Replace()方法用于将字符串中的内容替换为指定的内容，其参数一为字符串中需要替换的内容，参数二指定新的内容，两个参数可以同时使用Char类型或同时使用String类型。如下面的代码。


Dim str1 As String = "abcdefgabcdefg"
Console.WriteLine(str1.Replace("abc", "***"))  '***defg***defg
Console.WriteLine(str1.Replace("d"c, "*"c)) 'abc*efgabc*efg



IndexOf()方法用于返回与参数指定的字符或字符串相匹配的第一个索引位置，如：


Dim str1 As String = "abcdefg"
Console.WriteLine(str1.IndexOf("bcd")) '1



相应的，LastIndexOf()方法用于返回与参数指定的字符或字符串相匹配的最后一个索引位置。

StartsWith()方法用于判断字符串的开始部分（前缀）是否与参数指定的内容相匹配，返回类型为Boolean。请注意，字符串内容比较时会区分字母的大小写，如下面的代码。


Dim str1 As String = "abcdefg"
Console.WriteLine(str1.StartsWith("Abc")) 'False
Console.WriteLine(str1.StartsWith("abc")) 'True



EndsWith()方法用于判断字符串的结束部分是否与参数指定的内容相匹配。

15.2.5. 内容比较

Compare()定义为类方法（共享方法），用于比较两个字符串是否相等，其中，参数一和参数二指定需要比较的两个字符串，参数三指定是否忽略字母的大小写，默认为False，即区分字符的大小写。


Console.WriteLine(String.Compare("abc", "ABC")) '-1
Console.WriteLine(String.Compare("abc", "ABC", True)) '0
Console.WriteLine(String.Compare("abc", "abc")) '0



请注意，Compare()方法的返回值是Integer类型，只有返回值为0时，两个字符串的内容才相等。

15.2.6. 格式化字符串

Format()定义为类方法，用于格式化字符串及相关数据，其功能与Console.WriteLine()方法相似，我们可以对众多的数据进行格式化，动态生成字符串。如下面的代码：


Dim num1, num2 As Integer
num1 = 10
num2 = 99
Dim str1 As String
str1 = String.Format("{0}+{1}={2}", num1, num2, num1 + num2)
Console.WriteLine(str1)



实际上，以上代码可以使用Console.WriteLine()方法直接定义，我们这里只是为了演示String.Format()方法的使用。

格式化字符串时，我们除了{<索引>}来指定数据位，还可以同时指定数据的格式，一个数据位的格式定义如下：


{索引[,对齐][:格式字符串]}



其中，索引指定数据的位置。对齐为可选部分，指定数据占用的最小字符数量；正数为右对齐，负数为左对齐。格式字符串同样为可选部分，用于指定数据的显示格式，常用的格式包括：

数字格式，常用的格式化字符包括：d（Decimal类型）、e（科学记数法）、x（十六进制数）。此外，我们还可以自定义数字的格式，常用的格式化字符包括：0（0占位符）、#（数字占位符）、.（小数点位）、,（千分位）、%（显示为百分数），以及\指定的转义字符，如\n（换行符）、\t（制表符）、\（\字符）等等。

日期与时间。常用的格式格式式字符包括：d（短日期）、D（长日期）、f（长日期与短时间格式）、F（长日期与长时间格式）、g（标准的短日期和短时间格式）、G（标准的短日期和长时间格式）、t（短时间格式）、T（长时间格式）。此外，我们还可以使用一些自定义的日期与时间格式定义，如“yyyyMMdd”定义为八位的年月日、“HHmmss”表示六位的时（24小时制）分秒。下一章，我们将讨论更多的关于日期与时间处理的相关内容。

显示枚举值，默认情况下，当我们显示枚举类型的数据时，会显示数字。如果我们需要显示枚举值的文本格式，可以使用g或f格式化字符。此外，如果需要使用十六进制的枚举值，同样可以使用x格式化字符。 下面显示了一些常用的格式化字符串。


Dim num1 As Integer = 123
Dim num2 As Double = 200.123
Console.WriteLine("{0,-8},{1,-8}", num1, num2) '占8字符左对齐
Console.WriteLine("{0:x}", num1) '十六进制
Console.WriteLine("{0:#.##},{1:#.##}", num1, num2) '两位小数



代码执行结果如下图。

[image: 图像说明文字]


Trim()方法用于清除字符串前部和后部的指定字符，如果不指定参数，则清除空白字符（如空格、制表符等）；如果需要清除指定的字符，可以在参数中使用一个字符数组。

TrimStart()和TrimEnd()方法，其功能与Trim()相似，只是它们分别用于清除字符串前部或后部的指定内容。

PadLeft()方法用于指定字符串占用的最小字符数量，并且在内容的左边补齐填充字符；参数一用于指定字符宽度，参数二则指定空白位置的填充字符，默认为空格符。PadRight()方法功能与PadLeft()方法相似，只是在字符串的右边补齐填充字符。如下面的代码。


Dim str1 As String = "abc"
Console.WriteLine(str1.PadLeft(8))
Console.WriteLine(str1.PadRight(8, "*"c))



代码执行结果如下图。

[image: 图像说明文字]


15.3. StringBuilder类

StringBuilder类定义在System.Text命名空间，用于处理可变字符串对象。通过StringBuilder对象，如果我们需要处理后的字符串，可以使用ToString()方法获取。

创建StringBuilder对象时，我们可以使用以下一些常用的构造函数：


	New()，创建一个空白的StringBuilder对象。

	New(Integer)，创建StringBuilder对象，指定其初始的缓冲尺寸，即可容纳的字符数，使用StringBuilder对象时，如果指定的容量不够，StringBuilder对象会自动增长，但不会超过限定的最大字符数。

	New(String)，使用一个字符串初始化StringBuilder对象。

	New(Integer, Integer)，参数一指定StringBuilder对象初始缓存容量，参数二指定允许的最大尺寸。



接下来，我们来看一看StringBuilder类中有哪些常用的成员。


	Length属性，返回对象中的字符数量。

	Append()方法，在对象的末尾添加内容，参数可以任意类型的数据。

	AppendFormat()方法，在对象的末尾添加格式化字符串，可参考String.Format()方法的使用。

	AppendLine()方法，添加指定的内容（String类型）后会自动添加一个换行符，如果不指定参数，则直接添加一个换行符。

	Insert()方法，在指定索引位置添加内容。

	Remove()方法，从指定的索引位置（参数一）删除指定数量（参数二）的字符。

	Replace()方法，一般的使用方法是用新的内容（参数二）替换原有的内容（参数一）。

	ToString()方法，返回StringBuilder对象中内容的String类型对象。可以使用参数指定开始的索引位置（参数一）和需要的字符数量（参数二）。如果不指定参数，则返回所有的字符串内容。



下面的代码，显示了StringBuilder类的使用。


Dim sb As New StringBuilder(255)
sb.Append("abc")
sb.Append("def")
sb.Append("ghi")
Console.WriteLine(sb.ToString())



15.4. 散列算法

对字符串进行散列操作的目标是对其内容进行编码，从而达到保护字符串数据的目的；此外，编码的过程是不可逆的。

常规的散列算法主要有MD5和SHA1，在.NET Framework中的相关资源定义在System.Security.Cryptography命名空间，包括MD5CryptoServiceProvider类和SHA1CryptoServiceProvider类，它们都是对字节数据进行编码操作，所以，在对字符串进行编码前，我们需要将其转换为字节数组，如下面的代码（VBChef/MCommon.vb文件）。


Public Function ToByteArray(s As String) As Byte()
    Return Encoding.Default.GetBytes(s)
End Function



Encoding类定义在System.Text命名空间，用于对字符编码的处理。其中的Default共享属性返回当前系统中的默认编码类型对象，我们使用其中的GetBytes()方法将字符串转换为字节数组。

通过Encoding中的GetString()方法，我们可以进行逆向操作，即将一个字节数组转换为字符串，如下面封装的ToStr()方法（VBChef/MCommon.vb文件）。


Public Function ToStr(arrByte As Byte()) As String
    Return Encoding.Default.GetString(arrByte)
End Function



下面，我们就看看在VB.NET中如何对字符串进行MD5算法的编码操作，如下面封装的GetMd5()方法（VBChef/MCommon.vb文件）。


Public Function GetMd5(s As String) As String
    Dim arrByte() As Byte = ToByteArray(s)
    Using md5 = New MD5CryptoServiceProvider()
        Dim md5Hash() As Byte = md5.ComputeHash(arrByte)
        Return ToHexStr(md5Hash)
    End Using
End Function



代码中，我们首先使用ToByteArray()方法将字符串转换为字节数组，然后，我们使用MD5CryptoServiceProvider对象进行MD5编码工作，其中的ComputeHash()方法用于计算字节数组数据的MD5编码，并返回计算后的字节数组。最后，我们使用ToHexStr()方法将MD5编码字节数组格式化成十六进制的字符串形式，其定义如下（VBChef/MCommon.vb文件）。


Public Function ToHexStr(arrByte As Byte()) As String
    Dim result = New StringBuilder(arrByte.Length * 2)
    Dim maxIndex As Integer = arrByte.Length - 1
    For i = 0 To maxIndex
        result.Append(arrByte(i).ToString("x2"))
    Next
    Return result.ToString()
End Function



我们封装的ToHexStr()方法，会将字节数组中的每个字节数据转换为16位（两字节）形式的十六进制数据形式。

最后，我们看一看字符串SHA1编码的获取，如下面的代码（VBChef/MCommon.vb）。


Public Function GetSha1(s As String) As String
    Dim arrByte() As Byte = ToByteArray(s)
    Using sha1 = New SHA1CryptoServiceProvider()
        Dim sha1Hash() As Byte = sha1.ComputeHash(arrByte)
        Return ToHexStr(sha1Hash)
    End Using
End Function



实际应用中，我们可以将一些安全性较高的文本内容进行散列处理（如用户登录密码等），然后进行保存、传输等操作；由于散列算法是不可逆的，所以，我们无法获取它的原始数据，这也正是保密的好办法。不过，对于简单文本的散列编码，依然可以通过对比来确定（碰撞测试），所以，对于密码这样的内容，字符组成的复杂性依然是保证其安全的必要措施。

15.5. 获取GUID

GUID（Globally Unique Identifier）通过一定的算法，可以保证在任何时间、任何计算机中生成的内容都不会重复，在一些操作中，这个功能非常重要。使用.NET Framework类库，我们可以很方便地生成GUID，即使用Guid结构（System命名空间）来实现，如下面的代码。


'' 生成GUID
Public Function GetGuid() As String
    Return Guid.NewGuid().ToString("N")
End Function



使用Guid.NewGuid()方法，我们可以获取一个新的GUID，然后，我们可以通过ToString()方法返回GUID的字符串形式，在这个ToString()方法中，我们可以使用一个字符指定返回GUID字符串的形式，可用的字符包括：


	N，返回纯编码的形式，如我们封装的代码中使用的那样（VBChef/MCommon.vb文件）。

	D，包含连接符（-）的编码字符串。

	B，在一对花括号{}中，包含连接符（-）的编码字符串。

	P, 在一对小括号()中，包含连接符（-）的编码字符串。







16. 日期与时间

在.NET Framework类库中定义的DateTime结构类型，为我们提供了一个处理日期和时间的强大工具，而在VB.NET中使用Date类型，则可以认为它是DateTime的别名，它们的使用方式是一样的。

本章，我们就来讨论日期与时间的处理问题，主要内容包括：


	DateTime结构与Date类型

	处理中国农历

	VB中的日期时间处理方法



16.1. DateTime结构与Date类型

在VB.NET中的Date类型，其本质与DateTime结构类型（System命名空间）是相同的，所以，当我们在讨论日期和时间的处理时，就不再特意区分这DateTime结构和Date类型，如果没有特别说明，这两种类型的操作是相同的。

16.1.1. 日期与时间处理基础

在.NET Framework中，日期和时间处理的最小单位是Tick，它是100毫微秒（即1秒的万分之一）。

处理日期和时间，有一个基准时间，即0Tick时刻，在DateTime结构中定义为MinValue，其值为公元0001年1月1日0时0分0秒0000。我们可以通过如下代码查看这个值。


Console.WriteLine(DateTime.MinValue)



使用DateTime结构时，我们可以使用以下共享属性获取系统中的当前日期或时间。


	Now共享属性，包含了完整的日期和时间信息。

	Today共享属性，包括日期信息，而时间则为0时0分0秒。



此外，在VB.NET中，我们还可以直接使用日期和时间的直接量，其格式如下：


#月/日/年 时:分:秒 <AM或PM>#



请注意，我们可以只指定日期或时间，此时，另一部分就会变成默认值；指定时间信息时使用12小时制，需要在使用AM或PM来指定是午前或午后。如下面的代码：


Console.WriteLine(#12/31/2015 3:34:56 AM#)
Console.WriteLine(#12/31/2015 3:34:56 PM#)
Console.WriteLine(#12/31/2015#)
Console.WriteLine(#3:34:56 AM#)



当我们获取日期或时间信息后，可以使用以下属性获取具体的数据：


	Year属性，获取年份，Integer类型。

	Month属性，获取月份，Integer类型。

	Day属性，获取在当前月份中的第几天，Integer类型。如果需要获取日期在当年中位于第几天，可以使用DayOfYear属性。

	Hour属性，小时，Integer类型。

	Mintue属性，分钟，Integer类型。

	Second属性，秒钟，Integer类型。

	Millisecond属性，毫秒数，Integer。

	Ticks属性，Tick值，Long类型。

	UtcNow属性，根据系统中设置的时间和时区获取对应的格林尼治时间（DateTime类型数据）。如北京时间位于时区的正八区，减去8个小时就是格林尼治时间。



我们还可以使用DayOfWeek属性获取日期位于一周中的第几天，其返回值类型为DayOfWeek枚举，枚举值包括：


	Sunday，星期日（0）。

	Monday，星期一（1）。

	Tuesday，星期二（2）。

	Wednesday，星期三（3）。

	Thursday，星期四（4）。

	Friday，星期五（5）。

	Saturday，星期六（6）。



请注意，在DateTime没有获取本周位于本年第几周的属性，我们可以使用VB函数DatePart()来完成这项工作；如下面的代码将返回当前日期位于本年的第几周。


Console.WriteLine(DatePart(DateInterval.WeekOfYear, DateTime.Now))



DatePart()函数中，第一个参数用于指定需要获取日期或时间信息的类型，它定义为DateInterval枚举类型，第二个参数指定为日期时间，函数将这个参数中获取相应日期或时间的数据。

虽然VB函数有很多、很实用的功能，但是，使用通用的.NET Framework开发资源会使我们的学习成果更加保值；所以，我们更建议使用.NET Framework中的通用开发资源，而对于在通用资源中没有定义的内容，我们则可以使用Microsoft.VisualBasic命名空间中的相关开发资源或者自己动手封装这些代码。

关于日期位于本年或本月第几周的计算方法，我们同样进行了封装，大家可以根据具体情况选择使用。如下面的代码（VBChef/MDataTime.vb）。


'' 某一天是当年中的第几周
Public Function WeekOfYear(dt As DateTime) As Integer
    Dim firstDay As New DateTime(dt.Year, 1, 1)
    Dim firstSunday As Integer =  _
        (7 - CInt(firstDay.DayOfWeek)) Mod 7 + 1
    Dim curDayOfYear As Integer = dt.DayOfYear
    If curDayOfYear < firstSunday Then
        Return 1
    ElseIf firstSunday = 1 Then
        Return (curDayOfYear - firstSunday) \ 7 + 1
    Else
        Return (curDayOfYear - firstSunday) \ 7 + 2
    End If
End Function

'' 某一天是当月中的第几周
Public Function WeekOfMonth(dt As DateTime) As Integer
    Dim firstDay As New DateTime(dt.Year, dt.Month, 1)
    Dim firstSunday As Integer =  _
        (7 - CInt(firstDay.DayOfWeek)) Mod 7 + 1
    If dt.Day < firstSunday Then
                Return 1
    ElseIf firstSunday = 1 Then
        Return (dt.Day - firstSunday) \ 7 + 1
    Else
        Return (dt.Day - firstSunday) \ 7 + 2
    End If
End Function



16.1.2. 常用方法

接下来，我们再来看一看DateTime中有哪些常用的方法。

日期与时间计算

首先，有一系列的AddXXX()方法可以帮助我们进行日期的加减运算，如我们需要计算当天100天以后的日期，可以使用如下代码。


Dim dt As DateTime = DateTime.Now.AddDays(100)



代码中可以看到，当我们将AddDays()方法的参数设置为正数时，就可以计算当天以后相应天数的日期了。

相似的方法包括：AddYears()、AddMonths()、AddDays()、AddHours()、AddMinutes()、AddSeconds()、AddMilliseconds()、AddTicks()等方法。使用这些方法时，如果参数为正数，则计算当前时间以后的时间值；如果参数为负数，则计算当前时间以前的值。

日期与时间格式

ToLongDateString()和ToShortDateString()方法分别获取长日期格式和短日期格式的字符串。

ToLongTimeString()和ToShortTimeString()方法分别获取长时间格式和短时间格式的字符串。

ToOADate()方法获取相应的OLE自动化时间格式数据，它定义为Double类型。实际上，在VB6中和Access中的日期时间数据格式就是这种格式，如果我们使用Access数据库，就需要小心日期和时间的格式及相应的处理问题。

ToString()方法可以通过参数自定义日期和时间的输出格式，在讨论String.Format()方法时已有一些介绍，如：


Console.WriteLine(DateTime.Now.ToString("yyyyMMdd"))
Console.WriteLine(DateTime.Now.ToString("HHmmss"))



使用日期时间格式化字符时，请注意字母的大小写，因为它们代表了不同的显示方式和内容。

如果我们需要获取星期几的名称，可以使用如下代码。


Console.WriteLine(DateTime.Now.ToString("ddd"))
Console.WriteLine(DateTime.Now.ToString("dddd"))



代码会显示当前时间的星期几名称，请注意，当我们使用"ddd"格式化字符时，会显示星期短格式名称，如周日、周一等；使用"dddd"格式化字符时，会显示星期长格式名称，如星期日、星期一。

此外，在ToString()方法中，如果我们只使用第一个参数（"ddd"或"dddd"），则星期的名称会使用当前系统中区域设置的格式。如果我们需要某一区域的星期名称，可以使用第二个参数，它定义为IFormatProvider接口，代码中，我们可以使用以下代码来定义它。


Console.WriteLine(DateTime.Now.ToString("ddd",  _
     New CultureInfo("zh-CN").DateTimeFormat))



代码中，我们使用了CultureInfo类，它定义在System.Globalization命名空间，在这里定义了一系列用于处理区域性设置的资源，我们可以使用标识来初始化CultureInfo对象，如：zh-CN（中国大陆）、zh-TW（中国台湾）、ja-JP（日本）、en-US（美国）等等。

当我们创建CultureInfo对象以后，就可以获取区域文化中的相关信息，如DateTimeFormat属性就给出了日期和时间格式。

下面的代码，我们可以得到完整的文化列表。


Dim arr() As CultureInfo = _
    CultureInfo.GetCultures(CultureTypes.AllCultures)
For Each item As CultureInfo In arr
    Console.WriteLine(item.Name)
Next



16.2. 处理中国农历

DateTime和VB函数在处理日期和时间时，功能的确很强大；不过，作为中国的程序员，我们怎么获取一些具有中国特色的日期数据呢？如属相、农历日期，以及年份在甲子中的名称等，本节，我们就来讨论如何获取这些信息，并将相关代码封装在CDateTime类中。

在.NET Framework资源中，ChineseLunisolarCalendar类用于处理中国农历信息，它定义在System.Globalization命名空间，在定义CDateTime类时不要忘记导入它。

农历日期信息中包含以下重要的数据：


	甲子年份。我们知道一个甲子是60年，其年份名称由十天干和十二地支组合而成。

	月份。包括正月到腊月，并会有闰月的情况。

	日期。包括初一到三十的日期。

	生肖。十二生肖，只要能得到年份位于一个甲子中的位置，就可以很方便地计算。



下面，我定义的CDateTime类，包含了一些必须的数据变量，以及它们的计算方法（VBChef/CDateTime.vb）。


Public Class CDateTime
    ''当前日期（阳历）
    Private myDate As DateTime
    ''甲子中的年份
    Private mySexagenaryYear As Integer
    ''年份天干
    Private myCelestialStem As Integer
    ''年份地支
    Private myTerrestrialBranch As Integer
    ''农历月份
    Private myLunarMonth As Integer
    ''农历闰月
    Private myLunarLeapMonth As Integer
    ''农历日子
    Private myLunarDay As Integer

    '' 根据日期计算农历信息
    Private Sub CaleDate(dt As DateTime)
        myDate = dt
        Dim caleCn = New ChineseLunisolarCalendar()
        '计算农历年份
        mySexagenaryYear = caleCn.GetSexagenaryYear(myDate)
        '计算天干
        myCelestialStem = caleCn.GetCelestialStem(mySexagenaryYear)
        '计算地支
        myTerrestrialBranch = _
            caleCn.GetTerrestrialBranch(mySexagenaryYear)
        '计算月份
        myLunarMonth = caleCn.GetMonth(myDate)
        '闰月
        If myDate.Month < 3 And myLunarMonth > 9 Then
            myLunarLeapMonth = caleCn.GetLeapMonth(myDate.Year - 1)
        Else
            myLunarLeapMonth = caleCn.GetLeapMonth(myDate.Year)
        End If
        '计算日子
        myLunarDay = caleCn.GetDayOfMonth(myDate)
    End Sub

    '' 构造函数
    Public Sub New(dt As DateTime)
        Me.CaleDate(dt)
    End Sub
    Public Sub New()
        Me.New(DateTime.Now)
    End Sub
End Class



代码中，定义的各个私有变量的功能已经标明了；而最重要的则是CalcDate()方法，它的功能就是通过一个阳历日期计算相应的农历日期数据，在这个方法中，我们使用了ChineseLunisolarCalendar类中一系列相关的成员，主要包括：


	GetSexagenaryYear()方法，根据阳历数据计算农历年份，即位于甲子中的第几年，返回值为1到60之间。方法的参数是DateTime变量，即阳历日期。

	GetCelestialStem()方法，计算农历年份中的天干数值，返回值为1到10之间。方法参数为甲子中的年份。

	GetTerrestrialBranch()方法，用于计算农历年份中的地支数值，返回值为1到12之间。方法参数同样为甲子中的年份。

	GetMonth()方法，用于计算农历月份，返回值为1到13，请注意，农历月份包括闰月的情况，一年可能有12个月（没有闰月），也可能是13个月（有闰月）。方法参数为阳历日期（DateTime类型）。

	GetLeapMonth()方法，返回农历闰月是第几个月，如闰月是第8个月，则为“闰七月”。如果当年没有闰月则返回0值。请注意此方法的参数是阳历年份，而计算闰月时是按农历计算，如果阳历年份是一月或二月时，我们就要注意是否还是农历的冬月或腊月，如果是，则应使用阳历的前一年来计算农历闰月数据。

	GetDayOfMonth()方法，返回农历日期是当月的第几天，返回值为1到30。



代码中，我们还定义了两个构造函数，可以指定日期来初始化农历数据，或使用系统当前日期来初始化农历数据。此外，当我们修改阳历日期时，还可以自动重新计算相应的农历信息，如下面定义的DateValue属性。


'' 属性，阳历信息
Public Property DateValue As DateTime
    Get
        Return myDate
    End Get
    Set(value As DateTime)
        Me.CaleDate(value)
    End Set
End Property



有了前面这些农历相关的数据，我们就可以根据需要返回特定的数据格式了。不过，我们还需要一些关于农历信息的文本数据，下面的代码就给出了这些内容（VBChef/CDateTime.vb文件）。


''年份，天干
Private Shared arrCelestialStem() As String =
    {"", "甲", "乙", "丙", "丁", "戊",
     "己", "庚", "辛", "壬", "癸"}
''年份，地支
Private Shared arrTerrestrialBranch() As String =
    {"", "子", "丑", "寅", "卯", "辰", "巳",
    "午", "未", "申", "酉", "戌", "亥"}

''月份名称
Private Shared arrMonthName() As String =
    {"", "正月", "二月", "三月", "四月", "五月", "六月",
    "七月", "八月", "九月", "十月", "冬月", "腊月"}

''日子
Private Shared arrDayName() As String =
    {"", "初一", "初二", "初三", "初四", "初五",
    "初六", "初七", "初八", "初九", "初十",
    "十一", "十二", "十三", "十四", "十五",
    "十六", "十七", "十八", "十九", "二十",
    "廿一", "廿二", "廿三", "廿四", "廿五",
    "廿六", "廿七", "廿八", "廿九", "三十"}

''属相
Private Shared arrAnimalName() As String =
    {"猪", "鼠", "牛", "虎", "兔", "龙",
     "蛇", "马", "羊", "猴", "鸡", "狗"}



现在，我们就可以获取相应的农历信息了；下面就是关于年份信息的属性，包括甲子年份数值和名称，以及属相的相关信息（VBChef/CDateTime.vb文件）。


'' 农历年份
Public ReadOnly Property SexagenaryYearValue As Integer
    Get
        Return mySexagenaryYear
    End Get
End Property

'' 农历年份，天干+地支
Public ReadOnly Property SexagenaryYearName As String
    Get
        Return arrCelestialStem(myCelestialStem) & _
            arrTerrestrialBranch(myTerrestrialBranch)
    End Get
End Property

 '' 属相值
Public ReadOnly Property AnimalValue As Integer
    Get
        Dim m As Integer = mySexagenaryYear Mod 12
        If m = 0 Then
            Return 12
        Else
            Return m
        End If
    End Get
End Property

'' 属性名称
Public ReadOnly Property AnimalName As String
    Get
        Return arrAnimalName(mySexagenaryYear Mod 12)
    End Get
End Property



首先，我们看SexagenaryYearName属性，它返回了天干和地支的组合文本内容；请注意这两个数组的定义，它们的第一个成员都定义为空字符中，其功能是占位，我们知道天干和地支都是从1开始的，所以，我们的正式数据也从索引1开始，这样就以很小的存储空间方便了代码的编写工作。

再看AnimalValue属性，它的功能是返回生肖的值（1到12），我们只需要有甲子年份除以12的余数就可以得到生肖的值，而需要注意的就是第12位的生肖值会计算为0，所以，我们使用If语句进行判断，如果余数是0则返回12，否则返回余数值即可。

在AnimalName属性中，我们直接使用了甲子年份除以12的余数，但我们在arrAnimalName()数组中配合了这个数值，即，我们将生肖第12位的猪放在数组的第一个成员位置，即索引值为0的位置，这样就可以直接使用计算结果，不需要多余的代码了。

接下来，我们再来看看月份的相关属性，如下面的代码（VBChef/CDateTime.vb文件）。


'' 农历月份
Public ReadOnly Property LunarMonthValue As Integer
    Get
        Return myLunarMonth
    End Get
End Property

'' 农历月份名称
Public ReadOnly Property LunarMonthName As String
    Get
        If myLunarLeapMonth = 0 OrElse   _
                  myLunarMonth < myLunarLeapMonth Then
            Return arrMonthName(myLunarMonth)
        ElseIf myLunarMonth = myLunarLeapMonth Then
            Return "闰" & arrMonthName(myLunarMonth - 1)
        Else
            Return arrMonthName(myLunarMonth - 1)
        End If
    End Get
End Property



LunarMonthValue和LunarMonthName属性分别返回农历月份的数值和名称。请注意在LunarMonthName属性中，当前月份小于闰月或没有闰月的时候，直接返回相应的月份名称；如果当前月份即是闰月，则返回前一月名称加“闰”字，如第8个月为闰月，则返回“闰七月”；如果有闰月，并且当前月大于闰月，则返回前一月的名称。

请注意，月份名称数组arrMonthName中的第一个成员同样使用了空字符串进行占位。

最后，我们看看日期的获取，如下面的代码（VBChef/CDateTime.vb文件）。


 '' 日期
Public ReadOnly Property LunarDayValue As Integer
    Get
        Return myLunarDay
    End Get
End Property
''
Public ReadOnly Property LunarDayName As String
    Get
        Return arrDayName(myLunarDay)
    End Get
End Property



这两个属性比较简单，只是需要注意日期名称数组arrDayName中的第一个成员也使用了空字符串进行占位处理。

下面的代码，我们将显示系统当前日期的农历及生肖相关信息。


Dim dt As New CDateTime()
Console.WriteLine(dt.SexagenaryYearName)
Console.WriteLine(dt.AnimalName)
Console.WriteLine(dt.LunarMonthName)
Console.WriteLine(dt.LunarDayName)



16.3. VB中的日期时间处理方法

在使用VB进行开发时，我们可以使用一系列的日期与时间处理方法，在VB6及以前，这些开发资源功能强大，非常实用，这一传统在VB.NET中也被保留了下来，我们可以在Microsoft.VisualBasic命名空间中的DateAndTime类中看到这些方法的定义，大家可以在项目根据需要参考应用这些方法。

在VB.NET代码中，由于VB.NET项目已默认引用了Microsoft.VisualBasic命名空间，所以，我们可像VB6一样把这些方法当做全局方法来使用。

下面，我们介绍一些常用的方法，方便大家快速参考应用。

16.3.1. DateAndTime类共享属性

首先，我们看一看在DateAndTime类中定义的共享属性：


	Now属性，返回包含当前日期和时间的DateTime数据。

	DateString属性，返回当前日期的字符串，字符串格式为系统区域设置中的短日期格式。

	Today属性，返回包含当前日期的DateTime数据，时间部分设置为00:00:00。

	TimeOfDay属性，返回当前时间的DateTime数据，日期部分设置为0001年1月1日。

	Timer属性，返回一个Double数据，表示从当天零点以来的秒数。

	TimeString属性，表示当前时间的字符串，其格式为24小时制的时间，如“23:56:59”。



在VB.NET代码中，我们可以直接使用这些属性，如：


Console.WriteLine(DateString)
Console.WriteLine(TimeString)



16.3.2. DateAndTime类共享方法

接下来，我们再来看看DateAndTime类中，有哪些常用的共享方法。

获取日期与时间信息

首先，有系列的方法可以从DateTime数据中提取相应的信息，如Year()、Month()、Day()、Hour()、Mintue()和Second()方法，分别用于从DateTime数据中指取年、月、日、时、分、秒数据，它们的参数都是DateTime类型。MonthName()方法用于获取提定月份的名称，其中参数一为月份值（1到12），参数二指定是否使用月份名称的缩写，默认为False。

此外，我们可使用Weekday()方法获取某一日期是星期几（数值），默认情况下，星期日为0，星期一到星期六分别是1到6；如果需要修改这一规则，可以指定第二个参数，它定义为FirstDayOfWeek枚举类型；一般情况下，我们并不建议修改星期操作的默认设置。

与星期相关的另一个方法是WeekdayName()方法，它会返回指定星期数值的名称，它可以使用三个参数：第一个参数为整数，一般使用Weekday()方法返回的数值；参数二为Boolean类型，指定是否使用星期的缩写名称，默认为False；参数三是FirstDayOfWeek枚举类型，一般情况下，我们并不建议使用。

创建日期与时间信息

我首先看DateSerial()方法，其功能通过年（参数一）、月（参数二）、日（参数三）数据组合并返回一个DateTime结构数据，它的三个参数都定义为Integer类型。请注意年份的使用，如果定义三位数以上则直接使用此年份；如果定义为两位数以下则自动转换为1930到2029年之间的年份；如果为负数年份，则使用系统当前日期减去此数绝对值的年份。此外，请注意，此方法只能创建DateTime规定范围内的日期数据（DateTime.MinValue到Datetime.MaxValue之间）。

DateValue()方法，用于从一个字符串形式的日期数据中创建DateTime结构数据，如DateValue("2011-6-26")。 请注意，DateSerial()和DateValue()方法构成的DateTime结构数据中，时间设置为00:00:00。

此外，创建时间数据时，我们可以使用TimeSerial()和TimeValue()方法。其中，TimeSerial()方法通过时、分、秒数据构造DateTime类型数据；而TimeValue()方法则通过字符串形式的时间创建DateTime类型数据。使用两个方法构建的DateTime数据是的日期都被设置为0001年1月1日。

日期与时间计算

DateAdd()方法用于进行日期或时间的向后或向前计算，其中参数一为DateInterval枚举类型或字符串，用于指定计算的日期或时间的哪一部分，其定义如下表；参数二用于指定日期相加的数值，正数向后计算，负数向前计算；参数三为需要计算原始日期或时间值，可以是DateTime类型，也可以是其它形式的日期或时间数据形式。

DateDiff()方法用于计算两个日期之间的间隔，其中参数一为DateInterval枚举类型或字符串，指定计算的类型，其成员如上表；参数二和参数三指定用于计算的两个日期时间数据，可以是DateTime或其它形式。DateDiff()会使用参数三的数据减去参数二的数据，正数时则参数三晚于参数二，负数时则参数三早于参数二。

DatePart()方法用于获取日期或时间某一部分的数据。参数一为需要的数据类型，定义为DateInterval枚举类型或字符串；参数二为DateTime类型数据，方法会从此数据中获取相应的信息。参数三为可选，用于指定一周的第一天，定义为FirstDayOfWeek枚举。参数四也为可选，用于指定确认一年第一周的方式，定义为FirstWeekOfYear枚举类型，其成员定义如下：


	Jan1值，默认值，指定1月1日所在的周，对应VB常数vbFirstJan1。

	FirstFourDays值，指定至少包含四天的一周为一年中的第一周，对应VB常数vbFirstFourDays。

	FirstFullWeek值，指定一年中的第一个整周为第一周，对应于VB常数 vbFirstFullWeek。

	System值，用系统设置，对应VB常数 vbUseSystem。







17. 文件系统

本章，我们将讨论文件与目录处理的相关内容，主要包括：


	获取系统相关目录

	目录操作

	文件操作

	Path类

	VB中的文件系统处理方法



17.1. 获取系统相关目录

我们使用Environment类（System命名空间）中的GetFolderPath()共享方法，可以获取Windows系统中的一些常用目录，此方法包括一个参数，定义为Environment.SpecialFolder枚举类型，其常用成员包括：


	Cookies，浏览器Cookie 的储存目录。

	DesktopDirectory，桌面内容。

	Favorites，用户“收藏夹”目录。

	History，浏览器浏览历史记录存放目录。

	InternetCache，浏览器临时文件存放目录。

	MyDocuments，用户“文档”文件夹。

	MyMusic，用户“音乐”文件夹。

	MyPictures，用户“图片”文件夹。

	Personal，当前用户主目录，如Widnows7系统中的“C:\Users\Administrator”。

	ProgramFiles，程序文件（Program Files）目录。

	Programs，开始菜单中“程序”或“所有程序”目录。

	Recent，用户最近使用的文档目录。

	SendTo，菜单项“发送到”的目录。

	StartMenu，“开始”菜单项的目录。

	Startup，启动”程序组的目录。

	System，Windows系统目录，如“C:\Windows\System32”。

	Templates，公共文档模板的目录。



在VBChef项目中，我们创建了CSysFolder类用于封装获取Windows中的特殊目录，如下面的代码，就是获取“文档”目录路径的方法（VBChef/CSysFolder.vb文件）。


Public Class CSysFolder
Public Shared ReadOnly Property MyDocuments As String
    Get
        Return Environment.GetFolderPath( _
                   Environment.SpecialFolder.MyDocuments)
    End Get
End Property
End Class



其它目录的获取方法应该不难，但是，在这里没有Windows安装目录的获取方法；不过，我们可以使用其它方法获取Windows安装目录的路径，主要包括两个方法，即读取环境变量或使用WinAPI函数。

首先看读取环境变量的方法，如下面的代码，我们就通过环境变量WinDir读取Windows的安装目录。


'' Windows 安装目录
Public Shared ReadOnly Property Windows As String
    Get
        Return Environment.GetEnvironmentVariable("WinDir", _
                           EnvironmentVariableTarget.Machine)
    End Get
End Property



另外一个获取Windows安装目录的方法就是使用WinAPI中的GetWindowsDiretory()函数，如下面的代码。


Private Declare Function GetWindowsDirectory _
    Lib "kernel32.dll" Alias "GetWindowsDirectoryA" ( _
        sb As StringBuilder, length As Integer) As Integer

'' 缓存尺寸
Private Const BufferSize As Integer = 255

'' Windows 安装目录
Public Shared ReadOnly Property WindowsDirectory As String
    Get
        Dim sb As New StringBuilder()
        Dim result As Integer = GetWindowsDirectory(sb, BufferSize)
        If result <> 0 Then
            Return sb.ToString()
        Else
            Return ""
        End If
    End Get
End Property



17.2. 目录操作

在System.IO命名空间中，定义了一系列输入输出（如文件、流等）操作的资源，其中，Directory类为我们提供了一些共享方法来操作目录，而DirectoryInfo类则用于创建目录对象，然后对目录进行操作。

17.2.1. Directory类

在Directory类中，定义了一些共享方法用于目录的操作，常用的方法包括：


	Exists()方法，判断参数中的目录是否存在。方法会返回Boolean类型的值。

	Delete()方法，删除指定的目录，如果目录中存在子目录和文件，可以将第二个参数设置为True，强行删除目录。

	CreateDirectory()方法，创建参数中指定的目录，创建成功会返回新目录的DirectoryInfo对象。此方法可以创建多级目录。

	GetDirectories()方法，返回一个String数组，包括目录中所有子目录。目录中没有子目录则返回空数组。

	GetFiles()方法，返回String数组，包括目录中所有文件。目录中没有文件则返回空数组。

	Move()方法，移动一个目录，如果源目录和目标目录在同一位置，则可以实现目录改名的功能。

	GetCreationTime()方法，返回目录的创建时间（DateTime类型）。使用SetCreationTime()方法可以设置此数据。

	GetLastAccessTime()方法，返回目录的最后访问时间（DateTime类型）。使用SetLastAccessTime()方法可以设置此数据。

	GetLastWriteTime()方法，返回目录的最后修改时间（DateTime类型）。使用SetLastWriteTime()方法可以设置此数据。



下面的代码，我们使用了其中的几个方法的使用，大家可以将语句通过注释功能一步步查看执行的结果。


Dim sDir As String = "c:\temp"
'
If Directory.Exists(sDir) Then
     Console.WriteLine("{0}目录已存在", sDir)
Else
     Console.WriteLine("{0}目录不存在", sDir)
End If
'
'Console.WriteLine(Directory.CreateDirectory(sDir))
'Directory.Move(sDir, "c:\MyTemp")



17.2.2. DirectoryInfo类

DirectoryInfo类用于表示一个具体的目录对象，在构造函数中，我们需要一个目录的路径来初始化，接下来，我们继续使用前面创建的c:\MyTemp目录来初始化一个DirectoryInfo对象，如下面的代码。


Dim sDir As String = "c:\MyTemp"
Dim oDir As New DirectoryInfo(sDir)



接下来，我们就可以使用DirectoryInfo对象的一系列成员对目录进行相应的操作，常用的成员包括：


	Exists属性，判断目录是否存在。

	Name属性，返回目录的名称，不包括路径。

	Parent属性，返回目录的父目录信息，其类型同样为DirectoryInfo对象。

	Root属性，返回目录所在的根目录信息，返回结果同样为DirectoryInfo对象。

	Create()方法，创建目录。

	CreateSubdirectory()方法，创建子目录，可以同时创建多级目录，如oDir.CreateSubdirectory("a/b/c")。

	Delete()方法，删除目录，如果目录中包括子目录和文件，可以将参数设置为True执行强制删除。

	GetDirectories()方法，返回一个包含子目录列表的String数组。

	GetFiles()方法，返回一个包含目录中文件列表的String数组。

	MoveTo()方法，移动目录的位置，源目录和目标目录在同一位置时，可以修改目录的名称。



请注意，无论是Directory类，还是DirectoryInfo类，都包含了一些关于目录权限的设置操作；而对于普通的应用项目，我们会建议让系统自动来管理目录的权限，如果真的需要在项目中控制目录的权限，可以参考如下资源。


	GetAccessControl()方法。

	SetAccessControl()方法。

	DirectorySecurity类。



17.3. 文件操作

在System.IO命名空间中为我们提供的文件操作的相关资源有很多，本节我们将主要介绍其中的File类、FileStream类和FileInfo类。

17.3.1. File类

File类定义了一系列文件操作的共享方法，接下来，我们就来讨论一些常用的文件操作。

Exists()方法，判断路径中的文件是否存在，返回Boolean类型数据。 Copy()方法，用于拷贝文件，参数一指定源文件路径，参数二指定目标文件路径。如果目标文件已存在，可以将第三个可选参数设置为True，即允许重写目标文件。

Delete()方法，删除指定的文件。

Move()方法，移动文件，参数一指定源文件路径，参数二指定目标文件路径。如果源文件和目标文件在同一目录，可以完成文件的改名操作。 Replace()方法，使用参数一文件内容替换参数二文件内容，并将参数二文件进行备份，备份的文件使用参数三指定。此外，我们还可以使用第四个可选参数，它是一个Boolean类型数据，指定是否忽略写入操作时可能产生的错误。

WriteAllText()和AppendAllText()方法都用于将文本内容写入到文件中，它们也有着相同的参数，参数一指定写入内容的文件路径，参数二指定需要写入的内容。这两个方法的不同点在于，WriteAllText()将使用参数二的内容替换所有文件原有内容，而AppendAllText()方法则会在文件的尾部添加参数二指定的内容。此外，这两个方法还可以使用第三个参数，用于指定文本内容的编码，它定义为Encoding对象（参考System.Text.Encoding类）。

ReadAllText()方法从文本文件中读取并返回所有内容。参数一指定读取内容的文件路径，参数二为可选，指定文本的编码（Encoding对象）。此外，读取文件之前，应判断文件是否已存在。

WriteAllBytes()和ReadAllBytes()方法，用于向文件写入字节数组或从文件中读取字节数据，并以字节数组的形式返回。在第9章，我们讨论了对象的序列化操作，实际上，我们可以很方便的将复杂的对象写入到磁盘文件，在需要的时候快速恢复它，而不需要经过重复初始化，特别是在对象中的数据不易重复创建的情况下。

WriteAllLines()方法用于将一个字符串数组写入文本文件，每个数组成员作为一行，此方法会在第一行文本后自动添加换行符。ReadAllLines()方法从文本文件中读取所有的行，并返回这些行组成的字符串数组。

如果我们需要更灵活的读写文件，可以使用FileStream类，不过，在File类中，我们可以使用一些方法打开文件，并生成FileStream对象。这些打开或创建文件的方法包括：

Open()方法，打开文件，并返回一个FileStream对象；参数一指定要打开文件，参数二指定打开的模式，其定义为FileMode枚举类型，主要成员包括：


	Append，只能对文件进行追加操作，不能访问文件原有的内容。

	Create，如果文件不存在则创建，如果文件存在则改写。

	CreateNew，创建一个新的文件，如果文件已存在则会产生异常。

	Open，打开已存在的文件，如果文件不存在则会产生异常。

	OpenOrCreate，打开或创建文件。

	Truncate，为完全重写而打开文件，此方式打开的FileStream对象不能读取文件内容。



OpenRead()方法，以读取方式打开文件，并返回一个FileStream对象。

OpenWrite()方法，以写入方式打开文件，并返回一个FileStream对象。

OpenText()方法，打开文件并创建一个FileStream对象，文本内容以UTF-8编码的文本进行操作。

Create()方法，打开一个可以完全读写的FileStream对象，参数为操作的文件路径。

此外，在File类和FileInfo类中定义了一系列关于文件创建、访问和修改时间的读取和设置文件，它们与Directory类中定义的方法相拟，大家可以参考使用。

17.3.2. FileStream类

FileStream对象表示一个准备读或写的文件对象，在对文件进行读写操作之前，我们可以使用以下属性判断是否能够进行相应的操作：


	CanRead只读属性，文件是否可以读取。

	CanWrite只读属性，文件是否可以写入。

	CanSeek只读属性，文件是否可以指定访问位置。



此外，FileStream类还定义了一些比较常用的属性，如：


	Length只读属性，表示FileStream对象的字节数（Long类型）。

	Postion属性，表示FileStream对象中的操作位置（Long类型）。



接下来，我们再来看一看FileStream对象中的常用方法。


	SetLength()方法，用于重新指定FileStream对象的字节数。

	Write()方法，向文件流中写入字节数组，第一个参数为需要写入的字节数组；第二个参数为开始写入的位置（Long类型），第三个参数指定写入的最大字节数量（Long类型）。

	Read()方法，从文件流中读取字节数据，以字节数组类型通过第一个参数输出；第二个参数指定读取的开始位置（Long类型），而第三个参数指定读取的字节数量（Long类型）。



下面的代码，我们将在VBChef项目中测试。

首先，我们创建一个CDataItem对象，然后将其序列化并保存在c:\MyTemp\obj1.archive文件中。如下面代码。


Const file1 As String = "c:\MyTemp\obj1.archive"
Const file2 As String = "c:\MyTemp\obj2.archive"
'
Dim obj1 As New CDataItem("Name", "Tom")
Dim arrByte As Byte() = ToBytes(obj1)
File.WriteAllBytes(file1, arrByte)



第二步，我们通过两个FileStream对象来复制这个文件，如下面的代码。


Using fs1 As FileStream = File.OpenRead(file1)
    Using fs2 As FileStream = File.OpenWrite(file2)
        fs1.Read(arrByte, 0, arrByte.Length)
        fs2.Write(arrByte, 0, arrByte.Length)
    End Using
End Using



最后，我们通过读取c:\MyTemp\obj2.archive文件来还原对象，并显示其内容。如下面的代码。


arrByte = File.ReadAllBytes(file2)
Dim obj2 As CDataItem = ToObject(arrByte)
Console.WriteLine("{0} : {1}", obj2.Name, obj2.Value)



代码会显示如下图所示的结果。

17.3.3. FileInfo类

FileInfo类用于表示一个文件对象，可以使用一个文件路径初始化；其属性包括：


	Exists只读属性，判断文件是否存在（Boolean类型）。

	IsReadOnly属性，判断文件是否为只读（Boolean类型）。

	Length只读属性，返回文件的字节数（Long类型）。

	Name只读属性，返回不包含路径的文件名（String类型）。

	Directory只读属性，返回文件所在目录的信息（DirectoryInfo对象）。

	DirectoryName只读属性，返回文件所在目录的完整路径（String类型）。



接下来，我们来看一看FileInfo类中定义的常用方法。

CopyTo()方法，将文件复制到指定的位置（参数一），如果目标文件已存在，可以指定第二个可选参数为True，指定强制替换目标文件。

对于FileInfo对象中的其它方法，与File类中的方法定义非常相似，只不过在FileInfo对象中定义的方法都是实例成员，我们可以参考File类中介绍的方法来使用。

17.4. Path类

Path类主要定义了对于目录或文件路径字符串内容的相关操作，它的成员都定义为共享成员，常用的包括：


	ChangeExtension()方法，修改扩展名，如ChangeExtension(path,".jpg")；去除护展名，如ChangeExtension(path, "")。此方法会返回修改扩展名以后的路径（String类型）。

	Combine()方法，用于合并两个字符串，我们可以使用这个方法合并路径和文件名，如Combine("c:\", "MyTemp\temp.txt")会返回"c:\MyTemp\temp.txt"。

	GetDirectoryName()方法，返回路径中的目录信息（String类型）。

	GetExtension()方法，给出路径中包含的扩展名，包括圆点。没有扩展名返回空字符串（String.Empty）。

	GetFileName()方法，返回路径中的文件名内容，包括扩展名。

	GetFileNameWithoutExtension()方法，，返回路径中的文件名内容，不包括扩展名。

	GetFullPath()方法，返回路径的完整路径（从根目录开始的路径）。

	GetPathRoot()方法，返回路径中的根目录（String类型）。

	GetTempFileName()方法，返回一个临时文件名。在项目中需要操作临时文件时，我们可以使用这个方法，返回的临时文件包括完整的路径和文件名（.tmp）。

	GetTempPath()方法，返回系统临时目录的路径。

	HasExtension()方法，判断路径中是否包含扩展名。

	IsPathRooted()方法，判断路径是否是从根目录开始的完整路径。



17.5. VB中的文件系统处理方法

使用过VB6的朋友一定对其中关于文件处理的众多方法比较熟悉，在VB.NET可，同样提供了相似的方法，它们定义在Microsoft.VisualBasic命名空间中的FileSystem类，其成员都定义为共享方法，而我们则可以在VB.NET代码中直接使用这些方法，而无需显式地调用FileSystem类。

不过，在VB.NET中还保留这些方法的唯一目的就是对VB6代码保持一定程度上的兼容，而从性能及技术的通用性来讲，我们还是应该更多地使用.NET Framework中定义的资源。





18. 图形图像

在.NET Framework的System.Drawing命名空间中，为我们提供了大量的图形与图像处理资源。本章，我们就来了解基中一些资源，并对常用的功能进行封装。本章内容主要包括：


	绘制太极图

	图形绘制

	Font类与文本绘制

	截取图像

	转换尺寸与格式

	旋转与翻转

	线性渐变效果

	图片DPI与尺寸



18.1. 绘制太极图

本节，我们将从一个简单的图形绘制操作来了解基本的图形图像处理知识，而我们要绘制的图形就是太极图，如下图。

绘制这个图形的代码也非常简单，我们在CImage类中封装了它的绘制代码，如下面的Taiji()方法（VBChef/CImage.vb文件）。


Imports System
Imports System.Drawing
Imports System.Drawing.Imaging
Imports System.Drawing.Drawing2D
Imports System.Drawing.Text

Namespace vbchef
    Public Class CImage
        '' 演示，绘制太极图并返回Bitmap对象
        Public Shared Function Taiji(width As Integer) As Bitmap
            Dim w_2 As Integer = width \ 2
            Dim w_4 As Integer = width \ 4
            Dim w_8 As Integer = width \ 8
            Dim w_16 As Integer = width \ 16
            '
            Dim bmp As New Bitmap(width, width)
            ' bmp.SetResolution(300, 300)
            Dim g As Graphics = Graphics.FromImage(bmp)
            '底色大圆，白
            g.FillEllipse(Brushes.White, 0, 0, width, width)
            ' 左半圆，黑色
            g.FillPie(Brushes.Black, 0, 0, width, width, 90, 180)
            ' 下部分，黑圆
            g.FillEllipse(Brushes.Black, w_4, w_2, w_2, w_2)
            ' 上部分，白圆
            g.FillEllipse(Brushes.White, w_4, 0, w_2, w_2)
            ' 上小黑圆
            g.FillEllipse(Brushes.Black, w_2 - w_16, w_4 - w_16,  _
                           w_8, w_8)
            ' 下小白圆
            g.FillEllipse(Brushes.White,  _
                          w_2 - w_16, w_4 + w_2 - w_16, w_8, w_8)
            '边框
            g.DrawEllipse(Pens.Black, 0, 0, width - 1, width - 1)
            '
            Return bmp
        End Function
    End Class
End Namespace



Taiji()方法的参数为太极图的直径，返回值类型是Bitmap（位图）对象，也就是太极图形的位图数据；接下来，我们可以将这个位图保存到磁盘文件中，如下面的代码。


Dim bmp As Bitmap = CImage.Taiji(500)
bmp.Save("c:\taiji.png", Imaging.ImageFormat.Png)



执行代码，会将太极图以PNG格式保存到C:盘下的taiji.png文件。

接下来，我们再回过头看一看Taiji()方法中使用了哪些图形图像处理资源。


	Bitmap类（System.Drawing命名空间）表示一个位图对象，我们可以进行一系列的操作，然后保存到磁盘文件或其它数据流中，比如，从页面中返回到客户端。

	Graphics类（System.Drawing命名空间）主要用于图形的绘制，其中定义了大量的用于绘制线条图形或填充图形的方法。

	Brush类（System.Drawing命名空间）表示填充图形中的填充风格，如各种颜色或渐变效果等。

	Pen类（System.Drawing命名空间）表示线条绘制风格，如颜色、尺寸等。



在图像的处理过程中，我们使用的坐标及尺寸结构如下图所示。

我们可以看到，在矩形图像的左上角为原点（0,0），width所示的为X坐标方向，height所示的为Y坐标方向。

18.2. 图形绘制

本节，我们将讨论常用的图形图像处理类及其使用技巧。

18.2.1. Color结构

在图像处理过程中，关于颜色的操作是必不可少的了，我们可以使用Color结构来处理颜色，包括颜色中的红值（Red）、绿值（Green）、蓝值（Blue）和Alpha值。

我们通常可以使用如下共享方法创建一个Color结构：


	FromArgb(red, green, blue)，通过红、绿和蓝值创建颜色。

	FromArgb(alpha, red, green, blue)，通过透明度、红、绿和蓝值创建颜色。



方法中的参数都定义为Integer类型，取值也都是从0到255。

18.2.2. 图片格式

对于常用的图片格式，或者说在.NET Framework中支持的图片格式，我们可以在ImageFormat类（System.Drawing.Imaging命名空间）成员中找到。

如果我们的图片需要支持透明色，可以使用GIF或PNG图片格式；而其它常用的图片格式则包括位图（BMP）、JPEG等格式的图片。 下面就是ImageFormat类中定义的，用于获取各种图片格式对象的共享属性：


	Bmp，位图图像。

	Emf，增强型Windows图元图像。

	Exif，可交换图像文件。

	Gif，GIF（图形交换格式）图像。

	Icon，Windows图标图像。

	Jpeg，JPEG（联合图像专家组）图像。

	Png，PNG（W3C 可移植网络图形）图像。

	Tiff，TIFF（标签图像文件格式）图像。

	Wmf，WMF（Windows 图元文件）图像。



18.2.3. Bitmap类

Bitmap类表示一个位图对象，它是以像素为基础的图像处理对象；实际应用中，我们可以使用多种方式来创建Bitmap对象，常用的包括：


	使用一个图片文件，如bmp = New Bitmap(filename)。

	指定尺寸，如bmp = New Bitmap(width, height)。

	使用其它的Bitmap对象，如bmp = New Bitmap(bmpA)。此外，我们还可以同时指定新Bitmap对象的尺寸，如bmp = New Bitmap(bmpA, width, height)。



创建Bitmap对象后，我们还可以使用它的一些常用成员，如：


	GetPixel()方法，给出某一像素的颜色，返回为Color结构变量。

	SetPixel()方法，设置某个像素的颜色，如bmp.SetPixel(1,1,Color.FromArgb(255,0,0))就是将像素(1,1)设置为红色。

	SetResolution()方法，设置图像的水平分辨率和垂直分辨，一般来讲，如果没有特殊需求，两个方向的分辨率是相同的。稍后，我们会讨论图像分辨率的实际应用问题。

	Save()方法，将位图对象保存到磁盘文件或其它数据流中。其中，常用的保存操作中，参数一为保存到的目标，如文件、Response.OutputStream等；参数二可以指定图片的保存格式，我们可使用ImageFormat类（System.Drawing.Imaging命名空间）中的共享对象来确定，如ImageFormat.Jpeg、ImageFormat.Png、ImageFormat.Bmp等格式。



18.2.4. Brush和Brushes类

Brush类是定义格式刷的基类，它定义为必须继承的类型（抽象类），我们可以使用以下方式来使用各种类型的格式刷。


	使用一些Brush类的子类，如LinearGradientBrush、SolidBrush和TextureBrush。

	Brushes类中定义了各种颜色的格式刷，如Brushes.Red、Brushes.Black、Brushes.Transparent等等。



18.2.5. Pen和Pens类

Pen类用于定义线条绘制的格式，对于普通的线条，我们常用的属性包括：


	Width属性，指定线条的宽度（像素）。

	Color属性，指定线条的颜色。



当然，我们还可以创建各种虚线的线条，此时，我们需要关注以下属性：


	DashCap属性，设置为DashCap枚举类型，设置线条两端的样式，如Flat（默认值，方形）、Round（圆形）和Triangle（三角形）。

	DashOffset属性，设置起点到短线之间的距离，定义为Single类型。

	DashPattern属性，使用一个Single数组，指定实线与空白的尺寸。

	DashStyle属性，定义为DashStyle枚举类型，指定虚线的风格，如Dash、DashDot、DashDotDot、Dot等类型。



下面的代码，我们将绘制一条虚线，并保存到C:\dashline.png文件。


Dim p As New Pen(Color.Red, 5)
p.StartCap = Drawing2D.LineCap.Round
p.DashStyle = Drawing2D.DashStyle.DashDotDot
p.DashCap = Drawing2D.DashCap.Round
Dim bmp As New Bitmap(200, 30)
Dim g As Graphics = Graphics.FromImage(bmp)
g.DrawLine(p, 3, 15, 200, 15)
bmp.Save("c:\dashline.png", Imaging.ImageFormat.Png)



代码绘制的线条如下图所示。

此外，在Pens类中定义了一些颜色的Pen对象，如Pens.Black、Pens.Red等，它们的Width属性默认为1，如果我们需要特殊的Pen对象，就必须自己创建。

18.2.6. Graphics类

实际上，我们已多次使用Graphics对象了，它的功能就是绘制图形，主要包括三种类型：


	线条类图形，使用一系列的DrawXXX()方法绘制线条图形，如矩形、圆形及线条等。

	填充类图形，使用一系列的FillXXX()方法绘制填充图形，我们可以使用颜色、也可以使用纹理或其它效果进行填充。

	文本，主要使用DrawString()方法来绘制文本内容。



使用Graphics绘制图形，我们应该首先指定绘制的目标，如一个位图对象，下面的代码就是将位图对象与Graphics对象关联的基本代码。


Dim bmp As New Bitmap(300, 200)
Dim g As Graphics = Graphics.FromImage(bmp)



接下来，我们看看常用的图形绘制方法，这些方法可能有多个重载版本，大家可以在帮助中查询它们的完整定义，而下面的内容则讨论一些常用的绘制方法。

绘制矩形

使用Graphics对象绘制矩形时，我们使用其中的DrawRentangle()方法，如下面的代码。


Dim bmp As New Bitmap(300, 200)
Dim g As Graphics = Graphics.FromImage(bmp)
g.DrawRectangle(Pens.Black, 10, 10, 200, 100)
bmp.Save("c:\test.png", Imaging.ImageFormat.Png)



代码绘制的矩形如下图。

[image: 图像说明文字]


绘制线条矩形时，我们使用Pen对象创建线条的样式，其后的四个参数分别是矩形左上角X和Y坐标，以及矩形的宽度和高度。

当我们需要绘制填充矩形时，应使用FillRectangle()方法，它与DrawRectangle()方法的参数设置非常相似，只是第一个参数中，我们使用Brush对象定义填充的格式刷，如果只是简单的颜色填充，我们可以使用Brushes类中定义的一些共享成员，如下面的代码，我们就绘制了一个红色的矩形。


Dim bmp As New Bitmap(300, 200)
Dim g As Graphics = Graphics.FromImage(bmp)
g.FillRectangle(Brushes.Red, 10, 10, 200, 100)
bmp.Save("c:\test.png", Imaging.ImageFormat.Png)



如果我需要设置图像的底色，可以在其它操作之前使用类似如下的方法：


Dim bmp As New Bitmap(300, 200)
Dim g As Graphics = Graphics.FromImage(bmp)
g.FillRectangle(Brushes.Black, 0, 0, bmp.Width, bmp.Height)



绘制圆形和椭圆

绘制椭圆的方法参数设置，与绘制矩形方法的参数设置很相似，这是因为绘制的椭圆是基于一个矩形区域，如下图所示。

[image: 图像说明文字]


绘制线条椭圆的方法是DrawEllipse()，如下面的代码就绘制了一个宽200像素，高100像素的椭圆。


Dim bmp As New Bitmap(300, 200)
Dim g As Graphics = Graphics.FromImage(bmp)
g.DrawEllipse(Pens.Black, 10, 10, 200, 100)
bmp.Save("c:\test.png", Imaging.ImageFormat.Png)



相对应的，绘制填充椭圆的方法是FillEllipse()。此外，如果我们需要绘制一个圆形，只需要将最后两个参数设置为相同，即宽度和高度相同就可以了。

绘制弧线与扇形

我们绘制的弧线，实际上是椭圆线条的一部分，所以，我们使用Graphics对象中的DrawArc()方法绘制弧线时，首先需要指定它所在的椭圆，然后，我们通过两个参数来指定弧线位于椭圆上的位置，这两个参数分别是：

开始角度，在椭圆中，向右水平方向为0度，顺时针旋转为正数，逆时针旋转为负数，如下图所示。

[image: 图像说明文字]


另一个参数是弧线相对于圆心的角的度数。

下面的代码，我们首先绘制一个灰色的椭圆，然后绘制了一个红的弧线。


Dim bmp As New Bitmap(300, 200)
Dim g As Graphics = Graphics.FromImage(bmp)
g.DrawEllipse(Pens.Gray, 10, 10, 200, 100)
g.DrawArc(Pens.Red, 10, 10, 200, 100, 0, 60)
bmp.Save("c:\test.png", Imaging.ImageFormat.Png)



使用Graphics对象中的DrawPie()和FillPie()方法，我们可以分别绘制扇形或填充扇形，它们的参数与DrawArc()方法相似。如下面的代码，我们绘制了蓝色的填充扇形。


Dim bmp As New Bitmap(300, 200)
Dim g As Graphics = Graphics.FromImage(bmp)
g.DrawEllipse(Pens.Gray, 10, 10, 200, 100)
g.FillPie(Brushes.Blue, 10, 10, 200, 100, 0, 60)
bmp.Save("c:\test.png", Imaging.ImageFormat.Png)



代码绘制图形如下图所示。

[image: 图像说明文字]


绘制线段与多边形

绘制一条线段时，我们可以使用DrawLine()方法，其中，我们需要指定Pen对象作为绘制线段的样式，然后，我们需要指定线段的起始位置和结束位置。如下面的代代码，我们就绘制了一条黑色的线段。


Dim bmp As New Bitmap(300, 200)
Dim g As Graphics = Graphics.FromImage(bmp)
g.DrawLine(Pens.Black, 0, 0, bmp.Width, bmp.Height)
bmp.Save("c:\test.png", Imaging.ImageFormat.Png)



如果需要绘制各多风格的线段，我们需要自己创建Pen对象，如前面的内容中，我们讨论的绘制虚线的方法。

此外，如果我们需要同时绘制多条相连的线段，可以使用DrawLines()方法，其中，参数一为Pen对象，参数二为一个点坐标数组，我们可以使用Point或PointF数组结构。如下面的代码，我们通过四个端点的设置绘制了一个三角形，请注意其中第一个点和最后一个点坐标相同时，才能绘制一个完全封闭的多边形。


Dim bmp As New Bitmap(300, 200)
Dim g As Graphics = Graphics.FromImage(bmp)
Dim pts As Point() = {New Point(10, 10), New Point(100, 100),
                      New Point(50, 150), New Point(10, 10)}
g.DrawLines(Pens.Black, pts)
bmp.Save("c:\test.png", Imaging.ImageFormat.Png)



代码绘制结果如下图所示。

[image: 图像说明文字]


18.3. Font类与文本绘制

使用Graphics对象绘制文本时，我们应该首先定义字体，此时，我们可以使用Font类（System.Drawing命名空间），我们可以使用多个构造函数来创建Font对象，常用的有：


	Font(fontName, fontSize)

	Font(fontName, fontSize, fontStyle)



其中，fontName指定字体名称（String类型），fontSize指定字体和尺寸（Single类型），fontStyle定义字体风格，使用FontStyle枚举类型，我们还可以通过Or运算符同时指定多个字体风格；FontStyle枚举成员包括：


	Bold，文本加粗。

	Italic，使用斜体。

	Regular，普通文本。

	Strikeout，中间贯穿线。

	Underline，下划线。



创建Font对象后，我们就可以使用Graphics对象中的DrawString()方法绘制文本了。如下面的代码，我们将绘制蓝色的文本，并使用了Arial字体，以及加粗和斜体效果。


Dim bmp As New Bitmap(300, 200)
Dim g As Graphics = Graphics.FromImage(bmp)
Dim f As New Font("Arial", 16, FontStyle.Bold Or FontStyle.Italic)
g.DrawString("Hello VB.NET", f, Brushes.Blue, 1, 100)
bmp.Save("c:\test.png", Imaging.ImageFormat.Png)



代码绘制结果如下图所示。

[image: 图像说明文字]


18.4. 截取图像

前面，我们已经了解了关于图像绘制的基本资源和操作方法，接下来的内容，我们将讨论一些在实际应用中可能会需要的功能。本节，我们就从图像的截取功能开始。

18.4.1. 截取矩形图像

首先，我们讨论如何从一个图像中截取其中的一个矩形部分。这个功能实际上很简单，我们只需要使用Graphics对象中的DrawImage()方法就可以完成这项工作，我们所需要的DrawImage()方法的一个重载版本定义如下：


Public Sub DrawImage ( _
    image As Image, _
    destRect As Rectangle, _
    srcRect As Rectangle, _
    srcUnit As GraphicsUnit _
)



此方法的功能就是将图像的一部分绘制到Graphics对象关联的绘制对象中，其中：

image为截取的图像对象，由于Bitmap类是Image类的子类，所以，我们完全可以将此参数指定为一个Bitmap对象。

destRect参数指定截取部分要绘制到的矩形区域。

srcRect参数指定从image中截取的矩形区域。

srcUnit参数指定图像所使用的单位，它定义为GraphicsUnit枚举，其成员包括：


	Display，使用显示设备单位，如像素。

	Document，使用文档单位（1/300 英寸）。

	Inch，英寸。

	Millimeter，毫米。

	Pixel，设备像素。

	Point，点（1/72 英寸）。

	World，世界坐标系单位。



为了方便操作，我们在CImage类中封装了SnipRectangle()方法进行图像的截取操作，其定义如下面的代码（VBChef/CImage.vb文件）。


'' 截取矩形区域，返回Bitmap对象
Public Shared Function SnipRectangle(img As Image, _
              x As Integer, y As Integer, _
              width As Integer, height As Integer) As Bitmap
    Dim bmp As New Bitmap(width, height)
    Dim g As Graphics = Graphics.FromImage(bmp)
    g.DrawImage(img, New Rectangle(0, 0, width, height), _
                New Rectangle(x, y, width, height), _
                GraphicsUnit.Pixel)
    Return bmp
End Function



实际应用中，我们可以使用类型如下代码来操作。


Dim bmp1 As New Bitmap("c:\a.jpg")
Dim bmp2 As Bitmap = CImage.SnipRectangle(bmp1, 100, 100, 300, 200)
bmp2.Save("c:\b.jpg", Imaging.ImageFormat.Jpeg)



代码中，我们将c:\a.jpg图片中从(100,100)开始，宽300像素，高200像素的区域截取并保存到c:\b.jpg文件中。大家可以准备一张自己喜欢的图片，并命名为a.jpg，然后进行测试。

18.4.2. 截取圆形图像

载取圆形图像时，需要我们多做一些工作，在CImage类中，我们定义了SnipEllipse()方法来截取椭圆或圆形的图像，其定义如下面的代码（VBChef/CImage.vb文件）。


'' 截取圆形区域，返回Bitmap对象
Public Shared Function SnipEllipse(ByRef img As Image, _
              x As Integer, y As Integer, _
              width As Integer, height As Integer) As Bitmap
    Dim mask As New Bitmap(width, height)
    Dim g As Graphics = Graphics.FromImage(mask)
    g.FillRectangle(Brushes.White, 0, 0, width, height)
    g.FillEllipse(Brushes.Black, 0, 0, width, height)
    mask.MakeTransparent(Color.Black)
    Dim bmp As Bitmap = SnipRectangle(img, x, y, width, height)
    g = Graphics.FromImage(bmp)
    g.DrawImage(mask, 0, 0, width, height)
    Return bmp
End Function



在SnipEllipse()方法中，我们首先创建一个蒙板层（mask对象），在这个对象中，我们做了以下一些工作：


	将底色全部设置为白色。

	绘制了一个所需截取区域相同的黑色填充椭圆。

	使用Bitmap对象中的MakeTransparent()方法设置透明色，即在图像中不显示的颜色；我们可以看到，代码中，我们将黑色设置为透明色，这样，椭圆区域就会变成透明的。



然后，我们使用SnipRectangle()方法截取了椭圆图像所在的矩形区域图像，并将蒙板层绘制到这个矩形图像上；这样，只有其中的椭圆区域会因为透明而显示，其它区域则显示为白色（mask对象的底色）。

下面的代码，我们将截取c:\a.jpg图片文件中的一个圆形区域，并保存到c:\c.jpg文件中。


Dim bmp1 As New Bitmap("c:\a.jpg")
Dim bmp2 As Bitmap = CImage.SnipEllipse(bmp1, 100, 100, 500, 500)
bmp2.Save("c:\c.jpg", Imaging.ImageFormat.Jpeg)



SnipEllipse()方法中，如果需要将椭圆以外的区域设置为透明，我们同样可以调用bmp.MakeTransparent(Color.White)语句来实现，只是在椭圆区域内的白色也会变成透明；不过，只要我们再做一次加工就可以解决这个问题，接下来，我们将Return bmp语句修改为下面的代码即可。


bmp.MakeTransparent(Color.White)
Dim result As New Bitmap(width, height)
g = Graphics.FromImage(result)
g.FillEllipse(Brushes.White, 0, 0, width, height)
g.DrawImage(bmp, 0, 0, width, height)
Return result



实际上， 我们做的工作很简单，只是再通过一个Bitmap对象，将椭圆或圆形区域下面绘制成白色的底色，以此补上透明区域的白色。

18.5. 转换尺寸与格式

图片文件尺寸与格式的转换操作实际上非常简单，我们只需要使用Bitmap类的构造函数和Save()方法，以及ImageFormat类中相应的图片格式对象即可。如下面的代码，我将c:\a.jpg文件转换为PNG格式，宽度和高度都缩小为原来的50%，并保存在c:\a.png文件中。


Dim bmp1 As New Bitmap("c:\a.jpg")
Dim bmp2 As New Bitmap(bmp1, bmp1.Width \ 2, bmp1.Height \ 2)
bmp2.Save("c:\c.png", Imaging.ImageFormat.Png)



18.6. 旋转与翻转

对于图片的旋转（Rotate）与翻转操作（Flip），我们可以通过Bitmap对象的RotateFlip()方法来完成，此方法使用一个参数来指定进行的旋转和翻转操作，它定义为RotateFlipType枚举类型，其成员包括：


	Rotate180FlipNone，旋转180度，不进行翻转操作。

	Rotate180FlipX，旋转180度，并进行水平翻转操作。

	Rotate180FlipXY，旋转180度，并进行水平翻转和垂直翻转操作。

	Rotate180FlipY，旋转180度，并进行垂直翻转操作。

	Rotate270FlipNone，旋转270度，不进行翻转操作。

	Rotate270FlipX，旋转270度，并进行水平翻转操作。

	Rotate270FlipXY，旋转270度，并进行水平翻转和垂直翻转操作。

	Rotate270FlipY，旋转270度，并进行垂直翻转操作。

	Rotate90FlipNone，旋转90度，不进行翻转操作。

	Rotate90FlipX，旋转90度，并进行水平翻转操作。

	Rotate90FlipXY，旋转90度，并进行水平翻转和垂直翻转操作。

	Rotate90FlipY，旋转90度，并进行垂直翻转操作。

	RotateNoneFlipNone，不进行旋转和翻转操作。

	RotateNoneFlipX，进行水平翻转操作。

	RotateNoneFlipXY，进行水平和垂直翻转操作。

	RotateNoneFlipY，进行垂直翻转操作。



下面的操作，我将对c:\a.jpg图片进行水平翻转的操作，并将新的图片保存为c:\d.jpg文件。


Dim bmp As New Bitmap("c:\a.jpg")
bmp.RotateFlip(RotateFlipType.RotateNoneFlipX)
bmp.Save("c:\d.jpg", Imaging.ImageFormat.Jpeg)



18.7. 线性渐变效果

实现图像中的颜色线性渐变效果，我们需要关注的主要资源包括LinearGradientBrush类和LinearGradientMode枚举类型，它们都定义在System.Drawing.Drawing2D命名空间。

LinearGradientBrush类用于定义线性渐变对象，实际应用中，主要有两种使用方法，即使用点到点的渐变效果，以及矩形渐变效果，接下来，我们首先看一看前一种渐变效果的实现。

下面的代码，我们显示了一个使用点到点线性渐变效果填充的圆形绘制。


Dim lb As New LinearGradientBrush(New Point(30, 30), _
                                  New Point(270, 270), _
                                Color.White, Color.Black)
Dim bmp As New Bitmap(300, 300)
Dim g As Graphics = Graphics.FromImage(bmp)
g.FillEllipse(lb, 0, 0, 300, 300)
bmp.Save("c:\e.png", Imaging.ImageFormat.Png)



代码绘制结果如下图所示。

[image: 图像说明文字]


默认情况下，如果渐变区域不能完全填充绘制区域，则渐变会进行平铺，如果我们需要设置平铺的方法，可以设置LinearGradientBrush对象的WrapMode属性，它定义为WrapMode枚举类型，其成员包括：


	Clamp，不进行平铺操作。

	Tile，平铺渐变或纹理。

	TileFlipX，平铺该纹理或渐变，并对纹理或渐变进行水平反转操作。

	TileFlipXY，平铺该纹理或渐变，并对纹理或渐变进行水平和垂直反转。

	TileFlipY，平铺该纹理或渐变，并对纹理或渐变进行垂直反转操作。



下面，我们将使用区域渐变来绘制矩形。如下面的代码。


Dim lb As New LinearGradientBrush( _
    New Rectangle(0, 0, 50, 50), _
    Color.White, Color.Black, _
    LinearGradientMode.Vertical)
Dim bmp As New Bitmap(300, 200)
Dim g As Graphics = Graphics.FromImage(bmp)
g.FillRectangle(lb, 0, 0, 300, 200)
bmp.Save("c:\f.png", Imaging.ImageFormat.Png)



代码绘制结果如下图所示。

[image: 图像说明文字]


代码中，我们创建了一个50*50像素的垂直渐变区域，在绘制填充矩形时，我们可以看到，渐变区域进行了平铺。

创建线性渐变区域时，我们还可以设置线性渐变的方向，一种方法就是像代码中那样使用LinearGradientMode枚举类型，它的成员包括：


	BackwardDiagonal，从右上到左下。

	ForwardDiagonal，从左上到右下。

	Horizontal，从左到右，即水平方向。

	Vertical，从上到下，即垂直方向。



另一种设置线性渐变方向的方法是直接指定角度，我们可以在LinearGradientBrush类的构造函数中直接指定，如下面的代码。


Dim lb As New LinearGradientBrush( _
            New Rectangle(0, 0, 50, 50), _
            Color.White, Color.Black, 15)
Dim bmp As New Bitmap(300, 200)
Dim g As Graphics = Graphics.FromImage(bmp)
g.FillRectangle(lb, 0, 0, 300, 200)
bmp.Save("c:\f.png", Imaging.ImageFormat.Png)



代码中，我们在LinearGradientBrush()方法的最后一个参数指定线性渐变的方向为15度，输出效果如下图所示。

[image: 图像说明文字]


18.8. 图片DPI与尺寸

DPI（Dots Per Inch），即每英寸的点数，在我们处理的Bitmap对象中，就是每英寸的像素，根据这些，我们就可以通过设置图像的DPI和像素值来确定图像的真正尺寸，当我们需要在应用中直接打印图片时，这个功能就有了用武之地了。

如果图片的DPI能够达到300，图片的清晰度就应该能够满足一般的图像质量的要求了；当然，当我们了解了计算方法，设置更高的DPI也不是问题。下面，我们就以300DPI为例，介绍图像的尺寸问题。

如果我们需要100毫米*75毫米，DPI为300的图片，就可以通过下面的代码计算出图像所需要的像素值，并创建Bitmap对象。


Dim dpi As Single = 300
Dim mmPerInch As Single = 25.4F  '每英寸25.4毫米
Dim pixelPerMm As Single = dpi / mmPerInch  '每毫米多少像素
Dim width As Integer = CInt(100 * pixelPerMm)
Dim height As Integer = CInt(75 * pixelPerMm)
Dim bmp As New Bitmap(width, height)
bmp.SetResolution(dpi, dpi)



下面的代码，我们将通过PrintDocument类（System.Drawing.Printing命名空间）打印一幅图片。


Sub Main()
    Dim pd As New PrintDocument()
    AddHandler pd.PrintPage, AddressOf pd_PrintPage
    pd.Print()
End Sub

''
Private Sub pd_PrintPage(sender As Object, e As PrintPageEventArgs)
    Dim dpi As Single = 300
    Dim mmPerInch As Single = 25.4F  '每英寸25.4毫米
    Dim pixelPerMm As Single = dpi / mmPerInch  '每毫米多少像素
    Dim width As Integer = CInt(100 * pixelPerMm)
    Dim height As Integer = CInt(75 * pixelPerMm)
    Dim bmp As New Bitmap(width, height)
    bmp.SetResolution(dpi, dpi)
    Dim g As Graphics = Graphics.FromImage(bmp)
    g.FillRectangle(Brushes.Gray, 0, 0, bmp.Width, bmp.Height)
    ' 
    e.Graphics.DrawImage(bmp, 0, 0)
End Sub



代码会直接在默认的打印机中输出一个大约100毫米宽，75毫米高的灰色矩形。

在这里，我们主要使用了PrintDocument类中的PrintPage事件和Print()方法，其中，Print()方法用于启动打印，而打印的内容则是在PrintPage事件中定义。代码中，我们首先绘制了一个100毫米宽，75毫米高的灰色矩形图片，然后，通过PrintPage事件中第一个参数e的Graphics属性进行绘制，而此对象绘制的内容就是通过打印输出的内容。





19. 网络

本章，我们将讨论与网络操作相关的一些基本内容，主要包括：


	测试网络状态

	下载与上传

	发送邮件



19.1. 测试网络状态

网络相关的操作，最基本的一项工作就是检查网络的状态；接下来，我们就来讨论关于网络状态的测试工作。

19.1.1. 使用My.Computer.Network类

在VB.NET中，我们可以使用My命名空间中的Network类来进行一基本的操作，如我们可以使用IsAvailable属性测试计算机是否已连接到网络。

此外，如果我们需要测试某一服务器是否连通，可以使用Ping()方法，此方法的第一个参数就是服务器的主机（计算机名称、URL或IP），可以使用字符串或Uri对象。参数二可指定测试的超时时间（毫秒），定义为Integer类型。测试成功Ping()方法返回True，否则返回False。

使用Ping()方法时应注意，某些计算机启动了防火墙功能，当我们对这些计算机进行Ping操作时，可能不会得到响应，在本章稍后的内容中，我们可以看到相应内容的讨论。

19.1.2. 使用Ping类

NET Framework中，在System.Net命名空间定义了一系列网络操作的相关内容，下面，我们就讨论关于测试网络连接的资源应用。

用于Ping操作的功能，我们可以使用System.Net.NetworkInformation命名空间中的Ping类。如下面的代码，我们演示了Ping类的使用。


Using oPing As New Ping()
    Dim reply As PingReply = oPing.Send("www.ituring.com.cn")
    If reply.Status = IPStatus.Success Then
        Console.WriteLine("网络连接成功")
    Else
        Console.WriteLine("网络连接失败")
    End If
End Using



代码中，我们使用了Ping对象中的Send()方法，它的功能就是测试一个网络主机的连接，我们可以使用字符串形式，也可以使用IPAddress对象。此外，我们同样可以指定测试的超时时间（毫秒），以及测试时需要向主机发送的数据（字节数组）。

Send()方法返回的测试结果为PingReply对象，其中的Status属性可以获取测试的结果，它定义为IPStatus枚举类型，成员Success则表示网络连接正常。

实际应用中，我们可以通过测试一个互联网主机来测试计算机是否已连接到互联网。如果觉得上面的对Ping类的使用比较麻烦，我们还可以封装自己的Ping()方法，如下面的代码（VBChef/CNet.vb文件）。


Imports System.Net
Imports System.Net.NetworkInformation
Namespace vbchef
    Public Class CNet
        '' 封装Ping()方法
        Public Shared Function Ping(addr As String) As Boolean
            Using oPing As New Ping()
                Dim reply As PingReply = oPing.Send(addr)
                Return reply.Status = IPStatus.Success
            End Using
        End Function
        ''
    End Class
End Namespace



19.2. 下载与上传

本节，我们将讨论如果从网络计算机中下载内容，或者将本机内容上传到网络计算机，在VB.NET中，我们同样可以使用两种方式来完成这些工作。

19.2.1. 使用My.Computer.Network类

前面，我们已经了解了Network类中的IsAvailable属性和Ping()，本节，我们会使用它的另外两个成员，即DownloadFile()和UploadFile()方法。

DownloadFile()方法用于从网络计算机中下载文件，并保存到本地计算机中指定的位置。此方法有多个重载版本，我们可以考虑使用其中的几个，如：


DownloadFile(String, String)
DownloadFile(String, String, String, String)
DownloadFile(String, String, String, String, Boolean, Integer, Boolean)



在这些版本的DownloadFile()方法中：


	参数一指定网络文件的位置，可以使用String类型，也可以使用Uri对象类型。

	参数二指定下载文件保存到本地计算机的路径。

	参数三和参数四用于指定登录到网络计算机的用户名和密码。

	参数五指定是否显示登录界面。

	参数六指定连接超时时间，单位为毫秒。

	参数七指定本地计算机如果有同名的文件，是否使用下载的新文件替换它。



UploadFile()方法的功能是将本地计算机中的文件上传到指定的网络计算机，其中第一个参数为本地文件的路径，第二个参数为网络地址，而其它的参数则与DownloadFile()方法的参数设置相似。

19.2.2. 使用WebClient类

与My.Computer.Network类相比，WebClient类的功能会更强大，它定义在System.Net命名空间。

在WebClient类中，用于从网络计算机中下载内容的基本方法有三个：


	DownloadFile()方法，用于下载网络文件，并保存到本地计算机。参数一为网络文件的地址，可以使用String或Uri对象；参数二为保存到本地计算机文件的路径。

	DownloadString()方法，从网络资源中获取文本内容，参数为网络资源的地址，可以使用String或Uri对象。方法返回获取的文本内容。

	DownloadData()方法，从网络资源中获取数据，参数为网络资源的地址，可以使用String或Uri对象。方法返回字节数组形式的数据。



获取网络资源时，我们还可以使用WebClient对象中的QueryString指定查询参数，此属性定义为一个NameValueCollection类型，我们可以使用其Add()方法添加查询参数，如：


Dim client As New WebClient()
client.QueryString.Add("id", 1)
Dim str1 As String = client.DownloadString("hostname")



你可以在DownloadString()方法中使用自己需要的实际网络资源地址。

前面，我们提到过，对于启动了防火墙的服务器，我们使用Ping()方法可能不会得到正确的回复；但是，如果我们知道这是一台Web服务器，就可以通过DownloadString()来测试服务是否正在运行；为了方便使用，我们在CNet类中定义了一个共享方法PingHttp()，它的功能就是测试网站类的服务器是否畅通，其定义代码如下（VBChef/CNet.vb文件）。


'' 测试HTTP类服务器状态
Public Shared Function PingHttp(addr As String) As Boolean
    Try
        Using client As New WebClient()
            If client.DownloadString(addr).Length > 0 Then
                Return True
            Else
                Return False
            End If
        End Using
    Catch ex As Exception
        Return False
    End Try
End Function



然后，我们通过一个普通的URI地址来测试Web服务器是否畅通，如下面的代码。


Console.WriteLine(CNet.PingHttp("http://192.168.0.1/"))



实际应用中，可以将网址设置为所需要的Web服务器地址。

此外，在WebClient类中还有一些比较重要的属性需要我们注意，如：


	Encoding属性，指定下载或上传时使用文本编码，默认为系统中的默认字符集，即Encoding.Default，我们也可以根据实际情况设置，如Encoding.UTF8等。

	如果网络资源访问需要用户名和密码，我们需要设置Credentials属性，使用NetworkCredential对象创建。



与下载网络资源相对应的上传方法包括：


	UploadFile()方法，用于将本地计算机文件上传到指定的网络地址。参数一为网络地址，可以使用String或Uri对象；参数二为本地计算机文件的路径。

	UploadString()方法，将文本内容上传到网络地址，参数一为网络资源的地址，可以使用String或Uri对象，参数二为上传的文本内容。

	UploadData()方法，将字节数组形式的数据上传到网络地址。参数一为网络资源的地址，可以使用String或Uri对象。参数二为上传的数据。



19.3. 发送邮件

自动发送邮件是很多自动化系统的必备功能，在.NET Framework中，为我们提供了相关的操作方法，本节，我们就来讨论发送邮件的相关内容。

一般来讲，我们使用SMTP（简单邮件传输协议）发送邮件，在System.Net.Mail命名空间为我们提供了相关资源，如SmtpClient类就可以使用SMTP发送邮件。

为了简化邮件的发送工作，我们将封装一个CMail类，它的基本定义如下（VBChef/CMail.vb）：


Imports System
Imports System.Net
Imports System.Net.Mail

Namespace vbchef
    Public Class CMail

    End Class
End Namespace



接下来，我们来看看发送邮件都需要些什么。首先是SMTP服务器信息，主要包括包括了主机域名或IP、用户名和密码。然后，还需要定义需要发送的邮件相关信息，主要包括发件人、邮件主题、正文和附件等。在CMail类中，我们定义了一些字段来表示这些信息，如下面的代码（VBChef/CMail.vb文件）。


'' SMTP服务器
Protected mySmtpHost As String = ""
Protected myUserName As String = ""
Protected myPassword As String = ""
'' 发件人
Protected myFromAddr As String = ""
Protected myDisplayName As String = ""
'' 邮件
Protected mySubject As String = ""
Protected myContent As String = ""
Protected myAttachment As New List(Of String)



面对这些信息，如果我们分别定义为属性，使用起来并不方便，我们的方案是将这些信息进行归纳，通过方法来分别设置。首先是SMTP服务器的相关信息，如下面定义的SetSmtpHost()方法（VBChef/CMail.vb文件）。


 '' 设置SMTP服务器信息
Public Sub SetSmtpHost(sHost As String, sUser As String, sPwd As String)
    mySmtpHost = sHost
    myUserName = sUser
    myPassword = sPwd
End Sub



在SetSmtpHost()方法中，我们分别设置了SMTP服务器名称，以及登录所需要的用户名和密码。

然后是发件人的信息，包括发件人邮箱和显示的发件人名称，如下面的代码（VBChef/CMail.vb文件）。


'' 设置发件人信息
Public Sub SetFrom(sFromAddr As String, sDisplayName As String)
    myFromAddr = sFromAddr
    myDisplayName = sDisplayName
End Sub



接下来，我们设置邮件的内容，如下面的代码（VBChef/CMail.vb文件）。


 '' 设置邮件主内容
Public Sub SetMail(sSubject As String, sContent As String)
    mySubject = sSubject
    myContent = sContent
End Sub



最后是添加附件，如下面的代码（VBChef/CMail.vb文件）。


'' 添加附件
Public Sub AddAttachment(sFile As String)
    myAttachment.Add(sFile)
End Sub



那么，这些与邮件相关的信息如何使用呢？下面的SendTo()方法将通过这些信息来发送邮件（VBChef/CMail.vb文件）。


 '' 发送邮件
Public Function SendTo(ByRef sTo As List(Of String), _
           Optional isHtml As Boolean = False) As Boolean
    Try
        Dim mail As New MailMessage()
        mail.Subject = mySubject
        mail.Body = myContent
        mail.From = New MailAddress(myFromAddr, myDisplayName)
        mail.IsBodyHtml = isHtml
        Dim maxIndex As Integer = sTo.Count - 1
        For i = 0 To maxIndex
            mail.To.Add(sTo(i))
        Next
        maxIndex = myAttachment.Count - 1
        For i = 0 To maxIndex
            mail.Attachments.Add(New Attachment(myAttachment(i)))
        Next
        '
        Dim smtp As New SmtpClient()
        smtp.Host = mySmtpHost
        smtp.Credentials =  _
            New NetworkCredential(myUserName, myPassword)
        smtp.DeliveryMethod = SmtpDeliveryMethod.Network
        smtp.Send(mail)
        Return True
    Catch
        Return False
    End Try
End Function



其中，我们可以看到，在发送邮件时，我们主要使用了几个类型，如MailMessage、MailAddress、Attachment、SmtpClient和NetworkCredential类型；下面分别介绍。

MailMessage对象表示了一个邮件的内容，主要包括以下属性：


	Subject属性，String类型，指定邮件的主题。

	Body属性，String类型，指定邮件的内容。

	From属性，指定发件人信息，定义为MailAddress对象，代码中， 我们使用发件人邮箱地址和显示名称初始化MailAddress对象。

	IsBodyHtml属性，Boolean类型，指定邮件的内容是否为HTML格式，默认为False，即邮件内容为纯文本格式。

	Attachments属性，指定邮件的附件部分，它定义为一个集合，我们可以使用Add()方法添加一个或多个附件。附件使用Attachment对象，我们可以附件文件的路径来初始化它。



真正的发送邮件操作，我们需要使用SmtpClient类，常用成员包括：


	Host属性，指定SMTP服务器。

	Credentials属性，指定登录SMTP服务的凭证，我们可以使用用户名和密码创建一个NetworkCredential对象。

	DeliveryMethod属性，指定邮件的发送方式，定义为SmtpDeliveryMethod枚举类型，我们使用其中的Network指定邮件通过网络发送到SMTP服务器。

	Send()方法，执行发送邮件的操作，参数为MailMessage对象。



在SendTo()方法中，第一个参数为收件人的邮箱地址，我们使用了List(Of String)类型，如果只需要一个收件人，我们还可以重载一个SendTo()方法，如下面的代码（VBChef/CMail.vb文件）。


Public Function SendTo(ByRef sTo As String, _
           Optional isHtml As Boolean = False) As Boolean
    Dim lst As New List(Of String)
    lst.Add(sTo)
    Return Me.SendTo(lst, isHtml)
End Function



实际使用中，我们可以使用类似如下的代码来发送邮件。


Dim m As New CMail()
m.SetSmtpHost("SMTP主机", "用户名", "密码")
m.SetFrom("发件人邮箱", "发件人署名")
m.SetMail("邮件主题", "邮件正文")
Console.WriteLine(m.SendTo("收件人邮箱"))







20. 数据库应用基础

本章，我们将以SQL Server 2012 Express为例，介绍数据库应用的一些基本知识，并对一些基本的操作进行封装。

通过本章的学习，我们应该能够理解数据库操作的常用概念，并能够在VB.NET中对数据库进行操作。虽然本章以SQL Server数据库为例，但对于市面上大多关系型数据库系统来讲，它们在操作层面是非常接近的，特别在SQL的应用上；所以，只要我们理解了SQL Server数据库的操作，就可以举一反三，快速学习其它关系型数据库的应用。

本章的内容包括：


	关系型数据库应用基础

	ADO.NET应用基础

	封装CDataItem和CDataCollection类

	插入新记录（Insert语句）

	更新记录（Update语句）

	删除记录（Delete语句）

	查询记录（Select语句）

	存储过程（Stored Procedure）

	事务（Transaction）



20.1. 关系型数据库概述

简单来讲，关系型数据库管理系统（Relational Database Management System）就是以“表（Table）”作为保存数据的基本形式，配合各种表与表之间的关系，从而构架一个数据存储、管理、应用等操作的数据综合处理系统。

一个完整的数据库管理系统，包括了数据的保存、删除、查询、优化、安全性等一系列操作，也是应用程序管理数据的常见工作，在VB.NET项目中，我们可以通过.NET Framework或第三方开发的ADO.NET组件，方便地操作各种数据库，如SQL Server、MySQL、Oracle等，此外，如果我们开发的是数据量不大的单机版程序，还可以使用桌面级数据库Access，这样数据库文件就可以和应用程序一同发布和转移，软件的部属工作更加方便、灵活。

一名合格的数据库管理员并不是短时间内可以修炼完成的，需要对数据库的实现、内部原理、SQL等知识有着深入地学习和理解，并在长期地学习、工作实践中累积，最终，才能够成为一名合格地，能够应对各种突发事件的数据库专业管理员。而且，数据库管理员也绝对不是一项轻松的工作。

作为软件开发人员，也许我们并不需要像数据库管理员那样深入地去学习数据库的管理工作，但是，对于数据库基础知识的理解，以及常用操作的了解，可以帮助我们在软件中更加高效地使用数据库系统，从而能够更有效地处理软件中的业务数据。

20.1.1. SQL

SQL（Structured Query Language，结构化查询语言）是关系型数据库操作的语言标准，是对数据库进行管理、定义、查询等一系列操作的完整语言标准体系。

具体到某一种数据库管理系统，对于SQL标准的实现又不是完全一致，不过，当我们学习了一种数据库中的SQL（如SQL Server中的Transact-SQL，简称T-SQL）以后，再学习其它数据库中的SQL语法也不会太难，毕竟关系型数据库的基本概念是相通的。

稍后，我们会看到在SQL Server数据库中一些常用的SQL查询语法，并会对一些常用的SQL语句创建操作进行封装。

20.1.2. 表

在关系型数据库中，数据是以二维表的形式保存和管理时，下面就是二维表的结构示意图。

在二维表结构中，列（column）又称为字段（field），通过字段，我们可以定义数据名称和类型等信息。行（row）又称为记录（record），表示一行数据（一条记录），而记录中的每个数据规则是通过相应的字段设置（名称、类型等）来决定的。在定位一个数据项时，我们会说某条记录中的某一字段的值。

20.1.3. 字段数据类型

字段，也就是表中每一列的设置，主要包括了字段名称和数据类型，以及其它一些特性，如Identity、唯一键、数据规则检查等等。在这里，我们先来看一看常用的数据类型有哪些。

SQL Server数据库从2005版本以后就和.NET Framework平台有着密不可分的关系了，所以，使用的数据类型也非常接近，我们知道，.NET Framework中定义的数据类型位于System命名空间，而在.NET Framework相对应的SQL Server数据库数据类型则定义在System.Data.SqlTypes命名空间，大家可以在这里查看完整的数据类型信息。

本书中，我们会在SQL Server中更多地使用以下几种数据类型：


	int，32位整数，对应Integer类型，即Int32类型。

	bigint，64位整数，对应Long类型，即Int64类型。

	decimal，对应VB.NET中的Single、Double或Decimal。在SQL Server中，我们可以通过decimal(m,n)格式来定义字段的取值范围和小数的精度，如decimal(10,2)就是定义一个包含8位整数部分和2位小数和部分的数据类型（合计为10位数字）。

	datetime，日期与时间类型，对应VB.NET中的Date类型，即DateTime结构类型。

	nvarchar，用于保存可变长度的Unicode编码的文本内容，我们可以指定它的字符数量，如nvarchar(50)允许最多50个字符，而nvarchar(max)则允许平台所确定的最大数量的字符数量。



此外，在定义字段类型时，我们还可以使用一个关键字进行附加的设置，如：


	not null，不允许数据项为空值（NULL）。请注意，空值和0是两个不同的概念，数据库中的空值是指没有数据，在.NET Framework中使用DBNull.Value值来表示。

	identity，指定字段为Identity字段，默认情况下，此字段的值会从1自动开始，然后每次生成新的Identity都会加1。使用Identity字段时，我们还可以指定起始值，如identity(1,1)，其第一个1为Identity开始的数值，第二个1为每次增加的步长。请注意，如果生成的Identity没有使用，则会跳过，下一次会生成新的Identity。

	unique，指定字段拥有唯一键（unique key），即表示每条记录中的此字段值不能重复。



20.1.4. 主键与外键

主键（primary key）和外键（foreign key）是属于数据表的特性，它们一起使用，可以创建表与表之间的主子关系，以下，我们以常见的超市小票信息为例。

小票中，我们可以看到客户号、柜员号、总金额、打印时间，以及各种商品的名称、单价、数量及金额小计信息，这实际上就是一种典型主子表结构。其中客户号、柜员号、总金额、打印时间等信息是购买批次的总体信息，即主数据；而商品列表则是分项信息，即子数据。在数据库中，我们可以使用如下图中所示的结构来表示它们的关系。

其中，主表中的“购物批次”就是该表的主键，主键的数据和唯一键一样，是不能重复的；而子表中的“购物批次”则是该表的外键，它用于将购买的商品信息与相应的购物批次信息关联，每个购物批次会有一个或多个商品信息。

我们可以看到，主子关系实际上是一种“一对多”的关系。此外，应注意区分主键与唯一键的不同，唯一键是字段属性，而主键则表的特性，在一个表中，主键可以使用多个字段共同限定唯一的一条记录。

下面的代码，我们在SQL Server 2012 Express创建表示以上结构的两个表，分别是主表order_main和子表order_sub。


-- 创建数据库
create database cdb_demo;
go

-- 引用数据库
use cdb_demo;
go

-- 创建主表
create table order_main(
MRID bigint identity not null primary key,
OrderNumber nvarchar(15) not null unique,
CashDesk int not null check(CashDesk>0),
PrintTime datetime not null,
OrderSum decimal(8,2) not null
);

-- 创建子表
create table order_sub(
SRID bigint identity not null primary key,
MRID bigint not null 
    foreign key references order_main(MRID),
CommName nvarchar(10) not null,
CommPrice decimal(8,2) not null,
CommNumber decimal(8,2) not null,
CommSum decimal(8,2) not null
);



在order_main和order_sub表中，我们都定义了MRID（Main Record Identity）字段，在order_main表中，它定义为主键（PK），而在order_sub表中，它定义为外键（FK），通过两个表中的MRID字段，我们就创建了两个数据表的主子关系。

请注意，我们并没有使用OrderNumber字段作为主键，这是因为在数据库的处理中，整数要比大量的文本数据处理快的多，所以，我们应该优先考虑使用整数作为Identity或其它索引类的字段类型。

我们可以SQL Server Management Studio中执行此SQL代码，创建cdb_demo数据库和这两个数据表。

20.1.5. 视图

在数据库中，可以将视图（view）看作是一个查询模板，我们可以直接通过视图来查询数据；但有一点应注意，视图的内容并不是物理存在的，当我们调用视图内容时，数据库会执行查询操作，然后再从查询结果中返回数据，所以，视图并不会提高查询速度，但是，视图可以定义复杂的查询，这样就可以避免我们多次重写这些复杂的查询语句了。

在SQL Server数据库中，我们使用create view语句创建视图，它的基本应用格式如下。


create view <视图名称>
as 
<查询语句>



稍后会介绍常用的select语句，然后，我们就可以使用这些语句创建自己的视图了。

在数据库中，还有很多的概念需要我们了解和学习，在后面的讨论中，我们会学习一些常用的SQL语句，以及一些与软件开发相关的常用概念，如存储过程、事务等。不过，在这之前，我们先来看一看在.NET Framework中用于操作数据库的组件如何使用。

20.2. ADO.NET应用基础

在.NET Framework中，用于访问和操作数据库的一系列资源称为ADO.NET，它们都定义在System.Data命名空间中，其中，常用的资源包括：


	System.Data命名空间，用于定义数据操作组件的基本资源，如相关的操作接口等。

	System.Data.Common命名空间，用于定义数据操作组件的基类等资源。

	System.Data.SqlClient命名空间，定义了用于操作SQL Server数据库的相关资源。

	System.OleDb命名空间，定义了用于操作OLEDB数据源的相关资源，如果我们需要使用桌面数据库Access，则需要使用此命名空间中的相关组件。



对于一些数据库开发组件，如MySQL、Oracle等，我们可以使用数据库公司或组织开发的ADO.NET组件，相信这些组件对相应的数据库操作会更高效。

20.2.1. ADO.NET组件简介

在ADO.NET组件中，常用的资源按照其操作数据库时的连接状态可以分为两类，它们是：


	连接类组件，这类组件在使用时必须连接数据库服务器，如DbConnection、DbCommand、DbDataReader、DbDataAdapter等组件。请注意，这里指出的都是相关组件的基类，它们都定义在System.Data.Common命名空间。以下内容，我们在使用通用组件时也会使用基类的名称来表示。

	非连接类组件，又称为脱机组件。这些组件在使用时并不需要连接数据库。如DataSet、DataTable等资源，这些脱机组件都定义在System.Data命名空间。



接下来，我们就主要以SQL Server数据库为主，讨论一些基本的操作资源的应用问题。请注意，我们使用的用于操作SQL Server数据库的组件都定义在System.Data.SqlClient命名空间，如SqlConnection、SqlCommand等。

20.2.2. 连接SQL Server数据库

连接数据库的组件基于DbConnection类，而连接SQL Server数据库的具体类型则是SqlConnection，它定义在System.Data.SqlClient命名空间。接下来，我们就来看看如何使用SqlConnection对象连接SQL Server数据库。

首先，我们应该了解，使用ADO.NET组件连接数据库，我们需要使用数据库连接字符串，从.NET Framework 2.0开始为我们提供了SqlConnectionStringBuilder类，它的功能就是帮助我们动态创建SQL Server数据库连接字符串；另一方面，使用SqlConnectionStringBuilder类创建数据库连接字符串的好处就是提高了安全性，避免了动态创建数据库连接串可能造成代码注入危险。

下面就是SqlConnectionStringBuilder类中的常用成员：


	DataSource属性，String类型，指定SQL Server数据库的服务器，包括服务器名称、IP、端口，以及SQL Server实例信息。

	InitialCatalog属性，String类型，指定默认连接的数据库名称。

	IntegratedSecurity属性，Boolean类型，指定是否使用集成安全性登录。如果SQL Server数据库和应用在同一计算机，我们可以将此属性设置为True，即使用“Windows身份验证”以简化数据库的登录。如果数据库位于远程服务器，即将此属性设置为False，然后使用用户名和密码进行登录，即使用“SQL Server身份验证”进行登录，此时需要数据库管理员分配用户及相应的操作权限。

	UserID属性，String类型，用于指定远程登录的用户名。

	Password属性，String类型，用于指定远程登录的密码。

	AsynchronousProcessing属性，Boolean类型，用于指定是否启动异步处理，默认为False。

	Pooling属性，Boolean类型，指定是否启动数据库连接池，以便更有效地管理数据库连接。默认值为True。

	ConnectionString属性，String类型，设置或获取实际的数据库连接字符串。



在我们封装的CSql类中，封装了两个用于创建SQL Server数据库连接串的方法，如下面的代码（VBChef/CSql.vb文件）。


'' 创建SQL Server连接字符串，Windows安全登录
Public Shared Function CreateLocalCnnStr(ByRef sDbName As String, _
              Optional ByRef sServer As String = ".") As String
    Dim cnnBuilder As New SqlConnectionStringBuilder()
    With cnnBuilder
        .DataSource = sServer
        .InitialCatalog = sDbName
        .IntegratedSecurity = True
        .AsynchronousProcessing = True
        .Pooling = True
    End With
    Return cnnBuilder.ConnectionString
End Function

'' 创建SQL Server连接字符串，远程登录
Public Shared Function CreateRemoteCnnStr(ByRef sDbName As String, _
              ByRef sServer As String, _
              ByRef sUser As String, _
              ByRef sPwd As String) As String
    Dim cnnBuilder As New SqlConnectionStringBuilder()
    With cnnBuilder
        .DataSource = sServer
        .InitialCatalog = sDbName
        .IntegratedSecurity = False
        .UserID = sUser
        .Password = sPwd
        .AsynchronousProcessing = True
        .Pooling = True
    End With
    Return cnnBuilder.ConnectionString
End Function



代码中，我们分别定义了两个用于连接SQL Server数据库的方法。如：

CreateLocalCnnStr()方法用于创建本机的SQL Server数据库服务。其中包括两个参数，参数一用于指定连接的默认数据库名称，参数二指定连接的服务器名称和实例，默认为“.”，它的含义是指本机的默认实例。请注意，SQL Server Express的默认实例名为SQLEXPRESS，如果在安装时没有修改，则我们使用使用“.\SQLEXPRESS”作为第二个参数。本书中的演示数据库，我们可以使用如下代码生成它的连接串。


Dim dbCnnStr As String = CSql.CreateLocalCnnStr("cdb_demo")



CreateRemoteCnnStr()方法用于连接远程服务器中的SQL Server，方法包括以下参数：


	sDbName，指定连接的数据库名称。

	sServer，指定SQL Server服务器和实例。我们可以使用计算名称、IP地址，也可以包含端口号，如“192.168.0.1,1433”；此外，如果我们不是连接默认的实例，还需要指定相应的实例，如“192.168.0.1\SQLEXPRESS”。

	sUser，指定连接使用的用户名。

	sPwd，指定连接密码。



创建了数据库连接字符串，我们就可以使用以下代码连接SQL Server数据库了。


Using cnn As New SqlConnection(dbCnnStr)
    cnn.Open()
    '数据操作代码
End Using



请注意，我们使用了Using...End Using语句结构，这样，无论数据库操作是否正确，都可以正确地关闭数据库连接，因为SqlConnection实现了IDisposable接口。

20.2.3. 执行命令

在SQL Server数据库执行SQL或调用存储过程，并返回执行的结果，我们可以使用SqlCommand对象，它包括以下一些常用成员：


	CommandText，指定执行的SQL、存储过程名或表名。

	CommandType，定义为CommandType枚举类型，用于指定CommandText内容的类型，默认为CommandType.Text，即表示CommandText中的内容是SQL语句；如果指定调用的是存储过程，我们可指定为CommandType.StoredProcedure。另一个可能的值是CommandType.TableDirect，表示CommandText中指定的是数据表名称，用于直接返回数据表内容的操作。

	Parameters属性，定义为参数集合类型（SqlParameterCollection），我们可以使用AddWithValue(<参数名>,<参数值>)方法添加参数，也可以使用Clear()方法清除所有参数。



在一个数据库连接中执行命令，我们首先需要创建一个SqlCommand对象，并与数据库连接关联，如下面的代码。


Using cnn As New SqlConnection(dbCnnStr)
    cnn.Open()
    ' 创建SqlCommand对象
    Dim cmd As SqlCommand = cnn.CreateCommand()
    '数据操作代码
End Using



代码中，我们使用SqlConnection对象中的CreateCommand()方法创建SqlCommand对象。

当我们执行命令后，可以通过以下三个方法返回执行结果。


	ExecuteNonQuery()方法，返回SQL执行所影响的记录数，返回结果为Integer类型。此方法的异步版本为BeginExecuteNonQuery()和EndExecuteNonQuery()方法。

	ExecuteScalar()方法，返回查询结果中第一条记录第一个字段的值，返回结果为Object类型。

	ExecuteReader()方法，返回查询的记录集，返回结果为SqlDataReader类型。此方法的异步版本为BeginExecuteReader()和EndExecuteReader()方法。



SqlDataReader类是一个单向访问的数据集合，访问一条记录后，就不能再次访问。它的主要成员包括：


	Read()方法，向下读取一条记录，成功返回True，没有记录返回False。

	HasRows属性，表示对象中是否包含记录。

	FieldCount属性，返回对象中的字段数量。

	GetName()方法返回字段的名称，其参数为从0开始的索引。

	Item默认属性，可以使用数值或字段名索引获取当前记录中相应字段的数据。一般会在Read()读取成功时才会使用Item属性获取数据。

	此外，我们还可以使用一系列的GetXXX()方法获取当前记录指定字段的值，其中的XXX为.NET Framework数据类型名称，方法会返回相应类型的数据，如果字段数据不能成功转换为目标类型，则会发生异常。使用这些方法时，我们应该首先知道数据表中字段类型的定义。



20.2.4. 传递参数

使用参数向数据库发送数据比直接使用SQL语句有一个最重要的优势，即使用参数可以有效避免SQL注入操作，提高了数据库的安全性。

为了便于向SqlCommand对象中添加参数，我们在CSql类中定义了相关的方法，如下面的代码（VBChef/CSql.vb文件）。


''向SqlCommand对象中添加参数
Public Shared Sub AddCommandParameters(ByRef cmd As SqlCommand, _
                      ByRef dataColl As CDataCollection, _
                      Optional paramPrefix As String = "")
    If cmd Is Nothing OrElse dataColl Is Nothing  _
                      OrElse dataColl.Count < 1 Then
        Exit Sub
    End If
    '
    Dim maxIndex As Integer = dataColl.Count - 1
    Dim dItem As CDataItem
    For i As Integer = 0 To maxIndex
        dItem = dataColl(i)
        cmd.Parameters.AddWithValue( _
            String.Format("@{0}{1}", paramPrefix, dItem.Name), _
            dItem.Value)
    Next
End Sub



请注意参数中的CDataCollection类型，这是我们自己创建的一个数据集合类，稍后会讨论它的定义。

20.2.5. DataSet和DataTable类

DataSet是一个非常重要的脱机数据组件，它甚至可以定义一个小型的本地“数据库”。在DataSet对象中，我们可以定义数据表（DataTable）、视图（DataView），以及表之间的关系（DataRelation）。这些类型都定义在System.Data命名空间。

本书中，我们主要会应用DataSet及DataTable类，首先来讨论DataSet类的使用。

DataSet可以作为一个脱机的数据集，我们可以使用DbAdapter对象向DataSet对象中填充（Fill）数据，如下面的代码。


Using cnn As New SqlConnection(dbCnnStr)
    cnn.Open()
    Dim cmd As SqlCommand = cnn.CreateCommand()
    cmd.CommandText = sql
    Using ada As New SqlDataAdapter(cmd)
        Dim ds As New DataSet()
        ada.Fill(ds, "table1")
        ' 应用或返回ds对象
    End Using
End Using



代码中，可以看到SqlDataAdapter类的使用，首先，我们可以看到它的构造函数，我们使用了一个SqlCommand对象来初始化它。然后，我们使用了Fill()方法向DataSet对象中填充数据，Fill()方法的第二个参数是指定填充数据在DataSet中DataTable对象的名称（Name属性）。

在DataSet对象中，包含一个表的集合（DataTableCollection），我们可以通过DataSet对象的Tables属性获取。然后，我们可以通过数值索引或表名来获取一个DataTable对象，如下面的代码。


Dim table0 As DataTable = ds.Tables(0)
Dim table1 As DataTable = ds.Tables("table1")



接下来，我们来看看DataTable有哪些常用成员。


	Columns属性，返回表中字段集合，其中的Count属性可以获取字段的数量；此外，我们可以通过数值索引或字段名获取相应的DataColumn对象。

	Rows属性，返回表中数据记录的集合，使用Count属性可以获取记录数量。然后，我们可通过数值索引可以获取相应的记录（DataRow），通过DataRow对象的默认属性Item，我们可以通过数值索引或字段名获取相应的数据。



20.3. 封装CDataItem和CDataCollection类

我们可以看到，ADO.NET组件中，为我们提供了大量的数据库操作资源，而另一方面，在.NET Framework中也为我们提供了很多集合操作的类型，那么，我们为什么还要封装CDataItem和CDataCollection类呢？

实际上，我们的目的也很简单，即结合了数据的包装、转换和传递等功能；在后面的讨论中，我们可以看到这两个类的实际应用，来这里，我们先来看看它们的定义，就当大家回顾类的定义了。

20.3.1. CDataItem类

CDataItem类用来表示一个数据项，下面的代码就它的基本定义（VBChef/CDataItem.vb文件）。


<Serializable>
Public Class CDataItem
    Private myName As String
    Private myValue As Object
    ''
    Public Sub New(Optional sName As String = "", _
                   Optional oValue As Object = Nothing)
        myName = sName
        myValue = oValue
    End Sub
    ''
    Public Property Name As String
        Get
            Return myName
        End Get
        Set(value As String)
            myName = value
        End Set
    End Property
    '
    Public Property Value As Object
        Get
            Return myValue
        End Get
        Set(value As Object)
            myValue = value
        End Set
    End Property
'' 其它代码
End Class



代码中可以看到，定义CDataItem类时，我们使用了SerializableAttribute特性，也就是说，此类型的对象是可以进行序列化操作的。接下来，我们定义了一个构造函数，它的两个参数分别用于指定数据名称和数据值，而且都是可选的，如果不指定参数，而默认使用空字符串作为数据名称，使用Nothing作为数据值。然后，就是两个属性，即数据名称和数据值。

接下来，CDataItem另一个主要的功能就是数据的快速转换，我们设置了一系列的只读属性，用于直接返回相应类型的数据，下面的代码给出其中的几个（VBChef/CDataItem.vb文件）。


''
Public ReadOnly Property StrValue As String
    Get
        Return ToStr(myValue)
    End Get
End Property
''
Public ReadOnly Property IntValue As Integer
    Get
        Return ToInt(myValue)
    End Get
End Property
''
Public ReadOnly Property LngValue As Long
    Get
        Return ToLng(myValue)
    End Get
End Property
''
Public ReadOnly Property DblValue As Double
    Get
        Return ToDbl(myValue)
    End Get
End Property



我们可以看到这些属性的实现，它们会直接返回相应类型的数据，如果myValue不能转换为目标类型，则返回目标类型的默认值。完整的属性列表，大家可以查看下载的源代码。

此外，如果大家需要判断数据值是不是空值，或者其它功能，可以定义一系列的判断属性，如下面的代码（VBChef/CDataItem.vb文件）。


''
Public ReadOnly Property IsEmpty As Boolean
    Get
        Return (myName = "")
    End Get
End Property
''
Public ReadOnly Property IsNull As Boolean
    Get
        Return (myValue Is Nothing)
    End Get
End Property



如果大家在使用CDataItem类时还需要什么功能，还可以进行扩展。

20.3.2. CDataCollection类

CDataCollection类用于定义一个数据集合，在它的内部，每一个数据项都是一个CDataItem对象，我们使用泛型List类型来定义数据项列表，如下面的代码（VBChef/CDataCollection.vb文件）。


<Serializable>
Public Class CDataCollection
    Private myList As List(Of CDataItem)

    '' 构造函数
    Public Sub New(ByRef dItem As CDataItem)
        myList = New List(Of CDataItem)
        myList.Add(dItem)
    End Sub

    Public Sub New(sName As String, oValue As Object)
        Me.New(New CDataItem(sName, oValue))
    End Sub

    Public Sub New()
        myList = New List(Of CDataItem)
    End Sub
'' 其它代码...
End Class



代码中，在CDataCollection的内部，我们使用myList对象来保存数据项列表，它定义为List(Of CDataItem)类型，即列表成员都必须是CDataItem对象。随后的三个构造函数，相信并不难理解，此外，我们可以看到，无论使用哪个构造函数，我们都对myList对象进行了初始化，这样，在操作过程中就可以避免空对象的检查工作。

接下来是一个默认属性Item，它的功能是设置或访问一个数据项，其定义如下面的代码（VBChef/CDataCollection.vb文件）。


'' 默认属性，数据集合项目，数值索引
Default Public Property Item(index As Integer) As CDataItem
    Get
        If index >= 0 And index < myList.Count Then
            Return myList(index)
        Else
            Return New CDataItem()
        End If
    End Get
    Set(value As CDataItem)
        myList(index) = value
    End Set
End Property
''
Default Public Property Item(sName As String) As CDataItem
    Get
        Dim index As Integer = Me.Find(sName)
        Return Me(index)
    End Get
    Set(value As CDataItem)
        Dim index As Integer = Me.Find(sName)
        If index >= 0 And index < myList.Count Then
            myList(index) = value
        Else
            myList.Add(value)
        End If
    End Set
End Property



代码并不难理解，需要注意的是，当我们没有找到指索引的数据项时，没有返回Nothing，而是返回一个空的CDataItem对象。

此外，代码中使用了Find()方法还没有定义，它的功能是根据数据名称查找数据项，并返回数据项的索引值，如果没有找到则返回-1，它的定义如下面的代码（VBChef/CDataCollection.vb文件）。


 '' 查找指定数据名称的数据，返回其索引值，没有返回-1
Public Function Find(ByRef sName As String) As Integer
    Dim maxIndex As Integer = myList.Count - 1
    For index = 0 To maxIndex
        If String.Compare(myList(index).Name, sName, False) = 0 Then
            Return index
        End If
    Next
    Return -1
End Function



请注意，在Find()中比较数据项的名称时，我们将区分大小写，这也是在.NET Framework架构中其它集合类（如Hashtable）的默认处理方式。

接下来，我们定义了用于添加数据项的方法，有两个方法，即Add()和Append()方法，每个方法也分别有两个重载版本。Add()和Append()方法的区别在于，Add()方法会判断同名的数据项是否存在，如存在则替换，如不存在则添加；而Append()方法会直接将数据项添加到列表中，而不会判断数据项是否存在。这两个方法的定义如下面的代码（VBChef/CDataCollection.vb文件）。


'' 添加数据项，数据名称不能重复
Public Sub Add(ByRef dItem As CDataItem)
    Dim index As Integer = Me.Find(dItem.Name)
    If index >= 0 Then
        myList(index) = dItem
    Else
        myList.Add(dItem)
    End If
End Sub

Public Sub Add(ByRef sName As String, ByRef oValue As Object)
    Me.Add(New CDataItem(sName, oValue))
End Sub

'' 直接追加数据项
Public Sub Append(ByRef dItem As CDataItem)
    myList.Add(dItem)
End Sub

Public Sub Append(ByRef sName As String, ByRef oValue As Object)
    Me.Add(New CDataItem(sName, oValue))
End Sub



最后，我们看一看移除数据项的方法Remove()，它同样有两个重载版本，分别使用索引值和数据项名称删除指定的数据项，并返回删除的数据项，如果指定的数据不存在，则返回Nothing值。

Remove()方法的定义如下（VBChef/CDataCollection.vb文件）。


'' 删除指定索引的数据项，并返回删除的对象
Public Function Remove(index As Integer) As CDataItem
    If index >= 0 And index < myList.Count Then
        Dim dItem As CDataItem = myList(index)
        myList.RemoveAt(index)
        Return dItem
    Else
        Return Nothing
    End If
End Function

'' 根据数据项名称删除
Public Function Remove(ByRef sName As String) As CDataItem
    Dim index As Integer = Me.Find(sName)
    Return Me.Remove(index)
End Function



20.4. 插入新记录（Insert语句）

接下来的操作，我们会使用一个名为user_main的数据表，它的创建SQL如下面的代码。


use cdb_demo;
go

create table user_main(
UserId bigint identity(1,1) not null primary key,
UserName nvarchar(15) not null unique,
UserPwd nvarchar(40) not null check(Len(UserPwd)=40),
Email nvarchar(50),
Sex int not null default(0),
IsLocked int not null default(1)
);

insert into user_main(UserName,UserPwd,IsLocked)
values('','D033E22AE348AEB5660FC2140AEC35850C4DA997',1);
insert into user_main(UserName,UserPwd,IsLocked)
values('admin','D033E22AE348AEB5660FC2140AEC35850C4DA997',0);



代码中，我们在cdb_demo数据库中创建了user_main表，并添加了两条记录，即空用户和admin用户的信息。

向数据表中添加新记录时，我们使用Insert语句，其一般的应用格式如下：


insert into <表名>(<字段列表>) values(<值列表>);



其中，<表名>就是要添加新记录的数据表名；<字段列表>是需要添加数据的字段名列表，使用逗号（,）分隔；<值列表>为需要添加的数据列表，它们应该和<字段列表>中的字段一一对应，多个值之间同样使用逗号分隔。

如我们向测试用的user_main数据表添加admin用户的数据时就使用了如下代码。


insert into user_main(UserName, UserPwd, IsLocked)
values('admin', 'D033E22AE348AEB5660FC2140AEC35850C4DA997', 0);



此外，在SQL Server中，从2005版本开始，支持inserted表，这是一个系统临时表，当我们使用insert语句添加记录时，inserted表就保存了新插入的数据，我们可以在代码中使用这个表的全部内容或部分内容。如下面的代码。


use cdb_demo;
go
insert into user_main(UserName, UserPwd, IsLocked)
output inserted.UserId,inserted.UserName
values('user', 'D033E22AE348AEB5660FC2140AEC35850C4DA997', 0);



我们在SQL Server Management Studio中执行此代码，可以得到如下图所示的输出结果。

[image: 图像说明文字]


代码中，我们使用了output关键字用于输出内容，然后我们引用inserted表中的UserId和UserName字段，即输出这两个字段的数据；如果我们需要使用输出所有新添加的数据，可以使用inserted.*。

为了便于Insert语句的创建，我们在CSql类中定义了一个CreateInsertSql()方法，它的功能就是根据指定的信息创建运行于SQL Server数据库中的Insert语句，如下面的代码（VBChef/CSql.vb文件）。


'' 生成Insert语句
Public Shared Function CreateInsertSql(ByRef sTable As String, _
              ByRef sIdName As String, _
              ByRef data As CDataCollection) As String
    ' 不允许空数据
    If data Is Nothing OrElse data.Count < 1 Then Return ""
    Dim sb As New StringBuilder(255)
    sb.AppendFormat("insert into {0}(", sTable)
    Dim sbValues As New StringBuilder(255)
    sbValues.AppendFormat(") output inserted.{0} values(", sIdName)
    ' 第一个字段及数据
    Dim sName As String = data(0).Name
    sb.Append(sName)
    sbValues.AppendFormat("@{0}", sName)
    ' 其它字段及数据
    Dim maxIndex As Integer = data.Count - 1
    For i = 1 To maxIndex
        sName = data(i).Name
        sb.AppendFormat(",{0}", sName)
        sbValues.AppendFormat(",@{0}", sName)
    Next
    '
    sbValues.Append(");")
    sb.Append(sbValues.ToString())
    Return sb.ToString()
End Function



在CreateInsertSql()方法中，我们定义了三个参数，包括：


	sTable指定插入记录的表名。

	sIdName指定数据表中的Identity字段名称，我们会使用inserted表返回新插入的记录Identity值，下一章，我们可以看到这个Identity的用途。

	data定义为CDataCollection类型，用于指定插入的数据。请注意，在代码中，我们必须至少有一个字段数据时才能正确创建和执行Insert语句，所以，如果此参数为Nothing或没有成员时，就会返回空的字符串。此外，对于字段数据，我们会使用参数的形式进行传递，实际应用中，我们会使用CSql.AddCommandParameters()方法来设置Insert语句中所需要的字段数据。



下面的代码测试了这个方法的使用。


Dim data As New CDataCollection()
data.Append("UserName", "testuser")
data.Append("UserPwd", "D033E22AE348AEB5660FC2140AEC35850C4DA997")
Dim sql As String = CSql.CreateInsertSql("user_main", "UserId", data)
Console.WriteLine(sql)



此代码会显示下面的Insert语句。


insert into user_main(UserName,UserPwd) 
output inserted.UserId 
values(@UserName,@UserPwd);



20.5. 更新记录（Update语句）

更新数据表中已存在的数据，我们可以使用Update语句，它的基本应用格式如下。


update <表名> set <字段1>=<值1>, <字段2>=<值2>, ...<字段n>=<值n>
where <条件>;



请注意，如果没有使用where关键字指定<条件>，那么，就会更新数据表中所有的记录；大多数情况下，这是非常危险的操作！！！所以，大家在执行update语句之前还是应该检查<条件>的设置是否合理。

下面的语句，我们会更新用户user的性别值。


use cdb_demo;
go
update user_main set Sex=1
where UserName='user';



在CSql类中，我们定义了CreateUpdateSql()方法来创建Update语句，如下面的代码（VBChef/CSql.vb文件）。


 '' 生成Update语句
Public Shared Function CreateUpdateSql(ByRef sTable As String, _
              ByRef data As CDataCollection, _
              ByRef cond As CDataCollection) As String
    ' 不允许空数据和空条件
    If data Is Nothing OrElse cond Is Nothing OrElse _
        data.Count < 1 OrElse cond.Count < 1 Then Return ""
    Dim sb As New StringBuilder(255)
    sb.AppendFormat("update {0} set ", sTable)
    ' 第一个数据
    Dim sName As String = data(0).Name
    sb.AppendFormat("{0}=@{1}", sName, sName)
    ' 其它数据
    Dim maxIndex As Integer = data.Count - 1
    For i = 1 To maxIndex
        sName = data(i).Name
        sb.AppendFormat(",{0}=@{1}", sName, sName)
    Next
    ' 第一个条件
    sName = cond(0).Name
    sb.AppendFormat(" where {0}=@cond_{1}", sName, sName)
    ' 其它条件
    maxIndex = cond.Count - 1
    For i = 1 To maxIndex
        sName = cond(i).Name
        sb.AppendFormat(" and {0}=@cond_{1}", sName, sName)
    Next
    '
    sb.Append(";")
    Return sb.ToString()
End Function



在CreateUpdateSql()方法中，我们定义了三个参数：


	sTable指定需要更新记录的数据表名。

	data指定新的数据，使用CDataCollection对象。不能为Nothing，而且成员数量要大于0。

	cond指定更新记录的条件，同样使用CDataCollection对象。同样不能为Nothing，成员数量不能为0，即不允许无条件更新。



下面的代码演示了这个方法的使用。


Dim data As New CDataCollection()
data.Append("Sex", 1)
data.Append("IsLocked", 0)
Dim cond As New CDataCollection("UserName", "testuser")
Dim sql As String = CSql.CreateUpdateSql("user_main", data, cond)
Console.WriteLine(sql)



执行此代码，我们会看到如下输出。


update user_main set Sex=@Sex,IsLocked=@IsLocked 
where UserName=@cond_UserName;



请注意，在执行Update语句时，数据和条件的字段可能相同，但数据未必相同，所以，我们在条件的字段数据参数名称中使用了“cond_”前缀，在CSql.AddCommandParameters()方法中，我们看到相对应的参数定义（paramPrefix）。

20.6. 删除记录（Delete语句）

删除数据时，我们使用Delete语句，它的基本应用格式如下：


delete from <表名> where <条件>;



请注意，和Update语句一样，没有条件的删除语句是非常危险的！！！ 此外，在SQL Server 2005及更新的版本中包含了deleted表，它的功能是保存刚刚删除的数据。

如下面的代码，我们将删除用户名为User的记录：


use cdb_demo;
go
delete from user_main 
output deleted.*
where UserName='user';



代码执行会显示如下图所示的结果。

[image: 图像说明文字]


在CSql类中，我们定义了CreateDeleteSql()方法来完成创建Delete语句的工作，如下面的代码（VBChef/CSql.vb文件）。


'' 生成Delete语句
Public Shared Function CreateDeleteSql(ByRef sTable As String, _
               ByRef cond As CDataCollection) As String
    ' 不允许无条件删除
    If cond Is Nothing OrElse cond.Count < 1 Then Return ""
    Dim sb As New StringBuilder(255)
    sb.AppendFormat("delete from {0} where ", sTable)
    ' 第一个条件
    Dim sName As String = cond(0).Name
    sb.AppendFormat("{0}=@{1}", sName, sName)
    ' 其它条件
    Dim maxIndex As Integer = cond.Count - 1
    For i = 1 To maxIndex
        sName = cond(i).Name
        sb.AppendFormat(" and {0}=@{1}", sName, sName)
    Next
    '
    sb.Append(";")
    Return sb.ToString()
End Function



在CreateDeleteSql()方法中，我们只定义了两个参数：


	sTable指定了将要删除记录的数据表名称。

	cond指定删除条件，不能为Nothing，而且必须至少有一个条件。



下面的代码演示了此方法的使用。


Dim cond As New CDataCollection("UserName", "testuser")
Dim sql As String = CSql.CreateDeleteSql("user_main", cond)
Console.WriteLine(sql)



执行此代码，我们会看到如下的输出。


delete from user_main where UserName=@UserName;



我们知道，无条件的删除语句会删除数据表中所有的记录，一般来讲，这样很危险，不过，如果你的目的就是这样，使用truncate语句会更高效，它会将数据表初始到新创建的状态，连Identity字段也会重新从1开始。trancate语句的使用格式如下：


truncate table <表名>;



20.7. 查询记录（Select语句）

在数据库中，Select语句的使用是非常灵活的，本节，我们就以SQL Server 2012为例，讨论一些常用的Select语法。

20.7.1. 基本查询

使用select语句的基本格式如下：


select <字段> from <表或视图>;



其中的<表或视图>是我们需要查询数据的表或视图名称，而<字段>则是查询结果中需要返回的字段，我们可以使用一个字段，也可以是多个字段（使用逗号分隔），如果我们需要返回表或视图中的所有字段，可以使用通配符*。如：


select * from user_main;



此外，在select关键字后面，有两个常用的子句可以使用，即：


	top子句，用于指定查询结果返回的记录数量。在SQL Server中可以使用top(n)或 top n格式，而后者则是ASNI SQL的标准语法。

	distinct子句，指定查询结果中如果有重复的记录，只返回其中一条。



20.7.2. 查询条件

查询记录时，我们往往根据自己的所需要的条件来执行，此时，我们可以在select语句的后部使用where子句指定查询条件。

和编程语言中的条件表达式相似，在where子句中，我们可以指定一个条件，也可以使用多个条件，而条件之间的也可以使用标准的逻辑关系，如与（And）、或（Or）、非（Not）。下面，我们先来看一看每一个条件如指定。

首先，我们应该知道数据库中每种数据直接量的表示，如数字直接写，而字符串、日期和时间则需要使用一对单引号包含起来；然后，我们可以使用一个比较运算符，如：


	=运算符，判断字段数据是否等于一个值，如IsLocked=1、UserName='admin'等。

	<>运算符，判断字段数据是否不等于一个值。

	<运算符，判断字段数据是否小于一个值。

	>运算符，判断字段数据是否大于一个值。

	<=运算符，判断字段数据是否小于等于一个值。

	>=运算符，判断字段数据是否大于等于一个值。



此外，当我们判断字段数据是否为空值（NULL）时，应该使用以下方法：


	Is Null运算，判断字段数据是否为空值，如Email Is Null。

	Is Not Null运算，判断字段数据是否不是空值。



Like运算符一般用于对字符串内容的模糊查询，它的一般应用格式如下：


<字段名> Like '<匹配内容>';



在<匹配内容>中，我们可以使用以下一些通配符：


	?表示一个字符。

	%表示零个或多个字符。



此外，因为单引号用于指定文本内容，所以， 当我们查询单引号内容时可以使用两个连续的单引号进行转义，如：


UserName like 'Tom''%'



此语句实际匹配的内容是以Tom'开头的文本内容。

20.7.3. 排序（Order By子句）

使用Order By子句，我们可以对查询结果进行排序，如下面的代码。


select * from user_main order by UserName asc;



此语句就是对查询的用户记录按用户名（UserName）的内容进行排序，asc关键字指定进行升序排序，而实际上，默认就是进行升序排列，我们可以省略asc关键字，如下面的代码功能和上面的语句功能是一样的。


select * from user_main order by UserName;



如果我们需要将记录进行降序排列，可以使用desc关键字，如：


select * from user_main order by UserName desc;



此外，使用Order By子句时，我们还可以同时指定多个排序字段，如：


select * from user_main order by IsLocked, UserName;



代码会将查询结果首先按IsLocked字段排序，如果IsLocked字段数据相同，则按UserName字段数据排序。

20.7.4. 函数

在select语句中，我们还可以使用一些函数对数据进行一个简单的计算和汇总工作，常用的函数有：


	min()，求查询结果中指定字段数据的最小值。

	max()，求查询结果中指定字段数据的最大值。

	sum()，求查询结果中指定字段数据的和。

	count()，求查询结果中指定字段的记录数，实际上，它的功能就是统计查询结果的数量。

	avg()，求查询结果中指定字段数据的平均值。



如下面的语句，我们将从order_main数据表中查询交易金额的最大值：


select max(OrderSum) from order_main;



20.7.5. 分组（Group By子句）

Group By子句可以帮助我们进行简单的汇总统计，如下面的代码，就可以反映出锁定用户和没有锁定的用户数量。


select IsLocked as [锁定状态], count(IsLocked) as [数量]
from user_main group by IsLocked;



此代码会返回类似下图的统计表。

请注意，代码中的as关键字指定了字段的别名，如果别名包含容易引起歧义的内容，如特殊字符、空格、单引号等，可以像上面的代码那样将名称使用个对方括号包含起来。

20.7.6. 联合查询（Join子句）

前面的示例中，我们创建了order_main和order_sub表来演示“主/子”的结构，实际应用中，如果数据在两个表中查看不太方便，我们可以使用联合查询将查询结果形成一个二维表的形式，如下面的代码。


select M.*, S.* from 
order_main as M join order_sub as S on M.MRID = S.MRID;



执行此操作后，查询的结果就会包含order_main和order_sub两个表所有字段，如下图。

两个表中的数据则是通过MRID字段进行关联的，我们可以看到，查询结果中出现了两次MRID字段，如果我们只需要其中的一部分字段，则应该分别写出它们，而不是使用通配符*。此外，语句中的as关键字在这里是用于指定表的别名，起个简单的名称可以让我们在列出字段时少输入很多字符。^_^

下面的代码，我们将创建一个视图，而它的内容就是所有销售数据。


use cdb_demo;
go

create view v_order
as
select M.MRID,M.OrderNumber,M.CashDesk,M.PrintTime,M.OrderSum,
S.SRID, S.CommName,S.CommPrice,S.CommNumber,S.CommSum
from 
order_main as M join order_sub as S on M.MRID = S.MRID;



代码中，我们创建了一个名为v_order的视图，在接下来的查询，或者是在VB.NET项目中，我们就可以直接通过这个视图来查询所有的销售数据。如：


select * from v_order;



20.7.7. 子查询

我们知道，Select语句会返回二维表形式的数据；如果我们只选择了一个字段，那么，就可以将这个单字段的查询结果当作一个列表来使用，从而可以创建更加复杂的查询。如下面的代码，我通过子查询查找购买了“商品1”的主销售记录数据。


select * from order_main 
where MRID in (select MRID from order_sub where CommName = '商品1');



代码中，in关键字用于指定字段数据应该是列表中的内容，我们也可以自己指定查询的列表内容，如：


select * from user_main 
where UserName in ('admin', 'administartor', 'guest');



代码会返回用户名为admin、administartor或guest的用户记录。

20.7.8. 封装CSql.GetSelectSql()方法

数据应用开发中，大多数的数据查询都是比较简单的操作，我们完全可以通过标准化的封装来简化这些操作；另一方面，对于复杂的查询操作，我们应该充分发挥数据库系统的优势，如使用视图或存储过程来定义查询模板，而不是在VB.NET代码中书写复杂的查询代码。

在CSql类中，我们封装了CreateSelectSql()方法，它的功能就是使用标准的Select进行查询，而且，在查询条件中会使用严格的限定，即使用参数传递数据，并且查询条件之间使用与（And）关系。其定义代码如下（VBChef/CSql.vb文件）。


 '' 创建Select语句
Public Shared Function CreateSelectSql(ByRef sTable As String, _
              ByRef sField As String, _
              Optional ByRef cond As CDataCollection = Nothing, _
              Optional ByRef returnCount As Integer = -1, _
              Optional ByRef addCondition As String = "") As String
    Dim sb As New StringBuilder(255)
    sb.Append("select")
    If returnCount > 0 Then
        sb.AppendFormat(" top {0} ", returnCount)
    End If
    sb.AppendFormat(" {0} from {1} ", sField, sTable)
    '查询条件
    If cond IsNot Nothing AndAlso cond.Count > 0 Then
        ' 第一个条件
        Dim sName As String = cond(0).Name
        sb.AppendFormat(" where {0}=@{1}", sName, sName)
        '其它条件
        Dim maxIndex As Integer = cond.Count - 1
        For i As Integer = 1 To maxIndex
            sName = cond(i).Name
            sb.AppendFormat(" and {0}=@{1}", sName, sName)
        Next
        '附加条件
        If addCondition <> "" Then
            sb.AppendFormat(" and ({0}) ", addCondition)
        End If
    ElseIf addCondition <> "" Then
        sb.AppendFormat(" where {0}", addCondition)
    End If
    '
    sb.Append(";")
    Return sb.ToString()
End Function



在下一章中，我们会看到此方法的具体应用。

20.8. 存储过程（Stored Procedure）

存储过程就像批处理一样，我们可以将一系列的查询进行封装，必要时传入相应的参数，然后返回查询的结果。在数据应用开发中，我们可以通过存储过程充分利用数据库系统的查询功能，帮助我们执行一些复杂的查询，一方面可以减少应用程序代码中的大规模数据处理（毕竟这不是它们的强项），另一方面，也可以减少应用程序与数据库系统的交互，有效地提高了系统和网络资源的利用效率。

在SQL Server中创建存储过程，我们可以使用create procedure语句，对于我们自己创建的存储过程，习惯使用usp_作为其名称的前缀。如下面的代码，我们就创建了一个用于检测用户登录的存储过程usp_login，我们需要传入用户名和登录密码，然后，存储过程会返回执行的结果，NULL值表示登录失败，大于0的整数（用户ID）则是登录成功。


use cdb_demo;
go

create procedure usp_login
@username nvarchar(15),
@userpwd nvarchar(40)
as
begin
declare @userid bigint;
select UserId from user_main 
where IsLocked = 0 and UserName=@username and UserPwd=@userpwd;
end



在SQL Server数据库中，我们可以使用如下代码来调用这个存储过程。


use cdb_demo;
go

declare @username nvarchar(15);
declare @userpwd nvarchar(40);
set @username = 'admin';
set @userpwd = 'D033E22AE348AEB5660FC2140AEC35850C4DA997';

execute usp_login @username,@userpwd;



而在VB.NET项目中，我们可以使用如下代码来调用这个存储过程。


Dim dbCnnStr As String = CSql.CreateLocalCnnStr("cdb_demo")
Using cnn As New SqlConnection(dbCnnStr)
    cnn.Open()
    Dim cmd As SqlCommand = cnn.CreateCommand()
    cmd.CommandText = "usp_login"
    cmd.CommandType = CommandType.StoredProcedure
    cmd.Parameters.AddWithValue("@username", "admin")
    cmd.Parameters.AddWithValue("@userpwd", GetSha1("admin"))
    Console.WriteLine(cmd.ExecuteScalar())
End Using



我们可以在VBChef项目中测试此代码，因为在这里，我们使用了已经封装过的CSql.CreateLocalCnnStr()和MCommon.GetSha1()方法。

20.9. 事务（Transaction）

数据库中的事务处理最大的特点就是其任务具有“原子性”，也就是说，事务中的工作要不全部完成，要不就什么也不做。一个经典的例子就是账户的转账操作，假如我们需要从账户A中转账10000元到账户B，那么，我们就必须从账户A转出10000元，然后账户B存入10000元，只有这两项工作都顺利完成了，整个转账工作才算成功完成，否则，就会出问题了，10000元不翼而飞可不是什么好事，对吧？

我们先来看下面的代码，它会在SQL Server中执行一个事务，不过，这个事务并不会成功完成。


use cdb_demo;
go

begin transaction

insert into user_main(UserName,UserPwd)
values ('test_user','D033E22AE348AEB5660FC2140AEC35850C4DA997');

update user_main set IsLocked = 'a'
where UserName = 'test_user';

commit transaction



我们可以看到，在代码中，我们使用begin transaction语句开始一个事务，其中的第一条语句是在用户表中插入一条新记录，这条语句并没有问题；而问题就出在第二条语句，我们更新新记录的IsLocked字段的值，它应该是一个整数，而我们在代码中指定了一个字符串类型的数据，此语句就会出错，这样，最后的commit transaction语句就不会执行，也就是不会提交事务所做的操作，此时，第一条语句也不会在用户表中插入新记录，即事务中的所有操作都不会执行。

下面，我们修改update语句的内容，如下面的代码。


update user_main set IsLocked = 0
where UserName = 'test_user';



然后，再执行此事务，我们就会发现，新的记录被创建，而且IsLocked字段数据被修改为0。我们可以使用以下代码查询这条记录。


select * from user_main where UserName='test_user';



对事务的操作，还有一条常用语句，即rollback transaction，它的功能是对事务中的操作进行回滚，即还原操作。

请注意，并不是所有操作都能够完全恢复到事务执行前的状态，如Identity值，每次执行Insert语句时，都会生成新的Identity，如果语句执行失败，生成的Identity值也会被丢弃，而在下一次执行Insert语句时，会产生新的Identity值。

下一章的内容中，我们还可以看到在ADO.NET组件中使用事务的方法。





21. 创建数据层组件

本章，我们将综合代码结构、设计模式、数据库操作等开发技术和方法，创建一个用于数据库操作的通用组件库，在软件开发中，我们可以可以灵活应用，并根据需要进行修改和扩展。

上一章，我们了解了SQL Server 2012数据库的一些基本操作，而在数据应用项目中，这些操作也是最常用到的，在基本的软件开发架构，如三层架构（界面、逻辑、数据）中，对数据库的操作都是非常重要的一个环节。

在前面的讨论中，我们也可以看到，对于数据库的基本操作，有着非常多的、相似的代码，而在软件开发中，当我们看到大量的重复代码时，就应该考虑如何简化它们的使用。本章，我们就来讨论通用的数据库操作组件，它的功能就是在逻辑代码（业务代码）和数据库之间构建一个通用的操作“接口”，通过这一系列的组件，我们可以在众多逻辑代码中，使用相同的操作方法来访问不同类型的数据库，相信这对于软件的创建和维护工作都有着极大的帮助。

在《数据应用开发详解（C#版）》中，我们详细讨论了数据层组件的创建思路和过程，以及如果在多种项目类型、多种数据库之间进行操作的方法，有兴趣研究的朋友可以参考。

本章的讨论，我们主要还以SQL Server 2012为主，并讨论了扩展Access和MySQL数据库操作组件所需要的注意内容，在本书的源代码中，包含了这三种数据库操作的完整代码。

本章内容主要包括：


	定义数据操作接口

	实现数据引擎组件

	实现记录操作组件

	Access数据库组件

	MySQL数据库组件



21.1. 定义数据操作接口

本节，我们将讨论数据库的基本操作接口，并定义为VB.NET的接口类型。我们可以看到，在ASP.NET项目中，对于数据库的操作，主要有以下几种：


	连接数据库。如连接SQL Server数据库，我们会使用ADO.NET中的SqlConnection类、连接Access数据库可以使用OleDbConnection类、连接MySQL数据库则可以使用MySqlConnection类等等。

	查询。我们会通过一系列的DbCommand类型、DbParameter类型等组件来完成数据库的查询操作，并返回查询结果。如DbCommand中定义的ExecuteNonQuery()方法返回影响的数据记录数、ExecuteScalar()方法返回查询结果中第一条件记录第一个字段的数据、ExecuteReader()方法返回一个查询记录集等等。

	插入新记录。我们同样使用一系列的DbCommand类型执行Insert语句，并返回执行的结果。

	更新已有数据。我们使用一系列的DbCommand类型执行Update语句，并返回执行的结果。

	删除数据。使用一系列的DbCommand类型执行Delete语句，并返回执行结果。



针对以上常用操作，我们将创建两个主要的数据操作接口类型，即数据引擎接口和数据记录操作接口。以下分别介绍。

21.1.1. 数据引擎接口

数据引擎接口的功能主要是连接数据库和执行基本的数据查询工作。在VB.NET中，我们使用IDbEngine接口来定义这些工作，如下面的代码（VBChef/DataInterfaces.vb文件）。


'' 数据引擎接口，数据库连接及数据搜索
Public Interface IDbEngine
    Property CnnStr As String
    ReadOnly Property EngineType As EDbEngineType
    ReadOnly Property Connected As Boolean
    Function GetValue(ByRef sTable As String, _
                      ByRef sField As String, _
                      ByRef cond As CDataCollection) _
                  As Object
    Function GetRecord(ByRef sTable As String, _
                       ByRef sFieldList As String, _
                       ByRef cond As CDataCollection) _
                   As CDataCollection
    Function GetDataSet(ByRef sTable As String, _
                        ByRef sFieldList As String, _
                        ByRef cond As CDataCollection) _
                    As DataSet
    Function SpGetValue(ByRef spName As String, _
                        ByRef cond As CDataCollection) _
                    As Object
    Function SpGetRecord(ByRef spName As String, _
                         ByRef cond As CDataCollection) _
                     As CDataCollection
    Function SpGetDataSet(ByRef spName As String, _
                          ByRef cond As CDataCollection) _
                      As DataSet
End Interface



我们首先来看IDbEngine接口中定义的三个属性：


	CnnStr为可读写属性，其功能是获取或设置数据库连接字符串。

	EngineType为只读属性，它将返回实现数据引擎类中的数据库类型。

	Connected同样为只读属性，它将返回数据库是否可以正常连接。



其中的EngineType属性定义为EDbEngineType枚举类型，它的定义如下面的代码（VBChef/DataInterfaces.vb文件）。


 '' 数据引擎枚举类型
Public Enum EDbEngineType As Integer
    Unknow = 0
    Access = 1
    SqlServer = 2
    MySQL = 3
    Oracle = 4
End Enum



当然，你也可以根据需要扩展此枚举类型的成员。

接下来，我们回到IDbEngine接口，先看一个GetValue()方法的定义。它的功能是返回一个查询值（Object类型），其参数定义如下：


	sTable参数指定查询数据的表（或视图）。

	sField指定返回数据的字段名称。

	cond指定查询条件，我们使用CDataCollection对象来传递数据。



再看GetRecord()方法，它的功能是返回一条查询记录（CDataCollection对象），其参数与GetValue()方法相似，只是第二个参数应指定返回字段名的列表，而不是一个单一的字段名。

GetDataSet()方法会返回一个查询记录集（DataSet对象），查询结果会保存在DataSet中的第一个表对象Table(0)中。此方法的参数与GetRecord()方法完全一致。

接下来的SpGetValue()、SpGetRecord()和SpGetDataSet()方法，它们的功能是通过调用存储过程（Stored Procedure）返回相应的数据，如一个值（Object类型）、一条记录（CDataCollection对象）和一个记录集（DataSet）。这三个方法的参数定义是一样的，其中，参数一指定存储过程的名称，参数二则指定存储过程执行所需要的参数（CDataCollection对象）。

稍后，我们会看到如何实现这个接口的一系列组件。

21.1.2. 数据记录操作接口

在前一章的讨论中，我们可以看到，对于数据表中记录的操作，主要有插入、更新和删除操作，当然，我们也可以进行查询操作。

请注意，在我们的数据记录操作组件中，会基于一个重要的约定，即每个数据表都应该有一个Identity字段，可以是32位整数或64位整数。在Access中，Identity字段可以使用“自动编号”类型；在MySQL数据库中，可以在定义字段时使用auto_increment关键字或直接使用serial定义字段类型。

接下来，我们创建了IDbRecord接口来定义数据记录的相关操作，如下面的代码（VBChef/DataInterfaces.vb文件）。


'' 数据记录操作接口
Public Interface IDbRecord
    Property DbEngine As IDbEngine
    ReadOnly Property IdName As String
    ReadOnly Property TableName As String
    ' 记录查询
    Function Find(ByRef cond As CDataCollection, _
                  ByRef notEqualId As Long) As Long
    Function Find(ByRef cond As CDataCollection) As Long
    Function Find(ByRef idValue As Long) As Long
    ' 记录保存
    Function Save(ByRef data As CDataCollection, _
                  ByRef cond As CDataCollection) As Long
    ' 载入记录
    Function Load(ByRef sFieldList As String, _
                  ByRef cond As CDataCollection) _
              As CDataCollection
    Function Load(ByRef idValue As Long) As CDataCollection
    ' 删除记录
    Function Delete(ByRef cond As CDataCollection) As Long
    Function Delete(ByRef idValue As Long) As Long
End Interface



在IDbRecord接口中，我们同样定义了三个属性：


	DbEngine定义为可读写属性（IDbEngine接口类型），用于获取或设置记录操作的数据引擎。

	TableName和IdName属性定义为只读属性，用于获取数据表名称和Identity字段名称。



下面，我们再来看看关于数据记录的操作方法，首先是Find()方法，它的功能主要是进行记录的查询，即判断指定条件的记录是否存在，如果存在，则返回第一条记录的Identity值，如果不存在返回0，执行错误返回-1。Find()方法共有三个重载版本：


	Find(CDataCollection, Long)方法，第一个参数为查询的条件，请注意，这些条件将按与（And）关系使用；第二个参数一个Identity值，在查询结果中将不包含此Identity的记录，不过，此参数只在大于0的时候有效。

	Find(CDataCollection)方法，根据指定的条件查询记录。

	Find(Long)方法，根据指定的Identity查找记录。



在IDbRecord接口中，我们还定义了一个用于保存记录的方法，它包含两个参数，都定义为CDataCollection对象；第一个参数为要保存的数据，第二个参数为更新记录时的条件。关于这个方法的实现，我们需要以下约定：


	首先判断参数一中是否包含Identity信息，如果包含则按此Identity值更新（Update）相应的记录，并会忽略参数二指定的条件。

	参数一不包含Identity信息时，如果参数二指定了条件，则进行记录的更新（Update）操作，如果没有指定更新条件，则进行插入（Insert）新记录的操作。



Load()方法用于载入一条记录，返回格式为CDataCollection对象，如果没有找到满足条件的记录，则返回Nothing值；Load()方法有两个重载版本：


	Load(String, CDataCollection)方法，参数一指定需要返回的字段列表，参数二指定记录查询的条件，它们也会使用与（And）关系来使用。

	Load(Long)方法，载入指定Identity值的记录。



Delete()方法用于删除记录，请注意，我们不允许无条件的删除操作。此方法操作成员时会返回1，删除操作执行错误会返回-1。如果指定条件的记录不存在，则ExecuteNonQuery()方法会返回0，即没有真的删除记录；不过，我们的目的就是让它们不存在，所以，当Delete语句正确执行后，我们就约定删除操作成功了。Delete()方法同样有两个重载版本。


	Delete(CDataCollection)方法，根据指定的条件删除记录，请注意，我们不允许无条件的删除操作。

	Delete(Long)方法，删除指定Identity值的记录。



介绍了我们定义的数据操作接口类型，接下来的工作，我们就分别来实现数据引擎组件和数据记录操作组件。

21.2. 实现数据引擎组件

数据引擎组件的实现会基于IDbEngine接口，我们会首先使用一个基类CDbEngineBase，然后，具体的数据库操作类都会基于这个基类，其基本结构如下图。

21.2.1. CDbEngineBase基类

首先，我们来看CDbEngineBase基类，它实现了IDbEngine接口类型，其定义如下面的代码（VBChef/CDbEngineBase.vb文件）。


Public MustInherit Class CDbEngineBase
        Implements IDbEngine
    ''
    Private myCnnStr As String
    '' 构造函数
    Public Sub New(ByRef sCnnStr As String)
        Me.CnnStr = sCnnStr
    End Sub

    '' 数据库连接串属性
    Public Property CnnStr As String Implements IDbEngine.CnnStr
        Get
            Return myCnnStr
        End Get
        Set(value As String)
            myCnnStr = value
        End Set
    End Property

    '' 子类需要重写的成员
    Public MustOverride ReadOnly Property Connected As Boolean _
           Implements IDbEngine.Connected
    Public MustOverride ReadOnly Property EngineType As EDbEngineType _
           Implements IDbEngine.EngineType
    Public MustOverride Function GetDataSet(ByRef sTable As String, _
               ByRef sFieldList As String, _
               ByRef cond As CDataCollection) As DataSet _
           Implements IDbEngine.GetDataSet
    Public MustOverride Function GetRecord(ByRef sTable As String, _
               ByRef sFieldList As String, _
               ByRef cond As CDataCollection) As CDataCollection _
           Implements IDbEngine.GetRecord
    Public MustOverride Function GetValue(ByRef sTable As String, _
               ByRef sField As String, _
               ByRef cond As CDataCollection) As Object _
           Implements IDbEngine.GetValue
    Public MustOverride Function SpGetDataSet(ByRef spName As String, _
               ByRef cond As CDataCollection) As DataSet _
           Implements IDbEngine.SpGetDataSet
    Public MustOverride Function SpGetRecord(ByRef spName As String, _
               ByRef cond As CDataCollection) As CDataCollection _
           Implements IDbEngine.SpGetRecord
    Public MustOverride Function SpGetValue(ByRef spName As String, _
               ByRef cond As CDataCollection) As Object _
           Implements IDbEngine.SpGetValue
End Class



CDbEngineBase类的实现实际很简单，我们创建了一个构造函数，用于使用数据库连接串来创建数据引擎对象；然后，我们实现了CnnStr属性，因为它是字符串类型的，所以，在这里实现以后，可以被所有的子类直接使用。而其它的方法，我们知道它们实际都需要真正的数据库操作，所以，它们将在子类中实现。

接下来，我们就来看看SQL Server数据库的引擎类，即CSqlEngine类的实现。

21.2.2. SQL Server数据库组件（CSqlEngine类）

用于操作SQL Server数据库的引擎类为CSqlEngine，它将定义为CDbEngineBase类的子类，如下面的代码。


Public Class CSqlEngine
    Inherits CDbEngineBase

    '' 构造函数
    Public Sub New(ByRef sCnnStr As String)
        MyBase.New(sCnnStr)
    End Sub

    '' 测试数据库连接
    Public Overrides ReadOnly Property Connected As Boolean
        Get
            Try
                Using cnn As New SqlConnection(Me.CnnStr)
                    cnn.Open()
                    Return True
                End Using
            Catch
                Return False
            End Try
        End Get
    End Property

    '' 数据引擎类型
    Public Overrides ReadOnly Property EngineType As EDbEngineType
        Get
            Return EDbEngineType.SqlServer
        End Get
    End Property
'' 其它代码
End Class



代码中，我们定义的构造函数，其中直接调用了父类的构造函数。然后，在Connected属性中，我们测试了数据库的连接，而EngineType属性则返回EDbEngineType.SqlServer值。

接下来的工作，我们会逐一实现六个用于获取数据的方法，首先是GetValue()方法。

GetValue()方法

GetValue()方法用于返回查询的第一条记录第一个字段的值，如果没有则返回Nothing值，它在CSqlEngine类中的定义如下面的代码（VBChef/CSqlEngine.vb文件）。


''根据条件返回一个值
Public Overrides Function GetValue(ByRef sTable As String, _
        ByRef sField As String, _
        ByRef cond As CDataCollection) As Object
    Try
        Using cnn As New SqlConnection(Me.CnnStr)
            cnn.Open()
            Dim cmd As SqlCommand = cnn.CreateCommand()
            cmd.CommandText =  _
                 CSql.CreateSelectSql(sTable, sField, cond, 1)
            CSql.AddCommandParameters(cmd, cond)
            Return cmd.ExecuteScalar()
        End Using
    Catch
        Return Nothing
    End Try
End Function



相信代码中的内容没有大家不熟悉的了吧，都是我们在前面介绍过的内容。

GetRecord()方法

GetRecord()方法用于返回一条查询记录，并以CDataCollection对象的形式返回，如果没有查询结果，则返回Nothing值。在CSqlEngine类中GetRecord()的方法定义如下（VBChef/CSqlEngine.vb文件）。


'' 根据条件返回记录
Public Overrides Function GetRecord(ByRef sTable As String, _
        ByRef sFieldList As String, _
        ByRef cond As CDataCollection) As CDataCollection
    Try
        Using cnn As New SqlConnection(Me.CnnStr)
            cnn.Open()
            Dim cmd As SqlCommand = cnn.CreateCommand()
            cmd.CommandText =  _
                 CSql.CreateSelectSql(sTable, sFieldList, cond, 1)
            CSql.AddCommandParameters(cmd, cond)
            Dim ar As IAsyncResult = cmd.BeginExecuteReader()
            Using dr As SqlDataReader = cmd.EndExecuteReader(ar)
                If dr.Read() Then
                    Dim maxIndex As Integer = dr.FieldCount - 1
                    Dim dc As New CDataCollection()
                    For i As Integer = 0 To maxIndex
                        dc.Add(dr.GetName(i), dr(i))
                    Next
                    Return dc
                Else
                    Return Nothing
                End If
            End Using
        End Using
    Catch
        Return Nothing
    End Try
End Function



代码中，我们使用了SqlCommand对象中ExecuteReader()方法的异步版本，即BeginExecuteReader()和EndExecuteReader()方法来读取数据记录，然后，我们将第一条记录的数据通过CDataCollection对象返回。

GetDataSet()方法

GetDataSet()方法用于返回一个查询结果记录集，我们使用DataSet类型返回返回对象，CSqlEngine类中的GetDataSet()方法定义如下面的代码（VBChef/CSqlEngine.vb文件）。


 '' 根据条件返回记录集
Public Overrides Function GetDataSet(ByRef sTable As String, _
        ByRef sFieldList As String, _
        ByRef cond As CDataCollection) As DataSet
    Try
        Using cnn As New SqlConnection(Me.CnnStr)
            cnn.Open()
            Dim cmd As SqlCommand = cnn.CreateCommand()
            cmd.CommandText =  _
                CSql.CreateSelectSql(sTable, sFieldList, cond)
            CSql.AddCommandParameters(cmd, cond)
            Using ada As New SqlDataAdapter(cmd)
                Dim ds As New DataSet()
                ada.Fill(ds, sTable)
                If ds.Tables.Count > 0 AndAlso  _
                        ds.Tables(0).Rows.Count > 0 Then
                   Return ds
                Else
                    Return Nothing
                End If
            End Using
        End Using
    Catch
        Return Nothing
    End Try
End Function



SpGetValue()、SpGetRecord()和SpGetDataSet()方法

在CSqlEngine类中，使用存储过程返回查询结果的三个方法和前面的三个方法相比，最重要的，也是唯一的区别就在于SqlCommand对象的使用，首先，我们需要将CommandText属性设置为存储过程的名称（spName参数），然后，我们需要将CommandType属性设置为CommandType.StoredProcedure；而其它的代码就完全一样了，大家可以自己创建这三个方法，也可以在下载的源代码中查看。

此外，如果你的代码库中不需要这三个方法，可以在IDbEngine接口类型和CDbEngineBase类中将这三个方法删除。实际上，在软件开发中，会做减法也不是一件坏事；让简单的事保持简单依然是开发高质量软件的重要策略之一。

21.3. 实现记录操作组件

以IDbRecord接口类型为基础的数据记录操作组件，其主要功能就是对数据记录进行常规操作，如插入、更新、删除、载入和查找。下图就是这一系列组件的结构示意图。

接下来，我们还是从一个基类开始。

21.3.1. CDbRecordBase基类

CDbRecordBase类实现了IDbRecord接口类型，并作为具体的数据库类型的记录操作类的基类，它的定义如下（VBChef/CDbRecordBase.vb文件）。


Public MustInherit Class CDbRecordBase
        Implements IDbRecord
    Private myDbEngine As IDbEngine
    Private myTableName As String
    Private myIdName As String

    '' 构造函数
    Public Sub New(ByRef dbe As IDbEngine, _
                   ByRef sTableName As String, _
                   ByRef sIdName As String)
        myDbEngine = dbe
        myTableName = sTableName
        myIdName = sIdName
    End Sub

    ''
    Public Property DbEngine As IDbEngine _
        Implements IDbRecord.DbEngine
        Get
            Return myDbEngine
        End Get
        Set(value As IDbEngine)
            myDbEngine = value
        End Set
    End Property

    '' 数据表名称
    Public ReadOnly Property TableName As String _
           Implements IDbRecord.TableName
        Get
            Return myTableName
        End Get
    End Property

    '' Identity字段名称
    Public ReadOnly Property IdName As String _
               Implements IDbRecord.IdName
        Get
            Return myIdName
        End Get
    End Property
'' 其它代码...
End Class



代码中，我们首先定义了一个构造函数，需要使用IDbEngine对象、数据表名称和Identity字段名称来创建一个数据记录操作对象；然后，我们定义了DbEngine属性，以及两个只读属性TableName和IdName。

接下来，我们看一看记录操作方法的实现，首先是Find()方法，如下面的代码（VBChef/CDbRecordBase.vb文件）。


'' 查找
Public MustOverride Function Find(ByRef cond As CDataCollection, _
                            ByRef notEqualId As Long) As Long _
                            Implements IDbRecord.Find

Public Overridable Function Find(ByRef idValue As Long) As Long _
        Implements IDbRecord.Find
    Return Me.Find(New CDataCollection(Me.IdName, idValue), -1)
End Function

Public Overridable Function Find(ByRef cond As CDataCollection) _
        As Long Implements IDbRecord.Find
    Return Me.Find(cond, -1)
End Function



我们可以看到，在CDbRecordBase的子类中，我们只需要重写一个版本的Find()方法就可以了，而其它两个版本的Find()方法则可以通过调用这个方法来实现。

接下来，我们看一看Load()方法的实现，如下面的代码（VBChef/CDbRecordBase.vb文件）。


'' 载入记录
Public Overridable Function Load(ByRef sFieldList As String, _
        ByRef cond As CDataCollection) _
        As CDataCollection Implements IDbRecord.Load
    Return Me.DbEngine.GetRecord(Me.TableName, sFieldList, cond)
End Function

Public Overridable Function Load(ByRef idValue As Long)  _
        As CDataCollection Implements IDbRecord.Load
    Return Me.Load("*", New CDataCollection(Me.IdName, idValue))
End Function



Load()方法的定义可以更简单了，我们直接使用IDbEngine中的GetRecord()方法就可以完成这项工作了。

下面是Delete()方法的实现（VBChef/CDbRecordBase.vb文件）。


'' 删除
Public MustOverride Function Delete( _
           ByRef cond As CDataCollection) As Long _
       Implements IDbRecord.Delete

Public Overridable Function Delete(ByRef idValue As Long) As Long _
        Implements IDbRecord.Delete
    Return Me.Delete(New CDataCollection(Me.IdName, idValue))
End Function



关于Delete()方法，我们也只需要在子类重写一个版本即可。

最后是记录的保存方法，我们分为两种情况，即插入新记录和更新已有记录，这两个操作是在Save()方法中自动分析处理的，如下面的代码（VBChef/CDbRecordBase.vb文件）。


 '' 保存记录
Public Overridable Function Save(ByRef data As CDataCollection, _
           ByRef cond As CDataCollection) As Long  _
       Implements IDbRecord.Save
    '
    If data Is Nothing OrElse data.Count < 1 Then
        Return -1
    End If
    '
    Dim idField As CDataItem = data.Remove(Me.IdName)
    If idField Is Nothing Then
        If cond Is Nothing OrElse cond.Count < 1 Then
            Return Me.Insert(data)
        Else
            Return Me.Update(data, cond)
        End If
    Else
        Return Me.Update(data, New CDataCollection(idField))
    End If
End Function

Protected MustOverride Function Insert( _
        ByRef data As CDataCollection) As Long

Protected MustOverride Function Update( _
        ByRef data As CDataCollection, _
        ByRef cond As CDataCollection) As Long



我们可以看到，在Save()方法中，我们严格按照在介绍IDbRecord接口时的说明进行了处理，然后，根据是否包含Identity数据或查询条件来选择使用Insert()或Update()。在CDbRecordBase的子类中，我们并不需要重写Save()方法，但是，我们必须重写Insert()和Update()方法，而这两个只需要分别做好插入和更新记录的工作就好了。

21.3.2. CSqlRecord类

通过对CDbRecordBase类的定义，我们可以看到在它的子类中，我们需要重写的方法有四个，即：


	Find(CDataCollection, Long)方法。

	Delete(CDataCollection)方法。

	Insert(CDataCollection)方法。

	Update(CDataCollection, CDataCollection)方法。



我们首先来看一看CSqlRecord类的基本定义，如下面的代码（VBChef/CSqlRecord.vb文件）。


Public Class CSqlRecord
    Inherits CDbRecordBase
    '' 构造函数
    Public Sub New(ByRef dbe As IDbEngine, _
            ByRef sTableName As String, _
            ByRef sIdName As String)
        MyBase.New(dbe, sTableName, sIdName)
    End Sub
'' 其它代码...
End Class



在CSqlRecord类中，我们定义了一个构造函数，它同样是直接调用了其父类的构造函数来未完成对象的初始化工作。接下来，我们再看看需要重写的四个方法的具体实现。

Find(CDataCollection, Long)方法

Find()方法用于查询记录，并返回查询结果中第一条记录的Identity值，如果没有找到则返回0，执行错误返回-1。它的实现如下面的代码（VBChef/CSqlRecord.vb文件）。


'' 查询记录，返回第一条记录的ID值
Public Overloads Overrides Function Find( _
       ByRef cond As CDataCollection, _
       ByRef notEqualId As Long) As Long
    Try
        Using cnn As New SqlConnection(Me.DbEngine.CnnStr)
            cnn.Open()
            Dim cmd As SqlCommand = cnn.CreateCommand()
            cmd.CommandText = CSql.CreateSelectSql(Me.TableName, _
                Me.IdName, cond, 1,  _
                String.Format("{0}<>{1}", Me.IdName, notEqualId))
            CSql.AddCommandParameters(cmd, cond)
            Return ToLng(cmd.ExecuteScalar())
        End Using
    Catch
        Return -1
    End Try
End Function



Delete(CDataCollection)方法

Delete()方法会根据指定的条件删除记录，操作成功返回1，否则返回-1；如下面的代码（VBChef/CSqlRecord.vb文件）。


'' 删除记录
Public Overloads Overrides Function Delete( _
               ByRef cond As CDataCollection) As Long
    Try
        Using cnn As New SqlConnection(Me.DbEngine.CnnStr)
            cnn.Open()
            Dim cmd As SqlCommand = cnn.CreateCommand()
            cmd.CommandText = CSql.CreateDeleteSql(Me.TableName, cond)
            CSql.AddCommandParameters(cmd, cond)
            Dim ar As IAsyncResult = cmd.BeginExecuteNonQuery()
            If cmd.EndExecuteNonQuery(ar) >= 0 Then
                Return 1
            Else
                Return -1
            End If
        End Using
    Catch
        Return -1
    End Try
End Function



Insert(CDataCollection)方法

Insert()方法用于插入新的记录，请注意，由于我们在CDbRecordBase.Save()方法中已经对数据和条件进行了相应的处理，所以，在这里，我们只需要直接进行记录的插入操作就可以了。如下面的代码（VBChef/CSqlRecord.vb文件）。


'' 插入一条新记录
Protected Overrides Function Insert(ByRef data As CDataCollection) _
        As Long
    Try
        Using cnn As New SqlConnection(Me.DbEngine.CnnStr)
            cnn.Open()
            Dim cmd As SqlCommand = cnn.CreateCommand()
            cmd.CommandText =  _
                CSql.CreateInsertSql(Me.TableName, Me.IdName, data)
            CSql.AddCommandParameters(cmd, data)
            Return ToLng(cmd.ExecuteScalar())
        End Using
    Catch
        Return -1
    End Try
End Function



Update(CDataCollection, CDataCollection)方法

Update()方法的功能就是根据指定的条件更新记录数据，其定义如下面的代码（VBChef/CSqlRecord.vb文件）。


'' 更新记录
Protected Overrides Function Update(ByRef data As CDataCollection,  _
        ByRef cond As CDataCollection) As Long
    Try
        Using cnn As New SqlConnection(Me.DbEngine.CnnStr)
            cnn.Open()
            Dim cmd As SqlCommand = cnn.CreateCommand()
            Using tran As SqlTransaction = cnn.BeginTransaction()
                cmd.Transaction = tran
                '先查询指定条件的记录是否存在
                cmd.CommandText = _
                    CSql.CreateSelectSql(Me.TableName,  _
                    Me.IdName, cond, 1)
                CSql.AddCommandParameters(cmd, cond)
                Dim temp As Long = ToLng(cmd.ExecuteScalar())
                If temp <= 0 Then Return 0
                '更新数据
                cmd.CommandText =  _
                    CSql.CreateUpdateSql(Me.TableName, data, cond)
                cmd.Parameters.Clear()
                CSql.AddCommandParameters(cmd, data)
                CSql.AddCommandParameters(cmd, cond, "cond_")
                Dim ar As IAsyncResult = cmd.BeginExecuteNonQuery()
                If cmd.EndExecuteNonQuery(ar) >= 0 Then
                    tran.Commit()
                    Return 1
                Else
                    Return 0
                End If
            End Using
        End Using
    Catch
        Return -1
    End Try
End Function



相对Insert()或Delete()方法，Update()方法的实现可能稍微复杂了一点，而且，在代码中，我们还使用了事务（Transaction）。

首先，我们会判断指定条件的记录是否存在，如果记录不存在，我们就不可能成功地进行更新操作。如果指定条件的记录存在，我们就继续执行更新操作，SQL执行成功则返回1，否则返回0；执行出错会返回-1，数据表不会做任何修改。

21.3.3. CDbRecord类

与IDbEngine组件不同，我们在IDbRecord相关组件中还定义了一个CDbRecord类，它的作用就是作为一个通用类，比如，我们需要一个操作用户数据表的类，那么它就应该继承一个基于IDbRecord接口的类，而这个类是什么呢？如果是CSqlRecord类，那么它只能操作SQL Server数据库，这显然不是一个灵活的设计。

在这里，我们的CDbRecord类就可以完成这项任务，从逻辑（业务）代码的角度看，它与具体的数据库类型无关，这样，我们就可以在逻辑代码中放心地使用，而不必担心使用了什么数据库？或者数据库是不是会改变。那么，CDbRecord类是怎么实现这一功能的呢？

实际上，CDbRecord类是一个包装类，只是我们可灵活的应用或扩展它所支持的数据库类型。我们来看一看它的基本定义，如下面的代码（VBChef/CDbRecord.vb文件）。


Public Class CDbRecord
    Implements IDbRecord

    '' 用于创建数据记录操作对象的委托
    Public Delegate Function DDbRecordBuilder( _
                    ByRef dbe As IDbEngine, _
                    ByRef sTable As String, _
                    ByRef sIdName As String) As IDbRecord

    '' 默认的创建数据记录操作对象方法
    Public Shared Function CreateDbRecord(ByRef dbe As IDbEngine, _
                  ByRef sTable As String, _
                  ByRef sIdName As String) As IDbRecord
        Select Case dbe.EngineType
            Case EDbEngineType.SqlServer
                    Return New CSqlRecord(dbe, sTable, sIdName)
            Case Else
                    Return Nothing
        End Select
    End Function

    '' 默认的数据记录操作对象
    Public Shared DbRecordBuilder As New DDbRecordBuilder( _
            AddressOf CDbRecord.CreateDbRecord)

    '' 包装对象
    Private myDbRecord As IDbRecord = Nothing

    '' 构造函数
    Public Sub New(ByRef dbe As IDbEngine, _
                   ByRef sTable As String, _
                   ByRef sIdName As String)
        myDbRecord = CDbRecord.DbRecordBuilder(dbe, sTable, sIdName)
    End Sub
'' 其它代码...
End Class



代码中，首先定义了一个委托类型DDbRecordBuilder，我们可以从它的参数和返回值中看到，它的功能就是通过IDbEngine对象、表名和Identity字段名创建IDbRecord对象。

接下来的共享方法CreateDbRecord()同样也是这个功能，只是它更实际；不过，我们也可以看出，这个方法目前只可以创建CSqlRecord对象。不用着急，稍后我们就可以看到如何扩展CDbRecord类支持各种数据库的方法。

共享字段DbRecordBuilder定义为一个DDbRecordBuilder委托类型，它默认调用的方法就是前面所定义的共享方法CreateDbRecord()。那么，我们在什么地方调用这个委托对象呢？答案就在CDbRecord类的构造函数中，我们可以看到，最终创建的IDbRecord对象就是myDbRecord，这就是我们的包装对象。

另一个调用DbRecordBuilder委托对象的地方就是在DbEngine属性中，如下面的代码（VBChef/CDbRecord.vb文件）。


'' 数据引擎数据
Public Property DbEngine As IDbEngine Implements IDbRecord.DbEngine
    Get
        If myDbRecord Is Nothing Then
            Return Nothing
        Else
            Return myDbRecord.DbEngine
        End If
    End Get
    Set(value As IDbEngine)
        If myDbRecord IsNot Nothing Then
            myDbRecord = CDbRecord.DbRecordBuilder( _
                value, myDbRecord.TableName, myDbRecord.IdName)
        End If
    End Set
End Property



我们可以看到，在DbEngine属性的Set代码块中，依然是使用共享字段DbRecordBuilder委托对象来创建myDbRecord对象。

剩下的IDbRecord接口成员，它的的实现就比较简单了，我们一行操作数据库的代码也不需要，因为这些工作都会由myDbRecord对象来完成，当前，前提是它已经被成功创建。

下面的代码是TableName属性的实现（VBChef/CDbRecord.vb文件）。


Public ReadOnly Property TableName As String _
            Implements IDbRecord.TableName
    Get
        If myDbRecord Is Nothing Then
            Return ""
        Else
            Return myDbRecord.TableName
        End If
    End Get
End Property



再看一个Save()方法的实现，如下面的代码（VBChef/CDbRecord.vb文件）。


Public Function Save(ByRef data As CDataCollection, _
                         ByRef cond As CDataCollection) _
                     As Long Implements IDbRecord.Save
    If myDbRecord Is Nothing Then
        Return -2
    Else
        Return myDbRecord.Save(data, cond)
    End If
End Function



也许你已发现了这些成员实现的小秘密，那么，大家就可以自己动手写出其它成员的实现了。如果不想敲代码，下载的源代码里已经为大家写好了。^_^

21.4. Access数据库组件

如果需要在我们的数据操作组件中支持Access数据库，我们就必须创建它的基本支持组件，即CAccessEngine和CAccessRecord类。虽然Access是桌面级数据库，但它使用的依然是SQL Server引擎，所以，大部分的操作代码都与SQL Server非常相似，不过，Access还是有一些特殊的地方，我们以Access 2003为例讨论一些需要注意的问题。

使用Access数据库，我们常常使用OLEDB方式连接，即使用OleDbConnection对象来连接，而其它一系列的操作类也都定义在System.Data.OleDb命名空间中。连接Access数据库文件时，我们可以使用如下所示的数据库连接字符串。


Provider=Microsoft.Jet.OLEDB.4.0;Data Source=<文件名>;
Jet OLEDB:Database Password=<数据库密码>;



在Access数据库中，Identity字段是通过“自动编号”类型来定义的。在Access2003中，日期时间数据是以Double类型数据形式进行处理的。

Access并不支持SQL Server中所有的功能，如不支持inserted表。所以，当我们在CAccessRecord类实现Insert()方法时，就必须通过两条SQL来完成，一个就是Insert语句执行记录的插入操作，另一个就是通过“select @@IDENTITY;”语句返回新的Identity值；为了保证操作的原子性，我们还可以将这两条语句放在一个事务中来完成。

Access数据库不支持异步操作，所以，只能使用同步方法来完成相关操作，如ExecuteNonQuery()、ExecuteReader()等。

Access数据库不支持存储过程，所以，我们需要注意IDbEngine接口中SpGetValue()、SpGetRecord()和SpGetDataSet()方法的实现。在这里，我们的方案是在Access中定义相应名称（spName参数）的“查询（Query）”，然后，在这三个方法中使用Select语句从这些查询中载入数据。

21.5. MySQL数据库组件

在VB.NET项目中，如果需要操作MySQL数据库，我们建议使用MySQL所带的Connector.NET，可以和MySQL一起安装，也可以独立安装。本章的内容，我们只需要在项目中引用支持.NET Framework 4.5的MySql.Data.dll文件即可。

在VBChef项目中，我们通过菜单“项目”->“VBChef属性”，然后选择“引用”页，如下图。

然后，我们单击“添加”按钮，在窗口中单击下方的“浏览”按钮，找到并添加MySql.Data.dll文件的引用；然后，我们就可以项目中使用了。在项目文件中，我们应该使用Imports语句引用相关的命名空间，如MySql.Data、MySql.Data.MySqlClient命名空间。

下面的代码，就是通过MySqlConnectionStringBuilder类创建MySQL数据库连接串的封装方法（VBChef/CMySql.vb文件）。


'' 创建MySQL数据库连接串
Public Shared Function GetCnnStr(svr As String, _
              db As String, _
              userid As String, _
              pwd As Integer, _
              Optional port As UInteger = 3306) As String
    Dim cnn = New MySqlConnectionStringBuilder()
    cnn.Server = svr
    cnn.Database = db
    cnn.UserID = userid
    cnn.Password = pwd
    cnn.Port = port
    Return cnn.ConnectionString()
End Function



方法中共定义了五个参数：


	svr指定MySQL服务器。

	db指定连接的默认数据库。

	userid和pwd指定连接MySQL服务器的用户和密码。

	port为可选参数，用于指定MySQL服务器的连接端口，其默认值就是MySQL服务的默认端口3306。



通过CMySql.GetCnnStr()方法，我们可以很方便地创建MySQL数据库连接字符串，然后，我们通过MySqlConnection对象连接MySQL数据库。

一系列MySQL数据组件的定义，与SQL Server定义的接口非常相似，只是在操作数据时，我们要注意MySQL数据库中的SQL实现会有些不同。如：

在MySQL数据库的select语句中，没有top子句，而是select语句的最后使用limit子句实现相同的功能。

在MySQL数据库的SQL语句中传递参数时，使用?符号，而不是@符号。

使用insert语句插入新记录后中，应使用“select last_insert_id();”语句获取新创建的Identity值。请注意，在MySQL数据表中，Identity字段是通过auto_increment关键字，或直接使用serial类型定义的。

本书附带的源代码中给出了完整的CMySql类、CMySqlEngine类和CMySqlRecord为的定义，大家可以参考使用。





22. 获取系统信息

在一些特殊的应用中，我们可能需要根据不同的计算机软、硬件配置做出相应的处理，此时，我们就需要获取相关的计算机信息，本章，我们就来了解一些相关的内容，并对常用内容进行封装。

本章内容主要包括：


	使用系统变量

	获取CPU信息

	获取内存信息

	获取驱动器信息

	获取操作系统信息



22.1. 使用系统变量

在Windows操作系统中，定义了一系列的系统变量，这些变量内容可以帮助我们获取一些基本的系统信息。此时，我们需要使用的是Environment类，它定义在System命名空间，此类的成员都定义为共享成员，我们可以直接使用Environment类调用。

22.1.1. 读取环境变量

获取系统变量时，我们使用其中的GetEnvironmentVariable()方法，其参数就是指定的变量名称。如下面的代码就会显示系统变量Path的内容。


Console.WriteLine(Environment.GetEnvironmentVariable("Path"))



需要查看系统中有哪些系统变量，我们可以通过“计算机”右键菜单“属性”打开“高级系统设置”，单击其中的“环境变量”查看，如下图所示。

Environment.GetEnvironmentVariable()方法还可以使用第二个参数，它指定读取环境变量的方法，其定义为EnvironmentVariableTarget枚举，其成员包括：


	Machine，也就是我们在上图中看到的内容，它们都保存在注册表中。

	Process，默认值，当前进程关联的环境块中的环境变量。进程终止时，操作系统将销毁该进程中的环境变量。

	User，当用户环境中的环境变量。



此外，我们还可以使用Environment.GetEnvironmentVariables()方法一次获取所有的环境变量，它将以IDictionary接口类型的形式返回，我们可以通过类似Hashtable对象的形式来访问，如下面的代码，我们将显示所有环境变量的内容。


Dim dict = Environment.GetEnvironmentVariables()
For Each key As String In dict.Keys
    Console.WriteLine("{0}={1}", key, dict(key))
Next



GetEnvironmentVariables()方法同样可以使用一个参数来指定获取环境变量的位置。

22.1.2. 设置环境变量

除了读取环境变量的信息，需要时，我们还可以写入信息，不过，大家操作时要小心，不要拿已存在的系统环境变量来测试。

设置环境变量时，我们使用SetEnvironmentVariable()方法，它可以使用三个参数：


	参数一指定环境变量的名称。

	参数二指定环境变量的值。

	参数三为可选，指定环境变量保存的位置。如果需要将环境变量保存到“系统变量”，我们应该指定为EnvironmentVariableTarget.Machine枚举值。



如下面的代码，我们就创建了一个名为“SysVarTest”环境变量，设置它的值后，我们再调用它并显示。


Environment.SetEnvironmentVariable("SysVarTest", "自己的环境变量", _
    EnvironmentVariableTarget.Machine)
Console.WriteLine(Environment.GetEnvironmentVariable("SysVarTest", _
    EnvironmentVariableTarget.Machine))



22.2. 获取CPU信息

本节，我们将讨论如何获取计算机中安装的CPU的完整信息，包括CPU数量、物理核心数量、CPU类型等。

我们知道，使用环境变量NUMBER_OF_PROCESSORS或Environment.ProcessorCount属性，我们可以得到CPU的线程数量；请注意，这不是CPU真正的核心数量，如果我们需要获取真正的CPU核心及其它CPU信息，可以使用System.Management命名空间中的相关资源，如：


	ManagementClass类，获取WMI对象，

	ManagementObject类，表示一个WMI数据项目对象。

	ManagementObjectCollection类，表示一个ManagementObject对象的集合。



如果项目中没有引用System.Management命名空间，我们还需要在项目属性添加。

创建ManagementClass对象时，我们可以使用相关设备的名称，如CPU，我们可以使用“Win32_Processor”。

接下来，我们会在VBChef项目中创建一个CCpu类，它用于封装获取CPU信息的相关内容。下面的代码（VBChef/CCpu.vb文件）就是CCpu类的基本定义，并包含了获取安装CPU数量及各个CPU核心数的代码。


Imports System
Imports System.Management

Namespace vbchef
    Public Class CCpu
        '' CPU数量
        Private Shared myCpuCount As Integer
        '' CPU信息
        Private Shared myCpu() As ManagementObject

        '' 获取全部CPU信息
        Shared Sub New()
            Dim mc = New ManagementClass("Win32_Processor")
            Dim moc As ManagementObjectCollection = mc.GetInstances()
            myCpuCount = moc.Count
            ReDim myCpu(myCpuCount - 1)
            Dim index As Integer = 0
            For Each mo As ManagementObject In moc
                myCpu(index) = mo
                index += 1
            Next
        End Sub

        '' 安装CPU数量
        Public Shared ReadOnly Property Count As Integer
            Get
                Return myCpuCount
            End Get
        End Property

        '' CPU核心数量
        Public Shared Function NumberOfCores( _
                      Optional index As Integer = 0) As UInteger
            If index >= myCpuCount OrElse index < 0 Then index = 0
            Return myCpu(index).GetPropertyValue("NumberOfCores")
        End Function
        '' 其它代码...
    End Class
End Namespace



代码中，我们使用内部共享变量myCpuCount保存计算机中安装的CPU数量，数组myCpu来保存CPU的信息，它定义为ManagementObject对象数组。在共享构造函数中，我们使用ManagementClass、ManagementObjectCollection类获取了CPU的全部信息，并保存在myCpu数组中。

只读共享属性Count返回计算机中安装CPU的数量。而NumberOfCores方法则返回指定CPU的物理核心数量，它返回的是CPU真正的核心数量。其中的参数默认为0。

以下代码，我们可以获取计算机中安装的所有CPU物理核心数，当然，一般的个人计算机应该只安装了一个CPU，而多个CPU一般安装在专业的服务器中。


Dim maxIndex As Integer = CCpu.Count - 1
For i = 0 To maxIndex
    Console.WriteLine("CPU{0}", i)
    Console.WriteLine(CCpu.NumberOfCores(i))
    Console.WriteLine()
Next



实际应用中，我们可以通过迭代所有CPU信息，从而计算所有的物理核心数量，如下面的代码。


Dim maxIndex As Integer = CCpu.Count - 1
Dim coreCount As Integer = 0
For i = 0 To maxIndex
    coreCount += CCpu.NumberOfCores(i)
Next
Console.WriteLine("本计算机CPU共有{0}个物理核心", coreCount)



在Win32_Processor结构中，还有很多成员来描述CPU的信息，在CCpu类中，我们也封装了一些常用的同名方法，如：


	Name()，String类型，CPU的完整名称，也是设备管理器中显示的名称。

	AddressWidth()和DataWidth()，UShort类型，分别表示寻址和数据传输宽度。

	CurrentClockSpeed()和MaxClockSpeed()，UInteger类型，分别表示当前时钟频率和最大时钟频率，单位是MHz。

	L2CacheSize()和L3CacheSize()，UInteger类型，分别表示二级缓存和三级缓存的尺寸，单位是MB。

	Status()，String类型，CPU状态描述，如“OK”表示正常。

	NumberOfLogicalProcessors()，UInteger类型，返回CPU的线程数。



这些封装的方法都使用了可选参数index，默认值也都设置为0。

22.3. 获取内存信息

本节我们讨论如何获取内存容量及内存条的相关信息。

22.3.1. 使用My命名空间

在VB.NET项目中，我们可以使用My命名空间的资源获取一系列系统信息和相关操作，关于内存的基本情况定义在My.Computer.Info属性中，相关的属性包括：


	TotalPhysicalMemory属性，全部的物理内存。

	AvailablePhysicalMemory属性，可用的物理内存。

	TotalVirtualMemory属性，全部的虚拟内存。

	AvailableVirtualMemory属性，可用的虚拟内存。



请注意，这些属性都定义为ULong类型，返回的内存容量单位都是字节，所以，它的数值会很大，这和我们常说的内存容量单位是不一样，不过，我们可以进行相应的转换，如除以230就换算为GB。

接下来，我们再讨论两种获取内存容量的方法，分别是使用调用WinAPI函数GlobalMemoryStatus()和使用WMI资源。

22.3.2. 使用GlobalMemoryStatus()函数

GlobalMemoryStatus()函数是WinAPI中的一员，它的参数是一个MEMORYSTATUSEX结构，它的成员包含了内存相关的信息，其定义如下：


'' 内存信息结构
Public Structure MEMORYSTATUSEX
    Public dwLength As UInteger
    Public dwMemoryLoad As UInteger
    Public ullTotalPhys As ULong
    Public ullAvailPhys As ULong
    Public ullTotalPageFile As ULong
    Public ullAvailPageFile As ULong
    Public ullTotalVirtual As ULong
    Public ullAvailVirtual As ULong
    Public ullAvailExtendedVirtual As ULong
End Structure



其中，dwLength成员指定了MEMORYSTATUSEX结构的尺寸，通过上述成员数据类型的定义，我们可以计算出此结构的尺寸共计64个字节。

下面的代码（VBChef/CMemory.vb文件），我们定义了CMemory类，用于获取基本的内存信息。


Namespace vbchef
    Public Class CMemory
        '' 声明GlobalMemoryStatusEx()函数
        Private Declare Function GlobalMemoryStatusEx _
            Lib "kernel32.dll" (ByRef lpBuffer As MEMORYSTATUSEX)  _
            As Long

        '' 内存信息
        Private Shared myMemory As MEMORYSTATUSEX

        '' 共享构造构造函数
        Shared Sub New()
            myMemory = New MEMORYSTATUSEX()
            'myMemory.dwLength = 64 ''结构成员字节数
            myMemory.dwLength = _
                 CUInt(Marshal.SizeOf(GetType(MEMORYSTATUSEX)))
            Dim rc As Long = GlobalMemoryStatusEx(myMemory)
        End Sub

        '' 字节转换为MB
        Public Shared Function ByteToMb(ByRef byteValue As ULong)  _
                      As ULong
            Return byteValue \ 1048576
        End Function

        '' 全部物理内存
        Public Shared ReadOnly Property TotalPhysMb As ULong
            Get
                Return ByteToMb(myMemory.ullTotalPhys)
            End Get
        End Property

    '' 其它代码...
    End Class
End Namespace



首先，使用Declare关键字声明WinAPI函数，我们需要了解WinAPI的定义才能正确引用。然后，myMemory定义为MEMORYSTATUSEX结构，用于保存本机的内存信息。

在共享构造函数中，我们调用了GlobalMemoryStatusEx()函数读取内存信息，并保存在myMemory变量中。其中的Marshal.SizeOf()方法用于返回指定类型所占用的内存尺寸（字节）。

ByteToMb()函数则用于将字节（Byte）转换为兆字节（MB）。

最后是TotalPhysMb属性，它返回全部物理内存尺寸，单位为MB。接下来，大家可以参考此属性和MEMORYSTATUSEX结构成员，定义一系列内存和页文件尺寸的属性。

22.3.3. 使用WMI获取内存条信息

使用GlobalMemoryStatusEx()函数，我们可以获取基本的内存和页文件尺寸，而使用WMI，我们还可以获取计算机中安装的内存条的相关信息。

这里，我们同样使用ManagementClass、ManagementObject和ManagementObjectCollection类获取信息。此时，在创建ManagementClass对象时，我们使用参数"Win32_PhysicalMemory"。

如下面创建的CPhysicalMemory类（VBChef/CPhysicalMemory.vb文件），我们初始化了本机所安装的内存条的信息。


Public Class CPhysicalMemory
    Private Shared myMemoryCount As Integer '内存条数量
    Private Shared myMemory() As ManagementObject

    '' 共享构造函数
    Shared Sub New()
        Dim mc = New ManagementClass("Win32_PhysicalMemory")
        Dim moc As ManagementObjectCollection = mc.GetInstances()
        myMemoryCount = moc.Count
        ReDim myMemory(myMemoryCount - 1)
        Dim index As Integer = 0
        For Each mo As ManagementObject In moc
            myMemory(index) = mo
            index += 1
        Next
    End Sub

    '' 安装的内存条数量
    Public Shared ReadOnly Property Count As Integer
        Get
            Return myMemoryCount
        End Get
    End Property

    '' 其它代码...
End Class



代码中，myMemoryCount变量用于保存计算机中安装内存条的数量。myMemory()数组用于包含各个内存条的信息。在共享构造函数中，我们初始化了这两项内容。最后，我们创建了一个只读共享属性Count，它将返回计算机中安装的内存条数量。

接下来，我可以根据Win32_PhysicalMemory结构的定义来封装一系列的CPhysicalMemory类的成员。如下面的方法，我们分别封装了内存条制造商、工作频率和容量信息的获取方法。


'' 内存容量
Public Shared Function Capacity(Optional index As Integer = 0) As ULong
    If index >= myMemoryCount OrElse index < 0 Then index = 0
    Return myMemory(index).GetPropertyValue("Capacity")
End Function

Public Shared Function CapacityMb(Optional index As Integer = 0) As ULong
    If index >= myMemoryCount OrElse index < 0 Then index = 0
    Return myMemory(index).GetPropertyValue("Capacity") \ 1048576
End Function

 '' 速度
Public Shared Function Speed(Optional index As Integer = 0) As ULong
    If index >= myMemoryCount OrElse index < 0 Then index = 0
    Return myMemory(index).GetPropertyValue("Speed")
End Function

'' 制造商
Public Shared Function Manufacturer(Optional index As Integer = 0)  _
              As String
    If index >= myMemoryCount OrElse index < 0 Then index = 0
    Return myMemory(index).GetPropertyValue("Manufacturer")
End Function



请注意，这些方法都使用了同样的可选参数，即内存条的索引值，索引值应该大于等于0，并且小于myMemoryCount的值。

执行下面的代码，我们可以看到计算机中安装的内存条的基本信息。


Dim maxIndex As Integer = CPhysicalMemory.Count - 1
For i = 0 To maxIndex
    Console.WriteLine("内存条{0}", i)
    Console.WriteLine("制造商: {0}", CPhysicalMemory.Manufacturer(i))
    Console.WriteLine("速度: {0}Hz", CPhysicalMemory.Speed(i))
    Console.WriteLine("容量: {0}MB", CPhysicalMemory.CapacityMb(i))
    Console.WriteLine()
Next



22.4. 获取驱动器信息

获取硬盘的相关信息，我们同样可以不同的方法，如DriveInfo类可以帮助我们快速获取一些基本信息，而WMI则可以获取更多的硬盘底层信息。

22.4.1. 使用DriveInfo类

DriveInfo类定义在System.IO命名空间，它可以为我们提供一些基本的驱动器信息。

每一个DriveInfo对象都表示一个驱动器的信息，如果我们需要所有驱动器的信息，可以使用DriveInfo类中的GetDrives()共享方法，它会返回一个DriveInfo对象数组。如下面的代码。


Dim arrDrive As DriveInfo() = DriveInfo.GetDrives()
For Each drv As DriveInfo In arrDrive
    Console.WriteLine(drv.Name)
Next



此外，如果我们需要获取某一个驱动器的DriveInfo对象，可以在构造函数中使用盘符，而且只需要从a到z的字母即可，如：


Dim cDrv As DriveInfo("c")



接下来，我们可使用一系列的属性获取驱动器的信息。

IsReady属性

只读属性，Boolean类型，表示驱动器是否已准备好。请注意，驱动器准备好意味着你可以读取它的内容，而未必能够写入，只有AvailableFreeSpace属性的值大于0时，驱动器才是可写入的。

TotalSize属性

只读属性，Long类型，获取驱动器的总尺寸，单位为字节。

TotalFreeSpace属性

只读属性，Long类型，获取驱动器的空闲尺寸，单位为字节。请注意，驱动器中有空闲尺寸并不意味着可以写入，如没有装满数据的只读光盘。

AvailableFreeSpace属性

只读属性，Long类型，获取驱动器的可用尺寸，单位为字节。只有此属性值大于0时，驱动器才是可写入的。

DriveFormat属性

只读属性，String类型，获取驱动器文件系统的名称，例如NTFS、FAT32等。

DriveType属性

只读属性，DriveType枚举类型，获取驱动器的类型，包括：


	CDRom，光盘驱动器。

	Fixed，硬盘。

	Network，网络驱动器。

	NoRootDirectory，没有根目录的驱动器。

	Ram，RAM磁盘。

	Removable，可移动存储设备，如U盘。

	Unknown，未知设备。



Name属性

只读属性，String类型，获取驱动器的名称，如“C:\”。

VolumeLabel属性

String类型，用于获取或设置驱动器的卷标。如果驱动器不可写入，则不能修改它的卷标。

RootDirectory属性

只读属性，返回一个包括驱动器根目录信息的DirectoryInfo对象，在第17章，我们已经讨论过此对象的应用。

22.4.2. 使用WMI获取硬盘信息

WMI对象的应用，与获取CPU和内存信息的方法很相似，在VBChef项目中，我们创建了CDiskDrive类封装获取硬盘信息的资源，其定义如下面的代码（VBChef/CDiskDrive.vb文件）。


Imports System.Management

Namespace vbchef
    Public Class CDiskDrive
        Private Shared myDiskCount As Integer
        Private Shared myDisk() As ManagementObject

        ''共享构造函数
        Shared Sub New()
            Dim mc = New ManagementClass("Win32_DiskDrive")
            Dim moc As ManagementObjectCollection = mc.GetInstances()
            myDiskCount = moc.Count
            ReDim myDisk(myDiskCount - 1)
            Dim index As Integer = 0
            For Each mo As ManagementObject In moc
                myDisk(index) = mo
                index += 1
            Next
        End Sub

        '' 设备管理器显示内容
        Public Shared Function Caption( _
                      Optional index As Integer = 0) As String
            If index >= myDiskCount OrElse index < 0 Then index = 0
            Return myDisk(index).GetPropertyValue("Caption")
        End Function

        '' 硬盘序列号
        Public Shared Function SerialNumber( _
                      Optional index As Integer = 0) As String
            If index >= myDiskCount OrElse index < 0 Then index = 0
            Return myDisk(index).GetPropertyValue("SerialNumber")
        End Function

        '' 硬盘尺寸（Byte）
        Public Shared Function Size( _
                      Optional index As Integer = 0) As ULong
            If index >= myDiskCount OrElse index < 0 Then index = 0
            Return myDisk(index).GetPropertyValue("Size")
        End Function

        '' 硬盘尺寸（MB）
        Public Shared Function SizeMb( _
                      Optional index As Integer = 0) As ULong
            Return Size(index) \ 1048576UL
        End Function

        '' 硬盘尺寸（GB）
        Public Shared Function SizeGb( _
                      Optional index As Integer = 0) As ULong
            Return Size(index) \ 1073741824UL
        End Function

        '' 状态，如OK
        Public Shared Function Status( _
                      Optional index As Integer = 0) As String
            If index >= myDiskCount OrElse index < 0 Then index = 0
            Return myDisk(index).GetPropertyValue("Status")
        End Function
        ''
    End Class
End Namespace



22.5. 获取操作系统信息

本节，我们将讨论如何获取与Windows操作系统相关的信息。

22.5.1. 获取Windows版本

在VB.NET中获取操作系统版本，我们可使用My命名空间中的Computer.Info.OSVersion属性，它将返回操作系统版本的字符串形式，如下面的代码。


Console.WriteLine(My.Computer.Info.OSVersion)



当然，如果我们需要更详细的操作系统的版本信息，可以使用Environment类（System命名空间），其中的OSVersion.Version属性将返回完整的操作系统的完整版本信息，它是一个Version类型的对象，我们可以主要使用以下几个属性获取相应的信息，它们都定义为Integer类型。


	Major属性，主版本号。

	Minor属性，次要版本号。

	Build属性，构建号。

	Revision属性，修订号。



如下面的代码，演示了这四个属性的使用。


Dim winVer As Version = Environment.OSVersion.Version
Console.WriteLine("主版本:{0}", winVer.Major)
Console.WriteLine("次要版本:{0}", winVer.Minor)
Console.WriteLine("构建号:{0}", winVer.Build)
Console.WriteLine("修订号:{0}", winVer.Revision)



此外，如果我们需要获取操作系统的字符串形式，可以使用Version对象中的ToString()方法，我们可以使用一个整数参数来定义字符串中所包含的版本信息，如：


	1，只包含主版本号。

	2，“主版本.次要版本”格式。

	3，“主版本.次要版本.构建号”格式。

	4，“主版本.次要版本.构建号.修订号”格式。



此外，Environment类的共享属性Version实际返回的是CLR（公共语言运行库）版本信息，也就是.NET Framework的版本；这些版本信息同样是Version类型的对象。

22.5.2. 获取计算机与用户名称

在Environment类中，我们还可以使用以下共享属性获取计算机与用户相关的名称，如：


	MachineName属性，计算机名称。使用My.Computer.Name同样可以获取当前计算机的名称。

	UserName属性，当前进程中的用户名称。

	UserDomainName属性，当前用户登录到的域，如果用户没有登录到域，则返回的内容和ManchineName属性相同。







23. Windows窗体应用代码封装

本章就不过多讨论关于窗体和控件应用的基本内容了，我们将讨论一些常用的Windows窗体项目中的开发技术和技巧，主要包括：


	窗体

	对话框

	使用数据库操作组件



本章代码位于WinFormDemo项目，我们可以在VS Express 2013 for Desktop开发环境中进行相关测试。

23.1. 窗体

作为Windows窗体项目，窗体（窗口）一直是非常重要的界面载体和元素，通过完整版本的Visual Studio开发工具，我们可使用很多窗体模板和控件创建出各种功能的界面。本节，我们将自己动手编码实现一些窗体功能。

23.1.1. 异形窗体

一般的窗体，其形状都是方方正正的矩形，如果我们是在开发一个多媒体项目、或者是需要实现悬浮窗的功能，就可能需要使用异形窗体了。

实际上，我们通过设置窗体的几个属性就可以很方便地将其改变为异形窗体，比如，使用一个图像作为窗体的外观。还记得前面绘制的太极图吗？我们就创建一个这样的圆形窗体。

在WinFormDemo项目中添加一个新的Windows窗体，使用默认的名称Form2，然后，在项目属性中设置其为启动窗体，如下图所示。

[image: 图像说明文字]


然后，我们在Form2设计视图中双击窗体，打开代码编写视图，添加Form2_Load()方法中的代码如下（/Form2.vb）。


Imports vbchef

Public Class Form2
    Private Sub Form2_Load(sender As Object, e As EventArgs)  _
            Handles MyBase.Load
        Me.ControlBox = False
        Me.Size = New Size(500, 500)
        Me.FormBorderStyle = Windows.Forms.FormBorderStyle.None
        Me.BackgroundImage = CImage.Taiji(400)
        Me.BackgroundImageLayout = ImageLayout.Center
        Me.TransparencyKey = Me.BackColor
    End Sub
End Class



代码中使用了VBChef代码库中的内容，大家可以参考本章中的23.3.1小节相关内容编译并引用这些开发资源。

执行程序，我们就会看到如下图所示的窗体（的确是窗体）。

[image: 图像说明文字]


在创建图形窗体的代码中，我们需要注意以下几点：


	通过ControlBox属性设置为False将窗体的控制按钮隐藏。

	设置窗体的FormBorderStyle属性为None，则窗体会成为一个纯色的平面窗体，而不显示窗体的边框，便于我们设置窗体的透明色。

	BackgroundImage属性设置窗体的背景图像，我们使用了CImage.Taiji()方法创建太极图像。

	BackgroundImageLayout属性设置背景图像的排列方式或位置，代码中，我们将背景显示在窗体的中间。

	TransparencyKey属性设置窗体绘制时的透明色，我们设置为窗体原有的背景色，这样，显示的窗体的就只有背景图像的内容，从而也就达到了我们创建图形窗体的目的。



接下来，我们看一看如何使用鼠标移动这样的无标题栏的窗体。

23.1.2. 移动无标题窗体

也许你觉得使用鼠标移动窗体是一项非常复杂的操作，在Windows内部机制中的确是这样；不过，在这里，我们可以使用两个简单的鼠标事件配合完成这一操作，并不需要调用复杂的WinAPI函数。

首先，我们在Form2窗体中定义两个变量分别保存移动前的鼠标坐标，如下面的代码（/Form2.vb文件）。


Imports vbchef

Public Class Form2
    Private mouseX As Integer
    Private mouseY As Integer
    ' 其它代码
End Class



当鼠标按下时，记录坐标，如下面的代码（/Form2.vb文件）。


Private Sub Form2_MouseDown(sender As Object, e As MouseEventArgs) _
        Handles Me.MouseDown
    mouseX = e.X
    mouseY = e.Y
End Sub



当按下的鼠标左键，并在窗体中移动鼠标时（MouseMove事件），就进行窗体的移动操作，如下面的代码（/Form2.vb文件）。


Private Sub Form2_MouseMove(sender As Object, e As MouseEventArgs) _
          Handles Me.MouseMove
    ' 左键按下移动窗体
    If e.Button = Windows.Forms.MouseButtons.Left Then
        ' 计算移动的距离
        Dim offsetX As Integer = e.X - mouseX
        Dim offsetY As Integer = e.Y - mouseY
        ' 窗体定位到新的位置
        Me.Left = Me.Left + offsetX
        Me.Top = Me.Top + offsetY
    End If
End Sub



执行程序，我们就可以使用鼠标移动太极窗体了。

23.2. 对话框

在图形界面中，大家一定对一个个的对话框（Dialog Box）不会陌生，如输入对话框、提示信息、选择打开或保存文件、选择颜色、选择目录等等。本节，我们从提示信息类对话框开始，了解一些常用的技巧，并对常用的功能封装，便于在项目中大量重复使用。

23.2.1. 信息类对话框

使用过VB6的朋友一定对MsgBox()方法不会陌生，在VB.NET中，我们依照可以使用这个方法，不过，我们还可以使用在.NET Framework中更通用的MessageBox类，而最主要的就是其中的Show()方法，这是一个共享方法，如：


MessageBox.Show("这是一条信息")



当然，我们知道，常用的信息类对话框有四种，即信息（Information）、警告（Warning）、错误（Error）和问题（Question）。无论是MsgBox()方法，还是MessageBox类中的Show()方法，它们都是通用的方法，通过参数来指定显示的图标类型、按钮的类型等，如果在项目中需要大量的不同类型的对话框，使用起来还是比较繁琐一些，不如我们进行一些简单的封装来解决这一问题。

在我们的VBChef代码库中，对话框相关操作封装在vbchef.winx.CDlg类中，如下面的代码，就是其中定义一个ShowInformation()方法（vbchef.winx.CDialog.vb文件）。


Imports System.Windows.Forms
Imports System.Drawing

Namespace vbchef.winx
    '' 简写类名
    Public Class CDlg
        ' 信息对话框
        Public Shared Sub ShowInformation(ByRef msg As String)
            MessageBox.Show(msg, "", MessageBoxButtons.OK, _
                MessageBoxIcon.Information)
        End Sub
    ' 其它代码
End Class



下面的代码可以使用这个方法显示一个带有信息图标的消息对话框（/Form5.vb文件）。


CDlg.ShowInformation("信息类对话框")



代码执行结果如下图。

[image: 图像说明文字]


项目中，在提示信息中还可能包含一些实时的数据，如我们在使用Console.WriteLine()方法时进行的内容组合一样；通过封装，我们同样可以在自己的对话框中使用这个的内容组合，如下面的ShowInformation()方法（vbchef.winx.CDialog.vb文件）。


Public Shared Sub ShowInformation(ByRef msg As String,  _
        ParamArray args() As Object)
    ShowInformation(String.Format(msg, args))
End Sub



下面的代码演示了这方法的使用。


Dim counter As Integer = 90
CDlg.ShowInformation("执行成功，共完成{0}项工作", counter)



代码会显示如下图所示的对话框。

[image: 图像说明文字]


接下来，警告类和错误类信息对话框的封装与前面的ShowInformation()方法非常相似，只是使用的图标不同而已。而问题类的对话框就复杂一点，主要是需要设置需要显示的按钮，下面就是一个ShowQuestion()方法的的定义（vbchef.winx.CDialog.vb文件）。


Public Shared Function ShowQuestion(buttons As MessageBoxButtons,  _
        msg As String) As DialogResult
    Return MessageBox.Show(msg, "", buttons, MessageBoxIcon.Question)
End Function



下面的代码演示了这一方法的使用。


If CDlg.ShowQuestion(MessageBoxButtons.YesNo, "确定吗？") = _
            Windows.Forms.DialogResult.Yes Then
    CDlg.ShowInformation("确定了")
Else
    CDlg.ShowInformation("否定了")
End If



执行代码，我们会看到如下图所示的执行结果。

[image: 图像说明文字]


关于信息类对话框，最后需要说明的是，微软帮助文档中已经不建议使用Question图标，因为它和帮助图标会产生歧义；相信大家可以很方便地取消此图标。

23.2.2. 选择颜色

在窗体中选择颜色，我们可以使用ColorDialog类，其重要的几个成员包括：


	AllowFullOpen属性，是否允许打开调色板，即是否可以自定义颜色。

	FullOpen属性，默认是否打开调色板。

	AnyColor属性，是否允许使用所有颜色。

	SolidColorOnly属性，是否只能使用基本颜色。

	Color属性，设置或获取对话框中选中的颜色。

	ShowDialog()方法，显示选择颜色对话框。



下面的代码，就是我们封装的通用颜色选择对话框方法（vbchef.winx.CDialog.vb文件）。


Public Shared Function SelectColor( _
       Optional defaultColor As Color = Nothing) As Color
    Dim dlg As New ColorDialog()
    dlg.AllowFullOpen = True
    dlg.FullOpen = True
    dlg.AnyColor = True
    dlg.SolidColorOnly = False
    ' 设置默认颜色
    If Not IsNothing(defaultColor) Then
        dlg.Color = defaultColor
    End If
    ' 返回选择的颜色
    If dlg.ShowDialog() = DialogResult.OK Then
        Return dlg.Color
    Else
        Return defaultColor
    End If
End Function



使用SelectColor()方法时，我们需要注意参数与返回值的配合问题。代码实现中，当单击颜色选择对话框中的确定后，会返回选中的颜色。单击取消按钮时应注意，如果我们调用此方法时没有带入默认颜色，则返回Nothing值，此时需要使用IsNothing()方法判断返回结果；如果参数带入了默认颜色，则直接返回此颜色值，我们可以直接使用此返回值，而不会有什么副作用。

此外，如果我们需要列出Color结构（System.Drawing命名空间）中定义的命名颜色，可以使用类似下面的代码。


Dim t As Type = GetType(Color)
For Each p In t.GetProperties()
    If p.PropertyType.Equals(t) Then
        ListBox1.Items.Add(p.Name)
    End If
Next



代码会在ListBox1列表控件中显示Color中定义的所有命名颜色值，获取其中一个颜色时，可以使用类似下面的代码。


' 选取颜色
If ListBox1.SelectedIndex < 0 Then
    CDlg.ShowInformation("请在列表中选择颜色")
Else
    Dim t As Type = GetType(Color)
    Dim pi As System.Reflection.PropertyInfo = _
           t.GetProperty(ListBox1.SelectedItem.ToString())
    Me.BackColor = pi.GetValue(New Color)
End If



使用颜色时，请注意关于透明色（Transparent）颜色的使用，并不是所有对象都支持。

23.2.3. 选择打开文件

选择要打开的文件，我们会使用OpenFileDialog类，其中比较重要的一个属性就是对文件类型的过滤操作，我们使用Filter属性设置，这是一个字符串类型的属性。

设置每一种文件类型过滤的格式为“描述|扩展名过滤”，如“PNG图片|*.png”，同时，我们也可以设置多种文件格式，如“常用图片|*.png;*.bmp;*.jpg;*.gif”。如果同时设置多个过滤，每个过滤之间同样使用“|”分隔，如“所有文件|*.*|PNG图片|*.png”。

下面的代码就是我们封装的用于打开单个文件的对话框操作（vbchef.winx.CDialog.vb文件）。


Public Shared Function SelectOpenFileName( _
        Optional sFilter As String = "所有文件|*.*",  _
        Optional defaultFileName As String = "") As String
    Dim dlg As New OpenFileDialog()
    dlg.Filter = sFilter
    If dlg.ShowDialog() = DialogResult.OK Then
        Return dlg.FileName
    Else
        Return defaultFileName
    End If
End Function



如果需要同时打开多个文件，应该将OpenFileDialog对象中的Multiselect属性设置为True，下面的代码就是我们封装的用于选择打开多个文件的方法（vbchef.winx.CDialog.vb文件）。


Public Shared Function SelectOpenFileNames( _
        Optional sFilter As String = "所有文件|*.*") As String()
    Dim dlg As New OpenFileDialog()
    dlg.Filter = sFilter
    dlg.Multiselect = True
    Return dlg.FileNames
End Function



应用时，我们可以根据返回的数组成员数量（Count属性）判断打开的文件数量，类似下面的代码。


ListBox1.Items.Clear()
Dim files() As String = CDialog.SelectOpenFileNames()
Dim maxIndex As Integer = files.Count - 1
For i = 0 To maxIndex
    ListBox1.Items.Add(files(i))
Next



23.2.4. 选择保存文件

选择保存文件名时使用SaveFileDialog类，其应用方法与OpenFileDialog类比较相似，下面的代码就是我们封装的用于选择保存文件的对话框操作（vbchef.winx.CDialog.vb文件）。


Public Shared Function SelectSaveFileName( _
        Optional sFilter As String = "所有文件|*.*") As String
    Dim dlg As New SaveFileDialog()
    dlg.Filter = sFilter
    If dlg.ShowDialog() = DialogResult.OK Then
        Return dlg.FileName
    Else
        Return ""
    End If
End Function



方法中，如果选择了要保存的文件名，并单击确定按钮，则返回选择的文件名的完整路径，否则返回空字符串。

23.2.5. 选择目录

使用选择目录对话框时使用FolderBrowserDialog类，其中的SelectedPath属性用于设置或获取选择的目录路径；下面的代码就是我们封装的用于选择目录的方法（vbchef.winx.CDialog.vb文件）。


Public Shared Function SelectFolder( _
        Optional defaultPath As String = "") As String
    Dim dlg As New FolderBrowserDialog()
    If defaultPath <> "" Then dlg.SelectedPath = defaultPath
    If dlg.ShowDialog() = DialogResult.OK Then
        Return dlg.SelectedPath
    Else
        Return ""
    End If
End Function



23.2.6. 选择字体

选择字体对话框使用了FontDialog类，与其它类型对话框不同的是，当我们设置字体时，可能还需要选择颜色，而字体和颜色是由两个类型的数据来保存的，所以，对于选择字体对话框的封装也会有一些不同，特别是方法的返回值。如下面的代码（vbchef.winx.CDialog.vb文件）。


Public Shared Function SelectFont( _
        Optional defaultFont As Font = Nothing,  _
        Optional defaultColor As Color = Nothing) As FontDialog
    Dim dlg As New FontDialog()
    dlg.AllowScriptChange = True
    dlg.AllowSimulations = True
    dlg.AllowVectorFonts = True
    dlg.AllowVerticalFonts = True
    dlg.FontMustExist = True
    dlg.ShowColor = True
    If Not IsNothing(defaultFont) Then dlg.Font = defaultFont
    If Not IsNothing(defaultColor) Then dlg.Color = defaultColor
    If dlg.ShowDialog() = DialogResult.OK Then
        Return dlg
    Else
        Return Nothing
    End If
End Function



实际应用中，如设置TextBox1控件的字体及颜色，我们可以使用类似如下的代码来实现。


Dim dlg As FontDialog =  _
        CDlg.SelectFont(TextBox1.Font, TextBox1.ForeColor)
If dlg IsNot Nothing Then
    TextBox1.Font = dlg.Font
    TextBox1.ForeColor = dlg.Color
End If



23.3. 使用数据库操作组件

本章的最后，我们将讨论如何在Windows窗体项目中使用已经封装的VBChef代码库，特别是数据库操作组件在项目中的灵活应用。

23.3.1. 编译封装代码库

对于VBChef代码库源代码，我们已经单独放在vbchef-library目录中，其中的common包含了一些通用代码，但在这里并没有包含关于获取系统信息的代码，它们还在VBChef项目中，如果在项目中需要，可以进行相关的组合编译；不过有一点需要注意，如果代码中包含了太深入的系统资源调用代码，某些安全软件会说这是病毒（不认识真正的病毒，也不要乱说吗！^^），你可能需要对软件的用户说明这一点。

在vbchef-library\common目录中，我们定义了build.bat批处理文件来完成代码的编译工作，其定义如下。


%windir%\Microsoft.NET\Framework64\v4.0.30319\vbc.exe vbchef.*.vb -target:library -out:vbchef.dll -r:MySql.Data.dll
pause



第一个命令有点长，请注意它们是在一行的；第二命令是pause，即当编译完成时暂停，这样我们就可以看到编译结果了，理想的编译结果当然是没有任何警告和错误。

此命令会引用MySql.Data.dll库，并生成vbchef.dll代码库文件，应用中，我们只需要将这两个dll文件引用到项目中即可使用。

最后请注意的是版本问题：


	一是.NET Framework的版本问题，代码中，我们编译使用的是.NET Framework 4.0中的64位版本。你可能需要根据项目所基于.NET Framework版本的不同而重新编译。

	另一个问题是相关资源的版本问题，如MySQL数据操作库的使用。



23.3.2. 定义项目主数据引擎

在项目中，对于通用资源的封装及统一管理是一项非常重要的工作，这对于项目的开发、维护及扩展等工作都是非常关键的。在这里，可以定义一个CApp类来完成这一系列工作，如下面的代码，就是定义的一个基本的项目主类（CApp.vb文件）。


Imports vbchef
Imports WinFormDemo.vbchef.winx

Public Class CApp
    '' 本项目初始数据库类型为SQL Server，
    '' 数据库名称cdb_demo
    Public Shared Db As IDbEngine = _
        New CSqlEngine(CSql.CreateLocalCnnStr("cdb_demo"))

    '' 项目初始化
    Public Shared Function Init() As Boolean
        Try
            ' 初始化CDbRecord对象构造器
            CDbRecord.DbRecordBuilder = _
                      New CDbRecord.DDbRecordBuilder( _
                      AddressOf CApp.CreateDbRecord)
            '
            Return True
        Catch
            Return False
        End Try
    End Function

    '' 项目数据记录操作创建方法
    Public Shared Function CreateDbRecord(ByRef dbe As IDbEngine, _
                           ByRef sTableName As String, _
                           ByRef sIdName As String) As IDbRecord
        Return New CSqlRecord(dbe, sTableName, sIdName)
    End Function

    ''
End Class



（请注意，本部分代码测试将在Form6窗体中完成。）

首先，我们定义了Db对象，它就是项目的主数据库引擎，我们将其设置连接到本地SQL Server默认实现中的cdb_demo数据库；然后，我们可以使用如下代码进行数据库的连接测试。


TextBox1.Text = CApp.Db.Connected.ToString()



Init()方法用于项目的初始化操作，我们可以在Form6窗体启动（即项目启动时）执行，如下面的代码。


Private Sub Form6_Load(sender As Object, e As EventArgs) _
        Handles MyBase.Load
    If CApp.Init() = False Then
        CDlg.ShowError("项目初始化失败！")
        Application.Exit()
    End If
End Sub



项目初始化失败则会显示错误信息并退出程序。在这里，我们还可以对失败的原因做出更详细的描述，并给出解决的方法，比如，我们可以在项目启动时测试主数据库的连接，如果不能正确连接到数据库，则提示用户与数据库管理员联系。

由于我们的项目默认使用SQL Server数据库，所以，对于CDbRecord对象的创建，直接生成CSqlRecord对象就可以了，在第21章，我们就已经详细介绍了这些代码的原理，这里就不过讨论了。





24. ASP.NET应用开发基础

虽然在本章将要介绍的内容是ASP.NET，但我们更建议大家，学习Web开发时应该从基本的内容开始，如传统的HTML、CSS和JavaScript。

而本章，我们就将介绍一些基于ASP.NET的，基本及常用Web开发内容，这些内容主要包括：


	ASP.NET项目概述

	页面

	常用Web对象

	常用Web控件

	自定义Web控件

	使用数据库操作组件

	实现验证码功能

	文件上传

	页面缓存



24.1. ASP.NET项目概述

了便于初学者入门，我们先来了解一些关于ASP.NET项目的基本结构，特别是代码及相关文件的组织。

我们通过VS Express 2013 for Web的菜单项“文件”->“新建网站”创建一个新的ASP.NET项目，如下图所示。

[image: 图像说明文字]


然后，我们会在开发环境中的“解决方案资源管理器”中看到，在网站中，只有一个文件，即Web.config文件，这是ASP.NET网站的独立配置文件，现在，我们并不需要修改它的内容。

接下来，我们就来了解在ASP.NET项目中更多的文件类型及特殊目录。

24.1.1. 常用文件类型与目录

除了网站传统的HTML（.html）、CSS（.css）、JavaScript（.js）文件等类型的文件，我们还会使用一些特殊类型的文件，如前面项目创建的Web.config配置文件等。

在网站中添加内容时，我们应该注意，应该在“解决方案资源管理器”中选中内容将要添加到的位置，如果添加到根目录，可以选择网站的名称，如下图的示。

[image: 图像说明文字]


ASP.NET项目中，页面通常使用.aspx扩展名的文件，我们将在下一节详细讨论这种文件的应用。我们可以通过菜单项“网站”->“添加新项”创建.aspx文件，它又称为“Web窗体”，如下图所示。

[image: 图像说明文字]


默认情况下，会将Default.aspx文件作为ASP.NET网站的首页，现在，我们的项目中还没有这个文件，我们就使用默认的名称，创建其它页面时，我们可以根据实际需要修改文件名称。另一个需要注意的是，我们现在测试的内容都比较简单，可以将将VB.NET代码和HTML内容都放在一个文件中，此时，我们可以取消“将代码放在单独的文件中”选项。

在ASP.NET项目中，我们还可以通过.ascx文件创建自定义Web控件，在24.5节中，我们会讨论相关内容。

此外，在ASP.NET项目中，还有一个特殊文件，即Global.asax，它称为全局应用程序类文件。我们可以通过菜单项“网站”->“添加新项”创建它，如下图所示。

[image: 图像说明文字]


在Global.asax文件中，默认定义了五个方法，分别是：


	Application_Start()，在ASP.NET网站开始运行时执行。

	Application_End()，在ASP.NET网站结束运行时执行。

	Application_Error()，在遇到未处理的错误时执行。

	Session_Start()，在会话连接开始时执行。

	Session_End()，在会话结束时执行。



前面，我们了解了几种 ASP.NET项目中常用的文件类型，接下来，我们再来看看在ASP.NET项目中有哪些常用的目录，这些目录都在网站的根目录中；本章，我们主使用如下目录：


	app_code目录，用于存放源代码。请注意，我们可以在ASP.NET使用不同的编程语言来编写代码，如VB.NET和C#，在app_code目录中，只要将不同语言编写的代码放在不同的目录就可以了。

	app_data目录，用于存放项目的数据文件，如Access文件、LocalDB文件等。

	bin目录用于存放已编译的库文件，如MySql.Data.dll文件。 当然，在ASP.NET项目中还有很多专用的目录，我们在这里就不详细说明了。



此外，除了ASP.NET项目专用目录，我们还可以使用一些习惯目录来存放相应的资源，如本章中，我们就使用了如下目录，它们同样在网站的根目录中：


	img目录，存放项目中的公用图片资源。

	css目录，存放CSS（可层叠样式表）文件（.css）。

	js目录，存放通用的JavaScript脚本文件（.js）。



在前面创建的空网站中还没有一系列的ASP.NET专用目录和一些常用目录，我们可以选择网站，然后通过菜单项“网站”->“新建文件夹”来添加这些目录。

24.1.2. 网站测试与编译

当我们使用VS Express 2013 for Web进行ASP.NET项目开发时，可以很方便地使用IIS Express进行测试，并可以选择进行测试的浏览器，我们只需要在开发环境的工具栏中点击“>Internet Explorer(1)”即可。

当然，如果我们需要接近真实环境下的测试，可以在Windows中配置IIS进行测试，此时，我们需要包含IIS的Windows版本。

关于ASP.NET项目的发布，如果允许在Web服务器中操作，则可以使用完全预编译的方式进行发布，这样，在用户访问网站时，就不会因为编译而产生延时，这对于用户体验来讲，是非常重要的。

当我们需要进行网站的完全预编译时，可以使用相应.NET Framework版本中的aspnet_compiler.exe命令，其基本使用语法如下。


aspnet_compiler -p <源代码路径> <发布路径> -v <网站虚拟路径>



如网站的源代码位于d:\aspnet\wwwroot，我们可以使用如下代码进行发布前的完全预编译（请写在一行）。


%windir%\Microsoft.NET\Framework64\v4.0.30319\aspnet_compiler 
-p d:\aspnet\wwwroot d:\aspnet\r1-0 -v /



通过此命令，我们就在d:\aspnet\r1-0目录中创建了一个用于发布的ASP.NET网站，通过-v（虚拟目录）指定此目录为网站的根目录（/）；然后，在IIS中，我们就可以将网站的路径指定为此位置。

如果我们无法操作Web服务器，也可以将大量的VB.NET或其它.NET代码编译为.dll文件，然后复制到网站的bin目录中；只是，此方法操作发布的网站，在访问页面时还是会有短暂的编译操作；不过，我们可以通过利用静态页面或缓存技术进行相应的优化。在这里，我们就不进一步讨论了，接下来，我们来看看ASP.NET项目开发的一些基础知识。

24.2. 页面

网站中，页面是非常重要的元素，它和其它元素（如CSS、JavaScript、图片、视频等等）一起构成了丰富多彩的Web体验。

在ASP.NET项目中，除了传统的HTML页面，我们使用的页面（.aspx）定义为Page类（System.Web.UI）的子类；接下来，我们以ASP.NET网站中常用的首页Default.aspx文件为例来看看ASP.NET页面的基本应用。

24.2.1. 从Default.aspx页面开始

如果网站中还没有Default.aspx文件，我们可以在VS Express 2013 for Web开发环境中，选中“解决方案资源管理器”中的网站根目录，通过菜单项“网站”->“添加新项”，在模板中选择“Web窗体”，并命名为“Default.aspx”。

下面就是Default.aspx文件的初始代码，我们看一看有哪些内容。

代码中，我们指出了三个主要部分的内容，它们是ASP.NET页面与传统页面主要的区别：


	
<%@ Page %>
 指令，用于指定页面相关的参数，如是否进行调试等；我们还可以指定动态代码的语言类型，如本书中我们使用的VB.NET编程语言，在这里就指定为“VB”。除了<%@ Page %>
 指令，另一个在页面中常用的指令是<%@ OutputCache %>
 指定，它用于指定页面的缓存方式。

	
<script runat="server"></script>
 部分，了解传统页面开发的朋友一定知道，在页面中的JavaScript脚本内容同样使用<script></script>
 来定义，只不过在这里使用了runat="server"
 属性来指定代码是在服务器中执行的动态代码。

	与传统的HTML表单（form）相似，我们同样使用<form runat="server"></form>
 来定义ASP.NET中的Web表单，其最显著的特点同样是使用了runat="server"
 属性。请注意，如果页面中没有需要提交到服务器的内容，在ASP.NET页面中也可以省略Web表单部分。



除了以上列出的三个部分，页面中的其它内容与传统的HTML就基本一样了，只是需要注意<head runat="server">
 标记，如果在页面的“头”部分不需要执行动态代码，则此标记中的runat="server"属性可以省略。

此外，在页面中还应注意以下几个重要的部分。


	
<title></title>
 ，定义在<head></head>
 中，用于指定浏览器标题栏或页面标签中显示的页面名称。

	
<body></body>
 定义了页面的主体部分，也就是显示在浏览器中的页面内容。



24.2.2. Page类

如果想要有效地使用ASP.NET页面，充分了解Page类是非常重要的，它定义在System.Web.UI命名空间，下面，我们将讨论一些常用的成员，对于完整的定义，大家可以参考Help Library。


	IsPostBack属性，判断页面操作是否首次打开，当此属性为False时，则说明页面为第一次打开，如果为True值，则说明页面是用户进行了再次提交（如表单进行了提交操作）。

	Title属性，指定页面显示的标题，只有在<head>
 标记中使用了runat="server"属性时，我们才能动态修改页面标题。

	Response属性，获取当前会话上下文中向客户端返回响应数据的对象，完整定义可以参考System.Web.HttpResponse类。

	Request属性，获取当前会话上下文中客户端提交数据操作的对象，完整定义可以参考System.Web.HttpRequest类。

	Server属性，封装了一些常用的方法，如URL或HTML内容的编码与解码操作，完整定义可以参考System.Web.HttpServerUtility类。

	Session属性，封装了当前会话操作对象，定义为HttpSessionState类型（System.Web.SessionState命名空间）。



下面的代码（Default.aspx文件），我们在Web窗体中添加两个标签控件和一个按钮控件（用于模拟提交操作）。


<body>
    <form id="form1" runat="server">
    <div>
        <asp:Label ID="lblOpenTime" runat="server" /><br />
        <asp:Label ID="lblRefreshTime" runat="server" /><br />
        <asp:Button ID="btnSubmit" Text="提交" runat="server" />
    </div>
    </form>
</body>



下面的代码（Default.aspx文件），我们通过页面的Load事件来显示页面的打开时间和刷新时间。


<%@ Page Language="VB" %>

<!DOCTYPE html>

<script runat="server">
    Private Sub Page_Load()
        If IsPostBack = False Then
            lblOpenTime.Text = Now.ToLongTimeString()
        End If
        lblRefreshTime.Text = Now.ToLongTimeString()
    End Sub
</script>



打开页面，我们可以看到两个时间是一样的，当单击“提交”按钮时，只有第二个时间会变化，而第一个时间不会变化。

在使用ASP.NET页面，或者说是使用Web窗体时，理解IsPostBack属性所表示的状态是很重要的；比如，我们可以在第一次加载页面时做一些初始化工作，而在随后的提交表单操作就不需要执行这些代码了。如：


<script runat="server">
    Private Sub Page_Load()
        If IsPostBack = False Then
            ' 页面初始化
        End If
        ' 其它操作，每次提交执行
    End Sub
</script>



24.2.3. 页面与代码分离

前面我们提到过，在项目中创建Web窗体时，可以指定一个选项“将代码放在单独的文件中”，如下图所示。

[image: 图像说明文字]


在大型Web项目中，前端代码和后台代码可能需要不同的设计者来完成，此时，我们就可以将HTML与VB.NET（或C#）代码放在不同的文件中；然后，在ASP.NET项目中，它们会自动完成组合。

下面，我们就在网站根目录下创建一个HTML与VB.NET代码分开的页面，并命名为“PartialPage.aspx”，如下图所示。

这样，对于一个Web窗体，即ASP.NET页面来讲，就会生成两个文件，如本例就是PartialPage.aspx和PartialPage.aspx.vb文件。如果我们打开这两个文件，就可以发现它们之间关联的小秘密，即在PartialPage.aspx文件中，我们使用<%@ Page %>
 指令中的CodeFile属性指定对应的代码文件。而在PartialPage.aspx.vb文件中，我们定义的PartialPage类使用了Partial关键字，即指定此类为分部类。

实际开发中，前端设计师可以进行PartialPage.aspx文件的编辑，而同时，逻辑代码可以由VB.NET程序员来完成。

本书中，由于代码相对比较简单一些，我们会在一个文件中同时包含HTML和VB.NET代码，这样代码也会更直观。

24.3. 常用Web对象

在Web项目开发中，有一些常用的Web对象，它们功能强大，可以帮助我们获取很多有用的信息，也可以帮助我们进行一系列的客户端与服务器之间的交互操作。

本节，我们将常用的代码封装到CWeb类，它定义在/app_code/vbchef.webx.CWeb.vb文件中。

在使用这些对象之前，我们还需要理解一个概念，即当前上下文（Context）；每一个客户端到服务器的连接都是独立的上下文，在代码库中，我们可以使用HttpContext.Current对象获取当前连接中的上下文，在此基础上，我们就可以使用以下一些常用的对象了。而在页面中的VB.NET代码中，我们可以直接使用这些对象。

24.3.1. Request对象

前面我们说过，Request对象定义为HttpRequest类型，包含了用户向服务器发送的相关信息，我们可以通过一些常用的属性获取这些数据，如：


	Headers属性，获取HTTP头数据。

	HttpMethod属性，获取提交数据的方法，如GET、POST等。

	InputStream属性，用户提交的内容流（Stream类型）。

	IsLocal属性，判断请求是否来自当前计算机。

	IsSecureConnection属性，判断是否使用安全套接字连接，即使用HTTPS协议。

	Params属性，包含QueryString、Form、ServerVariables和Cookies数据的所有集合。

	RawUrl属性，返回不包含主机的完整资源路径，如http://127.0.0.1/index.html中，RawUrl属性返回/index.html。

	UserHostAddress属性，返回客户端的IP地址（String类型）。

	UserHostName属性，返回客户端的DNS名称（String类型）。

	UserLanguages属性，返回客户端支持的语言数组（String()类型）。



在CWeb类中，我们定义了获取用户IP的只读属性UserIp，如下面的代码（/app_code/vbchef.webx.CWeb.vb文件）。


Imports Microsoft.VisualBasic
Imports System.Web

Namespace vbchef.webx
    Public Class CWeb
        ' 当前用户IP
        Public Shared ReadOnly Property UserIp As String
            Get
                Return HttpContext.Current.Request.UserHostAddress
            End Get
        End Property
    End Class
End Namespace



代码中，我们可以使用如下代码快速获取当前上下文中的客户IP地址。


Dim strIp As String = CWeb.UserIp



另外，如果我们需要获取用户浏览器所支持的首选语言，可以在CWeb类封装相应的代码，如下面的代码（/app_code/vbchef.webx.CWeb.vb文件）


Public Shared ReadOnly Property UserLanguage As String
    Get
        Dim str() As String = HttpContext.Current.Request.UserLanguages
        If str.Length > 0 Then
            Return str(0).ToLower()
        Else
            Return ""
        End If
    End Get
End Property



24.3.2. Response对象

定义了一系列由服务器向客户端发送的数据及相关信息，常用的成员包括：


	Charset属性，设置字符集。

	ContentType属性，MIME信息，如“image/png”。

	Clear()方法，清除当前上下文缓存中的内容。

	Flush()方法，发送当前上下文缓存中的内容。

	Write()方法，向当前上下文缓存中写入内容。

	WriteFile()方法，发送服务器中的文件。

	Redirect()方法，跳转到指定的页面。



为了在代码库中简化这些资源的使用，我们可以同样将常用的方法封装在CWeb类中，如下面的代码（/app_code/vbchef.webx.CWeb.vb文件）。


''
Public Shared Sub Write(s As String)
    HttpContext.Current.Response.Write(s)
End Sub

''
Public Shared Sub Write(obj As Object)
    HttpContext.Current.Response.Write(obj)
End Sub

''
Public Shared Sub Flush()
    HttpContext.Current.Response.Flush()
End Sub

''
Public Shared Sub Clear()
    HttpContext.Current.Response.Clear()
End Sub



24.3.3. Server对象

Server对象定义为HttpServerUtility类，其中定义了一些很实用的工具。


	Execute()方法，处理指定的资源，在输出执行结果。

	MapPath()方法，返回网站虚拟路径在服务器中的物理路径。

	HtmlEncode()和HtmlDecode()方法，对于HTML内容的编码与解码。

	UrlEncode()和UrlDecode()方法，对于URL内容的编码与解码。



下面的代码会返回网站根目录在服务器中的实际路径，大家可以看一下，但不要在实际项目将类似的重要信息暴露给客户端。


Response.Write(Server.MapPath("/"))



24.3.4. Session对象

Session对象定义为HttpSessionState类型，可用于当前会话的管理工作，其常用成员包括：


	SessionID属性，返回会话的唯一ID，定义为String类型。

	Clear()方法，清楚所有会话集合内容。



此外，我们可以通过“键”保存当前会话的数据，如:


Session("user_name") = "admin"



在保存简单的用户信息时，我们可以直接使用此方法；如果是访问量较大，我们就应该考虑使用其它方法来保存会话数据，如保存到数据库中。

接下来，我们来开始介绍ASP.NET页面中的一些常用控件及其应用特点。

24.4. 常用Web控件

从代码上看，使用Web控件与使用WinForm中的控件很相似，但我们应注意，Web项目与Windows窗体项目的工作方式是完全不一样的，在Web项目中，我们需要从服务器、客户端（浏览器），以及网络状态等多方面来考虑项目的结构问题。

本节，我们就来讨论一个常用的Web控件的应用问题。

24.4.1. 添加Web控件

虽然在VS Express 2013 for Web开发环境中，我们可以在页面的“设计”视图中（通过右下角按钮切换页面视图模式）像设计Windows窗体那样添加控件，但是，如果想要深入学习Web开发，自己添加代码将是不可缺少的一项工作。

在ASP.NET页面中，我们应该将Web控件添加到包括runat="server"
 属性的<form>
 元素（Web窗体）中，如：


<form id="form1" runat="server">
    <div>
        <asp:Button ID="btnSubmit" Text="提交" runat="server" />
    </div>
</form>



在Web窗体中添加控件，有一些基本的注意事项：


	Web控件标签都使用 “asp:前缀+控件类型”的格式；并且同时需要开始标记和关闭标记。如<asp:Button>
 和</asp:Button>
 ，不过，如果在控件中不包括其它类型的组件，可以简化关闭标记，如<asp:Button />
 。

	ID属性定义控件的唯一的标识。

	Text属性设置控件中的默认显示内容。

	
runat="server"
 属性指定这是一个运行在服务器端的控件。对于Web控件来讲，这个属性是必不可少的。

	Visible属性，设置控件是否可见。



对于在Web控件中没有定义的属性，我们也可以使用，此时，ASP.NET会将这些属性内容直接发送到客户端（如浏览器）中处理。利用这一特性，我们可以在Web控件中使用传统的HTML元素属性，如使用style属性直接设置控件的CSS样式。

对于一些与HTML元素同名的Web控件属性（如onclick），我们在Web窗体中只能定义为Web控件属性；但对于一些特殊而又常用的HTML元素属性，在Web控件中也定义了对应的属性，如：


	CssClass属性，设置控件的CSS类名，可以与CSS样式表配合使用。对应HTML元素中的class属性。

	OnClientClick属性，设置HTML元素的单击事件的响应代码（JavaScript），对应HTML元素中的onclick属性。稍后，我们会在按钮类控件中看到此属性的应用。



和Windows窗体控件一样，在Web控件中，我们也可以使用控件的事件来响应操作，如按钮的单击事件，我们需要使用OnClick属性设置响应的方法，如：


<asp:Button ID="btnSubmit" OnClick="btnSubmit_Click" Text="提交" runat="server" />



然后，在<script runat="server"></script>
 中，我们需要定义这个方法，如下面的代码。


<script runat="server">
    Private Sub btnSubmit_Click(s As Object, e As EventArgs)
        lblRefreshTime.Text = "提交时间:" & Now.ToLongTimeString()
    End Sub
    ' 其它代码
</script>



请注意btnSubmit_Click()方法中参数的定义，这是与事件的委托类型定义相对应的，我们可以在Help中查看各种事件的相关定义。

ASP.NET中的基本控件类型都定义在System.Web.UI.WebControls命名空间，大家可以在这里查看这些控件的完整定义与使用说明，接下来，我们讨论一些常用的控件类型。

接下来的示例，大家可以在/Test.aspx页面中手工输入以便加深映像。

24.4.2. 文本类控件

和Windows窗体控件一样，文本类控件的作用就是用于显示或输入文本内容，在Web窗体中，也有一些常用的文本类控件，如：


	Literal控件和Label控件，都用于显示文件，它们的区别在于Literal控件中的文本不允许应用样式，并且会保留控件的位置。

	TextBox控件，用于显示或输入文本内容。



我们知道，使用Text属性可以设置文本类控件的显示内容；而使用TextBox控件时，还有以下一些特点需要注意，如TextMode属性，它包括三个可能的属性值，即：


	SingleLine，默认值，用于输入或显示单行文本。对应HTML中的<input type="text" />
 元素，实际上，在<input />
 元素中，单行文本同样是默认的表现，如<input />
 显示的元素。

	Password，内容显示为掩码。对应HTML中的<input type="password" />
 元素。

	MultiLine，用于输入或显示多行文本。对应HTML中的<textarea></textarea>
 元素。



下面的代码，我们定义了一个简单的提交文本内容的窗体，包括用户名、密码和提交的内容。


<form id="form1" runat="server">
<div>
<p>用户名
<asp:TextBox ID="txtUser" runat="server" /></p>
<p>密码
<asp:TextBox ID="txtPwd" TextMode="Password" runat="server" /></p>
<p>提交内容<br />
<asp:TextBox ID="txtContent" TextMode="MultiLine" runat="server" /></p>
</div>
</form>



页面显示类似下图。

我们必须承认，这个界面的确不太美观，不过，样式的设计主要是CSS的工作，在这里就不过多讨论了；还是那句话，如果想玩好Web开发，传统的HTML、CSS和JavaScript依然是非常重要的开发技术。

如果你需要使用在ASP.NET页面中定义HTML控件，可以参考System.Web.UI.HtmlControls命名空间中的相关内容。

24.4.3. 按钮类控件

OnClientClick属性

Web控件中，从操作类型上看，按钮类控件主要有两种，一种是链接类，比如，显示为为页面中的一个超链接（Hyperlink）；另一种是响应类，如表单提交按钮。

在System.Web.UI.WebControls命名空间中，常用的按钮类控件包括Button、LinkButton和ImageButton。

Button控件

我们首先看一看Button控件的应用。默认情况下，Button控件呈现为常见的按钮样式，下图从左到右分别是Chrome、Firefox和IE11浏览器中的默认样式。

下面，我们继续在Test.aspx页面中添加一个提交按钮，如下面的代码。


<p><asp:Button ID="btnSubmit" Text="提交" runat="server" /></p>



这里使用的属性相信大家都很熟悉了，而响应单击按钮的方法，我们也已经看到过，即通过OnClick属性设置其响应的方法，如：


<asp:Button ID="btnSubmit" Text="提交" OnClick="btnSubmit_Click"
 runat="server" />



然后，在<script runat="server"></script>
 中定义btnSubmit_Click()方法，如下面的代码。


<script runat="server">
    Private Sub btnSubmit_Click(s As Object, e As EventArgs)
        ' 单击响应代码
    End Sub
</script>



此外，当我们了解JavaScript相关知识以后，可能需要在客户端浏览器中确认表单的提交操作，如基本的数据检查或友情提示；此时，HTML元素中的onclick事件需要使用Web控件的OnClientClick属性来指定，如：


<asp:Button ID="btnSubmit" Text="提交" OnClick="btnSubmit_Click"
 OnClientClick="return confirm('确认要提交数据吗？');" 
 runat="server" />



当我们在页面中单击“提交”按钮时，会出现如下图中的JavaScript对话框。

单击“确定”，Web表单会提交到服务器，而单击“取消”则撤消单击操作，不会将表单数据提交到服务器。

在使用OnClientClick属性时，除了设置简单的JavaScript代码，我们还可以在页面中使用JavaScript函数，而函数则应该返回一个Boolean类型的数据，如true或false值。如下面的代码。


<asp:Button ID="btnSubmit" Text="提交" OnClick="btnSubmit_Click"
 OnClientClick="checkData();" 
 runat="server" />



然后，我们需要在页面中定义JavaScript函数checkData()，如下面的代码。


</html>
<script type="text/javascript">
    function checkData() {
        if (document.getElementById("txtUser").value == "" || 
            document.getElementById("txtPwd").value == "" ||
            document.getElementById("txtContent").value == "")
        {
            alert("请输入完整的用户及提交信息");
            return false;
        } else {
            return true;
        }
    }
</script>



在这里，请注意页面中添加JavaScript脚本代码需要注意的几点：


	JavaScript脚本放在页面的最后，即</html>
 关闭标记之后，这样的好处是当页面载入完成以后，才会执行脚本，可以有效防止脚本调用页面元素时可能出现的问题。

	同样使用<script></script>
 元素包含JavaScript脚本，但不使用runat属性，而是使用type属性指定脚本类型，虽然在HTML5标准中type属性已不是必须，但写明确点也没坏处。



LinkButton控件

LinkButton控件的外观默认会是超链接的样式，其它方面与Button控件的使用非常相似，这里就不多说了。

ImageButton控件

ImageButton控件使用图片作为按钮的样式，我们可以使用ImageUrl属性设置图片的位置，而其它应用方式也与Button控件相似。

24.4.4. 列表类控件

相对于Windows窗体中的列表类控件，在Web表单中的控件可能更易于使用，为什么这么说，我们慢慢来讨论。在Web控件中，常用的列表类控件包括：ListBox、DropDownList、RadioButtonList和CheckBoxList控件。我们首先看一看ListBox控件的使用。

ListBox控件

ListBox控件用于呈现标准的列表控件，其Rows属性设置显示的列表项数量，默认值是4。另外一个重要的属性是SelectionMode，它定义了列表项的选择模式，默认为Single，即单选模式；另一个可能的值是Multiple，即多选模式。

在使用列表类控件时，其每一个列表项都是一个ListItem对象，主要包括Value和Text属性。其中，Text属性表示列表项中显示的内容，而Value则表示列表项中包含的数据值，这两个属性都定义为String类型（因为网页内容本身就是基于文本的了^^）。

接下来，我们来看一看关于列表控件的两个主要的操作，即设置或绑定列表项，以及如何获取选中的内容。以下代码位于/ListBoxDemo.aspx页面。

我们首先看一看如何手工定义列表内容，如下面的代码。


<asp:ListBox ID="listbox1" runat="server">
    <asp:ListItem Text="男" Value="1" />
    <asp:ListItem Text="女" Value="2" />
    <asp:ListItem Text="保密" Value="0" Selected="True" />
</asp:ListBox>



代码中，我们分别定义了三个列表项，并且使用Text和Value属性设置了列表项显示的文本和对应的数据值；此外，我们还使用Selected属性将“保密”项设置为默认值，即默认选中的项目。

在项目中，如果列表项是固定的，那么，这种定义方法将是最有效的，但是，如果列表项是动态的，比如需要从数据表中绑定最新的数据，我们就需要使用动态代码来完成这些工作。前面，我们已经封装了关于数据库操作的代码，稍后，我们将使用这些代码简化列表数据的操作。

现在，我们来看一看如何获取列表选中的项目，相关的属性包括：


	SelectedIndex属性，已选择的列表项索引值，第一个选项索引值为0，第二个为1，以此类推。如果没有选中列表项，则返回-1，我们可以使用此属性值判断是否有选中的项目。

	SelectedItem属性，表示选中的ListItem对象，我们可以使用其中的Text属性获取项目的文本内容。

	SelectedValue属性，直接返回选中项目的值，此属性返回结果与SelectedItem.Value相同。



此外，如果需要在选中的列表中改变时做一些相关联的工作，可以有两种方法，一是在客户端，如浏览器中使用JavaScript来实现；另一种方法就是使用控件的OnSelectedIndexChanged事件，同时需要将AutoPostBack属性设置为True，这样，当选中的列表项改变时，就会自动回传到服务器做出相应的操作。如下面的代码，我们修改listbox1控件的属性设置。


<asp:ListBox ID="listbox1" AutoPostBack="True"
 OnSelectedIndexChanged="listbox1_SelectedIndexChanged"
 runat="server">
<!-- 其它代码 -->



然后，我们在<script runat="server"></script>
 中添加响应代码，如下面的代码。


<script runat="server">
Private Sub listbox1_SelectedIndexChanged(s As Object,e As EventArgs)
    If listbox1.SelectedIndex = -1 Then
        Response.Write("没有选中性别")
    Else
        Response.Write(String.Format("选中索引{0}，值{1},{2}", _
                                     listbox1.SelectedIndex, _
                                     listbox1.SelectedValue, _
                                     listbox1.SelectedItem.Text))
    End If
End Sub
</script>



当我们打开/ListBoxDemo.aspx页面，并选择不同的列表项时，可以看到自动显示相应的内容，如下图。

使用AutoPostBack属性时应注意，如果是并发操作不多时，可以考虑使用；而在并发访问量非常大的项目中，还是应该在客户端分担一些工作，使用JavaScript和HTML元素的onclick等事件来完成一些任务，而不是把所有的工作交由服务器来做。

关于判断ListBox列表的多选项，我们需要首先将ListBox控件的SelectionMode属性设置为Multiple；然后，使用一个循环来判断所有的项目，并通过ListItem对象中的Seleccted属性来判断此项是否被选中，如下面的代码。


Dim maxIndex As Integer = listbox1.Items.Count - 1
For i = 0 To maxIndex
    If listbox1.Items(i).Selected = True Then
        Response.Write(i.ToString() & "被选中")
    Else
        Response.Write(i.ToString() & "没有选中")
    End If
    Response.Write("<br />")
Next



不过，如果是提供一个多选项，使用CheckBoxList控件可能要比ListBox控件更加直观，稍后讨论。

DropDownList控件

DropDownList控件用于显示下拉列表，如下面的代码（/DropDownListDemo.aspx文件）。


<asp:DropDownList ID="drop1" AutoPostBack="True" 
        OnSelectedIndexChanged="listbox1_SelectedIndexChanged" 
        runat="server">
    <asp:ListItem Text="男" Value="1" />
    <asp:ListItem Text="女" Value="2" />
    <asp:ListItem Text="保密" Value="0" Selected="True" />
</asp:DropDownList>



控件会显示如下图所示的下拉列表。

关于DropDownList控件的其它操作，完全参照ListBox即可，只是需要注意，在DropDownList控件中不允许多项。

RadioButtonList控件

实际应用中，单个的RadioButton控件可能并不常用，而是使用RadioButtonList控件，它与DropDownList控件的功能基本相似，所不同的是，在使用RadioButtonList控件时，用户可以更清晰地看到可选项，当可选项不多时，使用RadioButtonList控件会比DropDownList更加直观。

RadioButtonList控件中的RepeatDirection属性用于设置单选按钮排列的方法，如Horizontal（水平排列）和Vertical（垂直排列，默认值）。下面的代码，我们将使用RadioButtonList控件设置性别的选项（/RadioButtonListDemo.aspx）。


<asp:RadioButtonList ID="rdoList1" RepeatDirection="Horizontal"
        AutoPostBack="True" 
        OnSelectedIndexChanged="listbox1_SelectedIndexChanged" 
        runat="server">
    <asp:ListItem Text="男" Value="1" />
    <asp:ListItem Text="女" Value="2" />
    <asp:ListItem Text="保密" Value="0" Selected="True" />
</asp:RadioButtonList>



打开/RadioButtonListDemo.aspx页面，我们可以看到如下内容。

获取选中的内容时，我们同样可以使用SelectedIndex、SelectedItem和SelecteValue等属性。

CheckBoxList控件

单个的CheckBox控件可以表示类似“真/假”、“是/否”的状态，此时，可以使用Checked属性来表示控件的状态值。而多个CheckBox组成的CheckBoxList控件，则可以更直观的表现出多选的意义。

如下面的代码，我们定义了几种爱好的选项（/CheckBoxListDemo.aspx文件）。


<asp:CheckBoxList ID="chkList1" RepeatDirection="Horizontal" 
        AutoPostBack="true"
        OnSelectedIndexChanged="chkList1_SelectedIndexChanged"
        runat="server">
    <asp:ListItem Text="读书" Value="1" />
    <asp:ListItem Text="音乐" Value="2" />
    <asp:ListItem Text="电影" Value="3" />
    <asp:ListItem Text="篮球" Value="4" />
    <asp:ListItem Text="足球" Value="5" />
    <asp:ListItem Text="乒乓球" Value="6" />
</asp:CheckBoxList>



然后，我们添加OnSelectedIndexChanged事件的响应代码（/CheckBoxListDemo.aspx文件）。


<script runat="server">
    Private Sub chkList1_SelectedIndexChanged(s As Object, _
            e As EventArgs)
        Dim maxIndex As Integer = chkList1.Items.Count - 1
        For i = 0 To maxIndex
            If chkList1.Items(i).Selected = True Then
                Response.Write(i.ToString() & "被选中")
            Else
                Response.Write(i.ToString() & "没有选中")
            End If
            Response.Write("<br />")
        Next
    End Sub
</script>



打开/CheckBoxListDemo.aspx页面，并选择其中的项目，我们可以看到如下图中的界面。

24.5. 自定义Web控件

在ASP.NET项目中需要自己创建的控件可能更多的是一些组合类控件，如页眉、页脚，以及一些常用的表单等，我们可以使用.ascx文件创建它们控件。

.ascx文件同样是基于页面内容的文本文件，只是我们需要使用<%@ Control%>
 指令来定义，如下面的控件定义（/controls/PageHeader.ascx文件）。


<%@ Control Language="VB" ClassName="PageHeader" %>
<script runat="server">
</script>
<header id="header1" class="pageHeader">
    <h1>页眉控件测试</h1>
</header>



在<%@ Control%>
 指令中，Lanauage属性同样指定动态语言类型，而ClassName则定义控件的类型名称。在.ascx文件中，同样可以包含动态代码部分和页面部分。本例中，我们并没有添加动态代码部分。

接下来，我们还需要在页面中使用这个控件，可以有两种方法注册它。第一种方法是当控件只在一个页面中使用时，可以使用<%@ Register %>
 指令在页面中注册，不过，只在一个页面中使用的控件似乎意义不大。

另一种注册Web控件的方法是在Web.config配置文件中注册，如下面的代码部分（/Web.config文件）。


<system.web>
  <pages>
    <controls>
      <add tagPrefix="chy" tagName="PageHeader" 
        src="/controls/PageHeader.ascx" />
    </controls>
  </pages>
</system.web>



请注意注册控件的所处的节点位置。在<add />
 标记中，我们使用了以下三个属性：


	tagPrefix属性，指定控件定义时的前缀，如asp:Button中的asp。

	tagName属性，指定控件名称，如asp:Button中的Button。

	src属性，指定.ascx文件的路径。



接下来，我们就可以在页面（/Test.aspx）中使用PageHeader控件了，如下面的代码。


<chy:PageHeader ID="pageheader1" runat="server" />



其显示如下图。

24.6. 使用数据库操作组件

对于ADO.NET基本组件的使用，无论是在Windows窗体项目或ASP.NET项目中，其操作是相同的，而现在，我们将前面编译的vbchef.dll和MySql.Data.dll文件拷贝到网站的bin目录，然后，在Web.config配置文件中可以引用一些常用的命名空间，这样就不需要在页面中单独引用了。如下面的代码（/Web.cofig.文件）。


<pages>
  <namespaces>
    <add namespace="vbchef"/>
  </namespaces>
</pages>



接下来，就可以使用已经封装的代码库了。

24.6.1. 定义项目主数据引擎

与Windows窗体项目类似，我们同样可以定义一个CApp类作为项目的主控类，如下面的代码（/app_code/CApp.vb文件）。


Imports Microsoft.VisualBasic
Imports vbchef

Public Class CApp
    '' 本项目初始数据库类型为SQL Server，
    '' 数据库名称cdb_demo
    Public Shared Db As IDbEngine = New CSqlEngine( _
     CSql.CreateRemoteCnnStr("cdb_demo", ".", "sa", "DEV_test123456"))

    '' 项目初始化
    Public Shared Function Init() As Boolean
        Try
            ' 初始化CDbRecord对象构造器
            CDbRecord.DbRecordBuilder = _
                      New CDbRecord.DDbRecordBuilder( _
                      AddressOf CApp.CreateDbRecord)
            '
            Return True
        Catch
            Return False
        End Try
    End Function

    '' 项目数据记录操作创建方法
    Public Shared Function CreateDbRecord(ByRef dbe As IDbEngine, _
                           ByRef sTableName As String, _
                           ByRef sIdName As String) As IDbRecord
        Return New CSqlRecord(dbe, sTableName, sIdName)
    End Function

    ''
End Class



你可能会发现，这里的CApp类与Windows窗体项目中定义的CApp类基本一样，只是项目初始化的Init()方法执行的位置不同，我们需要在Global.asax文件中的Application_Start()方法中执行，如下面的代码。


<%@ Application Language="VB" %>

<script runat="server">
    Sub Application_Start(ByVal sender As Object, ByVal e As EventArgs)
        ' Code that runs on application startup
        CApp.Init()
    End Sub
    ' 其它代码
</script>



24.6.2. 实现用户注册与登录

在System.Web.UI.WebControls命名空间中，定义了Login等类型的组件进行用户的注册和登录操作，不过，在这里，我们将自己编码来实现这些功能，并在稍后的内容中完善验证码功能的实现。

首先，对于用户的相关操作，我们封装为CUser类，其基本定义如下面的代码（/app_code/CUser.vb文件）。


Imports Microsoft.VisualBasic
Imports vbchef

Public Class CUser
    Inherits CDbRecord
    '
    Public Sub New()
        MyBase.New(CApp.Db, "user_main", "UserId")
    End Sub
    '
End Class



代码中，我们定义CUser继承于CDbRecord类，并在构造函数中定义了数据库引擎、数据表和Identity字段名称；这样，CUser也就拥有了项目主数据库中的用户数据表（user_main）的操作能力。

接下来，我们定义登录验证方法，如下面的代码（/app_code/CUser.vb文件）。


' 用户登录
Public Shared Function Login(sUser As String, sPwd As String) As Boolean
    Dim cond As New CDataCollection()
    cond.Append("UserName", sUser)
    cond.Append("UserPwd", GetSha1(sPwd))
    cond.Append("IsLocked", 0)
    Dim user As New CUser()
    Return (user.Find(cond) > 0)
End Function



我们对于登录操作的判断比较简单，根据用户名和密码（使用SHA1算法编码），当然还要是未锁定的用户；登录信息正确则返回True，否则返回False。

下面是/user/login.aspx页面的代码，它将使用CUser类进行用户的登录操作。


<%@ Page Language="VB" %>

<!DOCTYPE html>

<script runat="server">
    Private Sub btnLogin_Click(s As Object, e As EventArgs)
        ' 登录操作
        If CUser.Login(txtUser.Text, txtPwd.Text) Then
            lblMsg.Text = "登录成功"
        Else
            lblMsg.Text = "登录失败"
        End If
    End Sub
</script>

<html xmlns="http://www.w3.org/1999/xhtml">
<head runat="server">
<meta http-equiv="Content-Type" content="text/html; charset=utf-8"/>
    <title>用户登录</title>
</head>
<body>
    <form id="form1" runat="server">
    <div>
        <p>用户名<asp:TextBox ID="txtUser" 
                     MaxLength="15" runat="server" /></p>
        <p>密码<asp:TextBox ID="txtPwd" MaxLength="15" 
                   TextMode="Password" runat="server" /></p>
        <p><asp:Button ID="btnLogin" Text="登录" 
               OnClick="btnLogin_Click" runat="server" /></p>
        <p><asp:Label ID="lblMsg" runat="server" /></p>
    </div>
    </form>
</body>
</html>



打开/user/login.aspx页面，我们可以看到类似如下图所示的界面。

实际应用中，用户登录成功后，可以根据需要自动跳转到相应的页面。

接下来，我们看一下用户的注册。实际上，从我们定义的用户数据表的内容上来看，注册一个用户需要的信息非常简单，全部数据包括：用户名、密码（需输入两次确认）、E-mail和性别；从数据库的角度上看，后两项为可选，不过，保存用户的E-mail是很实用的，很多情况下，这可能是找回登录密码的唯一方法。下面的代码就是CUser类中定义的用户注册的方法（/app_code/CUser.vb类）。


Public Shared Function Register(sUser As String, _
                       sPwd As String, _
                       Optional sEmail As String = "", _
                       Optional iSex As Integer = 1) As Long
    ' 判断用户是否存在
    Dim data As New CDataCollection()
    data.Append("UserName", sUser)
    Dim user As New CUser()
    ' 用户已存在
    If user.Find(data) > 0 Then Return -2
    ' 添加用户记录
    data.Append("UserPwd", GetSha1(sPwd))
    data.Append("Email", sEmail)
    data.Append("Sex", iSex)
    data.Append("IsLocked", 0)
    Return user.Save(data, Nothing)
End Function



下面是注册页面的实现（/user/register.aspx文件）。


<%@ Page Language="VB" %>

<!DOCTYPE html>

<script runat="server">
    Private Sub btnRegister_Click(s As Object, e As EventArgs)
        ' 用户注册
        If txtUser.Text = "" Then
            lblMsg.Text = "用户名不能为空"
        ElseIf txtPwd.Text.Length < 6 Then
            lblMsg.Text = "密码最少需要6个字符"
        ElseIf String.Compare(txtPwd.Text, txtPwd.Text, False) <> 0 Then
            lblMsg.Text = "密码不能为空，且两次输入要一致"
        ElseIf CUser.Register(txtUser.Text, txtPwd.Text,  _
                      txtEmail.Text, rdoSex.SelectedValue) > 0 Then
            '注册成功
            Response.Redirect("/user/login.aspx")
        Else
            ' 注册失败
            lblMsg.Text="注册失败，请稍后重试！"
        End If
    End Sub
</script>

<html xmlns="http://www.w3.org/1999/xhtml">
<head runat="server">
<meta http-equiv="Content-Type" content="text/html; charset=utf-8"/>
    <title>用户注册</title>
    <style type="text/css">
        tr { height:2em; }
    </style>
</head>
<body>
    <form id="form1" runat="server">
    <table border="0">
        <tr><td>用户名</td><td><asp:TextBox ID="txtUser" 
            MaxLength="15" runat="server" /></td></tr>
        <tr><td>密码</td><td><asp:TextBox ID="txtPwd" 
            MaxLength="15" TextMode="Password" runat="server" />
                </td></tr>
        <tr><td>密码确定</td><td><asp:TextBox ID="txtPwd1" 
            MaxLength="15" TextMode="Password" runat="server" />
                </td></tr>
        <tr><td>E-mail</td><td><asp:TextBox ID="txtEmail" 
            MaxLength="50" Width="300" runat="server" /></td></tr>
        <tr><td>性别</td><td>
            <asp:RadioButtonList ID="rdoSex" 
                RepeatDirection="Horizontal" runat="server">
                <asp:ListItem Value="0" Text="保密" Selected="True" />
                <asp:ListItem Value="1" Text="男" />
                <asp:ListItem Value="2" Text="女" />
            </asp:RadioButtonList>
        </td></tr>
    </table>
    <p><asp:Button ID="btnRegister" Text="注册" 
        OnClick="btnRegister_Click" runat="server" /></p>
    <p><asp:Label ID="lblMsg" runat="server" /></p>
    </form>
</body>
</html>



打开/user/register.aspx页面，我们可以看到类似如下图的界面。

在register.aspx页面中，我们使用了HTML表格进行布局的设计，如果你是<div>+css
 控，可以修改相关的布局方法。另一方面，当用户注册成功后，页面会跳转到login.aspx页面，我们可以使用注册的用户进行登录操作测试。

下面，我们将看到验证码功能的实现。

24.7. 实现验证码功能

出于操作安全性的考虑，各种验证方法层出不穷，而传统的验证码依然是常见的方法，本节，我们就讨论在ASP.NET项目中自己动手实现验证码的方法。

验证码工作机制中，操作由服务器端发起，由客户端根据显示的内容输入验证码内容，最后再由服务器验证用户输入的正确性。如下图所示。

在ASP.NET项目中，我们封装了CCheckCode类用于验证码的基本操作，其中定义了三个方法，即：


	Create()方法，创建验证码，并将图片内容发送到客户端。

	Save()方法，验证码在服务器中的保存方法。

	Check()方法，对用户输入验证码的检查。



下面的代码就是我们定义CCheckCode类（/app_code/vbchef.webx.CCheckCode.vb文件）。


Imports Microsoft.VisualBasic
Imports System.Web
Imports System.Drawing
Imports System.Drawing.Imaging
Imports System.Text

Namespace vbchef.webx
    Public Class CCheckCode
        '' 颜色刷
        Protected Shared myBrushes As Brush() =
            {Brushes.Blue, Brushes.Black, Brushes.Red, 
             Brushes.DarkGreen,Brushes.DarkRed, Brushes.Purple, 
             Brushes.Navy, Brushes.DarkOrange}

        '' 随机数生成
        Protected Shared rnd As New Random()

        '' 创建验证码将发送到客户端
        Public Shared Sub Create(sKey As String,  _
                                  Optional iLen As Integer = 5)
            Const width As Integer = 100
            Const height As Integer = 40
            Const height_2 As Integer = 20
            Dim bmp As New Bitmap(width, height)
            Dim g As Graphics = Graphics.FromImage(bmp)
            Dim sb As New StringBuilder(iLen)
            '
            For i = 1 To iLen
                Dim s As String = rnd.Next(0, 10).ToString()
                sb.Append(s)
                '
                Dim f As New Font("Arial", rnd.Next(15, 26), _
                                     FontStyle.Italic)
                Dim tmp As Integer = (i - 1) * 16
                Dim x As Integer = rnd.Next(tmp - 3, tmp + 3)
                Dim y As Integer = rnd.Next(1, height_2)
                Dim b As Brush =  _
                       myBrushes(rnd.Next(0, myBrushes.Length))
                g.DrawString(s, f, b, x, y)
                g.DrawLine(New Pen(b, 2), x, y,  _
                       rnd.Next(0, width), rnd.Next(0, height))
            Next
            ' 保存验证码
            Save(sKey, sb.ToString())
            ' 发送验证码
            Dim hc As HttpContext = HttpContext.Current
            hc.Response.ContentType = "image/png"
            hc.Response.AppendHeader("Content-Disposition", _
                           "attachment;filename=checkcode.png")
            bmp.Save(hc.Response.OutputStream, ImageFormat.Png)
        End Sub

        '' 保存验证码
        Protected Shared Sub Save(sKey As String, sValue As String)
            HttpContext.Current.Session(sKey) = sValue
        End Sub

        '' 检查用户输入
        Public Shared Function Check(sKey As String, _
                                     sInputCode As String) As Boolean
            If sInputCode = "" Then Return False
            Dim code As String =  _
                    ToStr(HttpContext.Current.Session(sKey))
            Return (String.Compare(code, sInputCode, False) = 0)
        End Function
    End Class
End Namespace



代码中，我们生成的验证码默认为5位数字，你可以修改代码以生成更复杂的验证码，如使用数字与字母等字符的混合内容。

CCheckCode定义为通用的验证码操作类，我们可以在CUser类中创建调用它的方法，如下面的代码（/app_code/CUser.vb文件）。


' 登录验证码
Public Shared Sub CreateLoginCheckCode()
    CCheckCode.Create("user_login")
End Sub

' 检查登录验证码
Public Shared Function CheckLoginCheckCode( _
           sInputCode As String) As Boolean
     Return CCheckCode.Check("user_login", sInputCode)
 End Function



下一步，我们将创建一个生成登录验证码图片的页面，如下面的代码（/user/loginCheckCode.aspx）。


<%@ Page Language="VB" %>
<script runat="server">
    Private Sub Page_Load()
        CUser.CreateLoginCheckCode()
    End Sub
</script>



下面，我们就可以在登录页面中使用验证码，首先是添加显示和输入验证码的元素。如下面代码（/user/login.aspx文件）。


<p>验证码
<asp:TextBox ID="txtCheckCode" MaxLength="5" 
         Width="100" runat="server" />
<img id="checkcode" src="loginCheckCode.aspx" 
         style="border:none;vertical-align:top;" />
<a href="javascript:changeImage();">换一张</a>
</p>



代码中，我们定义了一个JavaScript函数changeImage()，它的功能是用户在看不清的情况下更换验证码图片，其定义如下面的代码（/user/login.aspx）。


</html>
<script type="text/javascript">
    function changeImage() {
        document.getElementById("checkcode").setAttribute("src",
            "loginCheckCode.aspx?ts" + new Date().getTime());
    }
</script>



在登录按钮的响应方法中，我们添加验证码的检查代码，如下面的代码（/user/login.aspx文件）。


Private Sub btnLogin_Click(s As Object, e As EventArgs)
    ' 检查验证码
    If CUser.CheckLoginCheckCode(txtCheckCode.Text) = False Then
        lblMsg.Text = "验证码输入错误"
        Exit sub
    End If
    ' 登录操作
    If CUser.Login(txtUser.Text, txtPwd.Text) Then
        lblMsg.Text = "登录成功"
    Else
        lblMsg.Text = "登录失败"
    End If
End Sub



最后，我们打开/user/login.aspx页面，看一看执行的效果，如下图所示。

24.8. 文件上传

在ASP.NET项目中，文件上传变得非常简单，我们只需要使用FileUpload控件就可以实现基本的文件上传操作，其常用成员包括：


	FileBytes属性，以字节数组格式返回上传文件的内容，定义为Byte()类型。

	FileContent属性，以流（Stream）格式返回上传的文件内容。

	FileName属性，返回上传的文件名，不包含路径。

	HasFile属性，判断是否真的上传了文件。

	SaveAs()方法，将上传文件保存到指定的位置。



在FileUpload类中定义的另一个属性是PostedFile（System.Web.HttpPostedFile类型），它表示一个已上传的文件对象。其常用成员包括：


	ContentLength属性，返回上传文件的字节数（Integer类型）。

	ContentType属性，返回上传文件的MIME类型标识（String类型）。

	FileName属性，返回上传文件在客户端的完整路径。

	InputStream属性，返回上传文件内容的流数据（Stream类型）。

	SaveAs()方法，将上传文件保存到指定的位置。



在ASP.NET项目中，另一个关于上传文件相关的问题是控制上传文件的尺寸，我们可以在Web.config文件中进行相关的配置，主要是<httpRuntime />
 标记中maxRequestLength属性，它定义了允许上传的尺寸，单位为KB，默认值为4096（即4MB）。

下面的代码（/uploadPicture.aspx）演示了如何上传一张图片，并重新设置格式，最后保存到服务器的操作。


<%@ Page Language="VB" %>

<!DOCTYPE html>

<script runat="server">
    Private Sub btnUpload_Click(s As Object, e As EventArgs)
        If file1.HasFile = False Then
            lblMsg.Text = "请选择图片文件"
        Else
            Dim ext As String =  _
                Path.GetExtension(file1.FileName).ToLower()
            If ext <> ".jpg" And ext <> ".bmp" And ext <> ".png" Then
                lblMsg.Text = "请选择JPEG、BMP或PNG格式图片"
            Else
                ' 保存文件
                Dim sPath As String = Server.MapPath("/upload")
                Dim sFile As String = GetGuid() & ".png"
                Dim bmp As New Bitmap(file1.FileContent)
                bmp.Save(Path.Combine(sPath, sFile), ImageFormat.Png)
                ' 保存数据
                Dim data As New CDataCollection()
                data.Append("UploadTime", Now)
                data.Append("UploadIp", CWeb.UserIp)
                data.Append("ClientFileName", file1.FileName)
                data.Append("ServerFileName", sFile)
                Dim up As New CUploadPicture()
                If up.Save(data, Nothing) > 0 Then
                    lblMsg.Text = "图片已成功上传"
                Else
                    lblMsg.Text = "图片上传数据保存错误"
                End If
            End If
        End If
    End Sub
</script>

<html xmlns="http://www.w3.org/1999/xhtml">
<head runat="server">
<meta http-equiv="Content-Type" content="text/html; charset=utf-8"/>
    <title>上传图片</title>
</head>
<body>
    <form id="form1" runat="server">
    <div>
        <p>请选择图片文件
        <asp:FileUpload ID="file1" runat="server" /></p>
        <p><asp:Button ID="btnUpload" Text="上传" 
            OnClick="btnUpload_Click" runat="server" /></p>
        <p><asp:Label ID="lblMsg" runat="server" /></p>
    </div>
    </form>
</body>
</html>



打开/uploadPicture.aspx页面，我们可以看到如下图所示的界面（Firefox浏览器）。

代码中，我们同时将图片相关信息保存到cdb_demo数据库中的upload_picture数据表，这样，可以便于图片的检索，保存图片信息的数据表定义如下，大家可以在SQL Server数据库中执行。


use cdb_demo;
go

create table upload_picture (
RID bigint identity(1,1) not null primary key,
UploadTime datetime,
UploadIp nvarchar(23),
ClientFileName nvarchar(255),
ServerFileName nvarchar(50),
);



用于操作upload_picture表的类定义为CUploadPicture，它的定义如下（/app_code/CUploadPicture.vb文件）。


Imports Microsoft.VisualBasic
Imports vbchef

Public Class CUploadPicture
    Inherits CDbRecord
    '构造函数
    Public Sub New()
        MyBase.New(CApp.Db, "upload_picture", "RID")
    End Sub
End Class



24.9. 页面缓存

合理地使用缓存机制对于网站的整体性能提高有很大的帮助，在ASP.NET的页面（.aspx）或控件（.ascx）中，我们可以使用OutputCache指令设置相关的缓存参数。

在OutputCache指令中，有以下几个常用的参数需要注意：


	Duration参数，指定缓存的时间，单位为秒。

	Location参数，设置缓存策略，定义为OutputCacheLocation枚举类型，默认值为Any。请注意，本参数不支持.ascx文件中的OutputCache指令。

	Shared参数，只用于.ascx文件中的OutputCache指令，指定控件缓存是否在所有引用的页面中都有效，对于静态的Web控件，我们可以将此参数设置为True。

	VaryByParam参数，设置根据页面参数不同时的缓存策略，默认为none，即不考虑参数的不同，我们也可以设置指定的参数名，多个参数使用逗号分隔，也可以使用通配符*指定对所有参数分别进行缓存。



最后，需要说明的是，OutputCache指令设置的缓存参数只是理想状态下的，服务器会根据运行状态随时对页面缓存进行清理，以维护服务器的高效运行。

